
Preface

Pharoとは?

Pharoは Smalltalkプログラミング言語・環境をフルに実装した、オープンソー
スによる新しい処理系です。 Pharoは古典的な Smalltalk-80システムの再実装
である、 Squeak1から派生しました。Squeakが主に実験的な教育向けソフト
ウェアを開発するためのプラットフォームとして作られたのに対して、Pharo
はプロフェッショナルなソフトウェアを開発するための、凝縮されたオープン

ソース・プラットフォームを目指しています。また、動的な言語・環境の研究

開発のための、頑丈で安定したプラットフォームの提供にも努めます。Pharo
はウェブアプリ開発フレームワーク Seasideのリファレンス実装用にも使われ
ています。

Pharoは Squeakにあったライセンスの問題を解決しています。Squeakの
以前のバージョンとは異なり、Pharoのコア部分はMITライセンスで寄贈され
たコードのみからできています。Pharoプロジェクトは 2008年 3月に Squeak
3.9からのフォークとして始まり、最初の 1.0ベータバージョンは 2009年 7月
31日にリリースされました。

Pharo は Squeak からパッケージをいくつも取り去った一方で、Squeak
ではオプションだった機能をたくさん取り入れています。例えば、Pharoには
True Typeフォントが入っています。Pharoは本物のブロッククロージャもサ
ポートしています (訳注:Squeakでは BlockContextという再帰できないクロー
ジャ実装でした)。ユーザインターフェースもシンプルで洗練されたものになっ
ています。

Pharo は極めて高い移植性を持っています。仮想マシンでさえもすべて
Smalltalk で書かれていて、デバッグ・解析・変更をしやすくなっています。
Pharoはマルチメディア・アプリケーションから、教育向けプラットフォーム、
商用ウェブ開発環境まで、広い範囲で革新的なプロジェクトの媒体となります。

1Dan Ingalls  et al., Back to the Future: The Story of Squeak, a Practical Smalltalk Written in
Itself. In Proceedings of the 12th ACM SIGPLAN conference on Object-oriented programming,
systems, languages, and applications (OOPSLA’97). ACM Press, November 1997 〈URL: http:
//www.cosc.canterbury.ac.nz/~wolfgang/cosc205/squeak.html〉.

http://www.cosc.canterbury.ac.nz/~wolfgang/cosc205/squeak.html
http://www.cosc.canterbury.ac.nz/~wolfgang/cosc205/squeak.html

Pharoは重要な視点を背景として持っています。「Pharoは過去の単なる複
製であってはならない。Smalltalkを再 発 明 すべきだ」というものです。ビッ
グバンアプローチは滅多に成功しません。Pharoでは進化的でインクリメンタ
ルな変化をさせる方法を取ります。重要な新機能やライブラリ開発を実験的に

進められるようにしたいのです。進化とは Pharoが誤りを受け入れていくこと
を意味します。Pharoでは一度で完璧に解決することを目指しません。そうし
たくなる気持ちをこらえてでも、です。小さな変化をいくつも積み上げていく

やり方が、Pharoではうまくいくでしょう。Pharoの成功はコミュニティから
の貢献にかかっているのです。

この本が想定している読者

この本はオープンソースで公開されている Squeakの入門書 Squeak by Example2

を元にしています。その名の通り、Pharoと Squeakの違いを反映して修正し
ました。Pharoの基本からはじまり、先進的な話題まで、内容は多岐にわたり
ます。

この本はどうやってプログラムするのかを教える本ではありません。この

本は既にプログラミング言語をいくつか知っている人に向けて書かれています。

オブジェクト指向プログラミングの経験も役に立つでしょう。

この本では Pharoのプログラミング環境 (言語と関連ツール)を紹介します。
一般的なイディオムや実例に触れてもらいますが、オブジェクト指向設計よ

りも技術の方に焦点を当てています。可能限り例をたくさん示そうと思います

(Alec Sharpによる素晴らしい Smalltalkの本3にインスパイアされています)。
ウェブ上には Smalltalk についての無料の本が他にもたくさんあります。

しかし Pharoに特化したものはありません。参照: http://stephane.ducasse.free.fr/
FreeBooks.html

読者に向けてのアドバイス

個々の Smalltalkの記述がすぐにわからないからといっていらだってはいけま
せん。すべてを知る必要はないのです! Alan Knightは、以下のようにこの原
則を表現しています。4:

2http://SqueakByExample.org
3Alec Sharp, Smalltalk by Example. McGraw-Hill, 1997 〈URL: http://stephane.ducasse.free.fr/

FreeBooks/ByExample/〉.
4http://www.surfscranton.com/architecture/KnightsPrinciples.htm

http://stephane.ducasse.free.fr/FreeBooks.html
http://stephane.ducasse.free.fr/FreeBooks.html
http://SqueakByExample.org
http://stephane.ducasse.free.fr/FreeBooks/ByExample/
http://stephane.ducasse.free.fr/FreeBooks/ByExample/
http://www.surfscranton.com/architecture/KnightsPrinciples.htm

気にしないようにする Smalltalkを学び始めたプログラマ
は、どうやってそれが動くのかを詳細まですべて理解しな

ければならないと考え、困ってしまうことがよくあります。

つまり、Transcript show: ’Hello World’ をマスターするまでに
かなり時間がかかるということです。オブジェクト指向の

優れた点は、「どうやって動くのだろう?」という問いに対
して「気にしない」と答えられることなのです。

オープンな本

この本は、以下の意味でオープンな本です。

• この本の内容はクリエイティブ・コモンズの表示—継承 (by-sa) ライセ
ンスで公開されています。つまり、以下の URLにあるライセンス条件
を尊重する限り、この本を自由に配布したり改変したりできます。URL:
http://creativecommons.org/licenses/by-sa/3.0/deed.ja

• この本は Pharoのコア部分のみを解説しています。理想を言うと、私た
ちが書かなかった部分も他の人が寄稿できるようにしたいと思っていま

す。この取り組みに参加したい人は連絡してください。私たちはこの本

が育つのを見たいのです!

詳しくは、http://PharoByExample.orgを見てください。

Pharoのコミュニティ

Pharoのコミュニティは親切で活動的です。役に立ちそうなリソースをいくつ
かここに挙げておきます。

• http://www.pharo-project.org Pharoのメインとなるウェブサイト。

• http://www.squeaksource.com Pharoプロジェクトにとっての SourceForge
的なもの。Pharo用の追加パッケージがここでたくさん生まれています。

例と練習問題

この本では特殊な記法を二つ決めました。

できる限りたくさんの例を示すようにしています。特に、実行可能な短い

コードで示した例がたくさんあります。式を選択して print itした結果を示すた
めに、 −→ 記号を使っています。

http://creativecommons.org/licenses/by-sa/3.0/deed.ja
http://PharoByExample.org
http://www.pharo-project.org
http://www.squeaksource.com

3 + 4 −→ 7 ”3+4を 選 択 し て ’print it’を 選 ぶ と 7が 得 ら れ る ”

Pharo で実際に試すときのために、この本のサイト (http://PharoByExample.org)
からすべてのコード例がテキストファイルでダウンロードできるようになって

います。

二つ目の決まりとして、みなさんにしてほしいことを示すため、 アイコ

ンを使います。

次 の 章 へ と 読 み 進 め ま し ょ う !

Part I

Getting Started





Chapter 1

Pharo早巡り

この章では、Pharo環境に親しんでもらうための、概要を解説します。Pharo
を実際に試す場面がたくさんあるので、この章はコンピュータを手元に置いて

読むとよいでしょう。

Pharoを動かして試してもらいたい箇所にはアイコン: で印を付けてお

きます。Pharoの起動方法、システムとやりとりする様々なやり方、基本的な
ツールについて知ることになるでしょう。また、新しいメソッドの定義の仕方

や、オブジェクトを作成してメッセージを送る方法についても学びます。

1.1 入門

Pharoは http://pharo-project.orgから自由にダウンロードできます。三つのパー
ツをダウンロードしなければなりません。それらは実際には 4個のファイルか
らなります (図 1.1)。

1. バ ー チ ャ ル マ シ ン (VM)。Pharoのシステムで唯一、オペレーティングシ
ステムとプロセッサごとに異なっているものです。主要なプラットフォー

ム用にコンパイル済みのバーチャルマシンが用意されています。図 1.1
にあるのは、某プラットフォーム用のバーチャルマシンで、Pharo.exeと
いう名前です。

2. ソ ー ス フ ァ イ ル 。ソースファイルには Pharo の中で、あまり頻繁
には変更されない部分のソースコードが入っています。図 1.1 では
SqueakV39.sourcesがソースファイルです。1

3. 仮想イメージ。シ ス テ ム 仮 想 イ メ ー ジ は、実行中の Pharoシステムのス
ナップショットです。このスナップショットは二つのファイルからなり

1Pharoは Squeak 3.9をベースに作られています。今のところバーチャルマシンは Squeakと
同じものを使っています。

http://pharo-project.org


4 Pharo 早 巡 り

Figure 1.1: Pharoがサポートするプラットフォーム用のダウンロードファイル。

ます。イメージファイル イ メ ー ジ フ ァ イ ル (.image)にはシステムのすべ
てのオブジェクト (クラスやメソッドもオブジェクトです)が入っていま
す。チェンジファイル チ ェ ン ジ フ ァ イ ル (.changes)にはシステムに加え
たソースコードの変更ログが入ります。図 1.1では、これらのファイル
は pharo.imageおよび pharo.changesです。

Pharo を ダ ウ ン ロ ー ド し て イ ン ス ト ー ル し ま し ょ う 。

Pharo by Exampleのウェブページにある仮想イメージを使うことをお勧め
します。2

本書のサンプルのほとんどは Pharo のどのバージョンでも動作するので、
もし Pharoを既にインストールしているのならそれを使ってもかまいません。
ただし、ここで説明することと、見た目や振る舞いが違ったからといって驚か

ないでください。

Pharoで作業しているとき、イメージファイルとチェンジファイルは常に
内容が更新されるため、これらは書き込み可能にしておいてください。そして、

二つのファイルは常に同じ場所に置いてください。テキストエディタで直接変

更してはいけません。Pharoは、オブジェクトを格納したりソースコードの変
更のログを書き出したりするのにこの二つのファイルを使います。ダウンロー

ドしたイメージファイルとチェンジファイルのバックアップを取っておくとよ

いでしょう。こうしておけば、いつでもまっさらな仮想イメージから Pharoを
開始し、自分で書いたコードを後からロードできます。

ソ ー ス フ ァ イ ル とバーチャルマシンは書き込み不可にして、ユーザ間で共

有できます。バーチャルマシン、ソースファイル、イメージファイル (チェン
2http://PharoByExample.org

http://PharoByExample.org

ジファイル)は同じ場所に置いてもかまいませんし、バーチャルマシンとソー
スファイルだけ別の共有ディレクトリに置いておくこともできます。オペレー

ティングシステムや作業スタイルに合わせて一番良い方法を選んでください。

Figure 1.2: http://PharoByExample.orgからダウンロードしたままのまっさらな仮
想イメージ。

起動する。 Pharoの起動は普通のアプリケーションと同じです: バーチャル
マシンのアイコンにイ メ ー ジ フ ァ イ ル をドラッグ&ドロップしたり、イ メ ー
ジ フ ァ イ ル をダブルクリックしたり、あるいはコマンドラインでバーチャルマ

シンの名前の後にイ メ ー ジ フ ァ イ ル へのパスを入力することになるでしょう。

使っているオペレーティングシステムに合わせてください (異なるバージョン
のバーチャルマシンがあると、オペレーティングシステムは正しいバーチャル

マシンを自動で選んでくれないかもしれません。そうしたときはドラッグ&ド
ロップ、あるいはコマンドラインから起動した方が安全でしょう)。

Pharoを起動すると、1個の大きなウィンドウが現れます。ウィンドウの中
にはワークスペースウィンドウがいくつか開いているかもしれません (図 1.2)。
しかしどうすればよいのか、これだけではわからないでしょう。メニューバー

が出ることもありますが、Pharoではもっぱらコンテキスト依存のポップアッ
プメニューを使います。

Pharo を 始 め ま し ょ う 。 ウ ィ ン ド ウ の 左 上 に あ る 赤 い ボ タ ン を ク リ ッ ク
す る と ワ ー ク ス ペ ー ス を 閉 じ る こ と が で き ま す 。

ウィンドウを最小化するにはオレンジ色のボタン、最大化するには緑色の

http://PharoByExample.org

ボタンをクリックします。

最初のやりとり。 図 1.3 (a) の ワールドメニュー から始めるのがいいでし
ょう。

メ イ ン ウ ィ ン ド ウ の 背 景 で マ ウ ス の ボ タ ン を ク リ ッ ク し て ワ ー ル ド メ ニ

ュ ー を 表 示 し ま し ょ う 。 そ し て Workspace を 選 択 し て 新 し い ワ ー ク ス ペ ー ス を
開 き ま し ょ う 。

(a) ワールドメニュー (b) コンテキストメニュー

(c) ハロー

Figure 1.3: ワールドメニュー (クリックすれば出る)、コンテキストメニュー
(アクションクリック)、ハロー (メタクリック)。

Smalltalkは元々 3ボタンマウスのコンピュータを想定して設計されてい
ます。マウスのボタンが足りないときは、修飾キーを押しながらマウスを ク

リックしなければなりません。2ボタンマウスでも十分 Pharoを使うことは
できますが、もし 1ボタンマウスしか持っていなら、クリックできるスクロー
ルホイールが付いた 2ボタンマウスを買うことを考えた方がいいでしょう。
そうすれば Pharoをいっそう楽しく使えるようになります。

様々なコンピュータ、マウス、キーボードそして個人設定があるため、Pharo
では「マウスの左ボタンをクリック」という表現は避けています。元々 Smalltalk
ではそれぞれのマウスボタンを色で表していました。3あることをするのにユー
ザが様々な修飾キー (control、ALT、metaなど)を使うので、この本では代わり
に以下の用語を用います:

クリック: 一番よく使われるマウスボタンで、通常は修飾キーを使わずに 1ボ
タンマウスをクリックするのと同じです。背景をクリックして「ワー

ルド」メニュー (図 1.3 (a))を表示してみましょう。

アクションクリック: 次によく使われるボタンです。このボタンでコンテキス
トメニューを表示します。コンテキストメニューが表示するアクション

の一覧は、マウスが指している場所によって異なります;図 1.3 (b)を見て
ください。マウスに該当するボタンがない場合、普通は control修飾キー
を使ってアクションクリックするように設定します。

メタクリック: 最後に、画面上の任意のオブジェクトをメタクリックすると、
そのオブジェクトの「ハロー」を表示させることができます。ハローは

画面上のオブジェクトを回転させたりリサイズしたりするためのハンド

ルの集まりです。図 1.3 (c)を見てください。4ハンドルにマウスをかざ
しておけば、バルーンヘルプが出てその機能について教えてくれます。

Pharoでは、メタクリックする方法はオペレーティングシステムに依存
します。shift ctrlあるいは shift optionを押しながらクリックすること
になるでしょう。

Time now と ワ ー ク ス ペ ー ス に 入 力 し ま し ょ う 。 そ し て ワ ー ク ス ペ ー ス で
ア ク シ ョ ン ク リ ッ ク し て 、 print it を 選 択 し ま し ょ う 。

右利きの人ならクリックは左ボタンに、アクションクリックは右ボタン

に、メタクリックはクリックできるスクロールホイール (あれば)に、それぞれ
設定することをお勧めします。Macintoshで 1ボタンのマウスを使っているの
なら、⌘ を押しながらマウスをクリックすることでアクションクリックか
メタクリックをシミュレートすることもできます。とはいえ、これから Pharo
を頻繁に使うのなら、最低でも二つ以上のボタンの付いたマウスに投資するこ

とをお勧めします。

オペレーティングシステムやマウスドライバの設定を変えることで、

マウスの動作を望み通りにできます。Pharo にはマウスやキーボードの
メタキーの設定を変更できる機能もあります。プリファレンス・ブラウザ

(System …. Preferences …. Preference Browser… ) では、 keyboard カテゴリの
swapControlAndAltKeys オプションを使うことで アクションクリック と メタ
クリックを入れ替えることができます。プリファレンス・ブラウザには他にも

様々なキーボードショートカットを設定するオプションがあります。

3マウスボタンの色は赤 、黄 そして青 です。著者はどの色がどのボタンを指していたのかさっぱ
り思い出せません。

4Pharoはデフォルトでハンドルを無効にしていますが、これらはプリファレンス・ブラウザを
使って有効にできます。プリファレンス・ブラウザについてはすぐ後で述べます。

Figure 1.4: プリファレンス・ブラウザ。

1.2 ワールドメニュー

Pharo の 背 景 で も う 一 度 ク リ ッ ク し て み ま し ょ う 。

再び World メニューが表示されるはずです。ほとんどの Pharoのメニュー
にはモードがありません。メニューは、右上のピンアイコンをクリックして、

画面上に残しておくことができます。やってみましょう。

ワールドメニューを使えば多くの Pharoのツールに簡単にアクセスするこ
とができます。

World メ ニ ュ ー と Tools … メ ニ ュ ー を 詳 し く 見 て み ま し ょ う (図 1.3 (a))。

ブラウザやワークスペースといった Pharoの主要なツールがリストされて
います。以降の章ではこれらのツールのほとんどを使うことになります。

1.3 メッセージを送る

ワ ー ク ス ペ ー ス を 開 け ま し ょ う 。 そ し て 以 下 の テ キ ス ト を 入 力 し ま し ょ う 。



メ ッ セ ー ジ を 送 る 9

BouncingAtomsMorph new openInWorld

ア ク シ ョ ン ク リ ッ ク し ま し ょ う 。 メ ニ ュ ー が 現 れ る は ず で す 。 do it (d) を
選 択 し ま し ょ う (図 1.5)。

Figure 1.5: 式を”do it”する。

Pharoの画面左上に、たくさんの原子が弾んでいるウィンドウが表示され
たはずです。

あなたはたった今、最初の Smalltalkの式を実行しました! BouncingAtoms-
Morphクラスへ newメッセージを送り、結果として BouncingAtomsMorphのイン
スタンスが生成され、さらに openInWorldメッセージがこのインスタンスに送
られました。BouncingAtomsMorphクラスは newメッセージを受け取った際、自
分がすることを決めました。つまり、bamクラスは newメッセージに対する
メ ソ ッ ド を探索して適切に反応しました。同様に BouncingAtomsMorphのインス
タンスも openInWorldに対するメソッドを探索して適切なアクションを取りま
した。

Smalltalkerとしばらく話せば、すぐに、彼らが「手続きをコールする」ま
たは「メソッドを呼び出す」といった表現を使わず、代わりに「メッセージを

送る」と言っていることに気づくでしょう。これは、オブジェクトが自身のア

クションに責任を持つという思想の現れです。決して、オブジェクトに何々を

しろと命 令 す る ことはありません。代わりにメッセージを送って、何かしてほ

しいと礼儀正しく頼 み ます。あなたではなくオブジェクトが、メッセージに反

応するための適切なメソッドを選ぶのです。



10 Pharo 早 巡 り

1.4 Pharoのセッションを保存、終了そして再開する

弾 む 原 子 の ウ ィ ン ド ウ を ク リ ッ ク し て 好 き な と こ ろ へ ド ラ ッ グ し ま し ょ

う 。 ウ ィ ン ド ウ は 「 意 の ま ま 」 で す 。 ク リ ッ ク し て 適 当 な 位 置 に 置 き ま し ょ う 。

Figure 1.6: BouncingAtomsMorph。 Figure 1.7: save as … ダイアログ。

World.Save as … を 選 択 し て 「 myPharo」 と 入 力 し 、 OK ボ タ ン を ク リ ッ
ク し ま し ょ う 。 そ し て World.Save and quit を 選 び ま し ょ う 。

オリジナルのイメージファイルとチェンジファイルがあった場所に

「myPharo.image」と「myPharo.changes」というファイルができているはず
です。これらのファイルには Save and quit する直前の Pharo仮想イメージの
動作中の状態が入っています。この二つのファイルはディスク上の好きな場所

に移動してかまいませんが、(オペレーティングシステムによっては)バーチャ
ルマシンとソ ー ス フ ァ イ ル も同じ場所へ移動、コピー (あるいはリンク)する必
要があるかもしれません。

今 作 っ た 「 myPharo.image」 フ ァ イ ル を 使 っ て Pharo を 起 動 し ま し ょ う 。

先ほど Pharoを終了したときとそっくりそのままの状態に戻ったことにお
気づきでしょう。BouncingAtomsMorphも同じところにあり、原子も弾み続けて
います。

Pharoを起動すると、Pharoバーチャルマシンは指定されたイメージファ
イルを読み込みます。このファイルにはたくさんのオブジェクトのスナップ

ショットが入っています。これらのオブジェクトには、既に書かれた大量の

コードやたくさんのプログラミングツール (これらはすべてオブジェクトです)
が含まれます。Pharoを使っていると、これらのオブジェクトへメッセージを
送ったり、新しいオブジェクトを作ったりすることになります。また、いくつ

かのオブジェクトは命を終え、オブジェクトに割り当てられていたメモリは回

収 (す な わ ち ガベージコレクション)されます。
Pharoを終了するとき、普通はすべてのオブジェクトのスナップショットを

保存することになるでしょう。ここで通常、古いイメージファイルは上書きさ

れることになりますが、先ほど行ったように新しく名前を付けてイメージファ

イルを保存することもできます。

イ メ ー ジ フ ァ イ ル の他にチ ェ ン ジ フ ァ イ ル もあります。このファイルは、標

準のツールを使っていったソースコードの変更をすべて記録しています。通常

このファイルはまったく意識する必要はありません。しかし後で見るように、

チ ェ ン ジ フ ァ イ ル はエラーから回復したり保存しそこなった変更を再現したり

するのに重宝します。これについては後ほど!
みなさんがここまで使ってきた仮想イメージは、1970年代後半に作られた

オリジナルの Smalltalk-80の仮想イメージの子孫です。この仮想イメージの中
には何十年と生き続けているオブジェクトもあります!
プロジェクトを保存し管理するのにも仮想イメージが基本的なメカニズム

として使われていると思うかもしれませんが、それは違います。すぐ後で見る

ように、ソースコードを管理し、ソフトウェアをチームで共有するにはもっと

良いツールがあります。仮想イメージは非常に便利ですが、Monticelloなどの
ツールを使えばバージョン管理やコードの共有がもっとうまくできるので、仮

想イメージに執着せず、無造作に作ったり捨てたりすることに慣れた方がよい

でしょう。

マ ウ ス (と 必 要 な ら ば 修 飾 キ ー ) を 使 っ て BouncingAtomsMorph を メ タ ク
リ ッ ク し ま し ょ う 。

5

色とりどりの円が表示されたはずです。これらをまとめて BouncingAtoms-
Morphのハローと呼びます。一つ一つの円をハ ン ド ル と呼びます。十字のピン
クのハンドルをクリックしましょう; BouncingAtomsMorphは消えるはずです。

1.5 ワークスペースとトランスクリプト

す べ て の ウ ィ ン ド ウ を 閉 じ 、 ト ラ ン ス ク リ プ ト と ワ ー ク ス ペ ー ス を 開 き

ま し ょ う (ト ラ ン ス ク リ プ ト は World.Tools ... サ ブ メ ニ ュ ー か ら 開 け ま す )。

ト ラ ン ス ク リ プ ト と ワ ー ク ス ペ ー ス の 位 置 や サ イ ズ を 変 え て 、 ワ ー ク ス

ペ ー ス を ト ラ ン ス ク リ プ ト に オ ー バ ー ラ ッ プ さ せ ま し ょ う 。

ウィンドウをリサイズするにはウィンドウの角をドラッグするか、メタク

リックしてハローを出して黄色の (右下の)ハンドルをドラッグします。
常にアクティブなウィンドウは一つだけです。そのウィンドウは前面にあ

り枠がハイライトされています。

トランスクリプトは、システムメッセージのログを取るのにしばしば使われ

るオブジェクトです。トランスクリプトは「システムコンソール」の一種です。

5うまくいかない場合はプリファレンス・ブラウザの halosEnabledオプションをチェックしてみ
てください。

ワークスペースは、試してみたい Smalltalkコードの断片を入力するのに
役立ちます。ワークスペースにはまた、任意のテキスト (TODOリストや仮想
イメージを使う人への手引きなど)を書き残しておくこともできます。ワーク
スペースはしばしば、保存した仮想イメージについてドキュメントを記すのに

使われます。先ほどダウンロードした標準仮想イメージがその例です (図 1.2)。

ワ ー ク ス ペ ー ス に 次 の テ キ ス ト を 入 力 し ま し ょ う :

Transcript show: ’hello world’; cr.

ワークスペースで、今入力したテキストの様々な箇所をダブルクリックし

てみましょう。単語の上、文字列の終わりあるいは式全体の終わりという具合

にクリックしてみて、単語全体、文字列全体あるいはテキスト全体が選択さ

れるのを確認しましょう。

入 力 し た テ キ ス ト 全 体 を 選 択 し て ア ク シ ョ ン ク リ ッ ク し 、 do it (d) を 選 択
し ま し ょ う 。

トランスクリプトウィンドウに「hello world」と表示されます。(図 1.8)。
もう一度やってみましょう。(メニュー項目 do it (d) の中の (d) は do itへのキー
ボードショートカットが cmd–dであることを示しています。これについて詳
しくは次の節で!)

Figure 1.8: オーバーラップしたウィンドウ。ワークスペースがアクティブ。

1.6 キーボードショートカット

式を評価したいとき、いつもアクションクリックする必要はありません。代わ

りにキーボードショートカットを使うこともできます。メニューの括弧書きの

部分が該当します。プラットフォームに応じて、いずれかの修飾キー (control、
alt、command、あるいは meta)を押すことになるでしょう (このようなキー
ボードショートカットを以後、cmd– キ ー と表記することにします)。

も う 一 度 ワ ー ク ス ペ ー ス の 式 を 評 価 し て み ま し ょ う 。 た だ し キ ー ボ ー ド

シ ョ ー ト カ ッ ト を 使 っ て く だ さ い (cmd–d)。

do it の他に print it、inspect it そして explore it にも気づいたでしょう。手短
に説明します。

3 + 4 と ワ ー ク ス ペ ー ス に 入 力 し ま し ょ う 。 そ し て キ ー ボ ー ド シ ョ ー ト カ ッ
ト を 使 っ て do it し ま す 。

何も起きなかったとしても驚かないでください! ここでは 3という数に +
というメッセージを 4という引数付きで送ったことになります。普通に 7が計
算されて返されたのですが、ワークスペースはこの答えをどうすべきかわから

ないので答えを捨ててしまいます。結果を見たければ、代わりに print it を使う
必要があります。 print it は式をコンパイル、実行して、その結果に printString
を送ってできた文字列を表示します。

3+4 を 選 択 し て print it し ま し ょ う (cmd–p)。

今度は期待通りの結果になります (図 1.9)。

Figure 1.9: ”do it”ではなく”print it”。

3 + 4 −→ 7

この本の約束事として、特定の Pharoの式を print it したらどうなるかを示す
のに −→ の表記を用います。

選 択 さ れ て い る 「 7」 を 削 除 し (Pharo は 既 に 「 7」 を 選 択 し て い る は ず で す 。
だ か ら delete キ ー を 押 す だ け で す )、 3+4 を も う 一 度 選 択 し て 今 度 は inspect it
(cmd–i) し ま し ょ う 。

タイトルに SmallInteger: 7と書かれたウィンドウが表示されるはずです。これ
をイ ン ス ペ ク タ と呼びます (図 1.10)。インスペクタは極めて便利なツールで、
これを用いればシステムのどんなオブジェクトもブラウズできますし、どんな

オブジェクトともやりとりすることができます。タイトルの意味は、7はクラ
ス SmallIntegerのインスタンスであるということです。左のペインを使ってオブ
ジェクトのインスタンス変数をブラウズすることができます。インスタンス変

数の値は右のペインに表示されます。下のペインにはオブジェクトへメッセー

ジを送るための式を書くことができます。

Figure 1.10: オブジェクトをインスペクトする。

7 の イ ン ス ペ ク タ の 下 の ペ イ ン に self squared と 入 力 し て print it し ま し ょ う 。

イ ン ス ペ ク タ を 閉 じ ま し ょ う 。 ワ ー ク ス ペ ー ス に Object と 入 力 し て 今 度
は explore it (cmd–I 、 大 文 字 の i) し ま す 。

今度は Objectのタイトルが付いたウィンドウ (オブジェクト・エクスプロー
ラ)が現れるはずです。このウィンドウの中には . root: Objectというテキスト
があります。三角をクリックして中身を開いてください (図 1.11)。
オブジェクト・エクスプローラはインスペクタと似ていますが、複雑なオ

ブジェクトのツリービューを提供する点で異なります。この例では、見ている

オブジェクトは Objectクラスです。ここではこのクラスに格納されている要素
を直に見ることができますし、さらに要素の内部構造をたどっていくことも容

易にできます。

Figure 1.11: Objectをエクスプロアする。

1.7 クラスブラウザ

クラスブラウザ6はプログラミングにとって重要なツールです。後で見るよう
に Pharoにはいくつかの興味深いブラウザがありますが、どの仮想イメージを
使う場合でも、クラスブラウザは最も基本的なブラウザです。

World .Class browser を 選 択 し て ブ ラ ウ ザ を 開 き ま し ょ う 。 7

図 1.12がブラウザです。タイトルバーは Objectクラスをブラウズしている
ことを示しています。

ブラウザを開くと、一番左のペイン以外、すべて空欄となっています。こ

のペインで全 パ ッ ケ ー ジ を一覧できます。各パッケージには関連するクラスが

グループ化されて入っています。

Kernel パ ッ ケ ー ジ を ク リ ッ ク し ま し ょ う 。

選択されたパッケージに含まれるクラスが 2 番目のペインにリストされ
ます。

6紛らわしいことに「システムブラウザ」や「コードブラウザ」などと呼ばれることがあります。
PharoではOmniBrowserというブラウザの実装が使われています。OmniBrowserは「OB」また
は「パッケージブラウザ」としても知られています。この本では単に「ブラウザ」としますが、曖

昧さを避けるために「クラスブラウザ」も用います。
7もしブラウザが図 1.12のようでなければ、デフォルトのブラウザを変える必要があるでしょ

う。FAQ 5, p. 326を見てください。

code of method 
"printString"

Packages

Classes

Protocols
Methods

name of method method 
comment 

Figure 1.12: Objectクラスの printStringメソッドを表示しているブラウザ。

Object ク ラ ス を 選 択 し ま し ょ う 。

今度は残りの二つのペインにテキストが表示されます。3 番目のペインに
は選択されたクラスのプ ロ ト コ ル が表示されます。プロトコルは、関連するメ

ソッドを扱いやすいようにグループ分けしたものです。プロトコルを何も選択

していなければ、すべてのメソッドが 4番目のペインに表示されます。

printing プ ロ ト コ ル を 選 択 し ま し ょ う 。

printingプロトコルを見つけるには下にスクロールしなければならないか
もしれません。今度は文字列表示に属するメソッドだけが 4番目のペインに表
示されます。

printString メ ソ ッ ド を 選 択 し ま し ょ う 。

今度は下のペインに printStringメソッドのソースコードが表示されます。こ
のメソッドはすべてのオブジェクトで共有されます (メソッドをオーバーライ
ドするものを除く)。

1.8 クラスを見つける

Pharoでクラスを見つける方法は複数あります。一つ目は、今見たようにブラ
ウザを使ってパッケージからたどっていく方法です。この場合パッケージ名は

最初からわかっているか推測することになります。

二つ目はクラスに browseメッセージを送って自身のブラウザを開いてもら
う方法です。例えば Booleanクラスをブラウズしたいとしましょう。

ワ ー ク ス ペ ー ス に Boolean browse と 入 力 し 、 do it し ま し ょ う 。

Booleanクラスのブラウザが開きます (図 1.13)。キーボードショートカッ
ト cmd–b (browse)もあります。このキーボードショートカットはクラス名が
現れる場所ならどのツールの中でも使えます。クラス名を選択して cmd–bを
タイプすればよいのです。

キ ー ボ ー ド シ ョ ー ト カ ッ ト を 使 っ て Boolean ク ラ ス を ブ ラ ウ ズ し ま し ょ う 。

Figure 1.13: Booleanクラスの定義を表示しているブラウザ。

Booleanクラスが選択されていて、プロトコルもメソッドも選択されていな
いときは、メソッドのソースコードの代わりにク ラ ス 定 義 が表示されることに

注意してください (図 1.13)。クラス定義は、親クラスにサブクラスの作成を依
頼するという、Smalltalkのメッセージ送信にすぎません。ここでは Objectクラ
スが Booleanという名前のサブクラスを作成するよう依頼されているのがわか
ります。Booleanクラスのインスタンス変数、クラス変数、プール辞書は空で、
カテゴリは Kernel-Objectsになっています。クラスペインの下の ? をクリック
すると、専用のペインでクラスコメントを見ることができます (図 1.14)。
クラスを見つけるには、大体は名前で検索するのが一番の早道です。例え

ば日付と時間を表すクラスを探しているとしましょう。

ブ ラ ウ ザ の パ ッ ケ ー ジ ペ イ ン に マ ウ ス を 置 き 、 cmd–f と タ イ プ す る か ア
ク シ ョ ン ク リ ッ ク し て find class …(f) を 選 択 し ま し ょ う 。 ダ イ ア ロ グ ボ ッ ク ス
に 「 time」 と 打 っ て く だ さ い 。

Figure 1.14: Booleanクラスのクラスコメント。

「time」を含んだクラス名のリストが表示されます (図 1.15)。Timeを選択しま
しょう。ブラウザが Timeクラスを表示します。Timeのクラスコメントには関
連するクラスが書いてあります。これらをブラウズしたければ、その名前を選

んで cmd–b をタイプします (実際はどのテキストペインでもこの方法は使え
ます)。

Figure 1.15: 名前でクラスを検索する。

検索ダイアログで完全なクラス名 (最初は大文字)を入力した場合、ブラウ
ザは候補リストを表示することなくそのクラスを直接開きます。

1.9 メソッドを見つける

メソッド名やその一部が、クラス名よりも簡単に推測できることもあります。

現在の時刻を知りたい場合、「now」というメソッドや「now」を含んだメソッ
ドがあるのではと思うでしょう。しかしどこにあるのでしょうか? そうしたと
きはメ ソ ッ ド ・ フ ァ イ ン ダ が助けてくれます。

World .Tools ... .Method finder を 選 択 し ま し ょ う 。 左 上 の ペ イ ン に 「 now」
と 入 力 し て accept し ま し ょ う (ま た は return キ ー を 押 し ま す )。

メソッド・ファインダは「now」を含むすべてのメソッド名のリストを表
示します。nowまでスクロールするには、リストにカーソルを移動し「n」をタ
イプします (ちなみにこの方法はすべてのスクロールするリストで使えます)。
「now」を選択しましょう。すると右側のペインにこの名前のメソッドを定義
しているクラスの一覧が表示されます (図 1.16)。クラス名を選択するとそのク
ラスのブラウザが開きます。

Figure 1.16: メソッド・ファインダ。nowの定義を含んだすべてのクラスをリ
ストしている。

時にはメソッドが存在することはわかっていても、何という名前なのか見

当がつかないこともあるでしょう。メソッド・ファインダはこのようなときに

も役立ちます! 例えば文字列を大文字化するメソッドを見つけたかったとしま
す。’eureka’を ’EUREKA’という具合にです。

’eureka’ . ’EUREKA’ と メ ソ ッ ド ・ フ ァ イ ン ダ に 入 力 し て return キ ー を 押 し
ま し ょ う (図 1.17)。

メソッド・ファインダはお望みのメソッドを勧めてくれます。8

8もしウィンドウが開いて推奨されないメソッドと警告されても、あわてないでください。メ

メソッド・ファインダの右側ペインの行頭のアスタリスクは、望んだ結果を得

るのに、そのメソッドが実際に使われたことを示します。今回 String asUppercase
の先頭にアスタリスクが付いているので、クラス Stringで定義された asUppercase
が ’eureka’ . ’EUREKA’の変換に使われたとわかります。アスタリスクが付いてい
ないメソッドは、単に名前が同じということで挙げられたものです。Character
»asUppercaseは実行されていません。’eureka’は Characterオブジェクトではな
いからです。

Figure 1.17: 例を使ってメソッドを見つける。

引数のあるメソッドについてもメソッド・ファインダを使うことができま

す。例えば二つの整数の最大公約数を求めるメソッドを探すときは、25. 35. 5
と入力します。また複数の例を使って検索の範囲を絞りこむこともできます。

下のペインのヘルプテキストに詳しい方法が書いてあります。

1.10 メソッドを新しく定義する

テスト駆動開発9 (TDD)が登場してから、コードを書く方法は一変しました。
TDDの背景にある思想は、振る舞いを期待するテストコードを、コード本体
より先に書くというものです。TDDではテストを書いてから、初めてそのテ
ストを満足させるコードを書きます。

ソッド・ファインダは候補すべてを表示しようとするので、推奨されないメソッドが含まれること

もあります。落ち着いて Proceed をクリックしてください。
9Kent Beck, Test Driven Development: By Example. Addison-Wesley, 2003, ISBN 0–321–14653–0.

「何かを大声で強調して言う」メソッドを書く、という課題があったとしま

しょう。これは正確にはどういう意味でしょう? そうしたメソッドの名前とし
てふさわしいのは何でしょう? 将来そのメソッドを保守するプログラマに、こ
のメソッドが何をするのかを確実に伝えるにはどうすればいいでしょう? 以下
の例で、こうした疑問に答えていきます。

文字列「Don’t panic」にメッセージ shoutを送ったときの結果は、
「DON’T PANIC!」となるべきである。

この例を、システムが理解できるようにテストメソッドに変換します。

Method 1.1: shout メ ソ ッ ド の テ ス ト
testShout
self assert: (’Don’’t panic’ shout = ’DON’’T PANIC!’)

Pharoで新しくメソッドを作るにはどうすればよいでしょうか? まずメソッ
ドがどのクラスに属するかを決めなければいけません。今からテストしようと

している shoutメソッドは Stringクラスのものとするので、テストは習慣とし
て StringTestクラス内に作成します。

Figure 1.18: StringTestの新規メソッドのテンプレート。

StringTest の ブ ラ ウ ザ を 開 き ま し ょ う 。 testShout の プ ロ ト コ ル と し て
tests - converting を 選 択 し ま し ょ う (図 1.18)。 下 の ペ イ ン で 選 択 さ れ て い る の
は メ ソ ッ ド の テ ン プ レ ー ト で す 。 こ れ を 見 れ ば Smalltalk の メ ソ ッ ド の 大 ま か
な 姿 が わ か り ま す 。 こ れ を 消 し て Method 1.1 の コ ー ド を 入 力 し ま し ょ う 。

ブラウザにテキストを入力すると下のペインが赤く縁取りされることに注

意してください。これにより、このペインの変更がまだ保存されていないこと

がわかります。それでは下のペインでアクションクリックして accept (s) を選
択するか、もしくは cmd–sをタイプし、メソッドをコンパイル・保存してく
ださい。

仮想イメージでコードをアクセプトするのが初めてだとしたら、名前を入

力するよう促されるはずです。仮想イメージはたくさんの人が書いたコードか

らできているので、メソッドを作成・変更した人を記録しておくことは重要で

す。ファーストネームとラストネームを、空白なし、またはドットで区切って

入力します。

shoutというメソッドはまだないので、ブラウザはこれが本当に意図した名
前であるかどうか確認します。そして他の可能性のある名前についても提案し

てきます (図 1.20)。単にミスタイプしたときにはこの機能は非常に助かります
が、今回は今から作ろうとするメソッドは shoutで間 違 い な い ので、確認用メ
ニューの一番上の項目を選択します (図 1.20)。

Figure 1.19: 名前を入力する。

今 作 っ た テ ス ト を 実 行 し ま し ょ う 。 World か ら SUnit TestRunner を 開 い
て く だ さ い 。

一番左に二つ並んだペインは、ブラウザの上部のペインに少し似ています。

左側のペインにはクラスカテゴリのリストが表示されますが、テストクラスを

含んだものに限られます。

Figure 1.20: StringTestの testShoutメソッドをアクセプトする。

CollectionsTests-Text を 選 択 し ま し ょ う 。 右 側 の ペ イ ン に は そ の カ テ ゴ リ の
す べ て の テ ス ト ク ラ ス が 表 示 さ れ ま す 。 そ の 中 に StringTest も あ り ま す 。 テ ス
ト ク ラ ス の 名 前 は も う 選 択 さ れ て い る の で 、 Run Selected を ク リ ッ ク す る と テ
ス ト が 実 行 さ れ ま す 。

テストを実行した結果が 図 1.21 のように表示され、テストの実行時にエ
ラーがあったことがわかります。エラーを起こしたテストのリストは右下の

ペインに表示されます。 見てわかるように、StringTest»#testShout が原因です
(Smalltalkでは StringTestクラスの testShoutメソッドを StringTest»#testShoutと
表記します)。StringTest»#testShoutをクリックすると再度そのテストだけが実
行され、「MessageNotUnderstood: ByteString»shout」というウィンドウが開きます。

エラーメッセージとともに開いたこのウィンドウは、Smalltalkのデバッガ
です (図 1.22)。デバッガとその使い方については第 6章で解説します。
このエラーは、もちろん期待した通りのものです。文字列が shoutするため

のメソッドをまだ書いていないので、テストを実行するとエラーが発生するの

は当然です。それでもなお、テストが失敗することを確認するのは良い習慣で

す。それによって、テストの仕組みが正しく設定され新しいテストが実行され

たことを確認できるからです。いったんエラーを確認したら、テストの実行を

やめて ( Abandon )デバッガウィンドウを閉じることができます。ちなみにたい
ていの Smalltalkでは、まだないメソッドを Create を使ってその場で書いて
しまうこともできます。デバッガの中で新しく作成されたメソッドを編集し、

テストを継続実行 ( Proceed )できます。

ではテストが成功するようにメソッドを作成しましょう!

Figure 1.21: StringTestを実行する。

Figure 1.22: (プリ)デバッガ

ブ ラ ウ ザ で String ク ラ ス を 選 択 し 、 converting プ ロ ト コ ル を 選 択 し ま す 。
Method 1.2 の テ キ ス ト で メ ソ ッ ド 作 成 テ ン プ レ ー ト を 上 書 き し accept し ま し ょ
う (注 : ↑ の と こ ろ に は ̂ を 入 力 し ま す )。

Method 1.2: shout メ ソ ッ ド

shout
↑ self asUppercase, ’!’

カンマは、文字列連結を意味します。つまりこのメソッドでは、shoutを受け
取った Stringオブジェクトを大文字にし、末尾に感嘆符を付け足します。Pharo
では ↑以下の式がメソッドの戻り値になります。ここでは戻り値は、大文字化
になり感嘆符が付いた新 し い 文字列です。

このメソッドは思い通り動くでしょうか? もう一度テストを実行してどう
なるか確認してみましょう。

テ ス ト ・ ラ ン ナ ー で Run Selected を 再 び ク リ ッ ク し ま し ょ う 。 今 度 は 緑 の
バ ー が 出 て 、 テ ス ト が す べ て 失 敗 も エ ラ ー も な く 実 行 で き た こ と を 示 す テ キ ス

ト が 表 示 さ れ ま す 。

緑のバーが出たら、成果を保存して一休みするとよいでしょう!

Figure 1.23: Stringクラスで定義された shoutメソッド。

1.11 まとめ

この章では Pharoの環境を紹介し、ブラウザ、メソッド・ファインダ、テスト・
ランナーと言った主要なツールの使い方を学びました。Pharoの文法について
も、全部ではありませんが少しだけ学びました。

• 実行中の Pharoシステムは バ ー チ ャ ル マ シ ン 、ソ ー ス フ ァ イ ル 、イ メ ー
ジ フ ァ イ ル および チ ェ ン ジ フ ァ イ ル からなります。この中で最後の二つ

だけが更新されます。これらは実行中のシステムのスナップショットを

記録します。

• Pharo仮想イメージをロードすると、実行中のオブジェクトも含め、保
存したときとまったく同じ状態が再現されます。

• Pharoは 3ボタンマウスでクリック、アクションクリック、メタクリッ
クするように設計されています。3ボタンマウスがなくても、キーボー
ドの修飾キーを使えば同じことができます。

• Pharoの背景部分をクリックするとワ ー ル ド メ ニ ュ ー が表示されます。
ワールドメニューからは様々なツールが起動できます。

• ワ ー ク ス ペ ー ス は、コードの断片を書いて評価するためのツールです。
ワークスペースに、任意のテキストを書いておくこともできます。

• コードを評価するのに、ワークスペースや他のツールのテキストペイン
上でキーボードショートカットを使うことができます。最も重要なのは

do it (cmd–d)、print it (cmd–p)、 inspect it (cmd–i)、explore it (cmd–I)そ
して browse it (cmd–b)です。

• ブ ラ ウ ザ は、Pharoのコードをブラウズしたり新たにコードを書いたり
するための最も重要なツールです。

• テ ス ト ・ ラ ン ナ ー は、ユニットテストを実行するためのツールです。テ
スト・ランナーはテスト駆動開発もサポートします。



Chapter 2

最初のアプリケーション

この章では、簡単なゲーム：Lights Out1を作成します。ゲームの作成を通じ
て、Pharoプログラマがプログラミングやデバッグに使用するツール群や、他
の開発者とプログラムをやりとりする方法について体験していきます。この

章で扱う開発ツールは、ブラウザ、オブジェクト・インスペクタ、デバッガ、

Monticello パッケージ ブラウザ です。Smalltalk による開発はとても効率的
です。Smalltalkはとても簡潔なプログラミング言語です。また、開発ツール
は言語と非常によく統合されています。そのため、開発の手順に手間取ること

なく、コードを書くことに多くの時間を使うことができるでしょう。

2.1 Lights Outゲーム

Figure 2.1: Lights Outのゲーム盤。ユーザは盤上のセルをマウスでクリック
するだけです。

Pharoの開発ツールの使い方を体験するために、これから Lights Outとい
う簡単なゲームを作っていきます。ゲーム盤は図 2.1のように、淡黄色をした
セ ル の四角形の配列で構成されます。セルをマウスでクリックすると、周囲の

1http://en.wikipedia.org/wiki/Lights_Out_(game)

http://en.wikipedia.org/wiki/Lights_Out_(game)

四つのセルが青色に変わります。もう一度クリックすると、それらはまた淡黄

色に戻ります。このゲームの目的は、できるだけ多くのセルを青色に変えるこ

とです。図 2.1で示すように、ゲームはゲーム盤と、100個のセルからなる 2
種類のオブジェクトで構成されています。Pharoのコードとしては、クラス二
つでゲームを実装しています。一つはゲームを表すクラス、もう一つはセルを

表すクラスです。それでは、Pharoの開発ツールを使い、どのようにこれらの
クラスを定義していくか、見ていくことにしましょう。

2.2 パッケージを作成する

第 1章のブラウザの紹介で、クラスやメソッドのブラウザの仕方や、新たなメ
ソッドを定義する方法を学びました。ここではパッケージ、カテゴリ、クラス

の作り方を学びます。

ブ ラ ウ ザ を 開 き パ ッ ケ ー ジ ペ イ ン で ア ク シ ョ ン ク リ ッ ク し 、 create package
を 選 択 し ま す 。

2

Figure 2.2: パッケージを追加します。 Figure 2.3: クラステンプレート。

ダイアログボックスに新たなパッケージ名 (PBE-LightsOut とします)を入力
し、 acceptをクリックしてください (または単にリターンキーを押してくださ

2パッケージ・ブラウザが標準ブラウザとしてインストールされているものとします。もし図 2.2
で示したブラウザでない場合は、デフォルトのブラウザを変更する必要があります。FAQ 5, p. 326
参照。

い)。すると新たなパッケージができあがり、パッケージ一覧にアルファベット
順となって表示されます。

2.3 LOCellクラスを定義する

当然ですが、新しいパッケージにはまだ何のクラスもありません。しかし、新

たなクラスを作りやすいように、クラスのテンプレートが、メインの編集ペイ

ンに自動的に表示されます (図 2.3参照)。
このテンプレートは、Objectクラスに NameOfSubClassというサブクラスを

作成するという、Smalltalkのメッセージ式になっています。新しいクラスは
何の変数も持たず、PBE-LightsOut カテゴリに属するようになっています。

カテゴリとパッケージについて

従来の Smalltalkではカ テ ゴ リ はありましたが、パッケージというものはあり
ませんでした。この二つの違いは何でしょうか。カテゴリとは Smalltalk関連
するクラスをイメージファイル内で集約しているだけのものです。パ ッ ケ ー ジ

とは関係するクラスと 拡 張 メ ソ ッ ド を集めたもので、Monticelloのようなツー
ルでバージョン管理できるものです。慣習としてパッケージ名とカテゴリ名に

は同じ名前を使います。通常は二つの違いを気にする必要はありません。しか

し、これらの用語の違いが重要なこともあるため、本書では注意深く使い分け

ています。Monticelloを使い始めるところで、より詳しく見ていくことにしま
しょう。

新しいクラスを作成する

テンプレートを修正し、目的に合ったクラスを作ってみましょう。

ク ラ ス 作 成 の テ ン プ レ ー ト を 修 正 す る 手 順 は 以 下 の 通 り で す :

• Objectを SimpleSwitchMorphに書き換えます。

• NameOfSubClassを LOCellに書き換えます。

• インスタンス変数のリストに mouseActionを追加します。

結果は Class 2.1の通りです。

Class 2.1: LOCellク ラ ス を 定 義 す る
SimpleSwitchMorph subclass: #LOCell
instanceVariableNames: ’mouseAction’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE -LightsOut’

この新しいクラス定義は、SimpleSwitchMorphという既存のクラスに対し、
LOCellというサブクラスを作るようにメッセージを送る Smalltalkの式になっ
ています (実際には LOCellクラスはまだ存在していないため、クラス名を表す
シ ン ボ ル #LOCellを引数で渡しています)。また、このメッセージには新たなク
ラスのインスタンスが mouseActionインスタンス変数を持つという指定も含ま
れています。mouseActionは、セル上でマウスがクリックされたときの振る舞い
を定義するための、インスタンス変数となります。

こ の 時 点 で は ま だ 何 も 作 ら れ て は い ま せ ん 。 クラステンプレートペインの

境界線が赤に変わっていることに注目してください (図 2.4)。これはまだ変 更
が 保 存 さ れ て い な い こ と を意味しています。実際にこのメッセージを送るには、

変更をacceptする必要があります。

Figure 2.4: クラス作成テンプレート。

新 し い ク ラ ス 定 義 を 「 了 解 (accept)」 し ま す

action-clickしてacceptを選択、もしくはショートカットの cmd–s (save)を

実行してください。SimpleSwitchMorphにメッセージが送られ、新しいクラスが
コンパイルされます。クラス定義を acceptすると、クラスができてブラウザ
のクラスペインに表示されます (図 2.5)。編集ペインにはクラス定義が表示さ
れ、その下にある小さなペインでクラスの目的を説明するように促されます3。
ここへ記述する内容はク ラ ス コ メ ン ト と呼ばれ、他のプログラマにクラスの

大まかな目的を伝える重要な手段となります。Smalltalkプログラマはコード
そのものの読みやすさを重視するため、メソッド中に詳細なコメントを書くこ

とはまれです。ここでの哲学は、コードの意図はコード自身に語らせよという

ことです (もしそうなっていない場合は、そうなるまでリファクタリングしま
しょう！ )。
コメントには、クラスの詳細すぎる説明ではなく、後でプログラマがこの

クラスを見ていくべきかを判断できる、大まかな目的を簡潔に記述するように

しましょう。

LOCell ク ラ ス の コ メ ン ト を 入 力 し 、 accept し て く だ さ い 。 後 に 内 容 を 変
更 す る こ と も で き ま す 。

Figure 2.5: 新しくできたLOCellクラス

2.4 クラスにメソッドを追加する

それではこのクラスにメソッドをいくつか追加してみましょう。

3訳者が確認した環境 (Pharo by Example Imageおよび 1.3環境)ではそのようなペインはあり
ません。代わりに、クラスブラウザ中にある「?」ボタンを押すことでクラスの目的を書くペイン
が表示されます。また、クラスコメントが書かれていない場合には、その「?」の箇所が赤くなり
クラスの目的を書くことを促す挙動となっています。

プ ロ ト コ ル ペ イ ン に あ る –all– プ ロ ト コ ル を 選 択 し ま す 。

編集ペインにメソッド作成用のテンプレートが表示されます。それを選択

し、Method 2.2で示した内容に書き換えてください。

Method 2.2: LOCellの イ ン ス タ ン ス を 初 期 化 す る
1 initialize
2 super initialize.
3 self label: ’’.
4 self borderWidth: 2.
5 bounds := 0@0 corner: 16@16.
6 offColor := Color paleYellow.
7 onColor := Color paleBlue darker.
8 self useSquareCorners.
9 self turnOff

3行目’’の文字は、間に何の文字も挟まない引用符です。二重引用符ではないこ
とに注意してください。’’は空の文字列を表します。

こ の メ ソ ッ ド の 定 義 を accept し ま す 。

上記のコードは何をしているのでしょうか？ ここでは詳細については触れ

ずに (本の残りの部分はそのためにあるのです！ )、簡単に内容を見ていきま
す。それでは 1行ずつ進んでいきましょう。

initializeというメソッドに注目してください。この名前はとても重要です。
initializeというメソッドは、慣習的にオブジェクトが生成された直後に呼び出
されます。つまり、LOCell newが評価されると initializeメッセージが新しく作ら
れたオブジェクトに自動的に送られます。initializeメソッドはオブジェクトの状
態、主にそのインスタンス変数を設定するために使われます。

このメソッドでは、スーパークラスであるSimpleSwitchMorphの initializeメソッ
ドをまず実行しています (2行目)。スーパークラスの initializeメソッドの実行に
より、継承した状態が適切に初期化されるということを期待しています。何か

の処理をする前には、スーパークラスの initializeメソッドを呼び出して、継承し
た状態を初期化しておくとよいでしょう。SimpleSwitchMorphの initializeメソッド
が実際に何をするかは知らなくともかまいません。不明瞭な状態のまま処理を

始める危険を冒すよりは、スーパークラスの initializeメソッドを呼び出すこと
で、インスタンス変数に妥当な初期値が設定されるようにした方がよいのです。

メソッドの残りの部分でこのオブジェクトの状態をさらに設定しています。

例えば、self label: ’’をオブジェクトに送ることで、オブジェクトのラベルに空文
字列を設定しています。

0@0 corner: 16@16については、おそらく説明が必要でしょう。0@0は、x,y
座標の両方が 0 に設定された Point オブジェクトを表しています。実際には、
0@0は、数値 0に@メッセージを引数 0で送るという、メッセージ送信になって
います。その結果、数値 0は、Pointクラスに座標 (0,0)のインスタンスを作るよ
う指示します。新たに作られた Pointオブジェクトに corner: 16@16メッセージ



オ ブ ジ ェ ク ト を イ ン ス ペ ク ト す る 33

が送られ、0@0と 16@16の角を持つ Rectangleオブジェクトが作られます。この
Rectangleオブジェクトは、スーパークラスから継承された bounds変数に代入さ
れます。Pharoの画面の座標系の原点は左 上 で、下 に 向 か っ て y座標の値が増
えることに注意してください。

メソッドの残りの部分について説明の必要はないでしょう。Smalltalkをう
まく書く秘訣は、コードが英語のように読めるように、良いメソッド名を付け

ることです。オブジェクトが自分自身に語りかけ、“自分は直角を使う!”, “自分
はスイッチを切る!”と言っているイメージを持つことができるはずです。

2.5 オブジェクトをインスペクトする

LOCellオブジェクトを作ってインスペクトすることで、先ほど書いたコードの
効果を確認できます。

ワ ー ク ス ペ ー ス を 開 い て 、 LOCell newと 打 ち 、 inspect it し て く だ さ い 。

Figure 2.6: LOCellオブジェクトを調べるために使用したインスペクタ。

inspectorの左側のペインには、インスペクトしたオブジェクトのインスタ
ンス変数のリストが表示されます。どれか一つ (例えばbounds)を選んでみてく
ださい。右側のペインにそのインスタンス変数の値が表示されます。

インスペクタの下側にあるペインは小さなワークスペースです。このワー

クスペースでは、選択中のオブジェクトが擬似変数 selfで参照できるため、と
ても便利です。

イ ン ス ペ ク タ ウ ィ ン ド ウ の ル ー ト に あ る LOCell を 選 択 し て く だ さ い 。 下
部 の ペ イ ン に 、 self bounds: (200@200 corner: 250@250)と 打 ち do it し ま し ょ う 。
イ ン ス ペ ク タ 上 で boundsの 値 が 変 化 し ま す 。 次 に ワ ー ク ス ペ ー ス 部 分 に self
openInWorldと 打 ち 込 み do it し て く だ さ い 。

画面の左上の端、つまり boundsが示している位置にセルが表示されます。セ
ル上でメタクリックをすると haloが現れます。右上の左隣にある茶色いハン
ドルでセルを移動したり、右下にある黄色のハンドルでセルをリサイズしたり

してみましょう。boundsの値の変化を、インスペクタから確認できます。(新
たな boundsの値を見るために refresh アクションクリックが必要かもしれま
せん。)

Figure 2.7: セルのリサイズ。

ピ ン ク 色 の ハ ン ド ル xを ク リ ッ ク し て セ ル を 消 し て く だ さ い 。

2.6 LOGameクラスを定義する

ゲームに必要となる、もう一つのクラスを作っていきます。クラス名は LOGame
とします。

ブ ラ ウ ザ の メ イ ン ウ ィ ン ド ウ に 、 ク ラ ス 定 義 テ ン プ レ ー ト を 表 示 さ せ ま す 。

パッケージ名をクリックすることで、クラス定義テンプレートを表示させ

ることができます。以下のようにコードを書いたら、acceptしてください。

Class 2.3: LOGame ク ラ ス の 定 義
BorderedMorph subclass: #LOGame
instanceVariableNames: ’’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE-LightsOut’

LOGameは BorderedMorphのサブクラスとしました。Morphは Pharoのグラ
フィック図形すべてのスーパークラスで、BorderedMorphは境界線を扱えるMorph
になります。2行目にある引用符の間にはインスタンス変数の名前を入れるこ
とができますが、ここでは空のままにしておきましょう。

LOGameの initializeメソッドを定義しましょう。

以 下 の 内 容 を LOGameの メ ソ ッ ド と し て Browser に 書 き 込 み 、 accept し て
く だ さ い 。

Method 2.4: ゲ ー ム の 初 期 化
1 initialize
2 | sampleCell width height n |
3 super initialize.
4 n := self cellsPerSide.
5 sampleCell := LOCell new.
6 width := sampleCell width.
7 height := sampleCell height.
8 self bounds: (5@5 extent: ((width*n) @(height*n)) + (2 * self borderWidth)).
9 cells := Matrix new: n tabulate: [ :i :j | self newCellAt: i at: j ].

Pharoはいくつかの用語が認識できないと警告をしてくるはずです。まず
Pharoは、cellsPerSideメッセージが認識できないため、スペルミスとして修正
案をいくつか提示してきます。

Figure 2.8: Pharo が未知のセレクタ
を検出。

Figure 2.9: 新しいインスタンス変数
を宣言。

しかし、cellsPerSideはスペルミスではありません。単にまだメソッドを定
義していないだけです。メソッドはこのすぐ後で定義することにしましょう。

メ ニ ュ ー か ら 最 初 の ア イ テ ム を 選 択 し 、 cellsPerSideが ス ペ ル ミ ス で は な い
こ と を 確 認 し ま す 。

次に Pharoは cellsが認識できないと警告してきます。Pharoはこの問題を
解消するための案をいくつか提示してきます。

cellを イ ン ス タ ン ス 変 数 と し て 扱 い た い の で 、 declare instance を 選 択 し ま す 。

最後に Pharoは、最終行で使われている newCellAt:at:メッセージについて警
告してきます。もちろんこれもスペルミスではないので、先ほどと同様に流し

てください。

さて、もう一度 LOGameのクラス定義を見てみましょう ( instanceボタンを
クリックします)。クラス定義が更新され、インスタンス変数 cellsが含まれるよ
うになったことを、ブラウザで確認できるはずです。

それでは、この initializeメソッドの中身を見ていきましょう。| sampleCell
width height n |と書かれた行では、四つの一時変数を宣言しています。ここで宣
言された変数は、スコープがこのメソッドに限定されるため、一時変数と呼ば

れます。説明的な名前を持った一時変数を使うことで、コードをより読みやす

くできます。Smalltalkでは定数と変数を区別するための特別な構文はありま
せん。また、ここで宣言された四つの”変数”は、実際には定数と変わりあり
ません。4-7行目で一度だけ値が設定されています。
ゲーム盤には必要な数だけのセルと、それらの境界を表示するための十分

な大きさが必要ですが、セルの数がどれぐらいになるかは今はまだ判断できま

せん。そこで、cellsPerSideというメソッドを別に定義して、セルの数を決定す
る責務をそちらに委譲することにします。メソッドを定義する前に cellsPerSide
をメッセージ送信するようにしたため、initializeメソッドを acceptするときに
Pharoから “confirm, correct, or cancel”と警告されることになってしまいま
したが、ここで避けてはいけません。まだ定義していないメソッドを使いなが

らメソッドを書いていくというのは、実のところとても良い習慣です。なぜな

ら、initializeメソッドを書くまで私たちはそうしたメソッドが必要となるかわか
りませんでしたし、その時点になって初めて、意味のある名前をメソッドに付

けることができ、思考を中断せずに進んでいくことができるからです。

4 行目でこのメソッドが使われています。Smalltalk 式 self cellsPerSideは
cellsPerSideメッセージを self 、つまりオブジェクト自身に送ります。この結
果、ゲーム盤の一辺に必要なセルの数が nに割り当てられました。

次の 3行では、新たな LOCellオブジェクトを作成し、ゲーム盤の幅と高さ
を、適切な一時変数に代入しています。

8行目では自身の boundsを設定しています。詳細は今は気にせず、括弧内
の式で原点 (す な わ ち 左上隅 (5,5))と、右下隅からなる、指定した数のセルを
置くのに十分な長方形が作られると理解してください。

最後の行では、LOGameオブジェクトのインスタンス変数 cellsに、適切な数
の行と列を持つMatrixオブジェクトを生成して代入しています。Matrixオブジェ
クトの生成は、new:tabulate:メッセージをMatrixクラス (このクラスももちろんオ
ブジェクトです。そのためメッセージを送ることができます。)に送ることで
実現しています。new:tabulate:は、二つのコロン (:)を持つことから二つの引数
を取ることがわかります。引数はコロンの後に書きます。もし、引数を括弧内

に一緒に書く言語しか使ったことがなければ、最初はこの書き方に戸惑うかも

しれません。しかし、これはただの構文です。あわてることはありません。そ

のうちに、メソッド名が引数の役割を説明してくれるわかりやすい構文だとい

うことがわかるでしょう。例えば次の式、Matrix rows: 5 columns: 2なら、Matrix
は 2行 5列ではなく 5行 2列であることは明白です。

Matrix new: n tabulate: [ :i :j | self newCellAt: i at: j ]によって n×nの行列とその要
素を初期化しています。各要素の初期値はその座標に依存します。(i,j)thの位置
にある要素は、self newCellAt: i at: j式を評価した結果で初期化されます。

2.7 メソッドをプロトコルにまとめる

メソッドの定義を続ける前に、ブラウザ上部にある 3番目のペインを見てみま
しょう。1番目のペインでクラスをパッケージに分類したように、3番目のペイ
ンではメソッドの分類ができます。これにより 4番目のペインが大量のメソッ
ドで埋もれてしまうのを防ぐことができます。こうしたメソッドの分類は「プ

ロトコル」と呼ばれます。

クラスにメソッドが少ししかないときは、プロトコルによる階層は必要あ

りません。そうしたクラスのために、ブラウザにはすべてのメソッドが含まれ

る仮想プロトコル–all–が用意されています。

Figure 2.10: 分類されていないメソッドを自動的に分類。



38 最 初 の ア プ リ ケ ー シ ョ ン

本書の例に従って作業をしているなら、三つ目のペインには as yet unclassified
(まだ分類されていない)というプロトコルが含まれているかもしれません。

プ ロ ト コ ル ペ イ ン を ア ク シ ョ ン ク リ ッ ク し て く だ さ い 。

various . categorize automatically を 選 択 す る と 、 initialization プ ロ ト コ ル が 新
し く で き 、 initializeメ ソ ッ ド が 移 り ま す 。

Pharoはどうやってプロトコルを定めるのでしょうか。一般的にはPharo側
で判断することはできません。しかしこの例では、スーパークラスにも initialize
メソッドが定義されているため、Pharoは initializeメソッドを、オーバーライド
元のメソッドと同じプロトコルに分類します。

表記規則 Smalltalkerは、メソッドが属するクラスを識別するために “>>”と
いう表記をよく使用します。例えば LOGameクラスの cellsPerSideメソッドは
LOGame>>cellsPerSideと表します。この表記が Smalltalk 式ではな い ことを明
示するため、本書では代わりに特殊なシンボル»を使うことにします。先ほど
のメソッドであれば、文中では LOGame»cellsPerSideと表します。

以後、本書ではこの記法でメソッド名を記述します。もちろん、実際に

Browser 上でコードを打ち込むときはクラス名や»を打つ必要はありません。
クラスペインで適切なクラスが選択されていることを確認するだけで十分です。

それでは、LOGame»initializeメソッドで使われている他の二つのメソッドを
定義していきましょう。両方とも initializationプロトコルに入れることにします。

Method 2.5: 定 数 メ ソ ッ ド
LOGame»cellsPerSide

”ゲ ー ム 盤 の 辺 に 並 ぶ セ ル の 数 ”
↑ 10

このメソッドは非常にシンプルで、単に 10という値を返すだけです。この
ように値をメソッドとして表現しておくと、プログラムが成長してこの値が単

純に定まらなくなった場合に、値を計算して返すようにメソッドを変更して対

処できるという利点があります。

Method 2.6: 初 期 化 補 助 メ ソ ッ ド
LOGame»newCellAt: i at: j

”位 置 (i,j)に セ ル を 作 成 し 、 適 切 な 画 面 の 位 置 に 追 加 。 新 し い セ ル を 返 す ”
| c origin |
c := LOCell new.
origin := self innerBounds origin.
self addMorph: c.
c position: ((i - 1) * c width) @ ((j - 1) * c height) + origin.
c mouseAction: [self toggleNeighboursOfCellAt: i at: j]

LOGame»cellsPerSideメ ソ ッ ド と LOGame»newCellAt:at:メ ソ ッ ド を 追 加 し て
く だ さ い 。

新たに出てきた toggleNeighboursOfCellAt:at:セレクタと mouseAction:セレクタ
のスペルをチェックしましょう。

Method 2.6 では、Matrix (行列) の (i,j)の位置に新しい LOCell
を 生 成 し て い ま す 。最 後 の 行 で は セ ル の mouseActionと し て
block[self toggleNeighboursOfCellAt: i at: j ]. を設定しています。これはマウス
をクリックしたときのコールバックとなります。対応するメソッドを定義しま

しょう。

Method 2.7: コ ー ル バ ッ ク メ ソ ッ ド
LOGame»toggleNeighboursOfCellAt: i at: j
(i > 1) ifTrue: [ (cells at: i - 1 at: j ) toggleState].
(i < self cellsPerSide) ifTrue: [ (cells at: i + 1 at: j) toggleState].
(j > 1) ifTrue: [ (cells at: i at: j - 1) toggleState].
(j < self cellsPerSide) ifTrue: [ (cells at: i at: j + 1) toggleState].

Method 2.7では、セルの (i, j)の東西南北の位置に隣接する四つのセルの状
態を切り替えます。ここで唯一難しいのはゲーム盤が有限ということです。そ

のため、隣接するセルが存在するかどうかを、状態を切り替える前に確認しな

ければなりません。

こ の メ ソ ッ ド を game logic と い う 新 た な プ ロ ト コ ル に 置 き ま す 。 (プ ロ ト
コ ル ペ イ ン 上 で 、 ア ク シ ョ ン ク リ ッ ク し て 新 た な プ ロ ト コ ル を 作 っ て く だ さ い )

メソッドを移動するには、名前の上でクリックし、新たに作成したプロト

コルまでドラッグします (図 2.11)。

Figure 2.11: メソッドをプロトコルへ持って行く。

Lights Outゲームを完成させるには、さらにマウスイベントを操作するた
めの二つのメソッドを、LOCellクラスに定義する必要があります。

Method 2.8: 典 型 的 な セ ッ タ ー メ ソ ッ ド
LOCell»mouseAction: aBlock
↑ mouseAction := aBlock

Method 2.8 ではセルの mouseAction変数に引数の値を設定して、値を返し
ているだけです。このようにインスタンス変数の値を変 更 す る メソッドはセ ッ

タ ー メ ソ ッ ド と呼ばれています。また、インスタンス変数の現在の値を返すメ

ソッドはゲ ッ タ ー メ ソ ッ ド と呼ばれます。

もし他のプログラミング言語でゲッターメソッドとセッターメソッドを使

うことに慣れていたら、メソッド名を setmouseActionや getmouseActionとしたく
なるかもしれません。しかし Smalltalkでは違い流儀です。ゲッターメソッド
の名前は常に取得するインスタンス変数と同じ名前にします。セッターメソッ

ド名も同じですが、mouseActionに対して mouseAction:のように末尾を “:”とし
ます。

セッターメソッドとゲッターメソッドをあわせてア ク セ サ メソッドと呼び

ます。また、これらは慣習として accessingプロトコルに置かれます。Smalltalk
ではすべてのインスタンス変数はプライベートなものとしてオブジェクトが保

持します。そのため、Smalltalkで、他のオブジェクトの変数を読み書きする
には、アクセサメソッドを介するしかありません4。

LOCellク ラ ス に 移 動 し 、 LOCell»mouseAction:を 定 義 し て く だ さ い 。 定 義 が
終 わ っ た ら 、 そ の メ ソ ッ ド を accessing プ ロ ト コ ル に 置 い て く だ さ い 。

最後に mouseUp:メソッドを定義する必要があります。このメソッドは画面
のセル上でマウスボタンを離したときに、GUIフレームワークから自動的に呼
び出されます。

Method 2.9: イ ベ ン ト ハ ン ド ラ
LOCell»mouseUp: anEvent
mouseAction value

LOCell»mouseUp:を 追 加 し 、 categorize automatically を 実 行 し て く だ さ い 。

このメソッドは valueメッセージを mouseActionインスタンス変数にバイン
ドされたオブジェクトに送ります。LOGame»newCellAt: i at: jで以下のコードを
mouseAction:に指定したことを思い出してください。

[self toggleNeighboursOfCellAt: i at: j ]

valueメッセージを送ると、この部分が評価されてセルの状態が切り替わること
になります。

2.8 コードを実行してみましょう

これで Lights Outゲームができあがりました！
手順に沿って進んできたのなら、二つのクラスと七つのメソッドで構成さ

れた、このゲームで遊ぶことができるはずです。

4厳密にはインスタンス変数はサブクラスからもアクセスできます。

workspace 上 で LOGame new openInWorldと 打 ち 、 do it し て く だ さ い 。

ゲームが始まります。セルをクリックすると動作を確認できるでしょう。

ところがクリックすると、PreDebugWindowと呼ばれるノ ー テ ィ フ ァ イ ア が、
エラーメッセージと共に表示されました。理論上は動くはずなのですが…...
図 2.12のように、このウィンドウは MessageNotUnderstood: LOGame»toggleState
と言ってきています。

Figure 2.12: cellをクリックしたときにゲームでバグが発生

何が起こったのでしょうか? 原因を探るため、Smalltalkではさらに強力なツー
ルとなっている、debuggerを使いましょう。

ノ ー テ ィ フ ァ イ ア に あ る debug ボ タ ン を ク リ ッ ク し て く だ さ い 。

デバッガが表示されるはずです。ウィンドウの上部には実行スタックが表

示されます。実行スタックには実行されたメソッドが表示されています。その

中の一つを選ぶと、真ん中のペインにメソッド内の Smalltalkコードが表示さ
れ、エラーを引き起こした部分がハイライトされます。

(上 部 の )LOGame»toggleNeighboursOfCellAt:at:と 表 示 さ れ て い る 行 を ク リ ッ
ク し て く だ さ い 。

デバッガ上で、エラーが発生したメソッドの実行コンテキストを確認する

ことができます (図 2.13)。
デバッガの下側には二つの小さなインスペクタが表示されています。左側

のペインでは、レシーバ、つまり選択中のメソッドを実行するためのメッセー

ジを受け取った、オブジェクトの中身をインスペクトできます。インスタンス

変数の値は、ここで確認できます。右側のペインでは現在のメソッドの実行状

態を示すオブジェクトをインスペクトできます。メソッドの引数や一時変数の

値などはこちらで確認します。

デバッガを使うことで、コードを 1行ずつ実行していくことができます。ま
た、引数や一時変数にバインドされたオブジェクトを見たり、ワークスペース

同様に式を評価したりすることも可能です。通常のデバッガと比べて最も驚く

のは、デバッグ中にコードを書き換えられることです。ほとんどの時間をブラ

Figure 2.13: デバッガ上でtoggleNeighboursOfCell:at:メソッドを選択する。

ウザよりもデバッガ上でプログラミングする Smalltalkerもいるくらいです。
デバッガ上でプログラミングをする利点は、作成中のメソッドが意図通りに動

くかを、実行コンテキストの中で、実際の引数値を使って確認できることにあ

ります。

今回の場合、toggleStateメッセージが、LOGameインスタンスに送られたと、
上部パネルの最初の行に表示されています。これは本来は LOCellのインスタン
スへ送るべきメッセージです。問題は cellsの初期化にあるように思われます。
LOGame»initializeのコードをブラウザで見ると cellsには newCellAt:at:の戻り値が
設定されていることがわかります。しかしメソッドの中身を確認すると、戻り

値として何も指定していません！ メソッドはデフォルトでは selfを戻り値と
して返します。従って newCellAt:at:の場合は、LOGameのインスタンスが返るこ
とになります。

デ バ ッ ガ ウ ィ ン ド ウ を 閉 じ て く だ さ い 。 そ の 後 で 、 “↑ c” と LOGame»
newCellAt:at:メ ソ ッ ド の 最 後 に 追 加 し て 、 cを 返 す よ う に し ま す 。 (Method 2.10
参 照 。 )

Method 2.10: バ グ の 修 正 。
LOGame»newCellAt: i at: j

”位 置 (i,j)に セ ル を 作 成 し 、 適 切 な 画 面 の 位 置 に 追 加 。 新 し い セ ル を 返 す ”
| c origin |
c := LOCell new.
origin := self innerBounds origin.
self addMorph: c.
c position: ((i - 1) * c width) @ ((j - 1) * c height) + origin.
c mouseAction: [self toggleNeighboursOfCellAt: i at: j].
↑ c

第 1章で、Smalltalkでは値を返す時には↑を使うと習ったのを思い出してく
ださい。^をタイプします。

デバッガ上で直接コードを修正した後、Proceedをクリックして、アプリ
ケーションを継続実行することもよくあります。今回の場合は、バグは、失敗

したメソッド中というより、オブジェクトの初期化処理にあったので、最も簡

単なのは、単に初めからやり直すことです。デバッガウィンドウを閉じ、(halo)
で実行中のゲームのインスタンスを破棄し、新たにインスタンスを生成します。

も う 一 度 LOGame new openInWorld を 実 行 し て く だ さ い 。

これでこのゲームはほぼ正常動作をするはずです。ただしマウスをクリック

してから離すまでの間にマウスを移動させると、マウス上のセルも切り替わっ

てしまいます。これは SimpleSwitchMorphを継承したことによる振る舞いです。
mouseMove:をオーバーライドして、何もしないように修正しましょう。

Method 2.11: マ ウ ス の 動 作 を オ ー バ ー ラ イ ド す る 。
LOGame»mouseMove: anEvent

ついに完成しました!

2.9 Smalltalkコードの保存と共有

現在、手元には動作する Lights Outゲームがあります。このゲームを友人と共
有するため、どこかに保存したいと思ったとします。もちろん、Pharoのイメー
ジファイルを丸ごと保存し、初めてのプログラムとして動かしてみせることも

できます。しかしもし友人が既に自分でコーディングしているイメージファイ

ルを持っていたとしたら、イメージファイル全体は欲しがらないかもしれませ

ん。そうしたときに必要なのは、Pharoのイメージファイルからソースコード
を取り出し、他のプログラマが自身のイメージファイルに取り込めるようにす

ることです。

最も簡単なやり方はコードを File outすることです。パッケージペインのメ
ニューをアクションクリックすると、PBE-LightsOut 全体をファイルアウトす

るメニュー項目 various .File outが出てきます。書き出されたファイルは人間が
読むことも可能ですが、本来はコンピュータが読むためのものです。このファ

イルを友人にメールで送れば、ファイルブラウザを使って、ソースコードを自

分の Pharoイメージに取り込むことができます。

PBE-LightsOut パ ッ ケ ー ジ を ア ク シ ョ ン ク リ ッ ク し 、 various .File out で フ ァ
イ ル ア ウ ト し て く だ さ い 。

“PBE-LightsOut.st”という名のファイルがイメージファイルと同じフォル
ダ内にできます。テキストエディタでこのファイルを見てみてください。

ま っ さ ら な Pharo の イ メ ー ジ フ ァ イ ル を 開 き 、 先 ほ ど 出 力 し た
PBE-LightsOut.st を 、 フ ァ イ ル ブ ラ ウ ザ を 使 っ て (Tools …. File Browser )、 フ ァ
イ ル イ ン File in し ま す 。 そ し て そ の イ メ ー ジ 上 で ゲ ー ム が 動 作 す る こ と を 確 認
し て く だ さ い 。

Figure 2.14: Pharoにソースコードをファイルインする。

Monticelloパッケージ

フ ァ イ ル ア ウ ト はコードを保存するには便利ですが、これは明らかに古いやり

方です。ほとんどのオープンソースのプロジェクトでは、コードをリポジトリ

で保守することができる CVS5や Subversion6といったツールを使っています。
Pharo プログラマも同様に、Monticello パッケージによるコードの管理の方
が、より便利だと思うことでしょう。Monticellonoパッケージではファイル名
の末尾に .mczが付きます。このファイルはパッケージ 内のコードを単に ZIP
で圧縮したものです。

Monticelloブラウザを使うことで、FTPやHTTPなどの様々なサーバーの
リポジトリに、パッケージを保存できます。もちろん自分で管理しているロー

カルディレクトリのリポジトリに、パッケージを保存することもできます。パッ

ケージのコピーは常にローカルのハードディスク上の package-cacheフォルダに
キャッシュされます。Monticelloを使うと、複数のバージョンのプログラムを
保存したり、マージしたり、古いバージョンに戻したり、バージョン間の違いを

比較したりすることができます。実際のところ、Monticelloは分散バージョン
管理システムです。つまり開発者の成果物は、CVSや Subversionのように一
箇所のリポジトリに置かれるのではなく、複数の異なった場所に保管されます。

もちろん.mczファイルを Eメールで送ることもできます。受け取った人は
それを package-cacheフォルダに置くことで、Monticelloで閲覧したりロード
したりすることができます。

World メ ニ ュ ー か ら Monticello ブ ラ ウ ザ を 開 い て く だ さ い 。

右側のペイン (図 2.15参照)はMonticelloリポジトリのリストです。イメー
ジ内にロードしたコードについてのリポジトリが、すべて並びます。

Figure 2.15: Monticello Browser。

Monticelloブラウザのリポジトリリストの一番上には、ローカルディレク
トリの package cacheというリポジトリが表示されます。ここにはネットワーク
を使ってロード・公開したパッケージのコピーがキャッシュされます。ローカ

ルキャッシュにはローカルでの自分の履歴が保存できるので非常に便利です。

つまりインターネットに接続できない場所や、回線が遅くて頻繁にリモートに

5http://www.nongnu.org/cvs
6http://subversion.tigris.org

http://www.nongnu.org/cvs
http://subversion.tigris.org

保存したくない状況でも作業できるということです。

Monticelloを使ったコードの保存と読み込み

Monticelloブラウザの左側には、イメージ内にロードしたパッケージ一覧が、
バージョンと共に表示されます。ロード後に修正されたパッケージには、アス

タリスクで印が付きます。(これらは時にダーティパッケージ dirtyと呼ばれま
す。)パッケージを選択すると、リポジトリリストは、選択したパッケージの
コピーを含むリポジトリのみを表示します。

Monticello ブラウザの +Package ボ タ ン を 押 し て 、 PBE-LightsOutと 打

ち 、 PBE-LightsOutパ ッ ケ ー ジ を 追 加 し ま す 。

SqueakSource : Pharoのための SourceForge

コードを保存、共有するのに一番良い方法は、プロジェクト用に SqueakSource
サーバーのアカウントを取ることです。SqueakSourceは SourceForge7のよう
なものです。SqueakSourceはMonticelloのHTTPリポジトリのwebフロント
エンドであり、各種のプロジェクトを管理できます。http://www.squeaksource.com
に、公開 SqueakSourceサーバーがあります。本書に関係したコードのコピーは
http://www.squeaksource.com/PharoByExample.htmlに保管されています。ウェブブ
ラウザからプロジェクトを見ることもできますが、PharoからMonticelloブ
ラウザを使ってパッケージを管理する方が効率よく作業できるでしょう。

ウ ェ ブ ブ ラ ウ ザ で http://www.squeaksource.com を 開 い て く だ さ い 。 ア カ ウ
ン ト を 作 り 、 Lights Out ゲ ー ム 用 の プ ロ ジ ェ ク ト を 作 っ て く だ さ い (つ ま り 登
録 し て く だ さ い )。

SqueakSource は、Monticelloブラウザ上にリポジトリを追加するための
情報を表示します。

SqueakSourceにプロジェクトを作成したら、Pharoにその設定情報を伝え
る必要があるのです。

PBE-LightsOutパ ッ ケ ー ジ を 選 択 し 、 Monticello ブ ラ ウ ザ の +Repository ボ タ
ン を ク リ ッ ク し て く だ さ い 。

利用可能な様々なリポジトリタイプが表示されます。SqueakSource のリ
ポジトリを追加するにはHTTPを選択してください。サーバー情報を入力する
ためのダイアログが表示されます。SqueakSourceプロジェクトを識別するた
め、サーバーから提示されたテンプレートをコピーし、Monticelloのダイアロ
グに貼り付け、自分のイニシャルとパスワードを記入する必要があります。

7http://sourceforge.net

http://www.squeaksource.com
http://www.squeaksource.com/PharoByExample.html
http://www.squeaksource.com
http://sourceforge.net

MCHttpRepository
location: ’http://www.squeaksource.com/YourProject’
user: ’yourInitials’
password: ’yourPassword ’

イニシャルとパスワードを空にした場合、プロジェクトを読むことはできても

更新はできません。

MCHttpRepository
location: ’http://www.squeaksource.com/YourProject’
user: ’’
password: ’’

このテンプレートを入力すると新たなリポジトリがMonticelloブラウザの
右側のリストに表示されます。

Figure 2.16: Monticelloリポジトリの閲覧

Save ボ タ ン を 押 し て 、 Light Out ゲ ー ム の 最 初 の バ ー ジ ョ ン を SqueakSource
に 保 存 し て く だ さ い 。

パッケージを自分のイメージファイルにロードするには、まずバージョン

を指定しなければいけません。リポジトリブラウザを使うことで、特定のバー

ジョンを選ぶことができます。Openボタンかメニューのアクションクリック
でリポジトリブラウザを開いてください。バージョン選択後、イメージファイ

ルにパッケージを読み込むことができます。

先 ほ ど 保 存 し た PBE-LightsOutリ ポ ジ ト リ を 開 い て く だ さ い 。

Monticello では非常に多くのことができますが、詳細は第 6 章で述べま
す。http://www.wiresong.ca/Monticello/でオンラインドキュメントを読むこともで
きます。

2.10 章のまとめ

この章ではカテゴリ、クラス、そしてメソッドの作成方法について学びました。

また、ブラウザ、インスペクタ、デバッガ、そしてMonticelloブラウザの使い
方についても学びました。

• カテゴリは関連するクラスをグループ化したものです。

• 既存のスーパークラスへメッセージを送ることで、新たなクラスを定義
できます。

• プロトコルは関連するメソッドをグループ化したものです。

• ブラウザでメソッド定義を編集して acceptすることで、新たなメソッド
を作ったり修正できます。

• インスペクタは、シンプルで汎用的なUIを提供し、ユーザは任意のオブ
ジェクトの値を見てやりとりすることができます。

• ブラウザは定義されていないメソッドや変数をチェックし、修正候補を
出してきます。

• Pharoでは initializeメソッドはオブジェクトが作られた直後に自動的に実
行されます。初期化のコードはそこに書きます。

• デバッガは動作中のプログラムの状態を見たり修正したりするための高
度な GUIを提供します。

• カテゴリを File outすることでソースコードを共有することができます。

• Monticelloを使って、SqueakSourceのような外部のリポジトリ上でコー
ドを共有するのが、より望ましい方法です。

http://www.wiresong.ca/Monticello/


Chapter 3

文法早わかり

Pharoの文法は、他のモダンな Smalltalkの方言と同様、 Smalltalk-80にとて
も近いものです。プログラムは一種のピジン英語のようで、声に出して読める

ように設計されています。

(Smalltalk includes: Class) ifTrue: [ Transcript show: Class superclass ]

Pharoの文法規則は必要最小限に抑えられています。本質的には、メ ッ セ ー ジ
を 送 信 す る (す な わ ち 式)ための文法しかありません。式は非常に少数の基本
要素から組み立てられます。キーワードは 6個しかありません。さらに、制御
構造のための文法も、新しいクラスを宣言するための文法もありません。代わ

りにオブジェクトにメッセージを送信することによりほとんどのことが達成さ

れます。例えば if-then-else制御構造の代わりに、Smalltalkでは Booleanオブ
ジェクトに ifTrue:といったメッセージを送信します。新しい (サブ)クラスを作
成するには、作成したいクラスのスーパークラスにメッセージを送信します。

3.1 文法要素

式は以下の構成要素でできています。(i) 6個の予約語、すなわち self、super、
nil、true、false、thisContextの擬 似 変 数 (ii)数値、文字、文字列、シンボル、配
列のためのリ テ ラ ル オ ブ ジ ェ ク ト 定数式 (iii)変数宣言 (iv)代入 (v)ブロック
クロージャ (vi)メッセージ送信
表 3.1は様々な文法要素の例です。

ローカル変数 startPointは変数名、つまり識別子です。規約により、識別子は
「キャメルケース」(す な わ ち 最初を除き単語の先頭を大文字にして連結)
で表します。インスタンス変数、メソッドとブロックの引数、一時変数

の先頭の文字は小文字でなければなりません。これにより変数の有効範

囲がプライベートということがコードの読み手に伝わります。

文法 意味

startPoint 変数名

Transcript グローバル変数名

self 擬似変数

1 10進整数
2r101 2進整数
1.5 浮動小数点数

2.4e7 指数表記

$a 文字 a
’Hello’ 文字列 Hello
#Hello シンボル #Hello
#(1 2 3) リテラル配列

{1. 2. 1+2} 動的配列

”a comment” コメント

| x y | 変数 xと yの宣言
x := 1 xに 1を代入
[ x + y ] x+yを評価するブロック
<primitive: 1> バーチャルマシンプリミティブもしくはアノテーション

3 factorial 単項メッセージ

3+4 二項メッセージ

2 raisedTo: 6 modulo: 10 キーワードメッセージ

↑ true 値 trueを返す
Transcript show: ’hello’. Transcript cr 式セパレータ (.)
Transcript show: ’hello’; cr メッセージのカスケード (;)

Table 3.1: Pharoの文法の要点

共有変数 大文字で始まる識別子はグローバル変数、クラス変数、プール辞書、

クラス名です。Transcriptはグローバル変数であり、TranscriptStreamクラ
スのインスタンスです。

レシーバ selfは、現在のメソッドを実行しているオブジェクト自身を指す予約
語です。selfは「レシーバ」とも呼ばれます。なぜならこのオブジェクト
は、メソッドを実行するきっかけになったメッセージを受け取っている

からです。selfは「擬似変数」と呼ばれます。selfには代入できないから
です。

整数 42のような通常の 10進整数に加えて、Pharoでは基数表記も使えます。
2r101は基数 2 (す な わ ち バイナリ)の 101であり、10進数の 5と同じです。

浮動小数点数 は 10 を基数とした指数 で指定することが可能です。 2.4e7 は
2.4× 107 です。

文字 文字リテラルにはドル記号を付けます: $aは aのリテラルです。非印字
文字のインスタンスは、Character spaceや Character tabのように Character
クラスに適切なメッセージを送信することにより得ることができます。

文字列 シングルクォートは文字列を定義するために使います。クォートを含

んだ文字列を表現したい場合は、’G’’day’のようにクォートを重ねます。

シンボル は文字の並びでできている点で文字列に似ています。ただし文字列と

異なり、シンボルはグローバルに一意であることが保証されます。#Hello
というシンボルオブジェクトは一つだけしか存在しません。一方で ’Hello’
という値の文字列オブジェクトは複数あるかもしれません。

コンパイル時配列 はスペースで区切ったリテラルを #( )で囲んで定義します。
括弧内のすべての要素はコンパイル時に定数でなければなりません。例

えば、#(27 (true false) abc)は要素が 3個のリテラル配列です。整数 27、
真偽値が 2 個入ったコンパイル時配列、そしてシンボル #abc が要素に
なります (これが #(27 #(true false) #abc)と同じということに注意してくだ
さい)。

動的配列 波括弧 { }は実行時に (動的)配列を定義します。動的配列の要素は、
ピリオドで区切った式です。従って、{ 1. 2. 1+2 } は 1、 2、そして 1+2
を評価した結果を要素とする配列を定義します。(波括弧による表記は、
Smalltalkの方言でも Pharoと Squeakに特有のものです! 他の Smalltalk
では、動的配列は明示的に組み立てなければなりません)。

コメント はダブルクォートで囲みます。”hello”は文字列ではなくコメントな
ので、Pharoコンパイラによって無視されます。コメントは複数行にま
たがってもかまいません。

ローカル変数宣言 ローカル変数を宣言するには、メソッド (およびブロック)
内で、ローカル変数群を縦棒 | |で囲みます。

代入 :=は変数にオブジェクトを代入します。

ブロック 角括弧 [ ]は、ブロッククロージャあるいはレキシカルクロージャと
も呼ばれるブロックを定義します。ブロックは関数を表す第一級のオブ

ジェクトです。後で見るように、ブロックは引数を取ることもローカル

変数を持つこともできます。

プリミティブ <primitive: ...>はバーチャルマシンプリミティブの呼び出しを表
します。(<primitive: 1>は SmallInteger»+のバーチャルマシンプリミティブ
です)。プリミティブの後ろにあるコードは、プリミティブが失敗した場
合にだけ実行されます。同じ文法はメソッドアノテーションにも使われ

ます。

単項メッセージ は単一の単語 (例えば factorial)からなり、レシーバ (例えば 3)
に送信されます。

二項メッセージ は演算子 (例えば +)のようなもので、単一の引数とともにレ
シーバに送信されます。3+4の場合、レシーバは 3で引数は 4です。

キーワードメッセージ は raisedTo:modulo:のように、複数のキーワードから
なるメッセージです。キーワードはコロンで終わり、それぞれ引数を取

ります。 2 raisedTo: 6 modulo: 10の式では、raisedTo:modulo:がメッセージ
セレクタで、コロンの後の 6と 10が引数です。メッセージは 2に送ら
れています。

メソッドリターン ↑はメソッドから値をリ タ ー ン するために使われます (↑を
入力するには ^とタイプしなければなりません)。

文の並び 文の区 切 り はピリオド (.)です。二つの式の間にピリオドを置くと、
それらは独立した文として扱われます。

カスケード セミコロンは、一つのレシーバにメッセージをカ ス ケ ー ド 送信する

ために使うことができます。Transcript show: ’hello’; crでは、まず Transcript
にキーワードメッセージ show: ’hello’を送り、次に同じレシーバ (Transcript)
に単項メッセージ crを送っています。

Number、Character、Stringおよび Booleanクラスについては第 8章でより
詳細に述べます。

3.2 擬似変数

Smalltalkには 6個の予約語として、擬 似 変 数 があります。nil、true、false、self
、super、 thisContextです。これらは事前に定義され、代入することができない
ため、擬似変数と呼ばれます。true、false、nilは定数です。一方 self、super、お
よび thisContextの値は、コードの実行中に動的に変化します。

trueと falseは、Booleanクラスである Trueおよび Falseの唯一のインスタン
スです。詳細については第 8章を参照してください。

selfは、現在実行しているメソッドのレシーバを指します。

superも現在実行しているメソッドのレシーバを指します。ただし superに
メッセージを送信する場合、メソッド探索はそのメソッドを定義しているクラ

スのスーパークラスから始まります。詳細については第 5章を参照してくだ
さい。

nilは未定義を表すオブジェクトです。クラス UndefinedObjectの唯一のイン
スタンスです。インスタンス変数、クラス変数およびローカル変数は、nilで初
期化されます。

thisContextは、実行時スタックのトップフレームを表す擬似変数です。言い
換えれば、 thisContextは現在実行している MethodContextあるいは BlockClosure
を指します。ほとんどのプログラマにとって、thisContextは意識する必要がな

いものですが、デバッガのような開発ツールを実装するには不可欠です。また、

例外処理と継続を実装するためにも使われます。

3.3 メッセージ送信

Pharoには 3種類のメッセージ (メッセージ送信)があります。

1. 単 項 メッセージは引数を取りません。1 factorialは、オブジェクト 1にメッ
セージ factorialを送信します。

2. 二 項 メッセージは引数を一つだけ取ります。1 + 2は、オブジェクト 1に
メッセージ +を送信します。引数は 2です。

3. キ ー ワ ー ド メッセージは一つ以上の引数を取ります。2 raisedTo: 6 modulo:
10は、オブジェクト 2に、メッセージセレクタ raisedTo:modulo:、引数 6、
引数 10からなるメッセージを送信します。

単項メッセージのメッセージセレクタは、英数字で記述します。先頭は小

文字になります。

二項メッセージのメッセージセレクタは、以下の文字を 1文字以上組み合
わせることでできます。

+ - / \ * ∼ < > = @ % | & ? ,

キーワードメッセージのセレクタは、一連のキーワードを連結したものです。

キーワードはそれぞれ小文字で始まりコロンで終わります。

単項メッセージの優先順位が最も高くなります。次に二項メッセージ、最

後にキーワードメッセージが評価されます。

つまり、

2 raisedTo: 1 + 3 factorial −→ 128

となります。(最初に 3に factorialが送られます。次に 1に + 6が送られます。
最後に 2に raisedTo: 7が送られます)。式を評価した結果を示すのに式 −→
結 果 という記法を使っているのを思い出してください。

優先度を別にすると、評価順は厳密に左から右となります。つまり、

1 + 2 * 3 −→ 9

です。7ではありません。評価順を変えるには括弧を使わなければなりません。

1 + (2 * 3) −→ 7

複数のメッセージ送信は、ピリオドやセミコロンを使って組み立てること

ができます。ピリオドで区切られた式の並びは、文として順々に評価されます。

Transcript cr.
Transcript show: ’hello world’.
Transcript cr

これは、Transcriptオブジェクトに crを送信し、次に Transcriptに show: ’hello world’
を送り、最後に Transcriptに再び crを送っています。

一連のメッセージを同 じ レシーバに送信する場合、カ ス ケ ー ド としてより

簡潔に表現することができます。レシーバを一度だけ指定し、メッセージの並

びをセミコロンで区切ります。

Transcript cr;
show: ’hello world’;
cr

これは前の例とまったく同じ意味になります。

3.4 メソッドのシンタックス

式は Pharoのあらゆる場所 (例えばデバッガ、ワークスペース、ブラウザなど)
で評価することができますが、通常、メソッドはブラウザやデバッガ内で定義し

ます。(メソッドは外部からファイルインすることもできますが、これは Pharo
のプログラミングにとって普通のやり方ではありません)。
プログラムは、クラスに属するメソッド単位で開発していきます。(クラス

は、既存のクラスに、サブクラスを作成するためのメッセージを送信して定義

します。クラスを定義するための特別な文法はありません)。
以下は String クラスの lineCount メソッドです。(慣習により、メソッドは

ClassName»methodNameと表記します。そのため String»lineCountと書き表せま
す)。

Method 3.1: Line count
String»lineCount

”レ シ ー バ の 行 数 を 答 え る 。
crに 出 会 う た び に 1行 増 や す 。 ”
| cr count |
cr := Character cr.
count := 1 min: self size.
self do:
[:c | c == cr ifTrue: [count := count + 1]].

↑ count

文法上、メソッドは以下のものから構成されます。

1. 名前 (す な わ ち lineCount) と引数 (この例では一つもない) からなるメッ
セージパターン

2. コメント (どこに書いてもかまいませんが、メソッドが何を行うかの説明
を一番上に書く慣習があります)

3. ローカル変数 (す な わ ち crと count)の宣言

4. ドットで区切られたメッセージ式。(ここでは式は 4個)。

↑(^とタイプする)が先頭に書いてある式は、評価されるとメソッドを終了
してその式の値を返します。明示的に値を返さずに終了するメソッドの場合、

暗黙的に selfをリターンします。

引数とローカル変数は常に小文字で始めなければなりません。大文字で

始まる名前はグローバルな変数であるとみなされます。例えば クラス名 (例:
Character)は、そのクラスを表すオブジェクトを参照するためのグローバル変
数です。

3.5 ブロックのシンタックス

ブロック は、遅延評価のためのメカニズムを提供します。ブロックは本質的

には無名の関数です。ブロックを評価するには、メッセージ valueを送ります。
ブロックは、明示的なリターン (↑)がなければ、最後の式の値を返します。
[ 1 + 2 ] value −→ 3

ブロックは引数を持つことができます。引数は先頭にコロンを付けて宣言し

ます。縦棒でブロック本体と引数の宣言とを区切ります。引数が 1個のブロッ
クを評価するには、ブロックに、引数一つのメッセージ value:を送信しなけれ
ばなりません。引数が 2個のブロックを評価するには、ブロックに value:value:
を送信しなければなりません。同様にして 4 個までの引数を渡すことができ
ます。

[ :x | 1 + x ] value: 2 −→ 3
[ :x :y | x + y ] value: 1 value: 2 −→ 3

引数の数が 4個より多いブロックには valueWithArguments:を使い、引数を
配列にして渡さなければなりません。(引数が多いブロックは、しばしば設計上
の問題がある兆候です)。
ブロック内でもローカル変数を宣言することができます。メソッドのロー

カル変数宣言と同様、縦棒で囲みます。ローカル変数は引数リストの後ろに宣

言します。

[ :x :y | | z | z := x+ y. z ] value: 1 value: 2 −→ 3

ブロックはそれを取り囲む環境の変数を参照することができるので、レキ

シカルク ロ ー ジ ャ です。次のブロックは、取り囲んでいる環境の変数 xを参照
しています。

| x |
x := 1.
[ :y | x + y ] value: 2 −→ 3

ブロックはクラス BlockClosure のインスタンスです。つまりブロックはオ
ブジェクトなので、他の普通のオブジェクトと同様、変数に代入したり引数と

して渡したりすることができます。。

3.6 条件式とループの要点

Smalltalkには制御構造のための特別の文法はありません。代わりに、真偽値
や数、コレクションに対し、ブロックを引数とするメッセージを送ることで制

御構造を表します。

条件式は、真偽値を返す式の値に、メッセージ ifTrue:、ifFalse: ifTrue:ifFalse:
などを送ることにより表現します。真偽値について、詳しくは第 8章を参照し
てください。

(17 * 13 > 220)
ifTrue: [ ’bigger’ ]
ifFalse: [ ’smaller’ ] −→ ’bigger’

ループは、ブロック、整数、コレクションなどにメッセージを送ることで

表します。ループの終了条件は、繰り返し評価されるかもしれないので、真偽

値そのものではなく真偽値を返すブロックでなければなりません。以下は非常

に手続き的なループの例です。

n := 1.
[ n < 1000 ] whileTrue: [ n := n*2 ].
n −→ 1024

whileFalse:では終了条件が逆になります。

n := 1.
[ n > 1000 ] whileFalse: [ n := n*2 ].
n −→ 1024

timesRepeat:により単純な固定回数の繰り返しができます。

n := 1.
10 timesRepeat: [ n := n*2 ].
n −→ 1024

数にメッセージ to:do:を送信することができます。するとその数 (レシーバ)
がループカウンターの初期値になります。to:の引数はループカウンターの上限
です。また do:の引数はブロックで、ループカウンターの現在の値を引数とし
て取ります。

result := String new.
1 to: 10 do: [:n | result := result, n printString, ’ ’].
result −→ ’1 2 3 4 5 6 7 8 9 10 ’

高階イテレータ コレクションは様々なクラスで実装されています。多くは

同じプロトコルをサポートします。コレクションの要素を繰り返し処理するた

めの、最も重要メッセージには、do:、collect:、select:、reject:、detect:および
inject:into:があります。これらは高レベルのイテレータを定義しているメッセー
ジであり、それによって非常にコンパクトなコードを書くことができます。

Intervalは、開始、終了を指定した数の並びを繰り返し処理できるコレクショ
ンです。1 to: 10は 1から 10までの Intervalを表します。Intervalはコレクション
なので、これにメッセージ do:を送信することができます。do:の引数はブロッ
クであり、コレクションの要素に対して繰り返し評価されます。

result := String new.
(1 to: 10) do: [:n | result := result, n printString, ’ ’].
result −→ ’1 2 3 4 5 6 7 8 9 10 ’

collect:は各要素を変換して同じサイズの新しいコレクションを作ります。

(1 to: 10) collect: [ :each | each * each ] −→ #(1 4 9 16 25 36 49 64 81 100)

select:と reject:は、真偽値ブロックの条件を満たす (あるいは満たさない)要
素からなる新しいコレクション (部分集合)を作ります。detect:は条件を満たし
た最初の要素を返します。文字列が文字のコレクションであることを覚えてお

いてください。つまり文字列内の文字はイテレートすることができます。

’hello there’ select: [ :char | char isVowel ] −→ ’eoee’
’hello there’ reject: [ :char | char isVowel ] −→ ’hll thr’
’hello there’ detect: [ :char | char isVowel ] −→ $e

最後に、コレクションは inject:into:メソッドにより、関数型の畳 み 込 み 演算
もサポートします。これを使うと、シードとなる値から開始し、コレクション

の各要素を注入 (inject)していった累積的な値を得ることができます。和や積
の累積は inject:into:の典型的な応用例です。

(1 to: 10) inject: 0 into: [ :sum :each | sum + each ] −→ 55

これは 0+1+2+3+4+5+6+7+8+9+10と等価です。

コレクションに関しては第 9章で詳しく述べます。

3.7 プリミティブとプラグマ

Smalltalkではすべてがオブジェクトです。そしてあらゆることがメッセージ
送信により起こります。しかしある地点で底に当たります。ある種のオブジェ

クトはバーチャルマシンプリミティブを呼び出さないと動作できません。

例えば、以下のメソッドはすべてプリミティブとして実装されています。メ

モリ割り当て (new、new:)、ビット操作 (bitAnd:、bitOr:、bitShift:)、ポインタお
よび整数演算 (+、-、<、>、*、/、=、==...)、および配列のアクセス (at:、at:put:)。
プリミティブは、文法 <primitive: aNumber>で呼び出します。プリミティブ

を呼び出すメソッドで、通常の Smalltalkコードがさらに続くことがあります。
そうしたコードは、プリミティブが失敗した場合にの み 評価されます。

SmallInteger»+ のコードを見てみましょう。プリミティブが失敗した場合、
式 super + aNumberが評価され値が返されます。

Method 3.2: プ リ ミ テ ィ ブ メ ソ ッ ド
+ aNumber
”プ リ ミ テ ィ ブ 。 レ シ ー バ と 引 数 で 加 算 し 、 結 果 が SmallInteger な ら そ れ を 返 す 。
引 数 か 結 果 が SmallInteger で な け れ ば 失 敗 す る 。
必 須 。 メ ソ ッ ド 探 索 な し 。 Object の ド キ ュ メ ン ト whatIsAPrimitive を 参 照 の こ
と 。 ”

<primitive: 1>
↑ super + aNumber

Pharoでは、山括弧を使った記法はメソッドのアノテーションにも使われ
ます。これはプラグマと呼ばれます。

3.8 まとめ

• Pharoには予約語が 6個 (だけ)あります。これらは擬 似 変 数 とも呼ばれ
ます。 true、false、nil、self、super、そして thisContextです。

• 5種類のリテラルオブジェクトがあります: 数 (5、2.5、1.9e15、2r111)、文
字 ($a)、文字列 (’hello’)、シンボル (#hello)、配列 (#(’hello’ #hi))です。

• 文字列はシングルクォートで、コメントはダブルクォートで囲みます。文
字列内にクォートを書くには、クォートを重ねます。

• 文字列と異なり、シンボルはグローバルに一意であることが保証されます。

• リテラル配列を定義するには #( ... )を使います。動的配列を定義するに
は { ... }を使います。#( 1 + 2 ) size −→ 3ですが、{ 1 + 2 } size −→ 1
なので注意してください。

• 3種類のメッセージがあります: 単 項 (例 え ば 1 asString、Array new)、二 項
(例 え ば 3 + 4、’hi’ , ’ there’)、キ ー ワ ー ド (例 え ば ’hi’ at: 2 put: $o)です。

• カ ス ケ ー ド によるメッセージ送信では、同じターゲットに連続してメッセー
ジを送れます。各メッセージはセミコロンで区切ります。OrderedCollection
new add: #calvin; add: #hobbes; size −→ 2。

• ローカル変数は縦棒で囲んで宣言します。代入には :=を使います。|x| x:
=1。

• 式は、メッセージ送信、カスケード、代入からなります。これらは括弧
でグループ化されることもあります。文 はピリオドによって区切られた

式です。

• ブロッククロージャは角括弧で囲まれた式です。ブロックは引数を取っ
てもよく、一時変数を使うこともできます。ブロック内の式は、ブロッ

クに、value...メッセージを、適切な個数の引数で送ったときに初めて評
価されます。

[:x | x + 2] value: 4 −→ 6。

• 制御構造用の文法はありません。条件によってブロックを評価させるメッ
セージを送ることで実現します。

(Smalltalk includes: Class) ifTrue: [ Transcript show: Class superclass ]。





Chapter 4

メッセージ構文を理解しよう

Smalltalkのメッセージ構文は非常に単純ですが、標準的なものではないので
慣れるのに少し時間がかかるでしょう。この章では、この特殊なメッセージ構

文に順応するためのガイダンスを提供します。もし既にこの構文に慣れている

のなら、この章は飛ばしてもかまいませんし後で読んでもよいでしょう。

4.1 メッセージを読み取る

Smalltalkでは、第 3章で紹介した構文要素 (:= ↑ . ; # () {} [ : | ])を除けばすべて
がメッセージ送信です。C++のように、+などの演算子を自作のクラス用に定
義することもできますが、すべての演算子の優先順位は同じです。さらに、あ

るメソッドが受け取る引数の個数は変えられません。『-』は常に二項演算子で
す;単項の『-』演算子を得る方法はありません。

Smalltalkでは、メッセージが送信される順番はメッセージの種類によって
決定されます。そしてメッセージには 3 種類しかありません: 単 項 、二 項 お
よびキ ー ワ ー ド メッセージです。単項メッセージが常に最初に送られ、その次

が二項メッセージ、そして最後がキーワードメッセージです。他の多くの言語

と同様に、括弧を使って評価順を変えることもできます。これらの規則によっ

て、Smalltalkのコードは可能な限り読みやすく、かつ規則のことで頭を悩ま
せなくてもよくなっています。

Smalltalk のほとんどの計算はメッセージパッシングによって行われるの
で、メッセージを正しく読み取ることが非常に重要です。次の用語と概念を理

解することが役に立つでしょう:

• メッセージは、セ レ ク タ およびオプションのメッセージ引数からなり
ます。

• メッセージは、レ シ ー バ に送信されます。

• レシーバとメッセージの組み合わせが、メ ッ セ ー ジ 送 信 と呼ばれます
(図 4.1)。

Figure 4.1: レシーバ、セレクタ、引数か
らなるメッセージ送信の例 (二つ)。

Figure 4.2: aMorph color: Color
yellow は二つのメッセージ送信の
組み合わせ: Color yellow および
aMorph color: Color yellow。

メッセージは常にレシーバに送信されます。レシーバは、

リテラル・ブロック・変数・他のメッセージを評価した結

果、のいずれかです。

図 4.2では、メッセージのレシーバが同定しやすくなるように、レシーバに
下線を引きました。また、それぞれのメッセージ送信を楕円で囲み、送信され

る順番に従って番号を付けました。

図 4.2 は Color yellowと aMorph color: Color yellow という二つのメッセージ送
信を表しています。従って楕円も二つあります。メッセージ送信 Color yellow
がまず実行されるので、それを囲む楕円に番号 1が付いています。全体ではレ
シーバが二つあります: 一つは aMorphでメッセージ color: ...が送られており、
もう一つは Colorでメッセージ yellowが送られています。レシーバにはそれぞ
れ、既に述べたように下線が引いてあります。

レシーバはメッセージ送信の最初の要素になることができます。例えば

100 + 200 の 100 や Color yellow の Color がそうです。しかし、レシーバは他の
メッセージ送信の結果であってもかまいません。例えば Pen new go: 100という
メッセージ送信では、メッセージ go: 100のレシーバは Pen newオブジェクト
(Pen newの結果として返されるオブジェクト)です。いずれにせよ、メッセー
ジはレ シ ー バ と呼ばれるオブジェクトに送信され、レシーバは他のメッセージ

送信の結果であってもよいということです。

表 4.1にいくつかのメッセージ送信の例を挙げました。必ずしもすべてのメ
ッセージ送信に引数があるわけではない、ということに注意してください。open
のような単項メッセージには引数はありません。go: 100 のような単独のキー
ワードメッセージや + 20のような二項メッセージは引数を一つ取ります。単純

メッセージ送信 メッセージの種類 結果

Color yellow 単項 色を表すオブジェクト。

aPen go: 100 キーワード レシーバであるペン・オブジェ

クトが 100ピクセル移動する。
100 + 20 二項 数値オブジェクト 100 にメッ

セージ + を数値 20 とともに
送信。

Browser open 単項 新しいブラウザを開く。

Pen new go: 100 単項およびキーワード ペン・オブジェクトが生成され

100ピクセル移動する。
aPen go: 100 + 20 キーワードおよび二項 レシーバであるペン・オブジェ

クトが 120ピクセル移動する。

Table 4.1: メッセージ送信と種類の例

なメッセージ・組み合わされたメッセージ、という分類もできます。Color yellow
や 100 + 20は単純です: 一つのメッセージが一つのオブジェクトに送信される
だけです。一方 aPen go: 100 + 20は二つのメッセージ送信が組み合わさったも
のです: + 20が 100に送信され、その結果を引数として go:が aPenに送信され
ます。オブジェクトを返す式 (代入、メッセージ送信およびリテラル)であれ
ばレシーバになることができます。Pen new go: 100では、メッセージ go: 100が
Pen newオブジェクト (Pen newの結果として返されるオブジェクト)に送信さ
れます。

4.2 3種類のメッセージ

Smalltalkではメッセージの送信順序を決定するいくつかの単純な規則が定義
されています。これらの規則は以下の 3種類のメッセージの別に基づきます:

• 単 項 メ ッ セ ー ジ は、あるオブジェクトに他の追加情報なしで送信される
メッセージです。例えば 3 factorialの factorialは単項メッセージです。

• 二 項 メ ッ セ ー ジ は、演算子 (しばしば算術的な)からなるメッセージです。
これらは、常にレシーバと引数という二つのオブジェクトが関与するた

め、二項メッセージと呼ばれます。10 + 20の場合、+が二項メッセージ
で引数 20と共にレシーバ 10に送信されています。

• キ ー ワ ー ド メ ッ セ ー ジ は、一つあるいはそれ以上のキーワードからなって
います。それぞれのキーワードの最後にはコロン (:)が付いており、キー
ワードごとに一つ引数を取ります。例えば anArray at: 1 put: 10の場合、キー
ワード at:は 1という引数を取り、キーワード put:が引数 10を取ってい
ます。



64 メ ッ セ ー ジ 構 文 を 理 解 し よ う

単項メッセージ

単項メッセージは引数を必要としないメッセージです。その構文は receiver
selectorという形式です。セレクタ (selector)は、:を含まない単純な文字列です
(例 え ば factorial、open、class)。

89 sin −→ 0.860069405812453
3 sqrt −→ 1.732050807568877
Float pi −→ 3.141592653589793
’blop’ size −→ 4
true not −→ false
Object class −→ Object class ”Object の ク ラ ス は Object class (!)”

単項メッセージは引数を必要とないメッセージです。

その構文は receiver selectorという形式です。

二項メッセージ

二項メッセージは引数を一つだけ取ります。ま た 二項メッセージのセレクタは、

次の文字セットからの 1文字以上の繰り返しでなければなりません: +、-、*、
/、\&、=、>、|、<、\∼、および@。構文解析時に問題が起こるので、--はセレ
クタとしては認められないことに注意してください。

100@100 −→ 100@100 ”Point オ ブ ジ ェ ク ト を 生 成 ”
3 + 4 −→ 7
10 - 1 −→ 9
4 <= 3 −→ false
(4/3) * 3 = 4 −→ true ”同 値 性 の テ ス ト も 、 Smalltalk で は 単 な る 二 項 メ ッ セ ー

ジ 。 分 数 オ ブ ジ ェ ク ト は 値 を 正 確 に 表 現 で き る ”
(3/4) == (3/4) −→ false ”別 個 に 作 ら れ た 二 つ の 分 数 は 、 同 じ オ ブ ジ ェ ク ト で

は な い ”

二項メッセージは引数を一つだけ取ります。ま た 二項メッ

セージのセレクタは、次の文字セットからの 1文字以上の
繰り返しでなければなりません: +、-、*、/、&、=、>、|、<、
∼、および@。--は認められません。
二項メッセージの構文は receiver selector argumentという形
式です。

キーワードメッセージ

キーワードメッセージは一つ以上の引数を取るメッセージです。そのセレクタは

一つ以上のキーワードを連結したもので、各キーワードは :で終わります。キー
ワードメッセージの構文は次の形式です: receiver selectorWordOne: argumentOne
wordTwo: argumentTwo

それぞれのキーワードが引数を一つずつ取ります。つまり r:g:b:というセレ
クタは 3引数であり、playFileNamed:や at:は 1引数、at:put:は 2引数です。Color
クラスのインスタンスを作るにはクラスに r:g:b:メッセージを送ります (例えば
Color r: 1 g: 0 b: 0によって赤色を表すオブジェクトを作ることができます)。こ
こではコロンもセレクタの一部であるということに注意してください。

Javaや C++であれば、Smalltalkにおける Color r: 1 g: 0 b: 0
は Color.rgb(1,0,0) というメソッド呼び出しに相当するでし
ょう。

1 to: 10 −→ (1 to: 10) ”区 間 オ ブ ジ ェ ク ト を 生 成 ”
Color r: 1 g: 0 b: 0 −→ Color red ”色 オ ブ ジ ェ ク ト を 生 成 ”
12 between: 8 and: 15 −→ true

nums := Array newFrom: (1 to: 5).
nums at: 1 put: 6.
nums −→ #(6 2 3 4 5)

キーワードメッセージは一つ以上の引数を取るメッセージ

です。そのセレクタは一つ以上のキーワードを連結したも

ので、各キーワードはコロン (:) で終わります。キーワード
メッセージの構文は次の形式です:
receiver selectorWordOne: argumentOne wordTwo: argu-
mentTwo

4.3 メッセージ式の組み合わせ

ここまでに述べた 3 種類のメッセージにはそれぞれ異なる優先順位があるの
で、これらをエレガントに組み合わせて使うことができます。

1. 単項メッセージが常に最初に送信され、次に二項、そして最後にキーワー
ドメッセージが送信されます。

2. 括弧 で囲まれたメッセージは他のメッセージよりも優先して送信され
ます。

3. 同じ種類のメッセージは左から右に評価されます。

この規則により、Smalltalkのプログラムは極めて自然に読み進められるよ
うになっています。メッセージが意図した通りの順番で送信されることを確実

にしたいのであれば、図 4.3にあるように括弧を足していくこともできます。
この図では、メッセージ yellowが単項メッセージであり color:がキーワードメッ
セージであるため、Color yellowがまず先に送信されます。しかし括弧で囲まれ
たメッセージがまず送信されるため、(冗長な)括弧を Color yellowの周りに書い
てこれがまず先に送信されるということを強調することができます。本節の残

りではこれらの点について説明します。

Figure 4.3: 単項メッセージが最初に送信されるので、Color yellowがまず送信さ
れ、返された色オブジェクトがメッセージ aPen color:の引数として渡される。

単項 >二項 >キーワード

単項メッセージがまず送信され、次に二項、そして最後にキーワードメッセー

ジが送信されます。単項メッセージの優先順位が他の種類のメッセージよりも

高いとも言います。

規則 1: 単項メッセージがまず送信され、次に二項、そして
最後にキーワードメッセージが送信されます。

単項 >二項 >キーワード

以下の例が示すように、Smalltalkの構文規則によれば、大概のメッセージ
送信は自然に読むことができます:

1000 factorial / 999 factorial −→ 1000
2 raisedTo: 1 + 3 factorial −→ 128

ただし残念ながら、算術演算の場合にはこの規則はやや単純すぎます。そ

のため二項演算子を複数組み合わせるときは、場合によって括弧を使って評価

順を強制する必要があります:

1 + 2 * 3 −→ 9
1 + (2 * 3) −→ 7

以下のさらに複雑な (!) 例は、込み入った Smalltalkの式も自然に読めると
いうことの良い例になっています。

[:aClass | aClass methodDict keys select: [:aMethod | (aClass>>aMethod) isAbstract ]]
value: Boolean −→ an IdentitySet(#or: #| #and: #& #ifTrue: #ifTrue:ifFalse:
#ifFalse: #not #ifFalse:ifTrue:)

ここでは Booleanクラスのどのメソッドが抽象メソッドであるかを調べようと
しています1。このブロックは、引数として渡されたクラス aClassにそのメソッ
ド辞書のキー集合を問い合わせ、その中から抽象メソッドを選択 (select:)して
います。ここでは aClassを Booleanにバインドしています。この式の中で括弧
は、二項メッセージ >>を単項メッセージ isAbstractより先に送信するためだけ
に必要でした (>>はメソッドをクラスから取り出す二項演算子です)。全体を評
価するとメソッド名の集合が返ります。これらのメソッドは Booleanの具象サ
ブクラス (Trueと False)で実装しなくてはなりません。

例。 メッセージ aPen color: Color yellowの中には、Colorに送信される単 項 メッ
セージ yellowと aPenに送信される キ ー ワ ー ド メッセージ color:があります。単
項メッセージがまず送信されるので、Color yellow がまず送信 (評価)され (1)、
その結果として返された色オブジェクトがメッセージ aPen color: aColorの引数
(aColor)となります (2)。Example 4.1を見てください。図 4.3は、これらのメッ
セージがどのように送信されるかをグラフィカルに表しています。

Example 4.1: aPen color: Color yellow の 評 価 順 を 分 解 す る
aPen color: Color yellow

(1) Color yellow ”単 項 メ ッ セ ー ジ が ま ず 送 信 さ れ る ”
−→ aColor

(2) aPen color: aColor ”キ ー ワ ー ド メ ッ セ ー ジ が 次 に 送 信 さ れ る ”

例。 メッセージ aPen go: 100 + 20の中には、二 項 メッセージ + 20とキ ー ワ ー
ド メッセージ go:があります。二項メッセージはキーワードメッセージより先
に送信されるので、100 + 20がまず評価されます (1): + 20がオブジェクト 100
に送信され、数値 120が返ります。次に aPenにメッセージ go: 120が送信され
ます (2)。Example 4.2は、このメッセージ送信がどのように実行されるかを示
しています。

Example 4.2: aPen go: 100 + 20 を 分 解 す る
aPen go: 100 + 20

(1) 100 + 20 ”二 項 メ ッ セ ー ジ が ま ず 送 信 さ れ る ”
−→ 120

(2) aPen go: 120 ”キ ー ワ ー ド メ ッ セ ー ジ が 次 に 送 信 さ れ る ”

1実を言うと、同等の式はさらに簡単に Boolean methodDict select: #isAbstract thenCollect: #selector
と書くこともできます。

Figure 4.4: 二項メッセージはキー
ワードメッセージより先に送信され

る。

Figure 4.5: Pen new go: 100 + 20 を分
解する

例。 練習としてメッセージ Pen new go: 100 + 20の評価順を分解してみてくだ
さい。このメッセージには単項メッセージが一つ、キーワードメッセージが一

つ、二項メッセージが一つあります (図 4.5)。

括弧優先

規則 2。括弧で囲まれたメッセージは他のメッセージより
も優先して送信されます。

(任意のメッセージ) >単項 >二項 >キーワード

1.5 tan rounded asString = (((1.5 tan) rounded) asString) −→ true ”括 弧 は 不 要 ”
3 + 4 factorial −→ 27 ”(5040 で は な い )”
(3 + 4) factorial −→ 5040

以下の例では、lowMajorScaleOn: を play よりも先に送信するために括弧 が
必要です。

(FMSound lowMajorScaleOn: FMSound clarinet) play
”(1) メ ッ セ ー ジ clarinet が FMSound ク ラ ス に 送 信 さ れ 、 ク ラ リ ネ ッ ト の 音 色 オ ブ

ジ ェ ク ト が 作 ら れ る 。

(2) こ の 音 色 オ ブ ジ ェ ク ト が lowMajorScaleOn: キ ー ワ ー ド メ ッ セ ー ジ の 引 数 と
し て FMSound に 送 信 さ れ る 。

(3) 結 果 の 音 オ ブ ジ ェ ク ト に play が 送 信 さ れ 、 演 奏 さ れ る 。 ”

例。 メッセージ (65@325 extent: 134@100) centerは、左上が座標 (65, 325)で大
きさが 134×100であるような長方形の、中央の座標を返します。Example 4.3
は、このメッセージがどのように分解されて送信されるかを示しています。ま

ず括弧で囲まれたメッセージが送信されます: 括弧の内側には二つの二項メッ
セージ 65@325と 134@100があるので、これらがまず送信 (評価)されて二つの
点オブジェクトが返ります。次に最初の点オブジェクトにキーワードメッセー

Figure 4.6: Pen new downを分解する

ジ extent:が送信され、長方形オブジェクトが返ります。最後に単項メッセージ
centerがその長方形オブジェクトに送信され、点オブジェクトが返ります。も
し括弧がなかったとすると、数値 100はメッセージ centerを理解できないため、
評価時にエラーとなります。

Example 4.3: 括 弧 の 例 。
(65@325 extent: 134@100) center

(1) 65@325 ”二 項 ”
−→ aPoint

(2) 134@100 ”二 項 ”
−→ anotherPoint

(3) aPoint extent: anotherPoint ”キ ー ワ ー ド ”
−→ aRectangle

(4) aRectangle center ”単 項 ”
−→ 132@375

左から右への評価

ここまでに、異なる種類のメッセージがどのような優先順位で処理されるかを

学びました。残る疑問は同じ優先順位のメッセージがどのような順番で送信さ

れるかということです。これらは左から右に送信されます。この振る舞いは実

は Example 4.3の中で確認しています。Example 4.3では二つの点オブジェク
トの生成メッセージ (@)が先に送信されていました。

規則 3。同じ種類のメッセージがある場合は、評価は左か
ら右の順で行われます。

例。 メッセージ送信 Pen new down では単項メッセージのみが使われている
ので、左端の Pen newがまず送信されます。その結果は新しいペン・オブジェ
クトで、このペン・オブジェクトに 2番目のメッセージ downが送信されます
(図 4.6)。

伝統的な算術記法との食い違いに関する問題

メッセージ式の組み合わせに関する規則はごく単純なものですが、算術演算が

二項メッセージとして記述されている場合には、伝統的な記法との食い違いが

問題になります。以下は、余計な括弧が必要となる例です。

3 + 4 * 5 −→ 35 ”(23 で は な い ) 二 項 メ ッ セ ー ジ が 左 か ら 右 に 実 行 さ れ
る た め ”

3 + (4 * 5) −→ 23
1 + 1/3 −→ (2/3) ”4/3 で は な い ”
1 + (1/3) −→ (4/3)
1/3 + 2/3 −→ (7/9) ”1 で は な い ”
(1/3) + (2/3) −→ 1

例。 メッセージ送信 20 + 2 * 5では、+と *という二項メッセージだけが使わ
れています。しかし Smalltalkでは +と *の間に特に優先順位の違いはありま
せん。どちらも単なる二項メッセージであって、*は +よりも優先順位が高い
わけではありません。Example 4.4のように、ここでは左端の +がまず送信さ
れ (1)、その結果に *が送信されます (2)。

Example 4.4: 20 + 2 * 5 を 分 解 す る
”二 項 メ ッ セ ー ジ 間 に 優 先 順 位 の 違 い が な い の で 、 算 術 の 約 束 と し て は * が ま ず

先 に 送 信 さ れ る べ き だ が 、 こ こ で は 左 端 の + が ま ず 評 価 さ れ る 。 ”

20 + 2 * 5
(1) 20 + 2 −→ 22
(2) 22 * 5 −→ 110

Example 4.4の結果は 30ではなく 110となります。これはもしかすると意図
したものと違うかもしれませんが、メッセージ送信の規則をそのまま適用した

結果であり、Smalltalkの簡素なモデルの代償と言えるかもしれません。正し
い (意図した)結果を得るためには括弧を使う必要があります。括弧で囲まれた
メッセージが先に評価されるため、メッセージ送信 20 + (2 * 5)は Example 4.5
に示されたような結果を返します。

Example 4.5: 20 + (2 * 5) を 分 解 す る
”括 弧 で 囲 ま れ た メ ッ セ ー ジ が 先 に 評 価 さ れ る の で * が + の 前 に 送 信 さ れ る 。 こ

れ は 正 し い 振 る 舞 い で あ る 。 ”

Figure 4.7: 括弧の有無に関わらない等価なメッセージ。

規則に基づく評価順 明示的に括弧を付けて書いた等価なメッセージ送信

aPen color: Color yellow aPen color: (Color yellow)
aPen go: 100 + 20 aPen go: (100 + 20)
aPen penSize: aPen penSize + 2 aPen penSize: ((aPen penSize) + 2)
2 factorial + 4 (2 factorial) + 4

Figure 4.8: メッセージ送信と、括弧を完全に付けた等価な式の対応例

20 + (2 * 5)
(1) (2 * 5) −→ 10
(2) 20 + 10 −→ 30

Smalltalkでは、+や *のような算術演算子に特別な優先順
位はありません。+や *も単なる二項メッセージで、*の優
先順位の方が +よりも高いということはありません。意図
した結果を得るには括弧を使ってください。

最初の規則は、単項メッセージは二項メッセージやキーワードメッセージに

優先する、というものでした。だから単項メッセージの周りに括弧を書く必要

はありません。表 4.8に、普通に書いたメッセージ送信と優先順位規則を無視
して明示的に括弧を付けて書いた等価なメッセージ送信を並べて書きました。

この表では、左右どちらのメッセージ送信も同じ実行順序となり同じ結果を返

します。

4.4 キーワードメッセージの切れ目を見つけるための

ヒント

初心者にとってしばしば問題なのは、いつ括弧が必要なのかがわかりにくいこ

とです。そこで、コンパイラがどのようにキーワードメッセージを認識するの

かを見てみましょう。

括弧か括弧なしか?

文字 [、]、(および)がコードの領域を区切るために使われます。このように囲
まれた領域の中では、:で終端された語 (キーワード)と引数の並びで、.や ;で
分断されていない最長のものがキーワードメッセージになります。つまりこの

ような並びが文字 [、]、(および)で囲まれた領域にある場合、その並びがその
領域内に局 所 的 な キーワードメッセージになります。

以下の例では、rotatedBy:magnify:smoothing: と at:put: という二つの独立した
キーワードメッセージが使われています。

aDict
at: (rotatingForm

rotateBy: angle
magnify: 2
smoothing: 1)

put: 3

文字 [、]、(および)がコードの領域を区切るために使われま
す。このように囲まれた領域の中では、: で終端された語
(キーワード)と引数の並びで、. や ;で分断されていない最
長のものがキーワードメッセージになります。つまりこの

ような並びが文字 [、]、(および) で囲まれた領域にある場
合、その並びがその領域内に局所的なキーワードメッセー

ジになります。

ヒント。 もしまだ優先順位規則がよくわからないのであれば、まずは括弧を

なるべくたくさん付けるようにして同じ優先順位を持つメッセージ同士であっ

ても確実に区別できるようにしてかまいません。

以下の例では、x isNilは単項メッセージでありキーワードメッセージ ifTrue:
よりも先に送信されるため、括弧は本来必要ありません。

(x isNil)
ifTrue:[...]

以下の例では、includes:と ifTrue:はどちらもキーワードメッセージなので、
括弧を使う必要があります。

ord := OrderedCollection new.
(ord includes: $a)
ifTrue:[...]

括弧を付けなかったとしたら、includes:ifTrue:という意味不明なメッセージがコ
レクション (ord)に送信されてしまいます!

[ ]と ( )のどちらを使うか

もしかすると、いつ鍵括弧を括弧の代わりに使うべきかよくわからないかもし

れませんね。基本原則としては、もしある式が何回 (もしかすると 0回かも)評
価されるかあらかじめわからないときにはその式を [ ]で囲みます。[expression]
は expressionからブロッククロージャ (す な わ ち オブジェクト)を生成します。
クロージャは、必要に応じて何回でも (0回の可能性も含めて)評価することで
きるオブジェクトです。[ ]の中にはメッセージ送信、変数、リテラル、代入あ
るいは別のブロックを書くことができます。

ifTrue:や ifTrue:ifFalse:のような条件分岐がブロックを必要とするのはこのた
めです。同じ原則に従い、whileTrue: メッセージのレシーバと引数も、何回評
価されるかはあらかじめ決定できないので、これらを鍵括弧で囲む必要があり

ます。

一方通常の括弧は、メッセージ送信の順序のみに影響を与えます。すなわ

ち (expression)では expressionは常 に 一回だけ評価されます。

[ x isReady ] whileTrue: [ y doSomething ] ”レ シ ー バ と 引 数 の 両 方 と も ブ ロ ッ ク で
な く て は な ら な い ”

4 timesRepeat: [ Beeper beep ] ”引 数 は 2 回 以 上 実 行 さ れ る の で ブ ロ ッ
ク で な く て は な ら な い ”

(x isReady) ifTrue: [ y doSomething ] ”レ シ ー バ は 1 回 だ け 評 価 さ れ る の で ブ
ロ ッ ク で は な い ”

4.5 式の並び

複数の式 (す な わ ち メッセージ送信や代入…)をピリオドで区切って並べると、
それらの式は順番に評価されます。ただし変数宣言とその直後の式の間にはピ

リオドは書かないので注意してください。このような式の並びにおいて、最後

の式の値が全体の値となり、これ以外の式が返した値は無視されます。ピリオ

ドは区切りであって終端子ではないということに注意してください。従って式

の並びの最後にはピリオドを付けても付けなくてもかまいません。

| box |

box := 20@30 corner: 60@90.
box containsPoint: 40@50 −→ true

4.6 カスケードメッセージ

Smalltalkには、複数のメッセージを同じレシーバに送信するためのセミコロ
ン (;)を使った記法があります。これは Smalltalk用語では カ ス ケ ー ド と呼ば
れます。

式メッセージ 1 ;メッセージ 2

Transcript show: ’Pharo is ’.
Transcript show: ’fun ’.
Transcript cr.

⇐⇒
等 価

Transcript
show: ’Pharo is’;
show: ’fun ’;
cr

カスケードメッセージのレシーバは、別のメッセージ送信の結果であるか

もしれないということに注意してください。具体的に説明すると、一連のカス

ケードの中で最初のカスケードメッセージを受け取ったオブジェクトが、カス

ケードされたすべてのメッセージのレシーバになります。以下の例では、カス

ケードの最初のメッセージは setX:setY:です。なぜなら、次にカスケードが続
いているのは setX:setY:だからです。カスケードメッセージ setX:setY:のレシー
バは Point newによって新たに作られた点オブジェクトであり、Point で は あ り
ま せ ん 。次のメッセージ isZeroは、setX:setY:と同じレシーバである点オブジェ
クトに送信されます。

Point new setX: 25 setY: 35; isZero −→ false

4.7 まとめ

• メッセージは常にレ シ ー バ と呼ばれるオブジェクトに送信されます。レ
シーバは他のメッセージ送信の結果であることもあります。

• 単項メッセージは、引数を必要とないメッセージです。
単項メッセージの形式は receiver selectorです。

• 二項メッセージは、レシーバと引数という二つのオブジェクトが関与す
るメッセージです。二項メッセージのセレクタはま た 、次の文字: +、-、
*、/、|、&、=、>、<、̃、および@からなります。二項メッセージの形式
は receiver selector argumentです。

• キーワードメッセージは、二つ以上のオブジェクトが関与し、そのセレ
クタにコロン (:)が使われているようなメッセージです。
キーワードメッセージの形式は: receiver selectorWordOne: argumentOne
wordTwo: argumentTwoです。

• 規則 1: 単項メッセージがまず送信され、次に二項、そして最後にキーワー
ドメッセージが送信されます。

• 規則 2: 括弧で囲まれたメッセージは他のメッセージよりも先に送信され
ます。

• 規則 3: 同じ種類のメッセージがある場合は、評価は左から右の順で行わ
れます。

• Smalltalkでは、+や *のような伝統的な算術演算子の優先順位はすべて
同じです。+や *も単なる二項メッセージで、*の優先順位の方が +より
も高いということはありません。異なる結果を得るには括弧を使わなけ

ればなりません。

Chapter 5

Smalltalkオブジェクトモデル

Smalltalkのプログラミングモデルは簡潔で一貫しています：すべてはオブジェ
クトで、オブジェクト間のやりとりはメッセージを送りあうことによってのみ

行われます。一方で、この簡潔さと一貫性が、他の言語に慣れたプログラマに

とって難しさの元になっています。この章では、Smalltalkのオブジェクトモ
デルの中心となる概念を示します。特に、クラスをオブジェクトとして表現す

ることが何をもたらすのかを議論します。

5.1 モデルの規則

Smalltalkのオブジェクトモデルは、統一的に適用できるいくつかの簡潔な規
則に基づいています。

Rule 1. すべてのものはオブジェクトである。

Rule 2. すべてのオブジェクトはクラスのインスタンスである。

Rule 3. すべてのクラスにはスーパークラスがある。

Rule 4. すべてはメッセージ送信で起こる。

Rule 5. メソッド探索は継承の連鎖をたどっていく。

では、これらの規則を一つ一つ詳細に見ていきましょう。

5.2 すべてのものはオブジェクトである

「すべてのものはオブジェクトである」という信念は、感染力が高いものです。

ほんの短い間でも Smalltalkを使えば、この規則がいかに物事を簡単にするか

に、驚くことでしょう。例えば、整数も本物のオブジェクトで、他のあらゆる

オブジェクトと同様、メッセージを送ることができます。

3 + 4 −→ 7 ”’+ 4’を 3に 送 る こ と で , 結 果 と し て 7が 得 ら れ ま す ”
20 factorial −→ 2432902008176640000 ”factorialを 送 る と , 大 き な 数 が 得 ら れ

ま す ”

20 factorialの内部表現と 7の内部表現は実際には異なっていますが、それら
はどちらもオブジェクトなので、コード上では— factorialの実装すら—そうし
たことを知る必要はありません。

また、この規則によって次の最も重要な帰結が生じます。

クラスもオブジェクトである。

もっと言えば、クラスはセカンドクラスオブジェクトではありません。クラス

は正真正銘のファーストクラスオブジェクトで、メッセージを送ったり、イン

スペクトしたりできます。つまり Pharoは本当の意味でリフレクション可能な
システムだということです。このことが開発者に強力な表現力を与えています。

Smalltalkの実装の深層部には、3種類のオブジェクトがあります。(1)参
照渡しされるインスタンス変数を持つ、通常のオブジェクト、(2)値渡しされ
る小 さ な 整 数 (SmallInteger)、(3)配列のように連続したメモリ領域を持つイン
デックス付きオブジェクトです。Smalltalkの美しさは、これら 3種類のオブ
ジェクトの違いについて通常は気にする必要がないというところにあります。

5.3 すべてのオブジェクトはクラスのインスタンスで

ある

すべてのオブジェクトはクラスを持っています。オブジェクトに classメッセー
ジを送るとわかります。

1 class −→ SmallInteger
20 factorial class −→ LargePositiveInteger
’hello’ class −→ ByteString
#(1 2 3) class −→ Array
(4@5) class −→ Point
Object new class −→ Object

クラスはインスタンス変数を通じてインスタンスの構 造 を定義します。そ

してメソッドよってインスタンスの振 る 舞 い を定義します。各メソッドの名前

はセ レ ク タ と呼ばれます。名前はそのクラスの中でユニークです。

ク ラ ス は オ ブ ジ ェ ク ト で、すべてのオブジェクトはクラスのインスタンス

であることから、クラスもまた、あるクラスのインスタンスということになり

ます。クラスをインスタンスとするクラスはメ タ ク ラ ス と呼ばれます。クラス

を定義するたびに、システムは自動的にメタクラスを生成します。メタクラス

は、インスタンスとしてのクラスの構造と振る舞いを定義します。99%は、メ
タクラスについて考える必要はないでしょう。当面は無視しても大丈夫です。

(第 13章でメタクラスについて詳しく見ていきます。)

インスタンス変数

Smalltalkでのインスタンス変数は、そのイ ン ス タ ン ス にとってのプライベー
トな変数です。これは、Javaや C++のインスタンス変数 (“フィールド”とか
“メンバ変数”とも呼ばれています)が同じクラスの他のインスタンスからのア
クセスを許しているのとは対照的です。Javaや C++ではオブジェクトのカ プ
セ ル 化 の 境 界 がクラスであるのに対し、Smalltalkではインスタンスになって
いると言えます。

Smalltalkでは、“アクセサメソッド”を定義しない限り、同じクラスの二
つのインスタンスはお互いのインスタンス変数にアクセスすることができま

せん。他のオブジェクトのインスタンス変数に直接アクセスするための言語構

文がないのです。(実際にはリフレクションという仕組みによって、他のオブ
ジェクトからインスタンス変数の値にアクセスすることもできます。こうした

メタプログラミングによって、他のオブジェクトの中身を見ることを目的にし

たツールであるオブジェクトインスペクタのようなツールを記述することがで

きます。)
インスタンス変数は、その変数を定義しているクラスやサブクラスのイン

スタンスメソッドから、名前でアクセスすることができます。このことから、

Smalltalkのインスタンス変数はC++や Javaのプ ロ テ ク ト 変数に似ていると言
えます。しかしながら、やはりプライベートと見た方が好ましいでしょう。と

いうのも、インスタンス変数をサブクラスからアクセスすることは、Smalltalk
では良くないスタイルと考えられているからです。

例

Point»dist: (Method 5.1)メソッドは、レシーバと引数で与えた点 (aPoint)との
距離を計算します。メソッド内から、レシーバのインスタンス変数 xと yに、直
接アクセスしています。しかし、もう一つの点である aPointのインスタンス
変数については、xメッセージや yメッセージを送ってアクセスしなければなり
ません。

Method 5.1: the distance between two points
Point»dist: aPoint

”aPointと レ シ ー バ の 距 離 を 答 え る 。 ”
| dx dy |
dx := aPoint x - x.
dy := aPoint y - y.
↑ ((dx * dx) + (dy * dy)) sqrt

1@1 dist: 4@5 −→ 5.0

クラスでのカプセル化ではなく、インスタンスでのカプセル化を選択する決

め手となったのは、インスタンスでのカプセル化が同じ抽象化の異なる実装を

共存させることができるからです。例えば、Point>>dist:メソッドは、引数 aPoint
がレシーバと同じクラスなのかどうか知る必要もありませんし、気にする必要

もありません。引数オブジェクトは極座標かもしれませんし、データベースの

レコードかもしれませんし、分散システムの他の計算機上にあるかもしれませ

ん。xメッセージや yメッセージに応えてくれる限り、Method 5.1のコードは
ちゃんと動作するでしょう。

メソッド

すべてのメソッドはパブリックです。1メソッドは、その目的に応じてプロト
コルにグループ分けされています。慣例により、よく使われるプロトコルとい

うものがあります。例えば、accessingはアクセサメソッド、initializationはオブ
ジェクト初期化のメソッドという具合になっています。privateプロトコルは、
外から見るべきでないメソッドをグループ化するのに使われます。しかし、そ

ういった “private”メソッドがあるからといって、実際にメッセージを送るこ
とを防ぐものではありません。

メソッドはそのオブジェクトのすべてのインスタンス変数にアクセスでき

ます。Smalltalk開発者の中には、アクセサを通してのみインスタンス変数にア
クセスすることを好む人もいます。このプラクティスは価値あるものですが、

クラスのインターフェースを乱雑にする側面があり、下手をすると外部の世界

にプライベートな状態を晒すことにもなります。

インスタンス側とクラス側

クラスはオブジェクトなので、クラス自身のインスタンス変数やメソッドを持

つことができます。これらはク ラ ス イ ン ス タ ン ス 変 数 やク ラ ス メ ソ ッ ド と呼ば

れていますが、実際のところ普通のインスタンス変数やメソッドと何も違いま

せん。クラスインスタンス変数は単にメタクラスで定義されたインスタンス変

1正確には、ほとんどすべて、です。Pharoでは、pvtで始まるセレクタを持つメソッドはプライ
ベートで、pvtメッセージは selfにの み 送ることができます。しかし、pvtメソッドはあまり使われ
ていません。

数にすぎませんし、クラスメソッドは単にメタクラスで定義されたメソッドに

すぎません。

クラスはメタクラスのインスタンスですが、クラスとそのメタクラスは別々

のクラスです。しかし、プログラマであるあなたにはおよそ関係ない話です。

プログラマが気にかけることは、オブジェクトの振る舞いやそれを生成するク

ラスを定義することなのです。

Figure 5.1: クラスとそのメタクラスをブラウズする。

こうしたことから、ブラウザは、クラスとメタクラスの両方を、一つのも

のの二つの側面として見ることができるようになっています。図 5.1に示され
ている、“インスタンス側”とクラス側”です。図は Colorクラスをブラウズし
ているところですが、 instanceボタンをクリックすると、Colorのインスタンス
(例えば青色)にメッセージが送られたときに実行されるメソッド一覧が表示さ
れます。 classボタンを押すと、Color classクラスのブラウズになり、Colorクラ
ス自体にメッセージが送られたときに実行されるメソッドを見ることになりま

す。例えば、Color blueは Colorクラスに blueメッセージを送ります。というこ
とは、この blueメッセージは Colorのクラス側に定義されているはずです。イン
スタンス側ではありません。

aColor := Color blue. ”ク ラ ス 側 の blueメ ソ ッ ド ”
aColor −→ Color blue
aColor red −→ 0.0 ”イ ン ス タ ン ス 側 の ア ク セ サ メ ソ ッ ド red”
aColor blue −→ 1.0 ”イ ン ス タ ン ス 側 の ア ク セ サ メ ソ ッ ド blue”

新たなクラスの定義は、インスタンス側で提供されるテンプレートを埋め

ることで行えます。このテンプレートをアクセプトすると、システムはクラス

だけでなく、対応するメタクラスも生成します。 classボタンをクリックする
とメタクラスをブラウズできます。メタクラス生成テンプレートで編集するの

は、クラスインスタンス変数のリストだけです。

いったんクラスが生成されると、 instanceボタンのクリックで、そのクラス
(およびそのサブクラス)のインスタンス用のメソッドを編集したりブラウズし
たりできます。例えば、図 5.1を見ると、Colorクラスのインスタンスに hueメ
ソッドが定義されています。対して、 classボタンを押すと、メタクラス (この
場合、Color class)のメソッドのブラウズ、定義に切り替わります

クラスメソッド

クラスメソッドは非常に便利なものです。Color classをブラウズすると良い例
が見つかります。メソッドには大きく分けて二種類あります。Color class»blue
のように、そのクラスのインスタンスを作るものと、Color class»showColorCube
のように、ユーティリティ的な機能を提供するものです。他の用途に使われて

いるクラスメソッドを見かけることもあるでしょうが、この二つが典型的な用

例です。

クラス側にユーティリティ的なメソッドを置くと、余計なインスタンスを

生成することなく実行できるので便利です。実際、そういったユーティリティ

メソッドの多くは、簡単に実行できるコメントを含んでいます。

Color class»showColorCubeメ ソ ッ ド を ブ ラ ウ ズ し て , コ メ ン ト ”Color
showColorCube”の 引 用 符 の 内 側 を ダ ブ ル ク リ ッ ク し て 、 cmd–d を 押 し て み て
く だ さ い 。

このメソッドが実行されたことがわかるでしょう。(元に戻すには
World . restore display (r)を実行してください。)

Javaや C++に慣れている人には、クラスメソッドは静的メソッドに似てい
るように見えるかもしれません。しかし、Smalltalkの一貫性の点で、いくぶ
んの違いがあります。Javaの静的メソッドは単に静的に名前解決される手続き
に過ぎないのですが、Smalltalkのクラスメソッドは動的に探索されるメソッ
ドです。つまり、Smalltalkではクラスメソッドを継承したりオーバーライド
したり superに送信したりできますが、Javaの静的メソッドではできません。

クラスインスタンス変数

通常のインスタンス変数では、あるクラスのインスタンスはすべて同じ名前の

インスタンス変数を持っていて、サブクラスのインスタンスもそれらの名前を

継承します。しかし、各インスタンスのインスタンス変数はそれぞれ個別の値

を持ちます。インスタンス変数もまったく同様です。各クラスはそれぞれ個別

のクラスインスタンス変数を持っています。サブクラスはクラスインスタンス

変数を継承しますが、サ ブ ク ラ ス は プ ラ イ ベ ー ト な コ ピ ー と し て ク ラ ス イ ン ス

タ ン ス 変 数 を 保 持 す る こ と に な り ま す 。 オブジェクトがインスタンス変数を共

有しないのと同様に、クラスとサブクラスもクラスインスタンス変数を共有し

ません。

あるクラスが生成したインスタンスの数を管理するために countというクラ
スインスタンス変数を使うことはできますが、そのサブクラスはどれも自分自

身の count変数を持つため、サブクラスによって生成されるインスタンスの数
は別々に数えられることになります。

例: クラスインスタンス変数はサブクラスと共有されない。 二つのクラス

Dog(犬)と Hyena(ハイエナ)を定義します。Hyenaは Dogからクラスインスタン
ス変数 countを継承します。

Class 5.2: 犬 ク ラ ス と ハ イ エ ナ ク ラ ス
Object subclass: #Dog
instanceVariableNames: ’’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE-CIV’

Dog class
instanceVariableNames: ’count’

Dog subclass: #Hyena
instanceVariableNames: ’’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE-CIV’

Dogの countを 0に初期化するクラスメソッドと、新しいインスタンスが生成
されたらそれに 1を足すクラスメソッドを定義しましょう：

Method 5.3: 新 し い 犬 の 数 を 数 え る
Dog class»initialize
super initialize.
count := 0.

Dog class»new
count := count +1.
↑ super new

Dog class»count
↑ count

これで、新しいDogのインスタンス(犬)が生成されるたびに、countに 1が足
されていきます。Hyena(ハイエナ)も同様ですが、別々に数えられていきます：

Dog initialize.
Hyena initialize.
Dog count −→ 0
Hyena count −→ 0
Dog new.
Dog count −→ 1
Dog new.
Dog count −→ 2
Hyena new.
Hyena count −→ 1

インスタンス変数がインスタンスごとにプライベートなのと同様に、クラ

スインスタンス変数もクラスごとにプライベートだということにも注意してく

ださい。クラスとそのインスタンスは別々のオブジェクトなので、以下のこと

が言えます。

クラスはそのインスタンスのインスタンス変数にアクセス

できない。

インスタンスは、そのクラスのクラスインスタンス変数に

アクセスできない。

このことから、インスタンスを初期化するメソッドは常にインスタンス側で定

義しなければなりません—クラス側からはインスタンス変数にアクセスでき
ないので、初期化できないのです。クラスができることは、インスタンスを新

しく生成するときに、アクセサなどを用いて初期化用のメッセージを送ること

だけです。

同様にインスタンスがクラスインスタンス変数にアクセスするには、クラ

スにアクセサメッセージを送り間接的に行うしかありません。

Java にはクラスインスタンス変数に相当するものはありません。Java と
C++の静的変数は、Smalltalkではどちらかというと 5.7節で説明するクラス
変数に似ています。すべてのサブクラスとインスタンスが同一の静的変数を共

有するというものだからです。

例: シングルトンを定義する。 シングルトンパターン2はクラスインスタンス
変数とクラスメソッドを使う典型例です。 ctWebServerクラスを実装してみ
ましょう。シングルトンパターンを使い、WebServerクラスには一つしかイン
スタンスができないようにします。

2Sherman R. Alpert, Kyle Brown and Bobby Woolf, The Design Patterns Smalltalk Companion.
Addison Wesley, 1998, ISBN 0–201–18462–1.

ブラウザの instanceボタンをクリックして、WebServerクラスを以下のよう
に定義します。(Class 5.4)

Class 5.4: シ ン グ ル ト ン ク ラ ス
Object subclass: #WebServer
instanceVariableNames: ’sessions’
classVariableNames: ’’
poolDictionaries: ’’
category: ’Web’

そして、classボタンをクリックして、クラス側でクラスインスタンス変数
uniqueInstanceを追加します。

Class 5.5: シ ン グ ル ト ン ク ラ ス の ク ラ ス 側
WebServer class
instanceVariableNames: ’uniqueInstance’

この結果、WebServerクラスは、新たなインスタンス変数を追加で持つこと
になります。superclassや methodDictといった変数もスーパークラスから継承し
ているからです。

さて、これでメソッド uniqueInstanceを定義できます。Method 5.6を見てく
ださい。このメソッドではまず、uniqueInstanceが初期化されているかをチェッ
クします。初期化されていない場合は、インスタンスを生成し、クラスインス

タンス変数 uniqueInstanceに代入します。最後に、uniqueInstanceの値を返してい
ます。uniqueInstanceはクラスインスタンス変数なので、このクラスメソッド内
で直接アクセスすることができます。

Method 5.6: uniqueInstance (ク ラ ス 側 )
WebServer class»uniqueInstance

uniqueInstance ifNil: [uniqueInstance := self new].
↑ uniqueInstance

Webserver uniqueInstanceが最初に実行されるときには、WebServerクラスのイ
ンスタンスが生成され、変数 uniqueInstanceに代入されます。以後は、新しいイ
ンスタンスを生成することなしに、前回生成されたインスタンスが返されます。

Method 5.6の条件分岐の内側にあるインスタンス生成のコードが self newと
いう形であって、WebServer newではないことに留意してください。何が違うの
でしょうか? uniqueInstanceメソッドはWebServer classに定義されているので、同
じことだと思うかもしれません。WebServerのサブクラスが作られるまでは確か
に同じなのです。しかし、もしWebServerのサブクラスとしてReliableWebServer
が作られ、uniqueInstanceメソッドを継承していたらどうなるでしょうか。普通
は ReliableWebServer uniqueInstanceの結果は ReliableWebServer のインスタンス
となってほしいはずです。こうしたときに、selfを使うことで、そのようにす
ることができます。self はメッセージを受け取るクラス自身だからです。な



88 Smalltalk オ ブ ジ ェ ク ト モ デ ル

お、WebServerと ReliableWebServerはそれぞれ自分のクラスインスタンス変
数 uniqueInstanceを持っていることにも留意してください。それぞれの変数は、
もちろん、別々の値になります。

5.4 すべてのクラスにはスーパークラスがある

Smalltalkでは、クラスは単一のス ー パ ー ク ラ ス から、振る舞いおよび構造の
記述を継承します。つまり Smalltalkは単一継承です。

SmallInteger superclass −→ Integer
Integer superclass −→ Number
Number superclass −→ Magnitude
Magnitude superclass −→ Object
Object superclass −→ ProtoObject
ProtoObject superclass −→ nil

伝統的には、Smalltalkの継承階層のルートはObjectクラスです。(なぜなら、
すべてのものはオブジェクトだからです。) Pharoでは、ルートは ProtoObject
と呼ばれるクラスになっています。しかし、通常はこのクラスを気にする必要

はありません。 ProtoObjectはすべてのオブジェクトが持ってい な け れ ば な ら な
い 最低限のメッセージをカプセル化しています。しかしながら、たいていのク

ラスは Objectからの継承となっています。Objectはほとんどすべてのオブジェ
クトが理解し対応すべきメッセージを追加でたくさん定義しています。特別な

理由がない限り、アプリケーションのクラスを生成する際には、Objectや、そ
のサブクラスを継承すべきです。

新 し い ク ラ ス を 生 成 す る に は 、 既 存 の ク ラ ス に subclass:
instanceVariableNames: ... と い う メ ッ セ ー ジ を 送 り ま す 。 他 に も い く つ か
ク ラ ス を 生 成 す る た め の メ ソ ッ ド が あ り ま す 。 Kernel-Classes .Class . subclass
creation プ ロ ト コ ル を 見 て 、 ど の よ う な メ ソ ッ ド が あ る か 見 て み て く だ さ い 。

Pharoには多重継承はありませんが、お互いに関係のないクラスで同じ振
る舞いを共有するためのト レ イ ト というメカニズムをサポートしています。ト

レイトはメソッドの集合で、継承関係にないクラス間で再利用されます。トレ

イトを使うと、同じコードを繰り返し定義することなく、別々のクラス間で共

有することができます。

抽象メソッドと抽象クラス

抽象クラスは、インスタンスを生成するためではなく、サブクラスを定義する

ために存在するクラスです。抽象クラスは通常は不完全なもので、メソッドの

中身すべてが実装されてはいるわけではありません。“空の”メソッド—他の
メソッドがそのメソッドがあることを想定しているのに、実装が定義されてい

ないメソッド—を抽象メソッドと呼びます。

Smalltalkには、抽象メソッドや抽象クラスを示す特別な構文はありません。
コード規約として、抽象メソッドの本文にself subclassResponsibilityと書きます。
これはいわゆる “マーカーメソッド”で、サブクラスでそのメソッドの具体的
な実装を定義する責任があることを示しています。self subclassResponsibilityを
書いたメソッドは必ずオーバーライドしなければならず、直接実行してはなら

ないものです。もしオーバーライドし忘れたら、メソッド実行の結果、例外が

発生してしまいます。

抽象メソッドを持つクラスが抽象クラスです。実際には抽象クラスのイン

スタンスを生成しないようにする仕組みはありません。抽象メソッドが実行さ

れるまでは、普通に動作します。

例: Magnitudeクラス

Magnitude は互いを比較するクラスを定義するのに役立つ抽象クラスです。
Magnitudeのサブクラスは<および=、hashの三つのメソッドを定義しなければ
なりません。Magnitudeでは、それらのメッセージを使って>および>=、<=、max:
、min:、between:and:等のメソッドを定義しています。こうしたメソッドはサブ
クラスで継承されて使われます。抽象メソッド<は、Method 5.7のようになっ
ています。

Method 5.7: Magnitude»<
Magnitude»< aMagnitude

”レ シ ー バ が 引 数 よ り 小 さ い か ど う か を 答 え る 。 ”
↑self subclassResponsibility

対照的に、>=は具象メソッドで、<を使って定義されています。

Method 5.8: Magnitude»>=
>= aMagnitude

”レ シ ー バ が 引 数 と 等 し い か そ れ 以 上 か ど う か を 答 え る 。 ”
↑(self < aMagnitude) not

他の比較メソッドも同様です。

Characterは Magnitudeのサブクラスで、subclassResponsibilityとなっていた<
メソッドをオーバーライドし、メソッドを再定義しています (Method 5.9 参
照)。Characterは=メソッドおよび hashメソッドも定義していて、>=および<=、
∼=等のメソッドを Magnitudeから継承しています。

Method 5.9: Character»<
Character»< aCharacter

”レ シ ー バ が aCharacterよ り 小 さ け れ ば 、 trueと 答 え る 。 ”
↑self asciiValue < aCharacter asciiValue

トレイト

ト レ イ ト は、継承なしにクラスの振る舞いに取り込むことができるメソッド

の集合です。これによって、単一継承の容易さを保ちながら、複数のクラスに

とって有用なメソッドを共通化することができます。

新しいトレイトを定義する方法は、単にサブクラス定義のテンプレートを、

Trait用のメッセージに置き換えるだけです。

Class 5.10: 新 し い ト レ イ ト を 定 義 す る
Trait named: #TAuthor
uses: { }
category: ’PBE-LightsOut’

ここでは、PBE-LightsOutカテゴリに TAuthorトレイトを定義しています。このト
レイトでは既存のトレイトを使 っ て いません。一般には、uses:の引数として、
他のトレイトのト レ イ ト 合 成 式 を使うことができます。ここでは単に空の配列

にしています。

トレイトはメソッドを持つことができますが、インスタンス変数は持ちま

せん。例として authorメソッドをクラス階層のどこに位置するかに関係なくい
ろいろなクラスに追加することを考えてみましょう。以下のように書きます。

Method 5.11: author メ ソ ッ ド
TAuthor»author

”著 者 の 頭 文 字 を 返 す 。 ”
↑ ’on’ ”oscar nierstraszの 頭 文 字 ”

クラスに既存のスーパークラスがあっても、このトレイトを使うことができま

す。第 2 章で定義した LOGameクラスを例にします。単に、LOGameクラスを
定義するテンプレートを書き変え、uses:の引数で TAuthorを使うように指定す
るだけです。

Class 5.12: ト レ イ ト を 使 う
BorderedMorph subclass: #LOGame
uses: TAuthor
instanceVariableNames: ’cells’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE-LightsOut’

LOGameのインスタンスを生成すると、期待通り authorメッセージに答えて
くれるようになります。

LOGame new author −→ ’on’

トレイト合成式では複数のトレイトを+演算子でつなげることができます。
もし複数のトレイトで同じ名前のメソッドを定義しているなどの不整合があっ

ても、-演算子でそうしたメソッドを明示的に除外したり、新規のクラスやト
レイト側で再定義したりすることで、解消することができます。@演算子でメ
ソッドにエ イ リ ア ス として別名を付けることも可能です。

トレイトはシステムの中核部分で使われています。Behaviorクラスが良い例
です。

Class 5.13: Behaviorの 定 義 で 使 わ れ て い る ト レ イ ト
Object subclass: #Behavior
uses: TPureBehavior @ {#basicAddTraitSelector:withMethod:->

#addTraitSelector:withMethod:}
instanceVariableNames: ’superclass methodDict format’
classVariableNames: ’ObsoleteSubclasses’
poolDictionaries: ’’
category: ’Kernel-Classes’

こ こ で は 、TPureBehaviorの addTraitSelector:withMethod:メ ソ ッ ド に 、

basicAddTraitSelector:withMethod:として別名を付けています。現在、トレイ
トをサポートする機能がブラウザに追加されているところです。

5.5 すべてはメッセージ送信で起こる

この規則は、Smalltalkプログラミングの本質を捉えています。
手続き的プログラミングでは、手続きが呼ばれたときにどのコードが使わ

れるかは、呼び出し側が決めます。呼び出し側が、名前で、静 的 に 、手続きや

関数を選んで実行します。

オブジェクト指向プログラミングでは、“メソッドを呼び出し” ま せ ん 。“メッ
セージを送信”します。この用語の違いは重要です。オブジェクトにはそれぞ
れの責任があります。オブジェクトが何をすべきか、その手続き指 示 す る よう

なことはしません。そうではなく、オブジェクトにメッセージを送ることで、

何かをするようにお 願 い するのです。メッセージはコードではあ り ま せ ん 。単

なる名前と引数の組なのです。メッセージを受けたレシーバは、頼まれたこと

を実行するため、メ ソ ッ ド を自ら選択して、どう対応するのかを決めます。オ

ブジェクトが異なれば、同じメッセージに対して反応するためのメソッドも異

なっていてよいのです。つまりメソッドはメッセージを受け取ったときに動 的

に 選択されることになります。

3 + 4 −→ 7 ”整 数 「 3」 に メ ッ セ ー ジ 「 +」 と 引 数 「 4」 を 送 信 す る ”
(1@2) + 4 −→ 5@6 ”点 「 (1@2)」 に メ ッ セ ー ジ 「 +」 と 引 数 「 4」 を 送 信 す る ”

結果として同 じ メ ッ セ ー ジ を異なるオブジェクトに送信することができ、オブ

ジェクトはそのメッセージに応えるために各 自 で メ ソ ッ ド を持つことができま

す。+ 4というメッセージにどう反応するのかを SmallInteger 3や Point 1@2に直
接指示しないのです。それぞれが+メソッドを持っていて、そのメソッドに従っ
て+ 4に反応するのです。

Smalltalkのメッセージ送信モデルの帰結として言えることがあります。オ
ブジェクトは非常に小さなメソッドのみを持ち、他のオブジェクトに仕事を移

譲する傾向となり、巨大で手続き的なメソッドを保持し、過大な責任をかかえ

ることを避けるようになります。ジョセフ・ペルリン (Joseph Pelrine)はこの
原則を次のように簡潔に表現しています。

誰かに任せられる仕事を自分でやらないこと。

多くのオブジェクト指向言語は、オブジェクトに対して静的・動的両方の

操作方法を提供していますが、Smalltalkには動的なメッセージ送信しかあり
ません。静的な操作の代わりに、例えばクラスがオブジェクトとなっており、

単にクラスにメッセージを送ればよいのです。

Smalltalkでは、ほ と ん ど すべてのことはメッセージ送信で起こります。し
かし時にはメッセージ送信ではない処理も発生します。

• 変 数 宣 言 はメッセージ送信ではありません。実際、変数宣言は実行可能
ですらありません。変数宣言は、オブジェクトの参照を格納する領域を

確保する意味しか持ちません。

• 代 入 はメッセージ送信ではありません。変数への代入はそれを定義した
スコープ内で名前を束縛します。

• リ タ ー ン はメッセージ送信ではありません。リターンは計算結果をメッ
セージの送り手に返します。

• プ リ ミ テ ィ ブ はメッセージ送信ではありません。プリミティブは仮想マ
シン側で実装されています。

これらのわずかな例外を除き、ほとんどすべてはメッセージ送信で起こり

ます。特に、Smalltalkには “パブリックなフィールド”はないので、他のオブ
ジェクトのインスタンス変数の値を変えるには、メッセージを送ってそのオブ

ジェクト自身のフィールドを更新するようお願いするしかありません。もちろ

ん、すべてのインスタンス変数についてセッターやゲッターを定義するのは、

オブジェクト指向として良いスタイルとは言えません。ジョセフ・ペルリンは

次のようにも言っています。

自身のデータを他人にいじらせないこと。

5.6 メソッド探索は継承の連鎖をたどっていく

オブジェクトがメッセージを受け取ると、何が起こるのでしょう?



メ ソ ッ ド 探 索 は 継 承 の 連 鎖 を た ど っ て い く 93

その過程はとてもシンプルです。メッセージを受けたレシーバのクラスは、

メッセージを処理するのに必要なメソッドを探索します。もしクラスがそのメ

ソッドを持っていなければ、自身のスーパークラスに尋ねます。継承の連鎖を

たどって、メソッドを探していくのです。メソッドが見つかれば、渡した引数

の値はそのメソッドの引数に束縛され、仮想マシンがメソッドを実行します。

メソッド探索は基本的には簡単なことなのですが、いくつか注意しなけれ

ばならない点もあります。

• メ ソ ッ ド が 明 示 的 に 値 を 返 し て い な い 場 合 は 、 ど う な る の で し ょ う か ?

• あ る ク ラ ス が ス ー パ ー ク ラ ス の メ ソ ッ ド を 再 実 装 し て い る 場 合 は 、 ど う
な る の で し ょ う か ?

• self 送 信 と super 送 信 は 何 が 違 う の で し ょ う か ?

• メ ソ ッ ド が 見 つ か ら な い 場 合 に は 、 ど う な る の で し ょ う か ?

ここでは、メソッド探索の概念的な規則を示します。仮想マシンを実装す

る人はありとあらゆるトリックや最適化を駆使してメソッド探索を高速化する

のですが、それは彼らに任せておきましょう。ここで示す規則から外れている

とはわからないように、やっているはずです。

まず、探索の基本戦略から見ていきましょう。その上で、上記の一連の疑

問について考察します。

メソッド探索

EllipseMorphのインスタンスを生成してみます。

anEllipse := EllipseMorph new.

ここで、このオブジェクトに defaultColorメッセージを送ると、Color yellowとい
う結果が得られます。

anEllipse defaultColor −→ Color yellow

EllipseMorphクラスは defaultColorを実装しているので、適切なメソッドが直ちに
見つかります。

Method 5.14: レ シ ー バ の ク ラ ス 自 体 に 実 装 さ れ た メ ソ ッ ド
EllipseMorph»defaultColor

”レ シ ー バ の デ フ ォ ル ト の 色 /塗 り 方 を 答 え る ”
↑ Color yellow

一方、anEllipseに openInWorldメッセージを送ると、メソッドはすぐには見つ
かりません。anEllipseクラスは openInWorldを実装していないからです。そのた



94 Smalltalk オ ブ ジ ェ ク ト モ デ ル

め、スーパークラスの BorderedMorphを探索していきます。探索は openInWorld
メソッドが Morphクラスで見つかるまで続きます。

Method 5.15: 継 承 さ れ た メ ソ ッ ド
Morph»openInWorld

”こ の モ ー フ を ワ ー ル ド に 追 加 す る 。 ”

self openInWorld: self currentWorld

Figure 5.2: メソッド探索は継承の連鎖をたどっていく

selfを返す

EllipseMorph»defaultColor (Method 5.14) は Color yellowを明示的に返している一
方で、Morph»openInWorld (Method 5.15)には戻り値を示すものがないことに注
目してください。

実際にはメソッドは常 に 値を返します。もちろん、値というのはオブジェ

クトです。答える値はメソッド中の↑で決まることもありますが、↑を実行する
ことなくメソッドの終わりまで来てしまったとしても、やはりメソッドは値を

返します。その場合、メソッドは、そのメッセージを受け取ったオブジェクト

自身を返すのです。これを、“selfを返す”と言います。Smalltalkでは擬似変
数 selfはメッセージを受けたレシーバそのものを指します。Javaの thisのよう
なものです。

つまり、Method 5.15はMethod 5.16と同じことです。

Method 5.16: 明 示 的 に self を 返 す
Morph»openInWorld

”こ の モ ー フ を ワ ー ル ド に 追 加 す る 。 ”

self openInWorld: self currentWorld
↑ self ”不 要 な の で 、 通 常 は わ ざ わ ざ 書 か な い こ と ！ ”

なぜ上のように↑ selfと書かない方がよいのかというと、明示的に値を返す
書き方は、メッセージを送った側に何か意味あるものを返していることを示す

ものだからです。selfを明示的に返す場合、センダ側がその戻り値を使うこと
を期待しているといった意味になるのです。上記のコードではそういう意図は

ないので、selfを返すことを明示しない方がよいでしょう。

これは Smalltalkでは日常的なイディオムで、ケント・ベックは “興味を引
くような戻り値 (Interesting return value)”3 と呼んでいます。

センダにその値を使ってほしいときにだけ、戻り値を示す

こと。

オーバーライドと拡張

図 5.2の EllipseMorphクラスの階層を見ると、MorphクラスとEllipseMorphクラス
のどちらも defaultColorを実装しています。実際、Morph new openInWorldで新し
いモーフを開くと、青い色になっています。楕円の場合には黄色になります。

これを、EllipseMorphが Morphから継承した defaultColorメソッドをオ ー バ ー
ラ イ ド している、と言います。継承したメソッドは、anEllipseからは見えなく
なります。

継承したメソッドをオーバーライドするのではなく、機能を追加して拡 張 し

たいこともあります。つまり、オーバーライドされるメソッドに、サブクラス

で新たな機能を付 け 加 え て 実行したいということです。Smalltalkでは、super
に対するメッセージ送信でこれを実現しています。単一継承をサポートする多

くのオブジェクト指向言語で典型的な方法です。

initializeメソッドは、この仕組みの最も重要な例です。あるクラスの新しい
インスタンスを初期化するときには、スーパークラスから継承したインスタン

ス変数も必ず初期化しなければなりません。しかし、その初期の仕方は、継承

の連鎖の中で各スーパークラスの initializeメソッドに既に記述されています。サ
ブクラスは継承したインスタンス変数の初期化には本来関わりがありません。

以上のことから、initializeメソッドを実装するときには、まず super initialize
を送信してから、他の変数の初期化 を書くようにするのが良いプラクティス

です。

3Kent Beck, Smalltalk Best Practice Patterns. Prentice-Hall, 1997.



96 Smalltalk オ ブ ジ ェ ク ト モ デ ル

Method 5.17: Super initialize
BorderedMorph»initialize

”レ シ ー バ の 状 態 を 初 期 化 す る ”
super initialize.
self borderInitialize

initializeメソッドは、super initializeで始めること。

self、superへのメッセージ送信

superへのメッセージ送信は、オーバーライドされてしまう振る舞いをサブク
ラス側で組み入れるのに使われます。継承されたものであろうとなかろうと、

複数のメソッドを組み合わせるには、通常 selfへのメッセージ送信を用います。

self 送信と super 送信の違いは何でしょうか? self と同様に、super はメッ
セージを受けたレシーバ自身を表します。違うのは、メソッド探索だけです。

super送信があると、レシーバのクラスから探索を始めるのではなく、superへ
の送信を行っているメソッドのクラスのスーパークラスから探索を始めます。

superはスーパークラスで は な い ことに注意してください! スーパークラス
と思うのは、よく見られる勘違いです。また、レシーバのスーパークラスから

探索を始めるというのも、間違いです。どういうことか、以下に例を示します。

constructorStringというメッセージを例にしましょう。これは、任意のモーフ
に送信することができます。

anEllipse constructorString −→ ’(EllipseMorph newBounds: (0@0 corner: 50@40)
color: Color yellow)’

戻り値は、評価することでモーフを復元できる文字列です。

self 送信と super 送信をどう組み合わせれば、この文字列が得られるの
でしょうか? まず、anEllipse constructorStringが実行されると、Morphクラスの
constructorStringが見つかります。図 5.3を見てください。

Method 5.18: self 送 信
Morph»constructorString
↑ String streamContents: [:s | self printConstructorOn: s indent: 0].

Morph»constructorString メソッドは printConstructorOn:indent:を self に送って
います。このメッセージもまた探索されますが、探索は EllipseMorph か
ら始まって、Morphで見つかります。そのメソッドの内部で、さらに
printConstructorOn:indent:nodeDict:が selfに送られます。それが fullPrintOn:を selfに
送ります。fullPrintOn:はEllipseMorphから探索を始め、fullPrintOn:がBorderedMorph
で見つかります。(もう一度、図 5.3を見てください)。重要なのは、selfへの送

Figure 5.3: self送信と super送信

信ではレシーバのクラス、この場合には anEllipseのクラスから、メソッド探索
が始まるということです。

self送信は、動 的 な メソッド探索を、レシーバのクラスから
開始します。

Method 5.19: super 送 信 と self 送 信 を 組 み 合 わ せ る

BorderedMorph»fullPrintOn: aStream
aStream nextPutAll: ’(’.
super fullPrintOn: aStream.
aStream nextPutAll: ’) setBorderWidth: ’; print: borderWidth;
nextPutAll: ’ borderColor: ’ , (self colorString: borderColor)

一方、BorderedMorph»fullPrintOn:では、superへメッセージ送信をして、スーパー
クラスから継承した fullPrintOn:の振る舞いを拡張しています。superへの送信な
ので、メソッド探索は super送信を行うクラスのスーパークラスである Morph
クラスから始まります。すると、Morph»fullPrintOn:が即座に見つかるので、それ
が評価されます。

super送信の探索はレシーバのスーパークラスから開始しているわけではな
い、ということに注意してください。もしレシーバのスーパークラスから探索

を開始すると、BorderedMorphから探索を始めるということになり、無限ループ
になってしまいます!

super送信は、静 的 な メソッド探索を、super送信を行って
いるメソッドを定義しているクラスのスーパークラスから

開始します。

super送信の解説を踏まえ、図 5.3を見ると、superは静的束縛だということ
がわかります。super送信をしているコードがどのクラスにあるかにより、すべ
て決まります。対照的に、self送信は動的です。今実行しているメッセージの
レシーバを表しています。つまり、selfに送られたす べ て の メッセージは、そ
のレシーバのクラスから探索を始めます。

理解されないメッセージ

探索してもメソッドが見つからなかったら、どうなるのでしょうか?
楕円に fooメッセージを送ってみましょう。まず、fooメソッドを探してObject

(正確には ProtoObject) まで継承の連鎖を一通りたどる、通常の探索が開始さ
れます。メソッドが見つからない場合、仮想マシン からそのオブジェクトに

self doesNotUnderstand: #fooメッセージが送信されます。(図 5.4を参照)

Figure 5.4: Message foo is not understood

これも通常の動的なメッセージ送信なので、また EllipseMorphから探索が
始まります。ただし、今度は doesNotUnderstand:メソッドを探します。結局
のところ、Objectが doesNotUnderstand:を実装しています。このメソッドでは、
MessageNotUnderstoodオブジェクトが生成されます。MessageNotUnderstoodオブ

ジェクトは、現在実行しているコンテキストでデバッガを立ち上げることがで

きます。

なぜ明らかなエラーに対してこんなに複雑なことをするのかというと、こ

うすると開発者がエラーを捕まえて別のアクションをさせることが容易になる

からです。Objectの任意のサブクラスは doesNotUnderstand:をオーバーライドす
ることで、別のやり方でエラー処理を行うことができます。

実際、あるオブジェクトから別のオブジェクトに自動的にメッセージを移

譲するように簡単に実装できます。Delegatorオブジェクトとして、実装してい
ないメッセージを別のオブジェクトに移譲して改めてメッセージを処理させた

り、エラーを発生させたりすることも可能です。

5.7 共有変数

ここでは Smalltalkの特徴の中で、五つの規則ではカバーされていないものを
見ていきましょう。共有変数のことです。

Smalltalkには 3種類の共有変数があります：(1)グローバルな共有変数 (2)
インスタンスとクラスの間で共有される変数 (ク ラ ス 変 数 ) (3)クラスのグルー
プの間で共有される変数 (プ ー ル 変 数 )です。これらの共有変数の名前は大文字
で始まります。これは、変数が複数のオブジェクトの間で共有されているとい

うことの警告となっています。

グローバル変数

Pharoでは、すべてのグローバル変数は Smalltalkという名前空間に格納されま
す。Smalltalk は SystemDictionary クラスのインスタンスとして実装されていま
す。グローバル変数はどこからでもアクセス可能です。各クラスはグローバル

変数で名付けられています。特別なオブジェクトや共通して使われるオブジェ

クトもグローバル変数で名前が付いています。

変数 Transcript (トランスクリプト)は TranscriptStreamのインスタンスを指し
ています。clsindTranscriptStreamはスクロール付きのウィンドウに書き込む
ストリームです。以下のコードは、Transcriptに情報を表示して改行します。

Transcript show: ’Pharoは楽しくてパワフル！’ ; cr

do itする前に、World .Tools ….Transcriptでトランスクリプトを開いておいて
ください。

Hint ト ラ ン ス ク リ プ ト へ の 書 き 込 み は 遅 い で す 。 ト ラ ン ス ク リ プ ト ウ ィ ン ド
ウ が 開 い て い る と き は な お さ ら で す 。 ト ラ ン ス ク リ プ ト に 書 き 込 ん で い て 遅 い

と 思 っ た と き は 最 小 化 し て み る の も 一 つ の 手 で す 。

他の便利なグローバル変数

• Smalltalkは SystemDictionaryのインスタンスです。Smalltalk自身を含むすべ
てのグローバル変数を定義してします。この辞書のキーはシンボルで、

Smalltalkのコード内でグローバル変数を示すのに使われます。例えば、

Smalltalk at: #Boolean −→ Boolean

Smalltalkはそれ自身がグローバル変数なので、

Smalltalk at: #Smalltalk −→ a SystemDictionary(lots of globals)

さらに、

(Smalltalk at: #Smalltalk) == Smalltalk −→ true

となります。

• Sensorは EventSensorのインスタンスで、Pharoへの入力を表してます。
例えば、Sensor keyboardはキーボードからの次の入力を答えます。そし
て、Sensor leftShiftDownは、左シフトキーが押されているときに trueと答
えます。さらに、Sensor cursorPointは現在のマウスの位置を示す Pointを返
します。

• Worldは PasteUpMorphのインスタンスで、スクリーンを表します。World
boundsはスクリーン全体の範囲を示す矩形を答えます。すべてのモーフ
はWorldのサブモーフです。

• ActiveHandは HandMorphの現在のインスタンスで、カーソルのグラフィッ
クを表します。マウスでモーフをドラッグしている間は ActiveHandのサ
ブモーフになっています。

• Undeclared も辞書です。宣言されていない変数群を保持しています。宣
言されていない変数を参照するメソッドを書くと、通常はブラウザが、グ

ローバル変数やインスタンス変数として宣言するよう促します。それで

も、変数の宣言を後で削除してしまった場合には、コードは未宣言の変

数を参照することになります。Undeclaredをインスペクトすることで、奇
妙な振る舞いの原因がわかることもあるでしょう。

• SystemOrganizationは SystemOrganizerのインスタンスです。クラスの構成
情報をまとめて保持しています。より正確に言えば、ク ラ ス 名 をカテゴ

リに分類します。

SystemOrganization categoryOfElement: #Magnitude −→ #’Kernel-Numbers’

グローバル変数は極力使わないというのが現在のプラクティスとなってい

ます。通常はクラスインスタンス変数やクラス変数を使い、クラスメソッドで

アクセスする方が望ましいです。もし Pharoを一から再実装するようなことが

あれば、グローバル変数は、クラス群を除いては、ほとんどシングルトンに置

き換えられるでしょう。

通常、グローバル変数を定義するには、do itで、大文字で始まる未宣言の識別
子に代入するだけです。すると、パーサがグローバル変数を定義するよう促し

ます。プログラムでグローバル変数を定義するには、Smalltalk at: #AGlobalName
put: nilを実行します。削除するには、Smalltalk removeKey: #AGlobalNameです。

クラス変数

クラスおよびクラスのすべてのインスタンスで、データを共有したいときがあ

ります。これはク ラ ス 変 数 で可能です。クラス変数という用語は、変数のライ

フタイムがクラスと同じということを示しています。しかし、その用語の字面

からは、クラス変数はクラスの間だけでなく、インスタンスとも共有されると

いうことはわかりません。図 5.5を見てください。実際のところ、共 有 変 数 と
いう名称の方が適切だったでしょう。役割がより明らかになりますし、使う際

(特に変更する場合)の危険性も伝わります。

Figure 5.5: インスタンスメソッドとクラスメソッドが別々の変数にアクセスし
ている。

図 5.5では、rgbと cachedDepthは Colorのインスタンス変数であり、そのた
め、Colorのインスタンスからアクセス可能です。また、superclassや subclass、
methodDict等はクラスインスタンス変数なので、Colorクラスからのみアクセス
できます。

新しく出てきたのは ColorNamesと CachedColormapsのク ラ ス 変 数 で、Colorに
定義されています。変数が大文字であることが、共有されているというヒント

を与えてくれます。実際、Colorの全インスタンスからこれらの共有変数にアク

セスでき、さらに Colorクラス自身とそ の す べ て の サ ブ ク ラ ス か ら も ア ク セ ス 可
能 で す 。インスタンスメソッドとクラスメソッドの両方からこれらの共有変数

にアクセスできるのです。

クラス変数はクラス定義テンプレートの中で宣言されます。例えば、Color
クラスでは大量のクラス変数を定義しており、色オブジェクトの生成を高速化

しています。その定義を下図 (Class 5.20)に示します。

Class 5.20: Color と ク ラ ス 変 数
Object subclass: #Color

instanceVariableNames: ’rgb cachedDepth cachedBitPattern’
classVariableNames: ’Black Blue BlueShift Brown CachedColormaps ColorChart
ColorNames ComponentMask ComponentMax Cyan DarkGray Gray
GrayToIndexMap Green GreenShift HalfComponentMask HighLightBitmaps
IndexedColors LightBlue LightBrown LightCyan LightGray LightGreen LightMagenta
LightOrange LightRed LightYellow Magenta MaskingMap Orange PaleBlue
PaleBuff PaleGreen PaleMagenta PaleOrange PalePeach PaleRed PaleTan
PaleYellow PureBlue PureCyan PureGreen PureMagenta PureRed PureYellow
RandomStream Red RedShift TranslucentPatterns Transparent VeryDarkGray
VeryLightGray VeryPaleRed VeryVeryDarkGray VeryVeryLightGray White Yellow’
poolDictionaries: ’’
category: ’Graphics-Primitives’

クラス変数 ColorNamesは、よく使われる色の名前を持つ配列です。この配
列は Colorとそのサブクラス TranslucentColorの、すべてのインスタンスで共有
されていて、インスタンスメソッドやクラスメソッドからアクセス可能です。

ColorNamesは Color class»initializeNamesで一度だけ初期化され、Colorのイン
スタンスからアクセスされます。Color»nameメソッドで、色の名前を探すため
にこの変数が使われています。このようになっているのは、ほとんどの色は名

前を持たないため、インスタンス変数 nameを色ごとに持つのは不適切と考え
たからです。

クラスの初期化

クラス変数があることで、どうやってそれを初期化するかが問題になります。

一つの解は、初期化を遅延させることです。これは、クラス変数へのアクセサ

メソッドを実装し、その中で変数が初期化されていなければ初期化を行うよう

に書くことで実現できます。この方法の場合、常にアクセサを使い、クラス変

数には絶対に直接アクセスしないようにすることになります。また、アクセサ

実行の際に、初期化の必要性をチェックするコストが上乗せされます。さらに

言えば、アクセサを使った場合、変数は事実上もはや共有されていないことに

なるので、クラス変数を使う意味も薄れます。

Method 5.21: Color class»colorNames
Color class»colorNames
ColorNames ifNil: [self initializeNames].

↑ ColorNames

もう一つの解は、クラスメソッド initializeをオーバーライドすることです。

Method 5.22: Color class»initialize
Color class»initialize
…
self initializeNames

この方法の場合、initializeメソッドを定義したら、Color initializeといった具合に
明示的に実行しておくのを忘れないようにします。クラス側の initializeメソッ
ドは、コードをメモリーにロードしたときには自動的に実行されますが、ブラ

ウザで入力してコンパイルしたときや編集して再コンパイルしたときには、自

動的には実行さ れ な い のです。

プール変数

プール変数は、直接継承関係にないクラスの間でも共有される変数です。プー

ル変数は元々はプール辞書に格納されていましたが、今では専用のクラス

(SharedPool のサブクラス) のクラス変数として定義しなければなりません。
ここで忠告しておきますが、なるべく使うのを避けてください。プール変数が

必要になることは非常にまれであり、特定の状況下だけです。以後は、プール

変数を使ったコードを読むのに必要十分な説明にとどめておきます。

プール変数にアクセスするにはクラス定義でそのプールを示さなければな

りません。例えば、Textクラスではプール辞書 TextConstantsを使うことを示し
ています。プール辞書 TextConstantsは CRや LFといった、テキストに関するす
べての定数を保持しています。#CRがキーで、値として Character cr、つまり改
行文字が結び付けられています。

Class 5.23: Textク ラ ス の プ ー ル 辞 書
ArrayedCollection subclass: #Text

instanceVariableNames: ’string runs’
classVariableNames: ’’
poolDictionaries: ’TextConstants’
category: ’Collections-Text’

こうすることで、Textクラスのメソッドは、メソッド本文で直 接 プール辞書
のキーにアクセスすることができます。つまり、明示的に辞書に聞かずとも、

変数を参照する構文でアクセスできます。例えば、以下のようにしてメソッド

を書けます。

Method 5.24: Text»testCR
Text»testCR

↑ CR == Character cr

繰り返します。プール変数やプール辞書はなるべく使わないことをお勧め

します。

5.8 章のまとめ

Pharo のオブジェクトモデルはシンプルで一貫したものです。すべてはオブ
ジェクトで、ほとんどすべてのことはメッセージ送信で起こります。

• すべてのものはオブジェクトである。整数のようなプリミティブな存在
もオブジェクトであり、クラスもファーストクラスのオブジェクトです。

• すべてのオブジェクトはクラスのインスタンスである。クラスは、プ ラ
イ ベ ー ト な インスタンス変数によってオブジェクトの構造を定義します。

そして、パ ブ リ ッ ク な メソッドによって振る舞いを定義します。各クラ

スはそれぞれのメタクラスの唯一のインスタンスです。クラス変数はク

ラスとそのすべてのインスタンスに共有されるプライベートな変数です。

クラスはそのインスタンスのインスタンス変数に直接アクセスすること

はできません。そしてインスタンスもクラスインスタンス変数にアクセ

スできません。そういったアクセスが必要ならば、アクセサを定義しな

ければなりません。

• すべてのクラスにはスーパークラスがある。単一継承の階層のルートは
ProtoObjectです。しかし、クラスを定義するときには、通常は Objectクラ
スかそのサブクラスから継承します。抽象クラスを定義する特別な構文は

ありません。単に抽象メソッドを持てば抽象クラスとみなされます。抽

象メソッドは self subclassResponsibilityのみからなるメソッドです。Pharo
は単一継承のみをサポートしていますが、共有したいメソッド群をト レ

イ ト としてパッケージングすることで、それらを簡単に共有できます。

• すべてはメッセージ送信で起こる。「メソッドを呼び出す」のではなく、
「メッセージを送信」します。するとレシーバがそのメッセージに対応す

るメソッドを選択します。

• メソッド探索は継承の連鎖をたどっていく。self送信は動的で、レシーバ
のクラスからメソッド探索を始めます。一方、super送信は静的で、super
送信を記述しているクラスのスーパークラスから探索を始めます。

• 3種類の共有変数がある。グローバル変数は、システムのどこからもアク
セス可能です。クラス変数はクラスとそのサブクラスとインスタンス間

とで共有されます。プール変数は指定したクラス間で共有されます。共

有変数の使用は、できるだけ避けるべきです。



Chapter 6

Pharoのプログラミング環境

この章の目的は、Pharoのプログラミング環境を使い、どのようにプログラム
を開発するかを示すことです。既にブラウザを使ってメソッドやクラスをどの

ように定義するかについて見てきたわけですが、この章ではブラウザのその他

の機能について示し、他のいくつかのブラウザについて紹介していきます。

もちろん作ったプログラムが期待したように動作しないのはよくあること

です。Pharoには優れたデバッガがあります。最も役立つツールではあります
が、最初に使うときは悩むかもしれません。デバッグの手順について説明し、

いくつかのデバッガの機能について実演します。

Smalltalk固有の特徴の一つは、プログラミングをしているとき、静止したプ
ログラムテキストの世界ではなく、動いているオブジェクトからなる世界にい

るということです。このことにより、プログラミング中に非常に素早いフィー

ドバックを得ることが可能で、このフィードバックによって生産性がより向上

します。動いているオブジェクトを見たり、実際に変更したりするために二つ

のツールがあります。それはイ ン ス ペ ク タ とエ ク ス プ ロ ー ラ です。

ファイルやテキストエディタではなく、動いているオブジェクトからなる

世界でプログラミングを行っているため、Smalltalkイメージファイルからプ
ログラムを取り出すために、何らかの明示的な作業をしなければいけません。

この作業のための昔からのやり方は、フ ァ イ ル ア ウ ト やチ ェ ン ジ セ ッ ト を作る

ことで、すべての Smalltalkの方言でサポートされています。これらは、基本
的に他のシステムで読み込めるようにコード化されたテキストファイルです。

Pharo上でプログラムを取り出すための新しい方法は、サーバー上のバージョ
ン管理されたリポジトリにコードをアップロードすることです。このために、

Monticelloと呼ばれるツールを使います。これは特にチームで作業する際に、
よりパワフルで効果的な手段となります。

6.1 概要

Smalltalkと現在のグラフィカルインターフェースは一緒に発展してきました。
Smalltalkの最初の公開リリースである 1983年以前でさえ、Smalltalkは独自
のグラフィカルな開発環境を備え、すべての Smalltalkの開発はその上で行わ
れてきました。では、Pharoの主なツールを見ていきましょう。

• Browser (ブラウザ)は、中心的な開発ツールです。クラスやメソッドの作
成、定義、構築を行う際にこのツールを使います。これを使うことですべ

てのクラスライブラリもブラウズできます。ソースコードを別々のファ

イルに保存する他の環境とは異なり、Smalltalkではすべてのクラスやメ
ソッドがイメージファイルに含まれます。

• Message Names (メッセージネーム)ツールを使うことにより、特定のセレ
クタもしくは部分文字列を含んだセレクタを持つすべてのメソッドを探

すことができます。(訳注：Pharo 1.4では Finderツールに統合されてお
り、ドロップダウンメニューで Selectorsを選択することで利用できます)

• Method Finder (メソッド・ファインダ)ツールは、メソッドを探すのに役
立ちますが、メソッド名と同様に、何をす る のかによりメソッドを検索

します。(訳注：Pharo 1.4では Finderツールに統合されており、ドロッ
プダウンメニューで Examplesを選択することで利用できます)

• Monticello Browser (Monticelloブラウザ)は、Monticelloパッケージから
コードを読み込んだり、保存したりするための出発点です。

• Process Browser (プロセスブラウザ)は、Smalltalk上で実行されているす
べてのプロセス (スレッド)を見るのに役立ちます。

• Test Runnerは、SUnitテストを実行、デバッグすることができます。詳
細については第 7章を参照してください。

• Transcript (トランスクリプト)は、Transcript出力ストリームに対するウィ
ンドウで、1.5 節で既に述べた通り、ログメッセージを書くのに役立ち
ます。

• Workspace (ワークスペース)は入力することができるウィンドウです。こ
れはいろいろな目的に使えますが、最もよく使われるのは Smalltalk式
を入力して、do itで実行することです。workspaceの使い方は 1.5節に
も図解されています。

Debugger (デバッガ)は名前からわかる通り明確な役割を持っています。し
かし、他のプログラミング言語のデバッガと比べて、より重要な役割を担って

いることに気づくでしょう。Smalltalkでは debugger上でプ ロ グ ラ ム できる

のです。デバッガはメニュー画面から起動されるのではなく、たいてい、失敗

するテストプログラムを実行したり、cmd– .と入力して起動しているプロセス
を中断したり、あるいは、コード中に self halt式を挿入したりすることで、起
動されます。

6.2 ブラウザ

Smalltalkの歴史と共に、長年にわたり多くの異なるクラスブラウザが開発さ
れてきました。Pharoでは、様々なビューを一つにまとめたブラウザを提供す
ることで、この物語を単純にしました。図 6.1は最初にブラウザを開いたとき
に表示されるものです。12

Figure 6.1: ブラウザ

ブラウザ上部の四つの小さなペインは、システム内メソッドの階層ビュー

を表しています。これは、NeXTstepの File ViewerやMac OS Xの Finderのカ
ラムモードといった、ディスク上のファイルを見るためのビューと同様のもの

です。左端のペインはクラスのパ ッ ケ ー ジ リストです。そのうちの一つ (例え

1もしブラウザが図 1.12のように表示されない場合、デフォルトのブラウザを変更する必要が
あるかもしれないことを思い出してください。FAQ 5, p. 326参照

2訳注：Pharo のデフォルトブラウザは何度も変更されています。バージョン 1.3 では、
OBSystemBrowser が使用されており、この章の説明とは異なる場合があります。バージョン
1.4 ではデフォルトブラウザは SystemBrowser であり、OBSystemBrowser は別途インストール
する必要がありました。バージョン 1.4 Summer では OBSystemBrowser がデフォルトブラウザ
になっています

ば Kernel)を選択すると、すぐ右隣にあるペインにパッケージ中のすべてのク
ラスが表示されます。

Figure 6.2: ブラウザでModelクラスを選択

同様に、左から 2番目のペインでクラスの一つ、例えばModel (図 6.2参照)を
選ぶと、その右隣の 3番目のペインにクラスに対して定義されたプ ロ ト コ ル が
表示され、デフォルトで仮想プロトコル–all–が選択されます。プロトコルとは
メソッドをカテゴリ化するためのもので、クラスの振る舞いを概念的に首尾一

貫して細分化することで、それらを探したり考えたりするのに役立ちます。左

から 4番目のペインは、選択したプロトコルで定義されているすべてのメソッ
ド名が表示されます。ここでメソッド名を選ぶと、そのメソッドのソースコー

ドがブラウザ下部にある大きなペインに表示されます。このペインではソース

コードのブラウズや編集、編集後の保存ができます。ブラウザでModelクラス、
dependentsプロトコル、そしてmyDependentsを選択していくと、図 6.3のよう
に表示されるはずです。

Mac OS Xの Finderとは違い、ブラウザ上部四つのペインは完全に同じも
のというわけではありません。クラスやメソッドは Smalltalk言語の一部であ
るのに対し、パッケージとプロトコルはそうではなく、これらはブラウザの各

ペインに表示すべき情報量を制限するために導入されたものです。例えば、も

しプロトコルがなければ、ブラウザは選択したクラスのすべてのメソッドのリ

ストを表示する必要があります。多くのクラスでは、このようなリストは大き

すぎて扱うのが不便になってしまいます。

このことから、新しいパッケージやプロトコルを作る方法は、新しいクラス

やメソッドを作る方法とは異なります。新しいパッケージを作るにはパッケー

ジペインでアクションクリックし new packageを選びます。新しいプロトコル

Figure 6.3: ModelクラスのmyDependentsメソッドをブラウザで表示

を作るにはプロトコルペインでアクションクリックし new protocolを選びます。
ダイアログの中に新しく作ったものの名前を入力すれば完了です。パッケージ

やプロトコルについては、名前や内容を入力する以外に必要な作業は何もあり

ません。

Figure 6.4: クラス作成テンプレートをブラウザに表示

これとは対照的に、新しいクラスやメソッドを作るには、実際に何らかの

Smalltalkコードを書く必要があります。(左端のペインで)現在選択されてい
るパッケージをクリックすると、ブラウザ下部のペインにクラス生成のテンプ

レートが表示されます (図 6.4)。このテンプレートを編集することで、新しい
クラスを作ります。Objectを作りたいサブクラスの親となる既存のクラス名で
置き換え、NameOfSubclassを新しいサブクラス名に置き換えます。既にわかっ
ているのであれば、インスタンス変数名を入力します。新しいクラスに対する

カテゴリは、デフォルトで現在選択しているパッケージのカテゴリとなります

が、3必要なら変更することもできます。既にサブクラス化の元となる既存ク
ラスをブラウザで選択しているのならば、クラスペインをアクションクリック

し class templates …. subclass templateを選択することにより、ほんの少し初期化
が異なる同様のテンプレートを得ることができます。また、既存のクラス定義

でクラス名を新しいものに変えることでもクラス定義を編集できます。どの場

合でも新しい定義をアクセプトすると、(#から始まる名前の)新しいクラスが
(つまり対応するメタクラスのインスタンスとして)作られます。クラスを作る
ことはそのクラスを参照するグローバル変数を作ることでもあるため、名前に

よって既存の全クラスを参照することができるわけです。

新しいクラス名が、(す な わ ち #で始まる)シンボルとしてクラスの作ったテ
ンプレートの中に現れる理由や、クラスが作られた後は (す な わ ち #のない)識
別子としてクラス名を使えば、そのクラスをコードから参照できるようになる

理由が理解できたでしょうか?
新しいメソッドを作るプロセスも同様です。最初にメソッドを作りたいク

ラスを選択し、次にプロトコルを選択します。図 6.5のように、ブラウザは編
集可能なメソッド作成テンプレートを表示します。

コード空間を渡り歩く

ブラウザはコードを探したり分析したりするためのいくつかのツールを提供し

ます。これらのツールは、様々なコンテキストメニューをアクションクリック

することで利用できます。また、特に最も頻繁に使われるツールなどにはキー

ボードショートカットが用意されています。

新しいブラウザウィンドウを開く

時折、複数のブラウザウィンドウを開きたくなる場合があります。コードを書

いているときは少なくとも二つ必要でしょう。一つは入力中のメソッドのため、

もう一つは物事がどう動いているのか調べるのにシステム周りを見るためです。

テキスト中でクラス名を選択した後にキーボードショートカット cmd–bを使
えば、そのクラスを表示するブラウザを起動することができます。

次 の こ と を 試 し て み て く だ さ い : ワ ー ク ス ペ ー ス ウ ィ ン ド ウ 内 で ク ラ ス 名

3パッケージとカテゴリが正確には同じものではないことを思い出してください。6.3節にてそ
れらの関係について詳しく見ていきます

Figure 6.5: メソッド作成テンプレートをブラウザに表示

(例 え ば Morph) を 入 力 し 、 そ れ を 選 択 し cmd–b を 押 し て く だ さ い 。 こ の 方 法 は
よ く 役 に 立 ち 、 ど の テ キ ス ト ウ ィ ン ド ウ で も 使 え ま す 。

メッセージのセンダとインプリメンタ

メソッドペインでbrowse …. senders (n)をアクションクリックすると、選択し
たメソッドを使用しているすべてのメソッドのリストが表示されます。ブラウ

ザを開いて Morphを選択した後、メソッドペインの中の drawOn:メソッドをク
リックすると、drawOn:本体がブラウザ下部に表示されます。 senders (n) (図 6.6)
を選ぶと、drawOn:を一番上の項目としたメニューが現れ、以下 drawOn:が送る
すべてのメッセージ (図 6.7)が順に表示されます。このメニューの項目を選ぶ
と、選択したメッセージ (図 6.8)を送る仮想イメージ内の全メソッドのリスト
がブラウザで表示されます。

senders (n)の中の“n”は、メッセージのセンダを探すためのキーボードショー
トカットが cmd–nであることを示しています。この方法は、ど の テキストウィ
ンドウでも使えます。

コ ー ド ペ イ ン に あ る テ キ ス ト “drawOn:” を 選 択 し て cmd–n を 押 せ ば 、 直
ち に drawOn:の セ ン ダ が 表 示 さ れ ま す 。

AtomMorph»drawOn:にある drawOn:のセンダを探せば、それが super send
であることがわかるでしょう。つまり、実行されるメソッドは AtomMorphの
スーパークラスにあることがわかります。ではいったいどのクラスでしょうか?



112 Pharo の プ ロ グ ラ ミ ン グ 環 境

Figure 6.6: senders (n)メニュー項目。

Figure 6.7: メッセージのセンダを選択。

browse .hierarchy implementorsをアクションクリックすれば、そのメソッドが
EllipseMorphクラスにあることがわかります。

図 6.8のリストの 6番目にあるセンダ Canvas»drawを見てみましょう。この
メソッドは、引数として渡されたどんなオブジェクトに対しても drawOn:を送る
ことがわかります。つまり、あらゆるクラスのインスタンスが引数となる可能

性があります。データフロー解析がいくつかのメッセージのレシーバクラスを

把握する助けにはなりますが、一般的には、ブラウザを使ってもどのメッセー

ジ送信がどのメソッドの実行につながるのかを知る簡単な方法はありません。

このような理由から “センダ”ブラウザは、メッセージが示唆するメソッドだ
け、つまり選択されたセレクタを持つすべてのメッセージセンダを表示します。

それにも関わらず、どのようにメソッドを使 う のかを理解する必要があるとき

には、すぐに使用例を導けるのでセンダブラウザが大いに役立ちます。同じセ

レクタを持つすべてのメソッドは同じように使われるはずなので、メッセージ

の使い方はどれも似ているはずです。

インプリメンタブラウザも似たような動作をしますが、メッセージセンダの

リストの代わりに、同じセレクタを持つメソッドを実装している全クラスのリ

ストを表示します。このことを確かめるために、メソッドペインにある drawOn:
を選び、 browse . implementors (m)を選びます (もしくはコードペイン中のテキ

Figure 6.8: センダブラウザは Canvas»drawメソッドが、引数に対して drawOn:
メッセージを送っていることを表しています。

スト “drawOn:”を選択し、cmd–mを押します)。すると、drawOn:メソッドを
実装する 90余りのクラスが、スクロール可能なメソッドリストのウィンドウ
として表示されます。これほど多くのクラスがこのメソッドを実装しているこ

とに驚く必要はありません。drawOn:は、画面上に自分自身を描画することがで
きるすべてのオブジェクトによって理解されるメッセージなのです。

メソッドのバージョン

新しいバージョンのメソッドを保存しても、古いものは失われません。Pharo
はすべての古いバージョンを保存し、異なるバージョンを比較したり、古いバー

ジョンに戻したり (“復帰”)することができます。 browse . versions (v)メニュー
項目により、選択したメソッドに加えられた一連の変更を調べることができま

す。図 6.9では、buildWorldMenu:メソッドの二つのバージョンを示しています。

上部のペインには、そのメソッドのバージョンごとに 1行ずつ表示されて
おり、このメソッドを書いたプログラマのイニシャル、保存した日時、クラス

とメソッドの名前、そして定義されているプロトコルが表示されます。現在の

(アクティブな)バージョンはリストの一番上にあります。選択されたバージョ
ンは、下部のペインに表示されます。選択されたメソッドと現在のバージョン

の違いを表示したり、選択したバージョンに復帰したりするためのボタンが用

意されています。

バージョンブラウザの存在は、必要なコードを残すべきか悩む必要がなく、

単に削除すれば良いことを意味しています。もしそのコードが本 当 に 必要だと

わかったら、いつでも古いバージョンに復帰することができますし、古いバー

Figure 6.9: バージョンブラウザが二つの TheWorldMenu»buildWorldMenu:メソッ
ドを表示

ジョンから必要なコードの断片を得て別のメソッドに貼り付けることもでき

ます。バージョンブラウザを使う癖を付けましょう。もはや必要のなくなった

コードをコメントアウトするのは悪い習慣です。なぜなら、そのことで現在の

コードが読みにくくなるからです。Smalltalkerは読みやすいコードを非常に
高く評価します。

Hint メ ソ ッ ド 全 体 を 削 除 し た 後 で 、 元 に 戻 し た い 場 合 は ど う す れ ば い い で
し ょ う ? チ ェ ン ジ セ ッ ト の 中 に 削 除 し た も の を 見 つ け る こ と が で き ま す 。 そ
こ で ア ク シ ョ ン ク リ ッ ク す る こ と に よ り 、 以 前 の バ ー ジ ョ ン を 確 認 で き ま す 。

チ ェ ン ジ セ ッ ト ブ ラ ウ ザ は 6.8 節 で 述 べ ら れ て い ま す 。

メソッドのオーバーライド

インヘリタンスブラウザは、表示されたメソッドによってオーバーライドさ

れたすべてのメソッドを表示します。どのように動作するかを見るために、

ブラウザで ImageMorph»drawOn: メソッドを選択してください。メソッド名の
隣にある三角形のアイコンに注目してください。(図 6.10)上向きの三角形は、
ImageMorph»drawOn:が継承されたメソッド (す な わ ち Morph»drawOn:)をオーバー
ライドしていることを示しており、さらに下向きの三角形は、サブクラスによって

オーバーライドされていることを示しています。(アイコンをクリックすること
で、これらのメソッドに移動することもできます。)ここで、browse . inheritance
を選択してください。インヘリタンスブラウザが、オーバーライドされたメソッ

ドの階層関係を示します。(図 6.10参照)

Figure 6.10: ImageMorph»drawOn:とそれがオーバーライドしたメソッド。選択
したメソッドの兄弟関係がリストに表示されている。

階層ビュー

標準では、ブラウザは左端のペインにパッケージのリストを表示します。しか

しクラスの階層ビューに切り替えることは可能です。関心のある特定のクラ

スを選択してください。例えば ImageMorphを選択し、 hier.ボタンをクリック
します。選択したクラスのすべてのスーパークラスとサブクラスを含むクラス

階層を、左端のペインで見ることができます。2番目のペインは選択したクラ
スのメソッドを実装するパッケージのリストが表示されます。4図 6.11 では、
ImageMorph のスーパークラスが Morph であることを階層ブラウザが示してい
ます。

変数への参照を見つける

クラスペインにあるクラスをアクションクリックし、browse . chase variables
を選択することで、インスタンス変数やクラス変数が使われている場所を見つ

けることができます。追 跡 ブ ラ ウ ザ を使うことで、すべてのインスタンス変数

やクラス変数のアクセサ、さらにこれらのアクセサを送るメソッドなどを探索

することができます (図 6.12)。

4訳注：バージョン 1.4 では階層ブラウザが表示されますが、2 番目のペインにパッケージのリ
ストは表示されません

Figure 6.11: ImageMorphの階層ビュー

Figure 6.12: Morphに対する追跡ブラウザ

ソース

メソッドペインをアクションクリック することで利用可能となる

various . view …メニュー項目は、“how to show” メニューを表示します

が5、これは選択されたメソッドをソースペインでブラウザがどのように表示
するかを選ばせるものです。 source（ソースコード）、prettyPrint（整形された
ソースコード）、byteCode（バイトコード）およびdecompile（バイトコードか
ら逆コンパイルされたソースコード）といった選択肢があります。

“how to show” メニューで選択した prettyPrintは、ソースを保存する前に
整形出力したメソッドと同じもの67ではないことに注意してください。このメ
ニューは、ブラウザの表示方法を制御するだけで、システムに格納されたコー

ドには何の影響も及ぼしません。二つのブラウザを開いて一方のブラウザで

prettyPrintを選び、もう一方の Browserで sourceを選ぶことで、このことを確
認できます。同じようにして、二つのブラウザで同じメソッドを表示させた状

態で、一方でbyteCodeを選択し、もう一方でdecompile選択することは、Pharo
の仮想マシンのバイトコードの命令セットについて学ぶ良い方法です。

リファクタリング

コンテキストメニューから多くの標準的なリファクタリングが行えます。四つ

のペインのどれでもアクションクリックすれば、現在リファクタリング可能な

操作を見ることができます。図 6.13を参照してください。
以前は、リファクタリングのためにリファクタリングブラウザと呼ばれる

特殊なブラウザを使う必要がありましたが、現在はどのブラウザからも利用で

きます。

ブラウザのメニュー

ブラウザのペインをアクションクリックすることで多くの追加機能を利用する

ことができます。たとえメニュー項目上のラベルが同じであっても、その意 味

は文脈に依存します。例えばパッケージペイン、クラスペイン、プロトコルペ

イン、そしてメソッドペインのすべてがFile outメニュー項目を持っています。
しかしそれらは異なる動作をします。パッケージペインのFile outメニューは、
パッケージ全体をファイル出力します。クラスペインのFile outメニューはクラ
ス全体をファイル出力します。プロトコルペインのFile outメニューはプロトコ
ル全体をファイル出力します。そしてメソッドペインのFile outメニューは表示
されたメソッドだけをファイル出力します。このことは明らかだと思えるかも

しれませんが、初心者にとっては混乱の元になります。

おそらくパッケージペインの中で最も便利なメニュー項目は find class…(f)で
す。カテゴリは活発に開発しているコードについては役立ちますが、ほとんど

5訳注：バージョン 1.4ではメソッドペインの下にある Viewボタンを押すと “Choose”メニュー
が表示されます

6 pretty print (r)はメソッドペインの最初のメニュー項目もしくはコードペインの真ん中下側にあ
ります。

7訳注：バージョン 1.4 では、メソッドペインおよびコードペインでの整形出力機能はありま
せん。

Figure 6.13: リファクタリング操作

誰もシステム全体のカテゴリ化について知りませんし、クラスがどのパッケー

ジにあるか推測するよりも、クラス名の先頭の数文字を入力した後で shortfと
入力した方がずっと早いのです。 recent classes…はたとえクラス名を忘れてし
まっても、最近ブラウズしたクラスへ素早く戻るのに役立ちます。

ブラウザの左上にある検索ボックスに名前を入力することで、特定のクラ

スやメソッドを探すこともできます。リターンキーを入力するとシステムに問

い合わせがなされ、その結果が表示されます。検索語の前に#を付けることで、
クラスへの参照や、メッセージのセンダを探せることに気をつけてください。

選択したクラスの特定のメソッドを探すのであれば、(デフォルトで表示され
る)–all–プロトコルをブラウズした方が速いでしょう。メソッドペインにマウ
スを移動し、探そうとしているメソッド名の最初の文字を入力してください。

メソッドペインがスクロールして、探しているメソッド名が見えるでしょう。

OrderedCollection»removeAt: で 二 つ の 操 作 方 法 を 試 し て く だ さ い 。

メニューには他の多くのオプションがあります。少々の時間を取って、ブ

ラウザ上のどこにどのような機能があるかを探しておくと、後ほど報われるこ

とでしょう。

ク ラ ス ペ イ ン メ ニ ュ ー の Browse Protocol , Browse Hierarchy , そ し て
Show Hierarchy の 結 果 を 比 較 し て く だ さ い 。

プログラムによるブラウズ

SystemNavigation クラスはシステム周りを操作するために役に立つユーティリ
ティメソッドのいくつかを提供しています。典型的なブラウザが提供する機能

の大半は、SystemNavigationによって実装されています。

drawOn::の セ ン ダ を 見 る た め に は 、 ワ ー ク ス ペ ー ス を 開 い て 以 下 の コ ー ド
を 評 価 し て く だ さ い 。

SystemNavigation default browseAllCallsOn: #drawOn:

センダの検索を特定のクラスのメソッドに制限するには以下のようにします。

SystemNavigation default browseAllCallsOn: #drawOn: from: ImageMorph

開発ツールもオブジェクトなので、プログラムから完全にアクセス可能であり、

自分自身のツールを開発することや既存のツールを必要に応じて変更すること

もできます。

implementorsメニュー項目と等価なプログラムは以下のようなものです。

SystemNavigation default browseAllImplementorsOf: #drawOn:

他に何ができるのかを知るには、ブラウザで SystemNavigationクラスを探し
てください。その他の操作例については、FAQ(付録 A)で見つけることができ
ます。

6.3 Monticello

2.9節において、Pharoのパッケージ作成ツールであるMonticelloの概要につ
いて説明しました。しかし、Monticelloは、その章で述べたこと以上の特徴を
備えています。Monticelloはパ ッ ケ ー ジ を扱うので、Monticelloについて説明
する前に、まずパッケージが何なのかということを正確に説明する必要があり

ます。

パッケージ: Pharoコードの宣言的なカテゴリ化

かなり前の 2.3節において、パッケージとは概ねカテゴリと同等のものであると
指摘しました。ここでは、その関係が実際には何であるか見ていきます。パッ

ケージシステムとは、Smalltalkソースコードを組織化するためのシンプルで
軽量な方法のことです。これは、カテゴリとプロトコルに対して簡単な命名規

則を利用します。

例を使って説明しましょう。あるフレームワークを開発していると仮定し

ましょう。これは Pharo から関係データベースを簡単に利用するためのもの

です。あなたはこのフレームワークを PharoLinkと呼ぶことにしました。そし
て、作成したクラスすべてを含むような一連のカテゴリを作ったとします。

例えば、’PharoLink-Connections’カテゴリは OracleConnection MySQLConnection
PostgresConnectionを含み、’PharoLink-Model’カテゴリは DBTable DBRow DBQuery
等を含みます。しかし、作成したコードのすべてがこれらのクラスの中にある

わけではありません。例えば、オブジェクトを SQL的な形式に変換するため
の一連のメソッドもあるかもしれません。

Object»asSQL
String»asSQL
Date»asSQL

これらのメソッドは、PharoLink-Connectionsや PharoLink-Modelカテゴリと同じ
パッケージに属しています。しかし、明らかにObjectクラス全体がそのパッケー
ジに属するわけではありません。つまり、クラスの残りが別のパッケージにあ

るとしても、特定のメ ソ ッ ド をパッケージに含める方法が必要になります。

それを行う方法は、(Object, String Date等)のメソッドを、prot*PharoLink(最
初のアスタリスクに注意してください) というプロトコルに置くことです。
PharoLink-…カテゴリと*PharoLink プロトコルの組み合わせが、PharoLinkパッ
ケージを形作ります。正確には、パッケージに置かれる規則は以下の通りです。

Fooと名付けられたパッケージは次のものを含みます。

1. Fooカテゴリもしくは Foo-で始まる名前のカテゴリ内にある、クラスの
すべてのク ラ ス 定 義

2. *Fooや*foo8という名前のプロトコルか、*Foo-や*foo-で始まる名前のプロ
トコルにある、任 意 の ク ラ ス 内のすべてのメ ソ ッ ド

3. Fooカテゴリか、Foo-で始まる名前のカテゴリのクラスにあるすべての
メ ソ ッ ド 。ただし* で始まる名前のプロトコルにあるメソッドは除 外 さ
れ る 。

これらの規則の結果、各クラスの定義と各メソッドが、ちょうど一つのパッケー

ジに属することになります。最後の規則にある例 外 は、それらのメソッドが他

のパッケージに属すために必要となります。規則 2にある、大文字小文字が無
視されるというのは、カテゴリ名がキャメルケース（空白を含まない）のに対

して、（必ずというわけではないが）典型的なプロトコル名が小文字である（空

白を含む場合がある）からです。

PackageInfoクラスがこれらの規則を実装しており、このクラスを試すこと
で規則の感触を掴むことができます。

ワ ー ク ス ペ ー ス 上 で 以 下 の 式 を 評 価 し て く だ さ い 。

mc := PackageInfo named: ’Monticello’

8比較の際に大文字か小文字かは無視されます。

Figure 6.14: Monticelloブラウザ

さて、このパッケージについて探索できるようになりました。例えば、ワー

クスペース上で mc classesを表示することで、Monticello パッケージを構成
するクラスの長いリストが得られます。mc coreMethods は、これらのクラス
にあるすべてのメソッドに対する MethodReferenceのリストが返されます。mc
extensionMethodsは、おそらく最も面白い問い合わせでしょう。この問い合わせ
は、Monticelloパッケージに含まれるすべてのメソッドのうち、Monticelloクラス
を除いたものを返します。

パッケージは比較的新しく Pharoに加えられたものです。しかし、パッケー
ジの命名規則は、既に使われていたものをベースとしていたため、パッケージ

の採用を意識していない古いコードを分析する際にも PackageInfoを使うことが
できます。

(PackageInfo named: ’Collections’) externalSubclassesを 表示し て く だ さ い 。 こ
の 式 は 、 Collectionパ ッ ケ ー ジ に 含まれていない Collectionsの す べ て の サ ブ ク ラ
ス の 一 覧 を 返 し ま す 。

Monticelloの基礎

Monticelloは、合衆国 3番目の大統領であり「バージニア信教自由法」の起草
者であるトーマス・ジェファーソンの山頂の家にちなんで名付けられました。

この名前はイタリア語で “小さな山”を意味し、イタリア語の発音に従い “c”
を chairの “ch”と発音して「モンティチェロ」といいます。

Monticelloブラウザを開くと、図 6.14のように、二つのペインと 1列に並
んだボタンが表示されます。左側のペインには、実行中の仮想イメージにロー

ドされたすべてのパッケージが表示され、名前の後の括弧の中にはパッケージ

を特定するバージョンが表示されます。

右側のペインには、Monticelloが知っているすべてのソースコードリポジ

Figure 6.15: リポジトリブラウザ

トリが表示されており、通常はそれらのリポジトリからコードがロードされま

す。左側のペインにあるパッケージを選ぶと、右側のペインには、選択したパッ

ケージのバージョンを含むリポジトリのみを表示するようにフィルタリングさ

れます。

リポジトリの一つは package-cache と呼ばれるディレクトリで、実行中の
イメージファイルのあるディレクトリのサブディレクトリです。遠隔地にあ

るリポジトリからコードを読み込んだり、書き込んだりする場合、コピーが

package-cache にも保存されます。ネットワークが利用できない場合にパッ
ケージへアクセスする必要のあるとき、package-cacheが役立ちます。また、
例えば電子メールの添付ファイルとしてMonticelloファイル (.mcz)が直接与
えられた場合、そのファイルを package-cacheディレクトリに置くのが一番手
軽なアクセス方法となります。

リストに新しいリポジトリを追加するには、 +Repositoryボタンをクリック
し、ポップアップメニューからリポジトリの種類を選んでください。ではHTTP
リポジトリを追加しましょう。

Monticello を 開 き 、 +Repository ボ タ ン を ク リ ッ ク し 、 HTTP を 選 び ま す 。 以
下 の よ う に ダ イ ア ロ グ 内 容 を 編 集 し て く だ さ い 。

MCHttpRepository
location: ’http://squeaksource.com/PharoByExample’
user: ’’
password: ’’

その後、このリポジトリをリポジトリブラウザで開くために、Openボタンを

クリックします。図 6.15のようなものが見えるでしょう。左側のペインには、
リポジトリ内のすべてのパッケージのリストがあります。その中の一つを選択

すれば、右側のペインにこのリポジトリ内にある選択されたパッケージのすべ

てのバージョンが表示されます。

その中の一つのバージョンを選択すると、Browseボタンでの (イメージファイ
ルに読み込むことなしに)ブラウズや Loadボタンでの読み込み、そして、Changes
ボタンでは、選択したバージョンを読み込んだ場合の変更点を見ることができ

ます。Copyボタンでは、任意のバージョンのパッケージのコピーを作成して、
他のリポジトリに書き込むこともできます。

見ればわかる通り、バージョン名には、パッケージ名、バージョン作成者

のイニシャル、およびバージョン番号が含まれています。バージョン名はリポ

ジトリ内のファイル名でもあります。これらの名前を変更してはいけません。

なぜならMonticelloの適切な操作は、これらの名前に依存しているからです!
Monticelloのバージョンファイルは単なる zipアーカイブであり、興味があれ
ば zipツールを使ってファイルを解凍できますが、その内容を見る最良の方法
は、Monticello自体を使うことです。

Monticelloでパッケージを作成するには、二つのことを行う必要がありま
す。一つはコードを書くこと、もう一つはそれをMonticelloに伝えることです。

PBE-Monticello と い う パ ッ ケ ー ジ を 作 成 し 、 図 6.16 の よ う に 二 つ の ク ラ ス
を そ の 中 に 置 い て く だ さ い 。 も ち ろ ん 、 Objectな ど の 既 存 の ク ラ ス の メ ソ ッ ド
を 作 り 、 120 ペ ー ジ に あ る 規 則 を 使 っ て こ の ク ラ ス と 同 じ パ ッ ケ ー ジ に 置 い て
く だ さ い — 図 6.17 を 参 照 の こ と 。

作成したパッケージを Monticelloに伝えるために、 +Packageボタンをク
リックし、パッケージ名、この場合は “PBE”を入力してください。Monticello
は PBEをパッケージリストに加えます。パッケージはアスタリスクでマークさ
れ、仮想イメージ内のバージョンがどのリポジトリにも書き込まれていないこ

とを示します。ここで、Monticelloに二つのパッケージを持つことに注意して
ください。一つは PBE、もう一つは PBE-Monticelloです。PBEは PBE-Monticello
や PBE-で始まる他のすべてのパッケージを含むため、これでよいのです。

最初はこのパッケージに関連している唯一のリポジトリは、図 6.18に示さ
れているようにパッケージキャッシュだけです。それで問題なく、コードを

保存することができますし、パッケージキャッシュに格納されます。 Saveボ
タンをクリックすると、図 6.19のように保存するパッケージのバージョンに
対するログメッセージを入力するよう求められます。このメッセージをアク

セプトするとMonticelloはパッケージを保存します。保存したことを示すた
め、Monticello のパッケージペインに付けられていたアスタリスクがなくな
り、バージョン番号が付与されます。

もしパッケージに変更を加えた— 例えばクラスの一つにメソッドを追加し

た — 場合、アスタリスクが再び現れて、変更が保存されていないことを示しま

す。パッケージキャッシュをリポジトリブラウザで開いて保存したバージョン

を選べば、Changesボタンや他のボタンを使うことができます。もちろんリポ

Figure 6.16: “PBE”パッケージの二つのクラス。

Figure 6.17: “PBE”パッケージに置かれる拡張メソッド。

ジトリに新しいバージョンを保存することもできます。Refreshボタンを押し
てリポジトリビューを一度更新すれば、図 6.20のように表示されるはずです。
新しいパッケージをパッケージキャッシュ以外のリポジトリに保存するた

めには、必要に応じてリポジトリを追加するなどして、まずMonticelloにその

Figure 6.18: Monticelloでまだ保存されていない PBEパッケージ。

Figure 6.19: パッケージの新バージョンに対するログメッセージを入力する。

Figure 6.20: 作成したパッケージの二つのバージョンがパッケージキャッシュ
にあります。

Figure 6.21: パッケージに関連づけられたポジトリのセットにリポジトリに追
加する。

リポジトリについて知らせる必要があります。その後、パッケージキャッシュ

上のリポジトリブラウザでCopyボタンを使って、コピー先のリポジトリを選
ぶことができます。図 6.21 のようにリポジトリ上でアクションクリック し、
add to package …を選ぶことで、お望みのリポジトリをパッケージと関連づける
こともできます。いったんパッケージがリポジトリに関連づけば、Monticelloブ
ラウザでリポジトリとパッケージを選んで Saveボタンをクリックすることで、新
しいバージョンを保存することができます。もちろんリポジトリへ書き込むため

のパーミッションは必要です。SqueakSource上の PharoByExampleリポジトリは
世界から読むことができますが、書き込むことはできません。そこに保存しよう

としてもエラーメッセージが表示されるでしょう。しかし、SqueakSourceにあ
なた専用のリポジトリを作成することは可能です。http://www.squeaksource.com
にあるWebインターフェースを使って、あなたのパッケージを保存するリポ
ジトリを利用できます。これは、友人とコードを共有したり複数のコンピュー

タを使ったりする場合などの仕組みとして特に有用です。

書き込みパーミッションを持たないリポジトリに対して保存しようとした

場合、ともかくそのバージョンはパッケージキャッシュに書き込まれます。つ

まり、リポジトリの情報を編集するか (Monticelloブラウザでアクションクリッ
ク)別のリポジトリを選んで、パッケージキャッシュブラウザからCopyボタン
を押せば、そのバージョンを復活させることができます。

http://www.squeaksource.com

Figure 6.22: TimeStamp nowをインスペクトする

6.4 インスペクタと [Explore]

Smalltalkが他の多くの言語と異なる特徴の一つは、静的なコードの世界では
なく、生きたオブジェクトの世界へ開かれた窓を提供しているということです。

そこではどのようなオブジェクトでもプログラマによって調べることが可能で

す。さらには変更することさえできます。— ただし、システムを構成してい

る基本的なオブジェクトを変更するようなときには注意が必要です。もちろん

言うまでもなく、最初にイメージを保存しておいてください！

インスペクタ

イ ン ス ペ ク タ を 使 っ て ど ん な こ と が で き る か 、 実 際 に 見 て み る た め に ワ ー

ク ス ペ ー ス で TimeStamp now と 入 力 し 、 ア ク シ ョ ン ク リ ッ ク で inpect it を 選 ん
で く だ さ い 。

(メニューを使う前にテキストを選択しておく必要はありません。テキスト
が選択されていない場合には、メニュー操作は現在の行すべてを対象にするか

らです。 inspect it を選ぶ代わりに cmd–iとタイプしても同じです)

図 6.22のようなウィンドウが現れます。これがインスペクタで、特定のオ
ブジェクトの内部に向かって開かれた窓と考えることができます。特定のオブ

ジェクトとはこの場合、 TimeStamp nowという式を評価することによって生成
された TimeStampのインスタンスオブジェクトのことです。

ウィンドウのタイトルバーには、インスペクトしているオブジェクトが印

字形式で表示されています。左側のペインの先頭の self を選択すると、右側
のペインにそのオブジェクトが文字列で表示されます。左側のペインはそのオ

ブジェクトのツリー構造ビューとなっており、selfをツリーの根としています。
インスタンス変数は、名前の横にある三角形を展開することで探求できます。9

インスペクタの下部にある横長のペインは小さなワークスペースウィンド

ウです。このウィンドウでは、擬似変数の selfが左側のペインで選択したオブ
ジェクトに束縛されている10ので便利です。そこで、

self - TimeStamp today

上記の式をワークスペースペインで inspect it すると、結果は Durationオブジェ
クトになります。このオブジェクトは、今日の午前０時と TimeStamp nowを評
価して得られた TimeStampオブジェクト（インスペクトしているもの）との、
時間間隔を表しています。さらに TimeStamp now - selfを評価すれば、あなたが
このセクションを読むのに費やした時間を教えてくれるでしょう！

selfだけでなく、オブジェクトのすべてのインスタンス変数も、ワークスペー
スペインの有効範囲にあります。つまり、変数に対して式で参照したり代入し

たりすることもできます。例えば、左のペインのツリーの根にあるオブジェク

トを選んでから、ワークスペースペインで jdn := jdn - 1を評価すると、インスタ
ンス変数 jdnの値が実際に変化することがわかりますし、TimeStamp now - self
の値も 1日分増えるでしょう。

DictionaryやOrderedCollectionやCompiledMethod、さらにいくつかの
クラスに対しては、インスペクタに特別な拡張がなされていて、これらのオブ

ジェクトの内容をより簡単に検査することができます。

オブジェクトエクスプローラ

オ ブ ジ ェ ク ト エ ク ス プ ロ ー ラ は概念的にはインスペクタとよく似ていますが、

情報の表し方が異なっています。この相違点を調べるために、先ほどまでイン

スペクトしていたものと同じオブジェクトをエ ク ス プ ロ ー ラ で探求してみま

しょう。

イ ン ス ペ ク タ の 左 側 の ペ イ ン で self を 選 び 、 ア ク シ ョ ン ク リ ッ ク で
explore (I) を 選 ん で く だ さ い

エクスプローラウィンドウは図 6.23のように表示されます。ツリー構造
の根の隣にある小さな三角形をクリックすると、ビューが図 6.24のように変
化します。このビューは、エクスプロアの対象とするオブジェクトのインスタ

ンス変数を表示します。offsetの隣の三角形をクリックすると、 そ の インスタ
ンス変数が見えます。エクスプローラは、複雑な階層構造を探求しなければな

らない場合に便利なため、この名前が付けられています。

オブジェクトエクスプローラのワークスペースペインの働きは、インスペク

タのものと若干異なります。selfはツリー構造のルートとなっているオブジェ
9訳注：Pharo 1.2 以降ツリーではなく一覧表示に変更されています

10訳注：Pharo 1.2 以降で self は常にインスペクトしているオブジェクトを束縛しています

Figure 6.23: TimeStamp nowを探求す
る

Figure 6.24: インスタンス変数を探求
する

クトではなく、現在選択されているオブジェクトに束縛されるようになってい

ます。また、選択されたオブジェクトのインスタンス変数も同様のスコープに

従います。

エクスプローラの真価を見るために、深い入れ子構造を持ったオブジェク

トを調べてみましょう。

ワ ー ク ス ペ ー ス で Object explore を 評 価 し て く だ さ い

これは Pharoの Objectクラスを表すオブジェクトです。メソッド辞書を表
すオブジェクトに加えて、そのクラスのコンパイル済みメソッドも直接ナビ

ゲートできることに注意してください（図 6.25参照）。

6.5 デバッガ

デバッガは間違いなく Pharoのツールの中で一番強力なツールです。ただデ
バッグのためというだけでなく、新しいコードを書くのにも利用されます。デ

バッガのデモンストレーションをするために、バグを作るところから始めま

しょう！

ブ ラ ウ ザ を 使 っ て 、 以 下 の メ ソ ッ ド を String ク ラ ス に 追 加 し て く だ さ い

Figure 6.25: Objectクラスを探求する

Method 6.1: バ グ が あ る メ ソ ッ ド
suffix
”自 分 自 身 を フ ァ イ ル 名 と 仮 定 し て 、 最 後 の ド ッ ト か ら 後 ろ の 部 分 を サ フ ィ ッ

ク ス と し て 返 す ”
| dot dotPosition |
dot := FileDirectory dot.
dotPosition := (self size to: 1 by: -1) detect: [ :i | (self at: i) = dot ].
↑ self copyFrom: dotPosition to: self size

もちろん、こんな簡単なメソッドは動くのが当たり前だろうと確信するので、

SUnitのテストも書かずにワークスペースで ’readme.txt’ suffixと入力し、print it(p)
を実行します。すると驚くことに、期待した答え ’txt’の代わりに、図 6.26の
ような PreDebugWindowがポップアップします。

PreDebugWindowはどんなエラーが起きたのかタイトルバーでわかるように
なっています。さらに、エラーに至るまでのメッセージの ス タ ッ ク ト レ ー ス

が表示されます。トレースの一番下を起点とし、 UndefinedObject>>DoItという
行が、ワークスペースで ’readme.txt’ suffixを選び、Pharoに print it させたとき
のコンパイルおよび実行されたコードを表しています。もちろん、このコード

は ByteStringクラスのオブジェクト (’readme.txt’)にメッセージ suffixを送ります。
これは Stringクラスから継承している suffixメソッドの実行を引き起こし、こ
のことがスタックトレースの次の行で ByteString(String)>>suffixとして表されて

Figure 6.26: PreDebugWindowによるバグの通知

います。スタックをさらにたどると、suffixメソッドは detect:メッセージを送
信し、…そして detect:ifNone:メソッドが errorNotFoundを送信していることがわ
かります。

な ぜ ドットが見つからなかったのかを調べるために、デバッガ本体が必要

なので、Debug ボタンをクリックします。

図 6.27にデバッガを示します。最初は威嚇的に見えますが、使うのはとて
も簡単です。タイトルバーと一番上のペインは PreDebugWindowにあるものと
とてもよく似ています。しかし、デバッガではスタックトレースはメソッドブ

ラウザと合体させており、スタックトレース上の行を選択すると対応するメ

ソッドが下のペインに表示されます。エラーを引き起こした実行はまだイメー

ジ内にありますが、中断状態にあることを意識することは重要です。スタック

トレースの各行は、実行を継続するのに必要なすべての情報を持った実行ス

タック上のフレームを表しています。これには、計算に伴うすべてのオブジェ

クトと各インスタンス変数、実行中のメソッドのすべての一時変数も含まれて

います。

図 6.27では、一番上のペインで detect:ifNone:メソッドが選択されています。
メソッドの本体は中央のペインに表示されています。valueメッセージが青く
ハイライトされているのは、このメソッドではメッセージ valueを送信し、そ
の結果を待っている状態であることを示しています。

デバッガの一番下の四つのペインは、実際には (ワークスペースペインを
持たない)二つの小さなインスペクタです。左側のインスペクタは現在のオブ
ジェクト、つまり、中央のペインで selfに相当するオブジェクトを表していま
す。別のスタックフレームを選ぶと、selfの示すオブジェクトが変わり、self-イ
ンスペクタの内容も変わります。左下のペインの selfをクリックすると、self
が Intervalオブジェクト (10 to: 1 by -1)であることがわかります。これは期待
通りのオブジェクトです。デバッガの小さなインスペクタにワークスペースペ

インは必要ありません。というのも、すべての変数がメソッドペインの有効範

囲内にあるからです。このペインに対する式の入力や選択、それらの式の評価

など自由に行うことができます。そうした変更は、いつでも cancel(l) メニュー

Figure 6.27: デバッガ

や cmd– lを使って取り消せます。
右側のインスペクタには、現在のコンテキストにおける一時変数が表示さ

れます。図 6.27では、valueが、引数 exceptionBlockに送られています。

スタックトレースで一つ下のメソッドを見ると、exceptionBlock は [self
errorNotFound: ...]です。エラーメッセージが表示されるのは驚くことではあり
ません。

ちなみに、ミニインスペクタに表示された変数に対して完全なインスペク

タやエクスプローラを開きたい場合は、単に変数名をダブルクリックするか、

変数名を選んで アクションクリック し、 inspect(i) もしくは explore(I) を選び
ます。こうしておくと、他のコードの実行中に変数の変化を見たい場合に便利

です。

メソッドウィンドウに戻って見てみると、メソッドの最後から 2番目の行
で、文字列 ’readme.txt’からドットを見つけようとしており、その実行が目的の
行に達していないことがわかります。Pharoは逆向きに実行させることはでき

Figure 6.28: detect: ifNone:メソッドを再実行した後のデバッガ

ませんが、メソッドを再実行させることは可能で、オブジェクト自体を変化さ

せず、新しいオブジェクトを生成するようなコードの場合にうまく働きます。

Restart を ク リ ッ ク し て く だ さ い 。 す る と 実 行 位 置 が 現 在 の メ ソ ッ ド の 最
初 の 文 に 戻 り ま す 。 次 に 送 信 さ れ る メ ッ セ ー ジ do: が 青 く ハ イ ラ イ ト し て 表 示
さ れ て い ま す ( 図 6.28 参 照 )。

Into と Over ボタンは二つの異なる方法でステップ実行します。Over ボタ
ンを押した場合、Pharoは現在のメッセージ送信 (この場合では do:)を、エラー
が発生しない限り 1ステップだけ実行します。つまり Over ボタンによって、
現在のメソッドの次のメッセージ送信へと移ります。この場合は valueとなり
ます。— これは、まさしく最初にいた場所なので、大した助けにはなりませ

ん。必要なことは、なぜ do:メソッドが求めている文字を見つけられないのか
を突き止めることです。

Over ボ タ ン を ク リ ッ ク し た 後 で 、 Restart ボ タ ン を ク リ ッ ク し て 図 6.28
の 状 態 に 戻 っ て く だ さ い 。

Into ボ タ ン を ク リ ッ ク し て く だ さ い 。 Pharo は ハ イ ラ イ ト さ れ て い た メ ッ
セ ー ジ 送 信 に 対 応 す る メ ソ ッ ド の 中 に 入 り ま す 。 こ の 場 合 は 、 Collection>>do:
で す 。

しかし、どちらにしてもこれはあまり助けにならないことがわかります。と

いうのも Collection>>do:が壊れていないことはかなり確信が持てます。Pharo
に対して 何 か を させているところにバグがあるはずです。この場合、Through
ボタンを用いるのが適しています。do:メソッド自身の詳細を無視して、引数
ブロックの実行に焦点を当てたいからです。

再 度 detect:ifNone: メ ソ ッ ド を 選 び 、 Restart ボ タ ン で 図 6.28 の 状 態 に 戻
し ま す 。 今 度 は Through ボ タ ン を 数 回 ク リ ッ ク し ま す 。 こ の と き 、 コ ン テ キ ス
ト ウ ィ ン ド ウ の each 変 数 を 選 択 し て お き ま す 。 do: メ ソ ッ ド の 実 行 に 従 っ て 、
each の 内 容 が 10 か ら カ ウ ン ト ダ ウ ン し て い く こ と が わ か り ま す 。

eachの値が 7のとき、ifTrue:ブロックが実行されることを期待しているの
ですが、そうなっていません。何が悪いのかを見るため、図 6.29で示すように
value:を Into ボタンでステップ実行します。

Into ボタンをクリックした後、図 6.30で示した位置にいることがわかりま
す。一見すると suffixメソッドに 戻 っ て しまったように見えますが、suffixメ
ソッドが detect:メソッドに引数として渡しているブロックを実行中なのです。
コンテキストインスペクタで dotを選ぶと、その値が ’.’であることがわかりま
す。ここに至って、なぜイコールとならなかったかわかりました。’readme.txt’
の 7番目の文字は Characterなのに、dotが Stringとなっています。

バグが判明したので修正方法も明白です。dotを探す前に、文字へ変換する
必要があります。

dot := FileDirectory dot first の よ う に 、 デ バ ッ ガ の 中 で コ ー ド を 正 し く 修 正 し 、
変 更 を accept し て く だ さ い 。

detect: の内側にあるブロック内のコードを実行しているため、この修正に
よっていくつかのスタックフレームが失われることになります。それでよいか

どうかを Pharoが確認します (図 6.31参照)、 yes をクリックして、新しいメ
ソッドを保存 (およびコンパイル)します。

’readme.txt’ suffix式の評価は完了し、’.txt’という答えを出力します。

この答えは正しいでしょうか？ 残念ながら、明確に答えることができませ

ん。サフィックスは .txtであるべきでしょうか？ それとも txt？ suffixメソッ
ドのコメントは全然明確ではありません。こうした問題を避ける方法は、答え

を定義する SUnitテストを書くことです。

Method 6.2: suffix メ ソ ッ ド の た め の 簡 単 な テ ス ト
testSuffixFound
self assert: ’readme.txt’ suffix = ’txt’

Figure 6.29: do: メソッドを数回 Through ボタンでステップ実行した後のデバ
ッガ

テストの記述に必要な労力は、同じテストをワークスペースで実行するよ

りも若干多くなりますが、SUnitを使うことで、テストを実行可能な文書とし
て保持し、他人が簡単にテストできるようになります。さらに、StringTestクラ
スにMethod 6.2を追加して SUnitでテストを実行すれば、エラーのデバッグ
に素早く戻ることができます。SUnitはアサーション違反があるとデバッガを
開きますが、図 6.32のように、スタックフレームを一つ遡り、Restart ボタン
でテストを再実行し、 Into ボタンで suffixメソッドに入ることでエラーを修正
できます。その後、SUnit Test Runnerで Run Failures ボタンをクリックして、
テストが通過することを確認するだけです。

以下はより良いテストコードです。

Method 6.3: suffix メ ソ ッ ド の た め の よ り 良 い テ ス ト
testSuffixFound
self assert: ’readme.txt’ suffix = ’txt’.
self assert: ’read.me.txt’ suffix = ’txt’

Figure 6.30: ’readme.txt’ at: 7が dotと等しくならない原因を示しているデバッガ

なぜこのテストがより良いものなのでしょう？ その理由は、対象となる文字

列に一つ以上のドットがある場合に、このメソッドがどのように振る舞うかを

読む人に教えてくれているからです。

エラーやアサーション違反の捕捉以外にも、デバッガに入るいくつかの方法

があります。コードの実行によって無限ループに陥ってしまった場合、 cmd–.
(あなたが英語を学んだ場所によりますが、終止符またはピリオドと呼ばれるも
の)を入力することで、割り込みをかけてその計算状況におけるデバッガを開
くことができます。11また、疑わしいコードの箇所に self haltと入れておくだけ
でもいいです。例えば、suffixメソッドであるならば、次のような感じです。

11同様に、どんなときでも cmd–shift–.の入力によって、緊急デバッガが起動することを覚えて
おくと便利です。

Figure 6.31: デバッガの中で suffix メソッドを修正する。内側のブロックから
抜けだすかどうかの確認

Method 6.4: suffix メ ソ ッ ド に halt を 入 れ る

suffix
”自 分 自 身 を フ ァ イ ル 名 と 仮 定 し て 、 最 後 の ド ッ ト か ら 後 の 部 分 を サ フ ィ ッ ク

ス と し て 返 す 。 ”
| dot dotPosition |
dot := FileDirectory dot first.
dotPosition := (self size to: 1 by: -1) detect: [ :i | (self at: i) = dot ].
self halt.
↑ self copyFrom: dotPosition to: self size

このメソッドを走らせると、 self haltの実行により pre-debuggerが開きま
すので、そこから、実行を続行させたり、デバッガに入って変数を見たり、ス

テップ実行したり、コードを編集したりすることができます。

デバッガについては以上ですべてですが、 suffix メソッドについてはまだ
です。最初のバグで気づいたでしょうが、対象となる文字列にドットがない場

Figure 6.32: デバッガの中で suffixメソッドを編集する: SUnitのアサーション
違反の後に Off-by-oneエラーを修正する

合、suffixメソッドはエラーを発生します。これは望ましい振る舞いではないの
で、こうしたケースで起きることを明示するために二つ目のテストを追加しま

しょう。

Method 6.5: suffix メ ソ ッ ド の た め の 二 つ 目 の テ ス ト : 対 象 が サ フ ィ ッ ク ス を 持
た な い

testSuffixNotFound
self assert: ’readme’ suffix = ’’

Method 6.5 を StringTest ク ラ ス の テ ス ト ス イ ー ト に 追 加 し 、 テ ス ト が エ
ラ ー を 起 こ す と こ ろ を 見 ま す 。 SUnit で エ ラ ー が 起 き た テ ス ト を 選 ん で デ バ ッ
ガ に 入 り 、 テ ス ト を 通 過 さ せ る た め に コ ー ド を 編 集 し ま す 。 最 も 簡 単 で 明 解 な

方 法 は 、 detect: メ ッ セ ー ジ の 代 わ り に detect:ifNone: を 用 い 、 2 番 目 の 引 数 を 単
に 文 字 列 の 大 き さ を 返 す ブ ロ ッ ク と す る こ と で す 。

SUnitについては第 7章でより詳しく学びます。

Figure 6.33: プロセスブラウザ

6.6 プロセスブラウザ

Smalltalkはマルチスレッドなシステムです。イメージ中では、たくさんの軽
量プロセス (スレッドとして知られる)が並行して走っています。将来的には
Pharoの仮想機械は可能ならばマルチプロセッサを利用するようになるかもし
れませんが、現時点では、平行性は時分割処理で実装されています。

プロセスブラウザは、デバッガの親類で、Pharoの中で稼働している様々
なプロセスを見せてくれます。図 6.33にスクリーンショットを示します。左
上のペインは Pharoの中のすべてのプロセスのリストで、優先度 80の割り込
みタイマー監視のプロセスから、優先度 10のアイドルプロセスに至るまでが
優先度順に並んでいます。もちろん、単一プロセッサ上ですので、見ていると

きに動いている唯一のプロセスはUIプロセスです。他のすべてのプロセスは、
何らかのイベントを待っています。デフォルトでは、プロセスの表示は静止し

ていますが、アクションクリック ingで turn on auto-update(a) を選ぶことで更
新することができます。

左上のペインでプロセスを選ぶと、ちょうど、デバッガのように、右上のペ

インにそのプロセスのスタックトレースが表示されます。スタックトレースを

選ぶと、対応するメソッドが下部のペインに表示されます。プロセスブラウザ

は selfや thisContextのためのミニインスペクタを備えてはいませんが、スタッ
クトレース上で アクションクリック ing することにより同様の機能を提供し
ます。

Figure 6.34: randomを部分文字列としてセレクタ名に持つすべてのメソッドを
表示しているメッセージネームブラウザ

6.7 メソッド検索

Pharo にはメソッド検索のための二つのツールがあります。両者はインター
フェースも機能性も異なっています。

メ ソ ッ ド ・ フ ァ イ ン ダ は 1.9節である程度説明したように、メソッドを名
前や機能性から探すことができます。しかし、メソッドの本体を見ようとする

と、メソッド・ファインダは新たにブラウザを開いてしまいます。これではす

ぐにウィンドウで溢れてしまいます。

メ ッ セ ー ジ ネ ー ム ブ ラ ウ ザ は、より限定的な検索機能を持っています。検

索用の入力欄にメッセージ・セレクタの一部を入力すると、図 6.34のように、
名前にその一部を含んだすべてのメソッドがブラウザに表示されます。さらに、

これは一人前のブラウザでもあります。左側のペインにある名前の一つを選択

すると、その名前を持つすべてのメソッドが右側のペインに表示され、下部の

ペインでブラウズできます。ブラウザと同様に、メッセージネームブラウザは

ボタン・バーを備えており、選択したメソッドやそのクラスを他のブラウザで

開くことができます。

Figure 6.35: チェンジソーター

6.8 チェンジセットとチェンジソーター

Pharoで作業をしている間、メソッドやクラスに対するどのような変更もチェ
ンジセットに記録されます。新しいクラスを作成したこと、クラス名を変更し

たこと、カテゴリを変更したこと、既存のクラスにメソッドを追加したことな

ど—重要な変更はすべて記録されます。しかし、きままな doitは含まれない
ので、例えばワークスペースで新しくグローバル変数を定義した場合には、変

数の作成はチェンジセットに記録されません。

多くのチェンジセットが存在していますが、常にそのうちの一つだけがイ

メージに対する変更を収集します。—これをカレントチェンジセットと呼び
ます— World .Tools ….Change Sorterを選ぶと現れるチェンジソーターを使え
ば、どれがカレントチェンジセットなのか確認したり、すべてのチェンジセッ

トについて調べたりすることができます。

図 6.35にチェンジソーターを示します。タイトルバーは、どのチェンジセッ
トがカレントチェンジセットなのかを表すとともに、カレントチェンジセット

が選択された状態でチェンジソーターが開きます。

左上のペインで他のチェンジセットを選ぶことができます。アクションク

リックメニューで他のチェンジセットをカレントチェンジセットにしたり、新

しいチェンジセットを作ったりすることができます。右隣のペインには、選択

したチェンジセット (とそのカテゴリ)が影響を与えるすべてのクラスがリスト
表示されます。クラスの一つを選ぶと、(クラスのすべてのメソッド で は な く )
そのチェンジセットに存在するメソッドの名前が左中央のペインに表示され、

さらにメソッド名を選ぶと下のペインにメソッドの定義が表示されます。チェ

ンジソーターは、クラスの生成自体がチェンジセットの一部であるかどうかを

示 さ な い ことに注意してください。なお、クラス生成に関する情報は、チェン

ジセットを表現するためのオブジェクト構造には格納されています。

チェンジソーターでは、クラスやメソッドの上で アクションクリック メ

ニューを使うことで、チェンジセットから削除することができます。

チェンジソーターは、同時に二つのチェンジセットを扱え、一つは左側で

もう一つは右側に表示されます。この配置により、チェンジソーターの主な特

徴である、一方のチェンジセットからもう一方へと変更の移動やコピーが支援

されます。図 6.35にアクションクリックメニューを示します。個々のメソッ
ドを一方からもう一方へコピーすることもできます。

なぜ、チェンジセットの合成に注意を払わなければならないのか、不思議に

思うかもしれません。その答えは、チェンジセットが Pharoからファイルシス
テムにコードを書き出すシンプルな仕組みを提供するからです。つまり、コー

ドを他の Pharoイメージに取り込んだり、Pharo以外の Smalltalkに取り込ん
だりすることができるからです。チェンジセットの書き出しは “ファイルアウ
ト”として知られているもので、どんなチェンジセットでも、どんなブラウザ
のクラスやメソッドでも、アクションクリックメニューにより行うことができ

ます。ファイルアウトを繰り返すと、ファイルの新しいバージョンが生成され

ますが、チェンジセットはMonticelloのようなバージョン管理ツールではな
いので、依存関係の追跡は行いません。

Monticelloの出現以前、チェンジセットは Smalltalkerの間でコードを交
換する主な手段でした。チェンジセットは、単純さ (ファイルアウトしたもの
は単なるテキストファイルですが、テキストエディタで編集するようなことは

お 勧 め し ま せ ん )と可搬性の高さという利点を持っています。
チェンジセットの主な欠点は、Monticelloパッケージに比べて依存関係の

概念をサポートしていないということです。ファイルアウトしたチェンジセッ

トは、読み込まれる際にイメージに変更を加える ア ク シ ョ ン の集合です。チェ

ンジセットを無事に読み込むためには、イメージはふさわしい状態にあること

が求められます。例えば、チェンジセットがあるクラスへメソッドを追加する

アクションを含んでいる場合、あらかじめイメージにそのクラスが定義されて

いるときのみ、読み込みを完了できます。同様にチェンジセットが、クラス名

の変更やカテゴリの変更を行う場合も、当然そのクラスがイメージに存在して

いるときのみ正しく動作します。ファイルアウト時には、クラスに定義されて

いるインスタンス変数をメソッドが参照しているのに、それを取り込もうとし

ているイメージには変数が存在していないこともあり得ます。問題は、チェン

ジセットがファイルインできる条件を明示していないことです。ファイルイン

のプロセスはうまくいくことが望まれますが、うまくいかない場合、たいてい

わけのわからないエラーメッセージとスタックトレースで終わることになり

ます。たとえファイルインが成功しても、あるチェンジセットが別のチェンジ

セットによる変更を勝手に元に戻してしまうことすらあります。

これとは対照的にMonticelloパッケージではコードを宣言的な流儀で表現
します。つまり、読み込みの完了後にイメージがどのような状態であるべきか

Figure 6.36: ファイルリストブラウザ

を記述します。これにより (二つのパッケージが最後の段階で矛盾した要求を
持つ場合など)コンフリクトについて警告し、依存関係の順序で一連のパッケー
ジをロードすることを提案します。

こういった欠点にも関わらず、チェンジセットは未だに使われています。特

にインターネット上では、参考にしたり場合によっては使いたいと思うチェン

ジセットが見つるかもしれません。さて、チェンジソーターを使ってチェンジ

セットをファイルアウトするところを見てきたので、今度はファイルインの方

法を説明しましょう。それには別のツール、ファイルリストブラウザが必要に

なります。

6.9 ファイルリストブラウザ

ファイルリストブラウザ は、実際のところ Pharo からファイルシステ
ム (FTP サーバーも含む) をブラウズするための汎用的なツールです。
World .Tools ….File Browser メニューで開くことができます。もちろん、みな
さんのローカルのファイルシステムの内容次第で見えるものは変わりますが、

典型的な見え方を図 6.36に示します。
最初にファイルリストブラウザを開くと、カレントディレクトリ、つまり

Pharoを起動したディレクトリが選択された状態になります。タイトルバーに
はこのディレクトリへのパスが表示されます。左側の大きなペインは、ごく一

般的なやり方でファイルシステムをたどるのに使えます。ディレクトリが選択

されると、そのディレクトリ内のファイル (ディレクトリは含まれない12)のリ
12訳注：実際には表示されます



144 Pharo の プ ロ グ ラ ミ ン グ 環 境

Figure 6.37: ファイルコンテンツブラウザ

ストが右側に表示されます。このファイルのリストは、Unix流のパターンを
ウィンドウの左上にある小さな入力ボックスに入力することでフィルタをかけ

ることができます。最初、このパターンは *であり、すべてのファイル名にマッ
チしますが、異なる文字列を入力してアクセプトすると、パターンが変更され

ます。(入力したパターンの前後には、暗黙的に *が付加されることに気をつけ
てください）ファイルの並び順は、name、date そして size ボタンで変更でき
ます。その他のボタンはブラウザで選択したファイル名に依存します。図 6.36
では、ファイル名のサフィックスが .csであると、ブラウザがチェンジセット
だと推測し、 install する（この場合、そのファイル名に基づいた名前のチェン
ジセットに フ ァ イ ル イ ン されます）ためのボタンや、ファイルの変更点を見

るための changes ボタン、ファイル内のコードを調べるための code ボタン、
カ レ ン ト チ ェ ン ジ セ ッ ト にコードを取り込むための filein ボタンを提供します。
conflicts ボタンは、イメージに既にあるコードとコンフリクトするようなチェ
ンジセットの変更を教えてくれるものと推測されるかもしれませんが、それは

違います。 その代わりに（改行の有無など）ファイルが正しく読み込まれる

かといった潜在的な問題をチェックします。

ボタン表示の選択はファイル内容ではなくファイルの 名 前 に依存している

ことから、時折期待するボタンが画面に現れないことがあります。しかし、す

べてのオプションは常にアクションクリックの more … から利用できるので、
簡単にこの問題へ対処できます。

code ボタンは、おそらくチェンジセットを扱う上で最も便利なもので、チェ
ンジセットファイルの内容をブラウザで開きます。例を 図 6.37 に示します。
このファイルコンテンツブラウザはブラウザによく似ていますが、クラスとプ

ロトコルとメソッドだけを表示し、カテゴリを表示しません。個々のクラスご

とに、そのクラスが既にシステムに存在しているかどうか、ファイルで定義さ

れているかどうか（定義が 同 一 か 否 か は示しません）がわかるようになってい

ます。個々のクラスのメソッドを表示した上で、さらに（図 6.37のように）現
在のバージョンとファイルのバージョンとの差分を表示します。上部の四つの

ペインのコンテキストメニューでは、チェンジセット全体をファイルインする

のか、それとも対応するクラスや、プロトコル、メソッドごとにファイルイン

するかどうかを指示できます。

6.10 Smalltalkでは、コードを失うことはありません

Pharoがクラッシュすることは十分あり得ることです。実験的なシステムであ
り、Pharoが機能するために必要なものを含め、すべてが変更可能になってい
るからです！

悪 意 を 持 っ て Pharo を ク ラ ッ シ ュ さ せ ま し ょ う 。 Object become: nil を 試 し
て み て く だ さ い

クラッシュしてイメージが何時間も前に保存したバージョンに戻ってしまっ

たとしても、良い知らせがあります。それはみなさんがやったことは決して失

われないということです。というのも、実行したあらゆるコードが .changes
ファイルに保存されているからです。あらゆるコードです! これにはワークス
ペースで評価した 1行プログラムから、プログラミング中にクラスへ追加した
コードまで含まれます。

ここでは、どのようにコードを回復するかという手順を説明します。必要

なときまで、特に読む必要はありませんが、そうなったときに見つけられるよ

う、ここでみなさんをお待ちしています。

最悪の場合では、.changesファイルをテキストエディタで開くこともできま
すが、数メガバイトに達するため遅くなりますし、お勧めできません。Pharo
は、より良い方法を提供します。

コードを回復する方法

最 後 の ス ナ ッ プ シ ョ ッ ト の Pharo を リ ス タ ー ト し て 、

World.Tools ….Recover lost changes を選びます。

すると、履歴をどれくらい遡ってブラウズしたいかを選ぶことができます。

通常、最後のスナップショットまで遡って変更をブラウズすることで十分で

す。（同様の効果を ChangeList browseRecent: 2000というコードを編集すること
で得ることができます。ただし 2000という数値は試行錯誤で決めることにな
ります）

recent changesブラウザが表示されますが、これは最後のスナップショット
の後の変更で、Pharoに対して行ったすべての操作のリストとなります。この

リストからアクションクリックメニューで項目を削除することができます。こ

れで十分と判断したら、残ったものをファイルインできます。このように、新

しいイメージに変更を加えます。ファイルインする前に、普通のチェンジセッ

トブラウザを使って新しいチェンジセットを作るのは良いアイデアです。これ

によって復元したすべてのコードが新しいチェンジセットの中に入るからです。

その後、チェンジセットをファイルアウトすることができます。

recent changesブラウザの便利な機能に remove doIts があります。たいてい、
do itの操作はファイルイン (つまり再実行)したくないでしょう。しかし、例
外もあります。クラス定義は doIt として表現されるからです。あ る ク ラ ス の メ
ソ ッ ド を フ ァ イ ル イ ン す る 前 に 、 ク ラ ス は 存 在 し て い な け れ ば な り ま せ ん 。 そ

こで、新しいクラスを定義する際には、最 初 に クラスを定義する do itをファ
イルインし、その後 remove doIts してからメソッドをファイルインします。

私がリカバリが完了するときは、新しいチェンジセットをファイルアウト

した後、イメージを保存することなく Pharoを終了します。再起動して、きれ
いな状態に戻ってから新しいチェンジセットを適用するようにしています。

6.11 まとめ

Pharoで開発作業を効果的に行うためには、開発環境が提供するツールについ
て学ぶ努力を惜しまないことが大切です。

• 標準的な ブ ラ ウ ザ は、既存のカテゴリやクラス、メソッドプロトコル
やメソッドを調べたり、新しく定義したりするための中心的なインター

フェースです。ブラウザは、メッセージのセンダやインプリメンタ、メ

ソッドのバージョンへ直接ジャンプするといった、いろいろな便利なボ

タンを提供します。

• （OmniBrowserやリファクタリングブラウザといった）様々なブラウザ
や、クラスやメソッドへの様々なビューを提供する（階層ブラウザなど）

特殊用途のブラウザがあります。

• どんなツールでも、クラスやメソッドの名前を選んだ後に cmd–bのキー
ボードショートカットを使えば、すぐにブラウザへジャンプできます。

• SystemNavigation defaultに対してメッセージを送ることでも、 Smalltalk
システムをプログラミング的にブラウズできます。

• Monticelloは、クラスやメソッドのパッケージを外部へ出力したり、取り
込んだり、バージョン管理や共有するためのツールです。Monticelloの
パッケージは、カテゴリやサブカテゴリと他のカテゴリに属する関連メ

ソッドプロトコルから構成されます。

• イ ン ス ペ ク タ と エ ク ス プ ロ ー ラ の二つのツールは、イメージの中で生
きているオブジェクトを調べたり対話したりするための便利なツールで

す。メタクリックすることによりMorphic Haloを出し、デバッグハン
ドル を選ぶと、ツールをインスペクトすることもできます。

• デ バ ッ ガ は、エラーが起きた際にプログラムの実行スタックを見せるだけ
でなく、ソースコードを含め、アプリケーションのすべてのオブジェクト

と対話することを可能にするツールです。多くの場合、デバッガでソース

コードを修正し、実行を継続することが可能です。デバッガは SUnit(第 7
章)と並んでテストファースト開発を支援する効果的なツールです。

• プ ロ セ ス ブ ラ ウ ザ は、イメージの中で現在実行中のプロセスを監視した
り問い合わせたり対話したりできます。

• メ ソ ッ ド ・ フ ァ イ ン ダ と メ ッ セ ー ジ ネ ー ム ブ ラ ウ ザ は、メソッドの場
所を突き止めるツールです。前者は、名前がはっきりわからないが期待

される振る舞いがわかってる場合に便利です。後者は、名前の一部しか

わからない場合に、ブラウズのための優れたインターフェースを提供し

ます。

• チ ェ ン ジ セ ッ ト は、イメージのソースコードに対するすべての変更を自
動的に記録されるログのことです。主にソースコードのバージョンの保

存や交換の手段としては、Monticelloに取って代わられていますが、ご
くまれに発生し得る壊滅的な故障からの復旧などでは未だに便利です。

• フ ァ イ ル リ ス ト ブ ラ ウ ザ はファイルシステムをブラウズするツールです。
これはファイルシステムからソースコードを filein することもできます。

• イメージを保存したりMonticelloでソースコードをバックアップしたり
する前にクラッシュしてしまった場合、チ ェ ン ジ リ ス ト ブ ラ ウ ザ を使う

ことで最新の変更から回復することができます。再実行したい変更を選

んで、最新のイメージのコピーに取り込むことができます。





Chapter 7

SUnit

7.1 はじめに

SUnitは小さいながらも強力な、テストの作成と運用のためのフレームワーク
です。その名前からわかるように、SUnitはユ ニ ッ ト テ ス ト を対象として設計
されていますが、統合テストや機能テストにも使えます。初期の SUnitはケン
ト・ベック (Kent Beck)が開発しました。その後ジョセフ・ペルリン (Joseph
Pelrine)を始めとする開発者たちが、SUnitにテストリソース (7.6節)の概念
を取り入れました。

テストとテスト駆動開発のメリットは、Pharoや Smalltalkにとどまりませ
ん。テストの自動化はアジャイルソフトウェア開発の特徴であり、ソフトウェ

アの品質を上げたい開発者なら採用するでしょう。ユニットテストの効果を高

く評価する開発者は多く、今では多くの言語に xUnit ライブラリがあります
(Java、Python、Perl、.Net、Oracleなど)。この章では SUnit 3.3 (原文の執筆
時のバージョン)について解説します。SUnitの最新版は、sunit.sourceforge.net
で確認できます。

テストをすることやテストスイートを作ることは、新しい考えではあ

りません。誰でもテストはエラーを発見する良い方法だと知っています。

エクストリーム・プログラミング (eXtreme Programming)ではテストが中心
的なプラクティスと位置付けられ、テストの自 動 化 が重視されました。その

結果、テストはプログラマにとって面倒な作業から生産的で楽しい作業に変

わりました。Smalltalk コミュニティは、長らくテストを重視してきました。
Smalltalkのプログラミング環境ではインクリメンタルな開発が容易だからで
す。Smalltalkの伝統的な開発では、プログラマはメソッドを実装するとすぐ
にワークスペースでテストします。メソッドのコメントにテストを書いたり、

セットアップが必要なテストはサンプルメソッドとして実装したりする場合も

あります。しかしこれらの方法には問題があります。まずワークスペースにテ

ストを書いた場合 (イメージファイルを共有しない限り)他のプログラマから使

sunit.sourceforge.net

えません。コメントやサンプルメソッドにした場合、状況は多少改善されます

がそれでも、メソッドの開発に合わせてテストを見直し、さらに自動的に実行

させるのは簡単ではありません。テストが実行されなければバグも見つけられ

ません! さらにサンプルメソッドは、何が期待された動作なのかについて何も
語りません。サンプルを実行して—意外な—結果が得られるかもしれません
が、それが正しい振る舞いかどうかはわからないのです。

SUnitは、テストの結果が正しいかどうかをテスト自身で検証できるとこ
ろに価値があります。またテストをグループ分けしたり、テストに必要なコン

テキストを準備したり、グループ単位でテストを自動実行することも可能に

なります。もうワークスペースに小さなコードを書かなくても、2分もあれば
SUnitでテストが書けるようになります。SUnitを使ってテストの自動化をし
ましょう。

この章では、テストを行う理由とより良いテストを書く方法の議論から始

めます。小さな例を挙げながら SUnitの使い方を紹介し、最終的には SUnitの
内部実装を見ていきます。SUnitを支えている Smalltalkのリフレクションの
力がわかるはずです。

7.2 なぜテストは重要か

残念ながら、テストを時間の無駄と思い込んでいる開発者もたくさんいます。

彼 ら はバグを作りません— 他 の プログラマだけがバグを作るのです。誰しも
一度くらいこう言ってしまった経験があります: 「時間があればテストを書き
ますよ。」絶対にバグのないコードが書けて、そのコードが将来にわたって絶

対に変更されないなら、確かにテストは時間の無駄です。しかし、そのような

コードはごく小さいか、あるいはあなたを含む誰も使わないようなものがほと

んどです。テストは将来への投資だと考えてください。テストはすぐにでも役

に立ちますが、将来アプリケーションや実行環境が変わったとき、絶 大 な効果

を発揮します。

テストにはいくつかの役割があります。一つは生きたドキュメントです。す

べてのテストに成功すれば、ドキュメントは最新版です。もう一つは、変更し

たコードが無事動くという確信です。もしシステムの一部が壊れたとしても、

すぐに原因を見つけることができます。最後に、コードを書くとき—あるい
はさらに進んでコードを書く前に—テストを書くことで、実装について考え
るよりもむしろ設計しようとしている機能について、そ し て そ れ を ど う ク ラ イ

ア ン ト コ ー ド に 見 せ る の か に つ い て 考えるようになります。コードを書く前に

テストを書くことで、コードが動作するのに必要なコンテキスト、コードとク

ライアントとのやりとり、期待される結果をまず明確にしなければならなくな

ります。きっとコードの品質が向上します。ぜひ試してみてください。

現実のあらゆるアプリケーションのすべての側面をテストするのは不可能

です。いくら良質のテストスイートを書いても、システムを壊す機会を窺って

いるバグが入り込みます。完全なテストを書くことを目標にするのではなく、

バグを見つけたらすぐにあぶりだすテストを書き、実行し、失敗を確認し、修

正を始めるようにしましょう。テストが成功すれば修正作業は終わりです。

7.3 良いテストを書くには?

良いテストが書けるようになるには、練習するのが一番です。テストのメリッ

トを最大限に活かすヒントを見ていきましょう。

1. テストは繰り返し実行可能であるべきです。実行したいときに実行でき
て、常に同じ結果を出すべきです。

2. テストは人の手を借りずに実行可能であるべきです。テストはたとえ夜
中でも実行可能であるべきです。

3. テストは物語であるべきです。各テストの目的を一つに絞るべきです。
テストをシナリオと考え、読めばアプリケーションの機能が理解できる

ようにすべきです。

4. テストの変更はなるべく控えるべきです。アプリケーションを変更する
たびにテストを変更する必要があっては大変です。テストを変更せずに

すませるには、主に公開インターフェースを使ってテストを書きましょ

う。プライベートな「ヘルパー」メソッドについては、それ自体が複雑

な処理をするようならテストを書いてもいいでしょう。ただし、公開イ

ンターフェースの内部実装が変わったときには、そのテストを変更する

か捨てる必要があるかもしれないことに気をつけましょう。

テストの目的を絞り (3)、テストすべき機能とテストの数とのバランスを取
りましょう。システムの一部を変更したときにすべてのテストではなく、限ら

れた数のテストだけが失敗するようにすべきです。なぜなら、100のテストに
失敗した方が 10のテストの失敗ですんだことより重大なメッセージだからで
す1。ただし、いつもこの通りにできるとは限りません。例えば変更によって
オブジェクトの初期化やテストのセットアップに問題が出れば、すべてのテス

トが失敗してしまいます。

エクストリーム・プログラミングでは、コードを書く前にテストを書くこ

と (テストファースト)を提唱しています。これはソフトウェア開発者の本能
と相反するように見えます。我々に言えるのは、「さあ一歩前に進んで試して

みよう」ということです。わかっているのは、テストを先に書くことで、どん

なコードを書くべきか、何をすべきか、どう要件をクラスに落とせばいいか、

どんな API にすればいいのかを考える手助けになることです。さらにテスト
ファースト開発は、開発速度を上げる自信にもつながります。開発を急ぐあま

りに肝心な点を見落とすかもしれない、と言った心配をする必要がなくなるか

らです。

1訳注: 一つの原因ですべてのテストに失敗するようでは、テストの失敗の数から問題の重大さ
がわかりません。

7.4 SUnit by example

SUnitの詳細に触れる前に、一つずつ例を挙げていきます。Setクラスのテス
トを例に進めます。例に従ってコードを入力してみましょう。

ステップ 1: テストクラスを作る

ま ず TestCase の サ ブ ク ラ ス ExampleSetTest を 作 り ま し ょ う 。 こ の ク ラ ス
に 二 つ の イ ン ス タ ン ス 変 数 full と empty を 追 加 し ま し ょ う 。

Class 7.1: ExampleSetTest ク ラ ス
TestCase subclass: #ExampleSetTest
instanceVariableNames: ’full empty’
classVariableNames: ’’
poolDictionaries: ’’
category: ’MySetTest’

ExampleSetTestクラスに Setクラスのテストをまとめます。このクラスには
テスト実行時のコンテキスト (テストの実行に必要な状況、状態)を定義しま
す。ここではコンテキストは二つのインスタンス変数 fullと emptyで表します。
fullは要素を持つ Setのインスタンス、emptyは空の Setのインスタンスを表し
ます。

クラスの名前は SUnitにとって重要ではありませんが、慣例により Testで
終わる名前にすべきです。Patternクラスのテストクラスなら、PatternTestとす
れば、ブラウザで見たときに (どちらも同じカテゴリであれば)両方のクラスが
アルファベット順に並びます。テストクラスが TestCaseのサブクラスであるこ
とが重要です2。

ステップ 2: テストのコンテキストを初期化する

setUpメソッドで、テスト実行時のコンテキストを定義します。ちょっと initialize
メソッドに似ていますね。setUpは各テストメソッド (後述)の実行前に実行さ
れます。

setUp メ ソ ッ ド を 次 の よ う に 定 義 し ま し ょ う 。 イ ン ス タ ン ス 変 数 empty に
空 の Set を 代 入 し 、 full に 二 つ の 要 素 を 持 つ Set を 代 入 し ま し ょ う 。

Method 7.2: フ ィ ク ス チ ャ を セ ッ ト ア ッ プ す る
ExampleSetTest»setUp
empty := Set new.
full := Set with: 5 with: 6

2訳注: 名前に関する慣例と違い、SUnitにとって重要です。

テスト用語では、コンテキストのことをテストのフ ィ ク ス チ ャ と呼びます。

ステップ 3: テストメソッドを書く

それではテストを書いてみましょう。ExampleSetTestクラスにテスト用のメソッ
ドを定義します。一つのメソッドで一つのテストを表します。テストメソッド

は、名前を『test』で始めます。SUnitは『test』で始まるメソッドを集めて、テ
ストスイートとしてまとめます。テストメソッドは引数を取りません。

次 の テ ス ト メ ソ ッ ド を 定 義 し ま し ょ う 。

最初のテスト testIncludesは、Setの includes:メソッドをテストします。こ
のテストは、5を要素に持つ Setにメッセージ includes: 5を送信すると trueが
返ると言っています。明らかにこのテストは、setUpメソッドが先に実行され
ていることを前提にしています。

Method 7.3: 要 素 を 含 む こ と を テ ス ト す る
ExampleSetTest»testIncludes
self assert: (full includes: 5).
self assert: (full includes: 6)

二つ目のテスト testOccurrencesは、既に 5が含まれている Setにさらにも
う一つの 5を追加しても、5の数は 1のままであることを検証しています。

Method 7.4: 要 素 の 数 を テ ス ト す る
ExampleSetTest»testOccurrences
self assert: (empty occurrencesOf: 0) = 0.
self assert: (full occurrencesOf: 5) = 1.
full add: 5.
self assert: (full occurrencesOf: 5) = 1

最後に、5を削除すると Setは 5を含まなくなることをテストします。

Method 7.5: 削 除 を テ ス ト す る
ExampleSetTest»testRemove
full remove: 5.
self assert: (full includes: 6).
self deny: (full includes: 5)

真でないことをアサート (assert:)するのに deny:メソッドを使っていることに
注意してください。aTest deny: anExpressionは aTest assert: anExpression notと等
価ですが、よりわかりやすくなります。

ステップ 4: テストを実行する

テストを実行するにはブラウザを使うのが一番簡単です。パッケージ、クラス

名、テストメソッドのいずれかをアクションクリックし、run the tests (t) を選択
するだけです。テストが成功したメソッドには緑のマークが、失敗したメソッ

ドには黄のマークが付きます。クラスにもマークが付きます。つまり、すべて

のテストに成功すると緑のマークが、いくつか失敗すると緑と赤の混じった

マークが、全部失敗すると赤のマークがクラスに付きます3。

Figure 7.1: ブラウザから SUnitのテストを実行する

SUnit テ ス ト ラ ン ナ ー (図 7.2) を使えば、実行するテストスイートを複数
選択してテスト結果の詳細なログを見ることができます。テ ス ト ラ ン ナ ー は

World .Test Runner で開きます。テ ス ト ラ ン ナ ー は、まとまったテストを簡単
に実行できるように設計されています。最も左にあるペインには、テストクラ

ス (す な わ ち TestCaseのサブクラス)を含むすべてのカテゴリが表示されます。
カテゴリを選択すると、そのカテゴリに含まれるテストクラスのクラス階層が

隣のペインに表示されます。抽象クラス (テストメソッドを含まないテストク
ラス)はイタリック体で表示され、テストクラスの階層はインデントによって
示されます。例えば、ClassTestCase のサブクラスは TestCase のサブクラスよ
り深くインデントされます4。

テ ス ト ラ ン ナ ー を 開 い て MySetTest カ テ ゴ リ を 選 択 し 、 Run Selected ボ タ
ン を ク リ ッ ク し ま し ょ う 。

3訳注: PBE.imageの場合。
4訳注: TestCaseが表示されるのは SUnit-Kernel カテゴリを選んだ場合か、どのカテゴリも選ば

ない場合です。

Figure 7.2: Pharo SUnitテストランナー

ExampleSetTest»testRemove に バ グ を 入 れ て 、 再 度 テ ス ト を 実 行 し て み ま
し ょ う 。 例 え ば 5 を 4 に し ま す 。

失敗したテスト (あれば)は、テ ス ト ラ ン ナ ー の右のペインに表示されます。
失敗の原因を知りたければ、表示されたテストメソッドをクリックしてくださ

い。ここからデバッグもできます。

ステップ 5: 結果を分析する

TestCaseクラスの assert:メソッドは、真偽値の引数を一つ取ります。通常はテ
スト対象となる式を指定します。引数が真ならテストに成功、偽なら失敗です。

テストの実行結果には 3通りあります。すべてのテストに成功する場合と、
テストに失敗する場合と、テストの実行中にエラーが発生してテストが実行で

きない場合です。すべてのテストに成功すると、テストランナーの右上のバー

が緑に変わります。テストに失 敗 するとバーが黄色に変わり、中段のペインに

失敗したテストが表示されます5。最後が、テストの実行中にエラー (例外)が発
5訳注: PBE.imageでは、エラーの発生とは無関係に、失敗したテストがあった場合必ずバーが

黄色になります。

生する場合です。例えばmessage not understoodや index out of boundsと言った
エラーが発生する可能性があります。エラーが発生するとバーが赤に変わり、

エラーの発生したテストが下段のペインに表示されます。

エ ラ ー が 発 生 し た り 失 敗 し た り す る よ う に テ ス ト を 変 更 し て み ま し ょ う 。

7.5 SUnitクックブック

この節では、SUnitのより詳しい使い方を解説します。JUnit6 などの他のテス
ティングフレームワークの経験があるなら、SUnit は親しみやすいでしょう。
どのフレームワークも SUnitにルーツがあります。通常 SUnitでは GUIでテ
ストを実行しますが、GUIを使いたくない場合もあるでしょう。

様々なアサーション

assert:や deny:の他にもアサーション用のメソッドがあります。

一つ目は assert:description: と deny:description: です。二つ目の引数には、テ
ストが失敗した理由を示す文字列を指定します。失敗の原因がわかりにくい場

合に使うといいでしょう。 7.7節でこれらのメソッドについて触れます。
次は should:raise: と shouldnt:raise: です。これらは、テストの実行時に適

切な例外が発生するかどうかをテストします。例えば self should: aBlock raise:
anExceptionを実行した場合、aBlockの実行中に例外 anExceptionが発生すれば
テストは成功です。Method 7.6に should:raise:を使った例を示します。

こ の テ ス ト を 実 行 し て み ま し ょ う 。

should:と shouldnt:の最初の引数は、評価する式を含 む ブ ロ ッ ク です。

Method 7.6: エ ラ ー の 発 生 を テ ス ト す る
ExampleSetTest»testIllegal
self should: [empty at: 5] raise: Error.
self should: [empty at: 5 put: #zork] raise: Error

SUnitは移植性が確保されていて、どの Smalltalkの処理系でも使えます。
SUnitの開発者は、処理系依存のコードを括り出すことで移植性を確保しまし
た。例えば TestResult class»error クラスメソッドは、処理系によるシステムエ
ラーのクラスの違いを吸収します。これを活かしてMethod 7.6のコードを次
のように書き換えると、同じテストを Smalltalkのあらゆる処理系で動かすこ
とができます。

6http://junit.org

http://junit.org

Method 7.7: 移 植 性 の 高 い エ ラ ー 処 理
ExampleSetTest»testIllegal
self should: [empty at: 5] raise: TestResult error.
self should: [empty at: 5 put: #zork] raise: TestResult error

試 し て み ま し ょ う 。

テストを一つだけ実行する

通常はテストランナーを使ってテストを実行しますが、もしテストランナーを

open… メニューから開くのが好みでければ、TestRunner openを print it するこ
ともできますね。

テストを一つだけ実行したい場合には、次のようにします。

ExampleSetTest run: #testRemove −→ 1 run, 1 passed, 0 failed, 0 errors

一つのテストクラスに定義されたすべてのテストを実行する

TestCaseのサブクラスに suiteメッセージを送信すると、名前が「test」で始まる
メソッドを集めてテストスイートを組み立てます。このテストスイートに run
メッセージを送信してテストを実行します。次に例を示します。

ExampleSetTest suite run −→ 5 run, 5 passed, 0 failed, 0 errors

必ず TestCaseクラスを継承しなければならないのでしょうか?

TestCaseクラスを継承しなくても TestSuiteの要素として使うことはできます。
ただし、自分で TestCaseクラスの主要なメソッド (assert:など)を実装したり、
テストスイートを自分で作ったりする必要が出てきます。TestCaseクラスを継
承することをお勧めします。フレームワークが既にあるのですから、それを使

いましょう。

7.6 SUnitフレームワーク

SUnitは次の四つのクラス: TestCase、TestSuite、TestResult、TestResource (図 7.3)
から構成されています。テ ス ト リ ソ ー ス は SUnit 3.1で導入された概念で、セッ
トアップのコストが高く、かつ一連のテスト全体で使われるようなリソースに

使われます。TestResourceの setUpメソッドは、TestCase»setUpと異なり、一連
のテストの前に一度だけ実行されます。TestCase»setUpは各テストの前に実行
されます。

Figure 7.3: SUnitを構成する四つのクラス

TestCase

TestCaseは抽象クラスです。TestCaseを継承して具体的なテストクラスを定義
します。テストメソッド 1個につき 1個、テストクラスのインスタンスが生成
されます。インスタンスにはまず setUpが送信され、次にテストメソッド自身
が実行され、最後に tearDownが送信されます。

テストのコンテキストは、テストクラスのインスタンス変数が保持しま

す。コンテキストを初期化するには setUpメソッドをオーバーライドします。
tearDownメソッドをオーバーライドすることもできます。tearDownはテストの
終了後に呼ばれるので、setUpで割り当てられたオブジェクトをこのメソッド
で開放することができます。

TestSuite

TestSuiteのインスタンス (テストスイート)は、テストや他のテストスイートの
集合です。つまり、テストスイートは TestCaseと TestSuiteのインスタンスを要
素として含みます。TestCaseと TestSuiteの間に継承関係はありませんが、どち
らも同じプロトコルを実装しているので同じように扱えます。例えば、どちら

も runメッセージを理解します。実際のところこれは、TestSuiteを枝、TestCase
を葉とする Compositeパターンの応用になっています—詳しくは デ ザ イ ン パ
タ ー ン を参照してください7。

TestResult

TestResult クラスはテストの実行結果を扱います。成功したテストの数、失敗
したテストの数、エラーが発生したテストの数を記録します。

7Erich Gamma  et al., Design Patterns: Elements of Reusable Object-Oriented Software. Reading,
Mass.: Addison Wesley, 1995, ISBN 0–201–63361–2–(3).

TestResource

一連のテストでは、各テストが互いに独立していることが重要です。一つのテ

ストの失敗が他のテストに波及しないように、そしてテストの実行順序が結果

に影響のないようにしなければなりません。setUpと tearDownがこの独立性を
高めます。

しかし、テストのたびにコンテキストを設定していては時間がかかりすぎる

場合があります。さらに、一連のテストが環境を壊さずにすむことが事前にわ

かっているのなら、一度作った環境を使い回せば十分です。例えば、一連のテ

ストがデータベースに問い合わせを行うなら、事前にセットアップしておいた

データベースに接続するだけで準備できます。コンパイルされたコードを分析

するテストなら、同様に事前にソースコードをコンパイルしておけば十分です。

こうした環境—リソースをどこにキャッシュすれば、一連のテストで共有
できるようになるでしょうか? テストクラスのインスタンス変数はキャッシュ
として使えません。テストクラスのインスタンスは 1回のテストごとに消滅し
てしまうからです。グローバル変数ならできますが、多用すると名前空間が汚

染されてしまいますし、各テストとグローバル変数の関係がわかりにくくなり

ます。必要なリソースを何らかのクラスのシングルトンオブジェクトに持たせ

るのが良い方法です。そのための抽象クラスとして、TestResourceが用意され
ています。つまり TestResourceのサブクラスのインスタンスをリソースとして
使います (テストリソース)。具体的なテストリソースを得るには TestResource
のサブクラスに current メッセージを送信します。するとサブクラスのシング
ルトンインスタンスが得られます。リソースの初期化と終了処理を行うには、

setUpと tearDownをオーバーライドします。

もう一つ、どうにかしてリソースとテストスイートを関連づける必要があり

ます。これにはまず、テストクラスで resources ク ラ ス メソッドをオーバーライ
ドし、テストクラスで使うリソース8を返すようにします。 TestSuite»resources
は、テストスイートに含まれるすべての TestCaseのリソースのクラスの和集合
を返します。

例を示しましょう。TestResource のサブクラス MyTestResource を定義しま
す。次に MyTestCaseのクラスメソッド resourcesをオーバーライドし、このク
ラスのテストで使うリソースのクラスの配列を返すようにします。

Class 7.8: TestResource の サ ブ ク ラ ス の 例
TestResource subclass: #MyTestResource
instanceVariableNames: ’’

MyTestCase class»resources
”リ ソ ー ス と テ ス ト ク ラ ス を 関 連 づ け る ”
↑{ MyTestResource }

8訳注: テストクラスで使うテストリソースのクラスのコレクション。

7.7 SUnitのより高度な機能

現在の SUnitは TestResourceに加えて、アサーションの詳細、ロギング、失敗
したテストの実行再開に対応しています。

アサーションの詳細

TestCaseのアサーションプロトコルには、アサーションの詳細を文字列 (String)
で知らせるメソッドが含まれています。テストに失敗すると、テストランナー

にアサーションの詳細が表示されます。もちろん詳細を表す文字列を動的に生

成することもできます。

| e |
e := 42.
self assert: e = 23
description: ’expected 23, got ’, e printString

関連する TestCaseのメソッドを次に示します:
#assert:description:
#deny:description:
#should:description:
#shouldnt:description:

ロギング

前節で触れたアサーションの詳細は、Transcript やファイルストリームなどの
Stream にも出力できます。出力の有無を設定するには、TestCase»isLogging を
オーバーライドして真偽値を返すようにします。出力するなら真を、しないな

ら偽を指定します。出力先を指定するには TestCase»failureLogをオーバーライ
ドし、出力したいストリームを返すようにします9。

テストの失敗後も実行を続ける

SUnitでは、テストの失敗後に実行を続けるかどうかも指定できます。Smalltalk
の例外処理の仕組みを利用した非常に強力な機能です。どのように動くのでしょ

うか。まず次のテストコードを見てください。

aCollection do: [ :each | self assert: each even]

このコードではコレクションの要素を順に調べていますが、偶数 (even)でない
要素があるとすぐにテストが終わってしまいます。しかし通常はテストを継続

9訳注: isLoggingのデフォルト値は false、failureLogのデフォルト値は Transcriptです。

して、偶数ではない要素がいくつあるのか調べたり、どれが偶数でないのかを

把握したりしたいでしょう。またログを残したいかもしれません。それには次

のようにします10。

aCollection do:
[:each |
self
assert: each even
description: each printString , ’ is not even’
resumable: true]

これで、失敗した要素についてのメッセージを failureLogに出力できます。実行
を続けても失敗回数は増えないので、たとえ 1回のテストを実行するうち 10
回以上アサーションに失敗しても、失敗の総数は 1回と数えられます11。これ
まで見てきた他のアサーション用のメソッドは、失敗後に実行を続けません。

assert: p description: sは、assert: p description: s resumable: falseと等価です。

7.8 SUnitの内部実装

SUnitの内部実装は、Smalltalkのフレームワークの良い事例でもあります。テ
ストの実行過程を通して、内部実装の鍵となる側面を見てみましょう。

一つのテストを実行する

一つのテストを実行するには、(aTestClass selector: aSymbol) runを評価します。

TestCase»runメソッドは、テスト結果を集計するために TestResultのインス
タンスを生成し、それを引数として自身に run:メッセージを送信します (図 7.4
を参照)。

Method 7.9: テ ス ト を 実 行 す る
TestCase»run
| result |
result := TestResult new.
self run: result.
↑result

TestCase»run:メソッドは、引数 (TestResultのインスタンス)に runCase:メッ
セージを送信します。

Method 7.10: TestCase を TestResult に 渡 す
TestCase»run: aResult

10訳注: PBE.imageではうまくいきません。
11訳注: 1回の TestResult»runCase: (7.8節)の実行で記録される失敗の回数は、たかだか 1回です。

Figure 7.4: 一つのテストを実行する

aResult runCase: self

TestResult»runCase:メソッドは、引数として渡された TestCaseに runCaseメッ
セージを送信して、テストでエラーが発生した回数と失敗した回数、成功した

回数を数えます。例外ハンドラを設定して、例外の発生とアサーションの失敗

に備えます。

Method 7.11: TestCase の エ ラ ー と 失 敗 を 捕 捉 す る
TestResult»runCase: aTestCase
| testCasePassed |
testCasePassed := true.
[[aTestCase runCase]

on: self class failure
do:
[:signal |
failures add: aTestCase.
testCasePassed := false.
signal return: false]]
on: self class error
do:
[:signal |
errors add: aTestCase.
testCasePassed := false.
signal return: false].

testCasePassed ifTrue: [passed add: aTestCase]

TestCase»runCaseは、自身に setUpと tearDownを送信します。

Method 7.12: テ ン プ レ ー ト メ ソ ッ ド と し て の TestCase»runCase12

TestCase»runCase
[self setUp.
self performTest] ensure: [self tearDown]

TestSuiteの実行

関連する複数のテストを実行するには、TestSuiteに runメッセージを送信しま
す。TestCaseクラスには、サブクラスのメソッドセレクタのコレクションからテ
ストスイートを組み立てる機能があります。MyTestCase buildSuiteFromSelectors
を評価すると、MyTestCaseに定義したすべてのテストを含むテストスイートを
返します。この処理の中心になるのは、以下のメソッドです。

Method 7.13: テ ス ト ス イ ー ト を 自 動 的 に 組 み 立 て る
TestCase class»testSelectors
↑self selectors asSortedCollection asOrderedCollection select: [:each |
(’test*’ match: each) and: [each numArgs isZero]]

TestSuite»runは TestResultのインスタンスを生成し、すべてのテストリソー
スが問題なく使えるかどうかを検証します。続けて自身に run:を送信し、テス
トスイートの中のすべてのテストを実行します (TestSuite»run:)。そしてテスト
の実行後にすべてのテストリソースを解放 (reset)します。

Method 7.14: テ ス ト ス イ ー ト を 実 行 す る
TestSuite»run
| result |
result := TestResult new.
self resources do: [ :res |
res isAvailable ifFalse: [↑res signalInitializationError]].

[self run: result] ensure: [self resources do: [:each | each reset]].
↑result

Method 7.15: TestResult を TestSuite に 渡 す
TestSuite»run: aResult
self tests do: [:each |
self changed: each.
each run: aResult].

12訳注: self performTestは、selfが知っているテストメソッドを間接的に呼び出します (perform)。

TestResourceクラスとそのサブクラスにはシングルトンインスタンスがあって、
currentクラスメソッドで取得できます。このインスタンスはテストリソースを
保持しており、テストの終了後に resetされます。

TestResource class»isAvailableクラスメソッドを使えばテストリソースが使え
るかどうか確認できますが、このとき必要に応じてテストリソースが再生成さ

れます。再生成されたインスタンスには、setUpが送信されます。

Method 7.16: テ ス ト リ ソ ー ス が 使 え る か ど う か の 確 認
TestResource class»isAvailable
↑self current notNil and: [self current isAvailable]

Method 7.17: テ ス ト リ ソ ー ス の 生 成
TestResource class»current
current isNil ifTrue: [current := self new].
↑current

Method 7.18: テ ス ト リ ソ ー ス の 初 期 化
TestResource»initialize
super initialize.
self setUp

7.9 テストのポイント

テストの仕組みは簡単ですが、良いテストを書くことは簡単ではありません。

テストを設計する上でのいくつかのポイントを紹介します。

Feathersのユニットテストの法則。 アジャイル開発コンサルタントであり「レ

ガシーコード改善ガイド」などの著作でも知られる TBD (Michael Feath-
ers)は、次のように書いています13。

次 の テ ス ト は ユ ニ ッ ト テ ス ト で は あ り ま せ ん :
• デ ー タ ベ ー ス と 対 話 し た り 、
• ネ ッ ト ワ ー ク に 接 続 し た り 、
• フ ァ イ ル シ ス テ ム に 触 っ た り 、
• 他 の ユ ニ ッ ト テ ス ト と 並 行 し て 実 行 で き な か っ た り 、
• 特 別 な 環 境 (設 定 フ ァ イ ル の 編 集 が 必 要 な ど ) を 準 備 し な
け れ ば な ら な い 。

13http://www.artima.com/weblogs/viewpost.jsp?thread=126923を参照。2005/9/9

http://www.artima.com/weblogs/viewpost.jsp?thread=126923

こ の よ う な テ ス ト も 悪 く は あ り ま せ ん 。 書 く 価 値 は あ り ま す

し 、 ユ ニ ッ ト テ ス ト の 補 助 に も な り ま す 。 し か し こ れ ら の テ ス

ト を 本 当 の ユ ニ ッ ト テ ス ト と は 分 け て お く こ と が 大 事 で す 。 そ

う す る こ と で 、 コ ー ド を 変 更 し た ら 即 刻 走 ら せ る こ と が で き る

テ ス ト を 確 保 す る こ と が で き る の で す 。

決して「ユニットテストするのに時間がかかるからやらないですませた

いな」と思ってしまうような状況を作ってはいけません。

ユニットテスト vs.承認テスト ユニットテストは機能の一部を単体でテスト

し、機能に関するバグの特定を容易にします。各メソッドのユニットテ

ストを失敗するまで可能な限り実行し、クラス単位で結果をまとめます。

しかし、複雑な環境をセットアップする必要があれば、より大きなアプリ

ケーションの中のシナリオに沿ってテストを書く方が簡単です。このよ

うなテストは承認テストまたは機能テストと呼ばれます。Feathersの法
則を守らないテストは、承認テストとしては良いテストである場合があり

ます。テストする機能に沿って承認テストをまとめましょう。例えばコ

ンパイラを書く場合、その言語のステートメントごとの生成コードを検

査する承認テストを書くことになるでしょう。たくさんのクラスがテス

トの実行に関与するかもしれません。そして、ファイルシステムにアク

セスするので時間がかかるかもしれません。そのようなテストを SUnit
で記述することはできますが、ほんのちょっとの変更のたびに実行した

いとは思わないでしょう。　ならば、そのようなテストは本当のユニッ

トテストとは分けておくべきなのです。

Blackのテストの法則。 システムのテストごとに、テストを行うことによっ

てシステムのどの性質についての確信が深まるのか、明らかにしなけれ

ばなりません。もちろん重要な性質は必ずテストしなければなりません

が、その一方、システムの性質についての確信を深めないようなテスト

はしてはいけません。そのようなテストはシステムに何ら価値を付加し

ません。例えば、同じ性質に関するテストを複数書いてもいいことはあ

りません。かえって二つのデメリットがあります。一つは、テストから

クラスの振る舞いが読み取りにくくなります。もう一つは、コードにい

くつバグが残っているのか見積もりにくくなります。たった一つのバグ

が複数のテストを失敗させるからです。テストを書くときは、システム

のどの性質についての確信を得るためなのか、常に考えましょう。

7.10 まとめ

この章では、なぜテストが将来のコードへの重要な投資であるのかを説明しま

した。Setクラスのテストを定義する方法について、一つ一つ手順を追って説
明しました。SUnitフレームワークの全体像を示すため、TestCase、TestResult、
TestSuite、TestResourceの四つのクラスを紹介しました。テストとテストスイー
トの実行過程の深層を見てきました。

• ユニットテストの効果を最大限に引き出すポイント: 高速、繰り返し可
能、人と直接やりとりしない、一つの機能のみをテストする。

• テストクラス名は「被テストクラス名 + Test」にします (MyClassのテス
トクラスなら MyClassTest)。テストクラスは TestCaseのサブクラスとし
て定義しましょう。

• setUpメソッドでテストデータを初期化します。

• 各テストメソッド名は「test」で始めます。

• TestCaseの assert:や deny:などのメソッドでアサーションを定義します。

• SUnitのテストランナーを使ってテストを実行します。



Chapter 8

基本的なクラス

Smalltalkの魔法の大部分は、言語そのものではなくクラスライブラリにあり
ます。Smalltalkで効率的にプログラミングするには、クラスライブラリがど
のように言語や環境を支えているのかを学ばなければなりません。クラスライ

ブラリはすべてが Smalltalkで書かれていて、パッケージ (クラス定義を含ん
でいる必要はありません)を使えば新しい機能をクラスに追加できるので、簡
単に拡張することができます。

この章の目的は、Pharoのクラスライブラリ全体について詳細で退屈な話
をすることではありません。代わりにここでは、効率よくプログラムするため

に利用したりオーバーライドしたりすべき、主要なクラスやメソッドを示しま

す。この章では、ほぼすべてのアプリケーションで必要になる、以下の基本的

なクラスを取り上げます: Object、Numberとそのサブクラス、Character、String、
Symbol、Boolean。

8.1 Object

何をするにも、Objectは継承階層のルートです。実際は、Pharoでは ProtoObject
が階層の真のルートです。ProtoObjectは、オブジェクトのふりをすることがで
きる最小の実体を定義するのに使用されます。しかし、今このことは気にしな

くていいでしょう。

Objectは Kernel-Objectsカテゴリの中にあります。驚いたことに、そこには
(拡張も含めると) 400ものメソッドがあります。これはつまり、新しいクラス
を定義すると、あなたがそれらについて知っていようといまいと、すべてに自

動的に 400個のメソッドが提供されるということです。ただ、削除されるべき
メソッドもいくつかあって、Pharoの新しいバージョンではそれらの不要なメ
ソッドのいくつかはおそらく削除されることを覚えておきましょう。

Objectのクラスコメントには次のように書かれています:

Object は ク ラ ス 階 層 内 の ほ ぼ す べ て の ク ラ ス の ル ー ト ク ラ ス で あ る 。
例 外 は ProtoObject (Object の ス ー パ ー ク ラ ス ) と そ の (直 接 の ) サ ブ
ク ラ ス で あ る 。 Object ク ラ ス は 、 通 常 の オ ブ ジ ェ ク ト の ア ク セ ス 、
コ ピ ー 、 比 較 、 エ ラ ー 処 理 、 メ ッ セ ー ジ 送 信 、 リ フ レ ク シ ョ ン な

ど の デ フ ォ ル ト の 振 る 舞 い を 提 供 す る 。 ま た 、 す べ て の オ ブ ジ ェ ク

ト が 反 応 す る 必 要 が あ る ユ ー テ ィ リ テ ィ メ ッ セ ー ジ も こ こ で 定 義 さ

れ る 。 Object は イ ン ス タ ン ス 変 数 を 持 っ て お ら ず 、 追 加 す べ き で も
な い 。 こ れ は 、 Object を 継 承 す る ク ラ ス に は 特 殊 な 実 装 を 持 つ も の
(例 え ば SmallInteger や UndefinedObject) が い く つ か あ っ て 、 こ れ ら
や VM が 、 あ る 標 準 的 な ク ラ ス の 構 造 と レ イ ア ウ ト を 知 っ て い て
そ れ に 依 存 し て い る た め で あ る 。

本節では、Objectのインスタンスボタン側のメソッドカテゴリ (プロトコル)
をブラウズしていきます。それに伴い、Objectが提供するいくつかの主だった
振る舞いが理解できるようになっていくでしょう。

printingプロトコル

Smalltalkのすべてのオブジェクトは、自分自身の表示用の形式を返すことが
できます。ワークスペースで好きな式を選び、print it メニューを選択してくだ
さい: print it は式を実行し、返されたオブジェクトに「あなたを表示してくだ
さい」と頼みます。実際には、返されたオブジェクトに printStringメッセージを
送信します。printStringはテンプレートメソッドで、その中で printOn:メッセー
ジをレシーバに送信します。printOn:はフックメソッドで、特殊化することが
できます。

おそらく Object»printOn:は、最も頻繁にオーバーライドされるメソッドの一
つになるでしょう。このメソッドは引数に Stream (ストリーム)を受け取り、そ
こにオブジェクトの String表現を書き込みます。printOn:のデフォルト実装は、
単に「a」または「an」の後にクラス名を書き込むだけです。Object»printString
は、ストリームに書き込まれたものを Stringにして返します。

例えば、Browserクラスは printOn:メソッドを再定義していないので、printString
メッセージを Browserのインスタンスに送信すると Objectで定義されたメソッ
ドが実行されます。

Browser new printString −→ ’a Browser’

Color クラスでは printOn: を特殊化した例が見られます。Color»printOn: は、
「クラス名色名 (例 え ば Color red)」と言った色を表すメッセージ式をプリント
します。

Method 8.1: printOn: の 再 定 義 。
Color»printOn: aStream
| name |
(name := self name) ifNotNil:
[ ↑ aStream
nextPutAll: ’Color ’;
nextPutAll: name ].

self storeOn: aStream

Color red printString −→ ’Color red’

printOn:メッセージは storeOn:とは異なる、ということに注意してください。
storeOn: メッセージは、レシーバを再生成するのに使える式を引数 (ストリー
ム)に流します。そしてその式は、ストリームが readFrom:メッセージによって
読み込まれるときに評価されます。一方 printOn:は、単にレシーバのテキスト
バージョンを返すだけです。もちろんそのテキスト表現が、レシーバを表す自

己評価型の式になっていることもあり得ます。

表現と自己評価型表現という単語について。 関数型プログラミングでは、式

は実行されると値を返します。一方 Smalltalk では、メッセージ (式) はオブ
ジェクト (値)を返します。オブジェクトの中には、値と自分自身が同じという
好ましい性質を持つものがあります。例えば、trueオブジェクトの値はそれ自
身、す な わ ち trueオブジェクトです。このようなオブジェクトを自 己 評 価 型 オ
ブ ジ ェ ク ト と呼びます。ワークスペースでオブジェクトを print it すると、オ
ブジェクトの値の表 示 用 バージョンを見ることができます。次のものが自己評

価型表現の例です。

true −→ true
3@4 −→ 3@4
$a −→ $a
#(1 2 3) −→ #(1 2 3)
Color red −→ Color red

配列のようなオブジェクトは、そこに含まれるオブジェクトによって自己

評価型であったりそうではなかったりすることに注意してください。例えば、

真偽値の配列は自己評価型ですが、personオブジェクトの配列はそうではあ
りません。次の例は、動的配列が自己評価型になるのは、要素も自己評価型の

場合だけであることを示しています:

{10@10. 100@100} −→ {10@10. 100@100}
{Browser new . 100@100} −→ an Array(a Browser 100@100)

リテラル配列はリテラルだけを保持できることに注意してください。その

ため次の配列は、二つの座標を保持するのではなく 6個のリテラル要素を持つ
ことになります。

#(10@10 100@100) −→ #(10 #@ 10 100 #@ 100)

printOn:メソッドの多くは、自己評価型の振る舞いを実装するように特殊化
されています。例えば Point»printOn: や Interval»printOn: の実装は、自己評価型
です。

Method 8.2: Point の 自 己 評 価 1

Point»printOn: aStream
”レ シ ー バ (の 表 現 形 式 )を 中 置 記 法 で aStream に 書 き 込 む ”
x printOn: aStream.
aStream nextPut: $@.
y printOn: aStream

Method 8.3: Interval の 自 己 評 価
Interval»printOn: aStream
aStream nextPut: $(;
print: start;
nextPutAll: ’ to: ’;
print: stop.

step ∼= 1 ifTrue: [aStream nextPutAll: ’ by: ’; print: step].
aStream nextPut: $)

1 to: 10 −→ (1 to: 10) ”Interval オ ブ ジ ェ ク ト は 自 己 評 価 型 ”

comparingプロトコル (同一性と同値性)

Smalltalkでは、=メッセージはオブジェクトの同 値 性 をテストします (す な わ
ち 二つのオブジェクトが同じ値を持つかどうか)。一方、==はオブジェクトの同
一 性 をテストします (す な わ ち 二つの式が同じオブジェクトを示すかどうか)。
オブジェクトの同値性テストのデフォルト実装では、オブジェクトの同一

性をテストしています:

Method 8.4: オ ブ ジ ェ ク ト の 同 値 性
Object»= anObject

”レ シ ー バ と 引 数 が 同 じ オ ブ ジ ェ ク ト を 示 す か ど う か 答 え る 。
も し = を サ ブ ク ラ ス で 再 定 義 す る な ら 、 hash メ ッ セ ー ジ の 再 定 義 も 検 討 す る
こ と 。 ”

↑ self == anObject

これは頻繁にオーバーライドすることになるメソッドです。Complex (複素
数)の場合を見てみましょう2:

1訳注: Pharo 1.3ではコードが変更されています。
2訳注: Complexクラスは Pharo 1.3に存在しません。

(1 + 2 i) = (1 + 2 i) −→ true ”同 じ 値 ”
(1 + 2 i) == (1 + 2 i) −→ false ”だ け ど 違 う オ ブ ジ ェ ク ト ”

このように動作するのは、Complexが =を次のようにオーバーライドしてい
るからです:

Method 8.5: 複 素 数 の 同 値 性
Complex»= anObject
anObject isComplex
ifTrue: [↑ (real = anObject real) & (imaginary = anObject imaginary)]
ifFalse: [↑ anObject adaptToComplex: self andSend: #=]

Object»∼=は、単に =を反転しているだけですが、通常は変更すべきではあ
りません。

(1 + 2 i) ∼= (1 + 4 i) −→ true

もし =をオーバーライドしたら、hashをオーバーライドしなければなりま
せん。クラスのインスタンスが Dictionaryのキーとして使用された場合に、同
値であるとみなされるインスタンスは、ハッシュ値が同じになるようにしてお

く必要があります:

Method 8.6: 複 素 数 の hash は 再 実 装 さ れ な け れ ば な ら な い
Complex»hash

”= を 実 装 し た の で hash も 再 実 装 す る 。 ”
↑ real hash bitXor: imaginary hash.

= と hash は同時にオーバーライドしなければなりませんが、== は決 し て
オーバーライドしてはいけません (オブジェクトの同一性に関する意味論は、す
べてのクラスで同じです)。==は ProtoObjectのプリミティブメソッドです。

Pharoには、他の Smalltalkと比べていくつか奇妙な振る舞いがあること
に注意してください: 例えばシンボルと文字列は同値とみなされることがあり
ます (これは機能ではなくバグではないかと著者らは考えています)。

#’lulu’ = ’lulu’ −→ true
’lulu’ = #’lulu’ −→ true

class membershipプロトコル

いくつかのメソッドを使って、オブジェクトのクラスに関する問い合わせをす

ることができます。

class. 任意のオブジェクトに classメッセージを送信して、それが属するクラ
スを答えさせることができます。

1 class −→ SmallInteger

反対に、オブジェクトが特定のクラスのインスタンスであるかどうかを尋

ねることもできます:

1 isMemberOf: SmallInteger −→ true ”ま さ に こ の ク ラ ス に 属 し て い な け れ ば
な ら な い ”

1 isMemberOf: Integer −→ false
1 isMemberOf: Number −→ false
1 isMemberOf: Object −→ false

Smalltalkのクラスライブラリは Smalltalk自身によって書かれているので、
スーパークラスとクラスメッセージを正しく組み合わせて、クラスライブラリ

の構造を自在にたどることができます (第 13章を参照)。

isKindOf: Object»isKindOf: は、レシーバのクラスが引数のクラスと等しいか、
またはそのサブクラスであるかを答えます。

1 isKindOf: SmallInteger −→ true
1 isKindOf: Integer −→ true
1 isKindOf: Number −→ true
1 isKindOf: Object −→ true
1 isKindOf: String −→ false

1/3 isKindOf: Number −→ true
1/3 isKindOf: Integer −→ false

1/3は Fractionで Numberでもあります。これは、Numberクラスが Fraction
クラスのスーパークラスだからです。しかし 1/3は Integerではありません。

respondsTo: Object»respondsTo:は、レシーバがメッセージセレクタ (respondsTo:
のシンボル引数)を理解するかどうか答えます。

1 respondsTo: #, −→ false

通常は、オブジェクトにクラスを問い合わせたり、オブジェクトがどのよう

なメッセージに応えられるかを聞いたりするのは良いやり方ではありません。

オブジェクトのクラスに基づいて判断する代わりに、オブジェクトにメッセー

ジだけを送信し、どう振る舞うかはオブジェクトの決定に任せるべきです。

copyingプロトコル

オブジェクトのコピーには、いくつかちょっとした問題があります。インスタ

ンス変数は参照によってアクセスされるので、あるオブジェクトのシ ャ ロ ー

コ ピ ー は元のオブジェクトとインスタンス変数の参照を共有することになり

ます。

a1 := { { ’harry’ } }.
a1 −→ #(#(’harry’))
a2 := a1 shallowCopy.
a2 −→ #(#(’harry’))
(a1 at: 1) at: 1 put: ’sally’.
a1 −→ #(#(’sally’))
a2 −→ #(#(’sally’)) ”内 側 の 配 列 が 共 有 さ れ て い る ”

Object»shallowCopyは、オブジェクトのシャローコピーを作成するためのプ
リミティブメソッドです。a2は a1の単なるシャローコピーなので、二つの配
列はネストされた配列の参照を共有しています。

Object»shallowCopyは Object»copyの「公開インターフェース」で、インスタ
ンスに一意性があればオーバーライドする必要があります。例えば Boolean 、
Character、SmalInteger、UndefinedObjectなどのクラスがそれに当たります。

Object»copyTwoLevelは、単純なシャローコピーではうまくいかないときに、
見たままのことを行ってくれます:
a1 := { { ’harry’ } } .
a2 := a1 copyTwoLevel.
(a1 at: 1) at: 1 put: ’sally’.
a1 −→ #(#(’sally’))
a2 −→ #(#(’harry’)) ”状 態 は 完 全 に 独 立 し て い る ”

Object»deepCopyは、任意の深さまでオブジェクトをディープコピーします。

a1 := { { { ’harry’ } } } .
a2 := a1 deepCopy.
(a1 at: 1) at: 1 put: ’sally’.
a1 −→ #(#(’sally’))
a2 −→ #(#(#(’harry’)))

deepCopyの問題は、相互再帰的な構造に対して使うと終了しなくなること
です:
a1 := { ’harry’ }.
a2 := { a1 }.
a1 at: 1 put: a2.
a1 deepCopy −→ ... does not terminate!

正しく動くように deepCopyをオーバーライドすることもできますが、Object
»copyがもっと良い解決策を提供してくれます:

Method 8.7: オ ブ ジ ェ ク ト を コ ピ ー す る テ ン プ レ ー ト メ ソ ッ ド
Object»copy

”レ シ ー バ と 同 じ よ う な も う 一 つ の イ ン ス タ ン ス を 返 す 。
サ ブ ク ラ ス は 通 常 、 postCopy を オ ー バ ー ラ イ ド す る 。
通 常 は shallowCopy は オ ー バ ー ラ イ ド し な い 。 ”
↑self shallowCopy postCopy

共有すべきではないインスタンス変数をコピーするには、postCopyをオー
バーライドしましょう。そして、postCopy の中では常に super postCopy しま
しょう。

debuggingプロトコル

デバッグする上で最も重要なメソッドは haltです3。メソッドにブレークポイ
ントを設定するには、メソッド本文の好きな場所に self haltというメッセージ
送信を挿入するだけでかまいません。このメッセージが送信されると、プログ

ラムのその位置で実行が中断され、デバッガが起動します (デバッガの詳細に
ついては第 6章を参照してください)。
次に重要なメソッドは assert:です4。これは引数としてブロックを取ります。

ブロックが trueを返すと実行は継続されます。そうでなければ AssertionFailure
例外が発生します。この例外がキャッチされなければ、例外の発生した位置で

デバッガが開きます。assert: は、契 約 に よ る 設 計 をサポートするのに特に便
利です。オブジェクトのパブリックなメソッドの重要な事前条件をチェックす

る、というのが最もよくある利用方法でしょう。Stack»popは、次のように簡
単に実装できます:

Method 8.8: 事 前 条 件 の チ ェ ッ ク
Stack»pop

”最 初 の 要 素 を 返 し 、 ス タ ッ ク か ら そ れ を 削 除 す る 。 ”
self assert: [ self isEmpty not ].
↑self linkedList removeFirst element

Object»assert:と TestCase»assert:を混同しないでください。後者は SUnitテ
スティングフレームワーク (第 7章を参照)の中で出てきます。前者は引数とし
てブロックを期待する5のに対し、後者は Booleanを期待します。共にデバッグ
のために有用ですが、それぞれ目的は大きく異なります。

error handlingプロトコル

このプロトコルには、ランタイムエラーを通知するのに便利なメソッドがいく

つか含まれます。

3訳注: haltは、実際には error handlingプロトコルに分類されています。
4訳注: assert:は、実際には error handlingプロトコルに分類されています。
5実際には valueメッセージを理解するあらゆる引数— Booleanを含む—を受け取れます。

self deprecated: anExplanationStringを送信すると、そのメソッドは今後は奨励
されないということが通知できます (プリファレンス・ブラウザの debugプロ
トコルで deprecationをオンにしているとき)。今後奨励される選択肢は、String
引数の形で提示しましょう。

1 doIfNotNil: [ :arg | arg printString, ’ is not nil’ ]
−→ SmallInteger(Object)»doIfNotNil: has been deprecated. use ifNotNilDo:

メソッド探索に失敗すると doesNotUnderstand: が送信されます。
doesNotUnderstand: のデフォルト実装、す な わ ち Object»doesNotUnderstand:
では、その位置でデバッガが起動します。doesNotUnderstand: をオーバーライ
ドして他の振る舞いをさせると便利かもしれません。

Object»errorと Object»error:は、例外を発生させるのに使える汎用的なメソッ
ドです (ただし独自に定義した例外を発生させる方が、一般的には望ましいで
しょう。自分のコードとカーネルクラスのどちらで発生したエラーなのかが区

別できるからです)。
Smalltalkの抽象メソッドは、規約によりメソッドの本文に self subclassRes-

ponsibilityと書いて実装します。従って抽象クラスが間違ってインスタンス化さ
れ抽象メソッドが呼び出されたとしても、Object»subclassResponsibilityが評価さ
れるだけです。

Method 8.9: メ ソ ッ ド が 抽 象 メ ソ ッ ド で あ る こ と を 通 知 す る 6

Object»subclassResponsibility
”こ の メ ッ セ ー ジ は サ ブ ク ラ ス の 振 る 舞 い に 関 す る 仕 組 み を 支 え て い ま す 。
サ ブ ク ラ ス は こ の メ ッ セ ー ジ を 実 装 す べ き だ と 忠 告 し ま す 。 ”
self error: ’My subclass should have overridden ’, thisContext sender selector

printString

Magnitudeと Numberと Booleanが、抽象クラスの古典的な例です。この章
の少し先で説明します。

Number new + 1 −→ Error: My subclass should have overridden #+

規約により、サブクラスが継承すべきではないメソッドを通知するには、サ

ブクラス側で self shouldNotImplement を送信します。これは、一般的にはクラ
ス階層の設計が何かうまくいっていないことのサインです。しかし単一継承の

制限があるため、どうしてもこのような回避策を取らざるを得ない場合もあり

ます。

典型的な例が Dictionary»remove: です。ここでは、Collection»remove: を継
承しないで shouldNotImplement フラグを立てています (Dictionary は代わりに
removeKey:を用意しています)。

6訳注: この例は Pharo 1.3では変更されています。



176 基 本 的 な ク ラ ス

testingプロトコル

testing プロトコルに分類されるメソッド (testing メソッド)は、SUnit のテス
トとは何の関係もありません! testingメソッドは、レシーバの状態を聞かれて
Booleanを答えるメソッドです。

たくさんの testingメソッドが、Objectによって提供されています。isComplex
は既に見ました。他にも isArray 、isBoolean 、isBlock 、isCollection などがあり
ます。一般的には、これらのメソッドの使用は避けるべきです。オブジェクト

に自身のクラスを問い合わせるのは、カプセル化を破ることだからです。オブ

ジェクトのクラスをテストする代わりに要求だけをオブジェクトに送信し、後

の処理はオブジェクトに任せるべきです。

とは言っても、これらの testingメソッドのいくつかは間違いなく有用です。
最も有用なのは、おそらく ProtoObject»isNilと Object»notNilでしょう (Null Object7
デザインパターンを使ってこれらのメソッドを不要にすることもできますが)。

initialize-releaseプロトコル

Objectではなく ProtoObjectにある最後のキーメソッドは、initializeです。

Method 8.10: 空 の フ ッ ク メ ソ ッ ド と し て の initialize

ProtoObject»initialize
”サ ブ ク ラ ス は こ の メ ソ ッ ド を 再 定 義 し て 、 イ ン ス タ ン ス 生 成 時 に 初 期 化 を 実
行 す る ”

これが重要なのは、Pharoではすべてのクラスの newメソッドのデフォル
ト実装が、新しく生成されたインスタンスに initializeを送信するように書かれ
ているからです。

Method 8.11: ク ラ ス 側 の テ ン プ レ ー ト メ ソ ッ ド と し て の new

Behavior»new
”添 字 付 き の 変 数 を 持 た な い レ シ ー バ (ク ラ ス )の 初 期 化 さ れ た 新 し い イ ン ス
タ ン ス を 返 す 。

ク ラ ス が 添 字 付 き の 場 合 は 失 敗 す る 。 ”
↑ self basicNew initialize

これは、initializeフックメソッドをオーバーライドしさえすれば、あなたの
クラスの新しいインスタンスが自動的に初期化されるということを意味してい

ます。通常 initializeメソッドは、継承されたすべてのインスタンス変数につい
てクラスの不変条件を確立するために、super initializeを実行すべきです (これ
は他の Smalltalkでは標準の振る舞いで は な い ことに注意してください)。

7Bobby Woolf, Null Object. In Robert Martin, Dirk Riehle and Frank Buschmann, editors,
Pattern Languages of Program Design 3. Addison Wesley, 1998.

8.2 Number

驚くべきことに、Smalltalkでは数値はプリミティブなデータではなく、本物
のオブジェクトです。もちろん数値はバーチャルマシンで効率的に実装されて

いますが、Numberの階層は、Smalltalkのクラス階層の他の部分と同じく、完
全にアクセス可能で拡張可能です。

Figure 8.1: Numberのクラス階層

Number は Kernel-Numbers カテゴリの中にあります。この階層の抽象的な
ルートは、Magnitudeです。このクラスは、比較演算子をサポートするあらゆる
種類のクラスを表します。Numberは、様々な算術演算子やその他の演算子を
追加します。これらの演算子のほとんどが抽象メソッドです。Floatと Fraction
は、それぞれ浮動小数点数と分数を表します。Integerも抽象クラスです。抽象
クラスの Integerは、従って、具象サブクラスの SmallInteger、LargePositiveInteger
、LargeNegativeIntegerとは区別されます。ただし、必要に応じて自動的に値が
変換されるので、ほとんどのユーザは三つの Integerクラスの違いを気にする必
要はありません。

Magnitude

Magnitudeは Numberの親であるだけでなく、Character、Duration、Timespanな
どのような比較演算をサポートするその他のクラスの親クラスでもあります

(Complexは比較できません。従って Complexは Numberを継承していません8)。
<と =は抽象メソッドです。それ以外の演算子は汎用的な形で定義されて

います。例えば:
8訳注: Complexクラスは Pharo 1.3に存在しません。

Method 8.12: 比 較 用 の 抽 象 メ ソ ッ ド
Magnitude» < aMagnitude

”レ シ ー バ が 引 数 よ り 小 さ い か ど う か 答 え る ”
↑self subclassResponsibility

Magnitude» > aMagnitude
”レ シ ー バ が 引 数 よ り 大 き い か ど う か 答 え る ”
↑aMagnitude < self

Number

同様に Numberでは、+、-、*、/などが抽象メソッドとして定義されており、
それ以外のすべての算術演算子は汎用的な形で定義されています。

すべての Numberオブジェクトは、asFloatや asIntegerのような様々な変 換
演算子をサポートします。また、それ以外にもたくさんのシ ョ ー ト カ ッ ト コ ン

ス ト ラ ク タ メ ソ ッ ド

9つまり、Numberを実部がゼロの Complex10 のインスタン
スに変換する iや、Numberから Durationオブジェクトを生成する hour、day、
weekなどがあります。

Numbersは、sin、log、raiseTo:、squared、sqrtなどのよくある数 学 関 数 を
直接サポートします。

Number»printOn:は、抽象メソッドの Number»printOn:base:を使って基数 10
の場合として実装されています。

testingメソッドとしては、even、odd、positive、negativeなどがあります。
Numberが isNumberをオーバーライドしているのは驚くほどのことではありま
せん。もっと興味深い点として、isInfinite は false を返すように定義されてい
ます。

切 り 捨 て 系のメソッドとしては、floor、ceiling、integerPart、fractionPartな
どがあります。

1 + 2.5 −→ 3.5 ”二 つ の 数 の 和 ”
3.4 * 5 −→ 17.0 ”二 つ の 数 の 積 ”
8 / 2 −→ 4 ”二 つ の 数 の 商 ”
10 - 8.3 −→ 1.7 ”二 つ の 数 の 差 11”
12 = 11 −→ false ”二 つ の 数 の 同 値 性 ”
12 ∼= 11 −→ true ”二 つ の 数 が 異 な る か ど う か を テ ス ト ”
12 > 9 −→ true ”よ り 大 き い ”
12 >= 10 −→ true ”以 上 ”
12 < 10 −→ false ”よ り 小 さ い ”
100@10 −→ 100@10 ”座 標 の 生 成 ”

9訳注: 便利コンストラクタとも呼ばれます。
10訳注: Complexクラスは Pharo 1.3に存在しません。
11訳注: Pharo 1.3では表示が異なる場合があります。

次の例は Smalltalkでは驚くほど正しく動作します:

1000 factorial / 999 factorial −→ 1000

1000 factorialは、他の多くの言語では計算するのが非常に難しいのですが、実
際に計算できていることに注目してください。これは、数値に関する「自動型

変換と正確な演算」の非常に良い例になっています。

1000 factorial の 結 果 を 表 示 し て み ま し ょ う 。 計 算 よ り も 表 示 に 長 い 時 間 が
か か り ま す !

Float

Floatは、Numberの抽象メソッドを浮動小数点数用に実装します。

さらに興味深いことに、Float class (す な わ ち Floatのクラスボタン側)は、次
の定 数 を返すメソッドを提供しています: e、infinity、nan、pi。

Float pi −→ 3.141592653589793
Float infinity −→ Infinity
Float infinity isInfinite −→ true

Fraction

Fractionは、分子と分母に当たるインスタンス変数によって表されます。分子
と分母は Integerであることが期待されます。Fractionは、通常は Integerの割り
算によって生成されます (Fraction»numerator:denominator:というコンストラクタ
メソッドによってではなく)。

6/8 −→ (3/4)
(6/8) class −→ Fraction

Fractionと Integerまたは Fractionと他の Fractionを掛け合わせると、Integer
になることがあります:

6/8 * 4 −→ 3

Integer

Integer は三つの具象整数クラスの親に当たる抽象クラスです。Integer は、多
くの Numberの抽象メソッドの具体的な実装を提供するだけでなく、factorial、
atRandom、isPrime、gcd:などの整数特有のメソッドやその他多くのメソッド
を追加します。

SmallInteger は、インスタンスがコンパクトに表現される点で特殊です —
SmallIntegerは、値を参照として格納する代わりに、通常は参照を保持するため
に使われるビット領域を使って直接値を表現します。この領域の最初のビット

が、オブジェクトが SmallIntegerかどうかを示しています。

minValと maxValというクラスメソッドで、SmallIntegerの範囲がわかります:

SmallInteger maxVal = ((2 raisedTo: 30) - 1) −→ true
SmallInteger minVal = (2 raisedTo: 30) negated −→ true

SmallInteger がその範囲に収まらなくなると、必要に応じて自動的に
LargePositiveIntegerまたは LargeNegativeIntegerに変換されます:

(SmallInteger maxVal + 1) class −→ LargePositiveInteger
(SmallInteger minVal - 1) class −→ LargeNegativeInteger

大きな整数も同様に、適切なタイミングで小さな整数に変換されます。

多くのプログラミング言語同様、整数は繰り返しを指定するのに便利なこ

とがあります。ブロックを繰り返し評価することに特化した timesRepeat:とい
うメソッドもあります。第 3章で既に同様の例を見ました:

n := 2.
3 timesRepeat: [ n := n*n ].
n −→ 256

8.3 Character

Character は、Collections-Strings カテゴリ内で Magnitude のサブクラスとして定
義されています。表示できる文字は、Pharoでは $〈char〉と表します。例えば:

$a < $b −→ true

表示できない文字は、様々なクラスメソッドを使って得ることができます。

Character class»value: は、Unicode (または ASCII) の整数値を引数として受け
取って、対応する文字を返します。accessing untypeable charactersプロトコルに
は、backspace、cr、escape、euro、space、tabなどのたくさんの便利コンス
トラクタ12が含まれています。

Character space = (Character value: Character space asciiValue) −→ true

printOn:メソッドは十分に賢く、文字を表現する三つの方法のうちどれが最
も適切かを知っています:

12訳注: 正確にはこれらはコンストラクタではありません。

Character value: 1 −→ Character home
Character value: 2 −→ Character value: 2
Character value: 32 −→ Character space
Character value: 97 −→ $a

Character には、様々な便利な testing メソッドが組み込まれています:
isAlphaNumeric、isCharacter、isDigit、isLowercase、isVowelなど。

Characterを、その文字だけを含む文字列に変換するには、asStringを送信し
ます。この場合 asStringと printStringとでは結果が異なります。

$a asString −→ ’a’
$a −→ $a
$a printString −→ ’$a’

8ビットの範囲内の Characterはそれぞれ一意のインスタンスで、クラス変
数である CharacterTableの中に保持されています:

(Character value: 97) == $a −→ true

しかし、0から 255の範囲外の Characterは一意ではありません:

Character characterTable size −→ 256
(Character value: 500) == (Character value: 500) −→ false

8.4 String

String クラスも Collections-Strings カテゴリで定義されています。String は、
Characterだけを保持する添字付きの Collectionです。

実際は、Stringは抽象クラスです。本当は Pharoの Stringは、具象クラスで
ある ByteStringか WideStringのインスタンスです。

’hello world’ class −→ ByteString

Stringのもう一つの重要なサブクラスは、Symbolです。主な違いは、Symbol
のインスタンスは一つの値につきたった一つしか存在しないということです (こ
れはしばしば「一意インスタンス性」と呼ばれます)。一方、別々に生成された
たまたま同じ文字の並びを持つ Stringは、異なるオブジェクトであることもよ
くあります。

’hel’,’lo’ == ’hello’ −→ false

(’hel’,’lo’) asSymbol == #hello −→ true

Figure 8.2: Stringのクラス階層

もう一つの重要な違いは、Stringが変更可能 (mutable)なのに対して Symbolは
変更不能 (immutable)なことです。

’hello’ at: 2 put: $u; yourself −→ ’hullo’

#hello at: 2 put: $u −→ エラー

見落としがちなのは、文字列はコレクションなので他のコレクションと同

じメッセージを理解することです:

#hello indexOf: $o −→ 5

Stringは Magnitudeを継承していませんが、<、=などの通常の comparingメ
ソッドをサポートしています。さらに String»match: は、いくつかの基本的な
globスタイルのパターンマッチを行うのに便利です。

’*or*’ match: ’zorro’ −→ true

さらに進んだ正規表現のサポートが必要なときは、Vassili Bykov による
Regex パッケージを検討してください。

String は比較的多くの変換メソッドをサポートしています。それらの多く
は、asDate、asFileNameなどのような他のクラスへのショートカットコンスト
ラクタメソッドです。また capitalizedや translateToLowercaseなどのような、文
字列から他の文字列に変換する便利なメソッドもたくさんあります。

文字列とコレクションに関するさらに進んだ話題については、第 9章を参
照してください。

8.5 Boolean

Booleanクラスを見れば、Smalltalkの言語機能のどれほど多くがクラスライブ
ラリ側に押しやられているかということについて、非常に興味深い洞察を得る

ことができます。Booleanは、Singletonクラスである Trueと Falseの抽象スー
パークラスです。

Figure 8.3: Booleanのクラス階層

Booleanの振る舞いのほとんどは、ifTrue:ifFalse:メソッドを注意深く調べれ
ば理解できるでしょう。このメソッドは、引数として二つの Block (ブロック)
を取ります。

(4 factorial > 20) ifTrue: [ ’bigger’ ] ifFalse: [ ’smaller’ ] −→ ’bigger’

このメソッドは、Booleanの中では抽象メソッドです。具象サブクラスでの
実装は、いずれも簡単なものです:

Method 8.13: ifTrue:ifFalse: の 実 装
True»ifTrue: trueAlternativeBlock ifFalse: falseAlternativeBlock
↑trueAlternativeBlock value

False»ifTrue: trueAlternativeBlock ifFalse: falseAlternativeBlock
↑falseAlternativeBlock value

実際、これが OOP の本質です: メッセージがオブジェクトに送信された
ら、オブジェクト自身が、どのメソッドを使って反応するかを決めるのです。

ifTrue:ifFalse: では、True のインスタンスは true 側の選択肢 (ブロック) を評価
し、反対に Falseのインスタンスは false側の選択肢 (ブロック)を評価します。
Booleanのすべての抽象メソッドは、Trueと Falseの中でこのような方法で実装
されています。例えば:

Method 8.14: 否 定 を 実 装 す る
True»not

”否 定 --レ シ ー バ が true な の で false と 答 え る 。 ”
↑false

Booleanには、ifTrue:、ifFalse:、ifFalse:ifTrueなどのような便利メソッドがい
くつかあります。また強欲・怠惰な論理積・論理和から好きなものを選ぶこと

ができます。

(1>2) & (3<4) −→ false ”両 辺 を 評 価 す る ”
(1>2) and: [ 3<4 ] −→ false ”左 辺 (レ シ ー バ )だ け を 評 価 す る ”
(1>2) and: [ (1/0) > 0 ] −→ false ”右 辺 (引 数 ブ ロ ッ ク )は 評 価 さ れ な い の で 、 例

外 は 発 生 し な い ”

最初の例では、&の両辺の Booleanの部分式が評価されます。&は (評価済
みの) Boolean引数を受け取ります。二つ目と三つ目の例では、and:は引数とし
て Block を取り、結果的に左辺 (レシーバ)しか評価されません。右辺 (Block)
は、レシーバが trueのときにだけ評価されるからです。

and: と or: が ど の よ う に 実 装 さ れ て い る か 想 像 し て み ま し ょ う 。 そ し て
Boolean と True と False で の 実 装 を 確 認 し て み ま し ょ う 。

8.6 まとめ

• =をオーバーライドしたら hashもオーバーライドしましょう。

• 自作のオブジェクトのコピーを正しく実装するには、postCopyをオーバー
ライドします。

• ブレークポイントを設定するには self haltを送信します。

• メソッドを抽象メソッドにするには、本体に self subclassResponsibilityと
書きます。

• オブジェクトの String表現を提供するには、printOn:をオーバーライドし
ましょう。

• インスタンスを適切に初期化するには、フックメソッドである initializeを
オーバーライドします。

• Numberの一連のメソッドは、自動的に Float、Fraction、Integer間の変換を
行います。

• Fractionは、浮動小数点数ではなく本物の有理数を表します。

• (0から 255の範囲の) Characterのインスタンスは一意です。

• String は変更可能 (mutable) ですが、Symbol はそうではありません。し
かし文字列リテラルを変更しないように気をつけましょう!

• Symbolは一意ですが、Stringはそうではありません。

• Stringと Symbolは Collectionなので、通常の Collectionのメソッドをサポー
トします。





Chapter 9

コレクション

9.1 はじめに

コレクションクラスは、Collectionと Streamの汎用のサブクラスのグループで
大まかに定義されます。「ブルーブック」1 には、こうしたグループの要素と
して Collectionの 17個のサブクラスと Streamの 9個のサブクラス、合計 28個
のクラスが取り上げられています。これらは、Smalltalk-80システムがリリー
スされる前に何度か再設計されたものでした。これらのクラスのグループは、

しばしばオブジェクト指向設計の模範的な実例としてみなされます。

Pharo では、抽象クラスである Collection には 101 個のサブクラスが、同
じく抽象クラスの Streamには 50個のサブクラスがありますが、これらのうち
多く (Bitmap、FileStream、CompiledMethodなど)はシステムやアプリケーション
で使われるように作られた専用のクラスなので、「Collection」カテゴリ内には
ありません。この章では「コレクション階層」という言葉は、Collectionと特 に
Collections-* というカテゴリ内にあるその 47個のサブクラスを指します。同様
に「ストリーム階層」という言葉は、 Streamと特 に Collections-Streamsカテゴ
リ内にあるその 9個のサブクラスを指します。これら 56個のクラスは、982
個のメッセージに反応し、合計 1609個のメソッドを定義しているのです2!
この章では、図 9.1に示したコレクションクラスのサブセットに焦点を当

てます。ストリームについては別に第 10章で論じます。

1Adele Goldberg and David Robson, Smalltalk 80: the Language and its Implementation. Read-
ing, Mass.: Addison Wesley, May 1983, ISBN 0–201–13688–0.

2訳注: Pharo 1.3では Collectionには 75個のサブクラス、Streamには 39個のサブクラスがあり
ます。また「コレクション階層」と「ストリーム階層」の合計 54個のクラスは 1660個のメソッ
ドを定義しており、993個のメッセージに反応します。

Figure 9.1: Pharoにおける主要なコレクションクラス

9.2 コレクションの種類

コレクションクラスをうまく使うには、幅広い種類のコレクションの実装とそ

れぞれの共通点と違いについて、少なくともうわべだけは理解しておかなけれ

ばなりません。

個別の要素についてではなくコレクションについてプログラムすることは、

プログラムの抽象度を高める上で重要な手法の一つです。Lispの map関数は、
このようなスタイルを実現した初期の例です。mapは、関数とリストを引数に
取り、リストの各要素に対して関数を適用してその結果をリストにして返しま

す。Smalltalk-80はこのようなスタイルを推し進め、「コレクションに基づく」
プログラミングを中心に据えたのです。MLやHaskellのようなモダンな関数
型プログラミング言語は、Smalltalkの後継者であると言えます。
このアイデアがいいのはなぜでしょう? 例えば学生についてのレコードの

集合と言ったデータ構造があったとして、何かの基準に一致するすべての学生

のレコードについて同じアクションを取りたかったとします。命令型言語に慣

れたプログラマなら、ループを使うことをすぐに思いつくでしょう。しかし、

Smalltalkプログラマならこう書くでしょう:

students select: [ :each | each gpa < threshold ]

これは「studentsの要素の中で、ある条件を満たすもの」からなる新しいコレ

プロトコル メソッド

accessing size, capacity, at: anIndex, at: anIndex put: anElement
testing isEmpty, includes: anElement, contains: aBlock,

occurrencesOf: anElement
adding add: anElement, addAll: aCollection
removing remove: anElement, remove: anElement ifAbsent: aBlock,

removeAll: aCollection
enumerating do: aBlock, collect: aBlock, select: aBlock, reject: aBlock,

detect: aBlock, detect: aBlock ifNone: aNoneBlock,
inject: aValue into: aBinaryBlock

converting asBag, asSet, asOrderedCollection, asSortedCollection,
asArray, asSortedCollection: aBlock

creation with: anElement, with:with:, with:with:with:,
with:with:with:with:, withAll: aCollection

Figure 9.2: 標準的なコレクションプロトコル

クションを返します。角括弧で囲まれた式が trueを返すことが条件です3。こ
の Smalltalkコードには、ドメイン特化言語4的なシンプルさとエレガンスがあ
ります。

select:メッセージは Smalltalkのす べ て の コレクションが理解します。学生
のデータ構造が配列なのか連結リストなのかを調べる必要はありません。どち

らも select:メッセージを理解します。ループを使った場合、studentsが配列な
のか連結リストなのかをあらかじめ知っておかなければなりませんが、それと

は大きく異なるということに注意しましょう。

Smalltalkでは、特定のコレクションを指定せずに単に「コレクション」と
言った場合には、要素に関する次のプロトコルが明確に定義されたオブジェク

トを指します: ある要素が含まれているかテスト (test)するプロトコル、要素
の列挙 (enumerate)を行うプロトコル。す べ て の コレクションは、testingメッ
セージ: includes:、isEmpty、occurrencesOf: を理解します。す べ て の コレクショ
ンは、enumerationメッセージ: do:、select:、reject: (select:の反対)、collect: (Lisp
の mapと同じ)、detect:ifNone:、inject:into: (左畳込みを行う)などを理解します。
このプロトコルの普遍性とバラエティが、コレクションをとても強力なものに

しています。

図 9.2は、コレクション階層内のほとんどのクラスが標準でサポートして
いるプロトコルをまとめたものです。それぞれのメソッドは、Collectionのサブ
クラス内で定義または再定義され、最適化され、時にはその使用を禁じられる

こともあります。

このような基本的な一貫性を前提として、サポートするプロトコルが異なっ

3角括弧内の式は無名関数 λx.x gpa < thresholdを定義する λ式だと考えることができます。
4訳注: domain-specific query language

たり、同じリクエストに対しての振る舞いが異なる様々なコレクションが存在

しています。これらの違いのうち重要なものについて見ていきましょう。

• 順序付き (Sequenceable): SequenceableCollectionのすべてのサブクラス
のインスタンスは、first要素から始まり、last要素まで明確に決まった順
序で並んでいます (sequenceable)。一方で、Set、Bag、Dictionaryのイン
スタンスは順序付きではありません。

• ソート済み: SortedCollectionは、その要素を常にソートされた状態に保ち
ます。

• 添字参照/添字付き: ほとんどの順序付きコレクションは添字付きです。
つまり at:で要素を取り出せます。Array (配列)は添字付きのデータ構造の
中でもとても身近なもので、固定されたサイズを持ちます。anArray at: n
は anArrayの n個目の要素を取り出し、anArray at: n put: vは n個目の要素
の値を vに変えます。LinkedListと SkipListは、順序付きですが添字付きで
はありません。つまり firstと lastは理解しますが、at:は理解しません5。

• キー参照: Dictionaryとそのサブクラスのインスタンスは、添字の代わり
にキーで要素を参照できます。

• 変更可能性 (Mutable): ほとんどのコレクションは変更可能 (mutable)
ですが、Intervalと Symbolは例外です。Intervalクラスは、Integerが扱える
範囲の大きさを持つ変更不能 (immutable)なコレクションです。例えば
5 to: 16 by: 2は、要素 5、7、9、11、13、15を含む Intervalオブジェクトで
す。これは at:を使って添字で参照できますが、at:put:で値を変更するこ
とはできません。

• 動的拡張: Intervalや Arrayのインスタンスは、常にサイズが固定されていま
す。他の種類のコレクション (SortedCollection、OrderedCollection、LinkedList
は、生成後にサイズを拡張することができます。

OrderedCollectionクラスは、Arrayより汎用的です。OrderedCollectionは動
的にサイズが拡張され、addFirst: や addLast: と言ったメソッドが、at: や
at:put:メソッドの他にあります。

• 重複の許容: Setは要素の重複を取り除きますが、Bagはそういったこと
をしません。重複の判定に、Dictionary、Set、Bagは、各要素の =メソッ
ドを用います。これらのクラスの Identityな変種は、二つの引数が同じオ
ブジェクトであるかをテストする ==メソッドを用い、Pluggableな変種で
あれば、コレクションの作成時に与えられた任意の同値関係を用います。

• 異種性 (Heterogeneous): ほとんどのコレクションは、どんな種類の要
素も格納できます。一方で String、CharacterArray、Symbolは、Character
だけしか入れられません。Arrayには様々なオブジェクトが混在できます
が、ByteArrayはバイトだけを、IntegerArrayは Integerだけを、FloatArray

5LinkedList»at:put:はありませんが、LinkedList»at:はあります。SkipListというクラスはありません。

Figure 9.3: 実装テクニックにより分類されたコレクションクラス

は Floatだけを格納できます。LinkedListの要素には、必ず Link .accessing
プロトコルを受け付けなければならないという制約があります。

9.3 コレクションの実装

機能による分類だけではなく、どのようにコレクションクラスが実装されてい

るかについても知っておく必要があります。図 9.3に示すように、主要な実装
テクニックとしては五つのものがあります。

1. 配列は、要素をコレクションオブジェクト自身の (添字付きの)インスタ
ンス変数に格納します。結果として配列のサイズは固定され、その代わ

りに配列は、一回のメモリ割り当てで生成できます。

2. OrderedCollectionと SortedCollectionは要素を配列に格納し、インスタンス
変数でその配列を参照します。こうすることで、内部の配列のサイズよ

りもコレクションが大きくなったらより大きい配列と入れ替えることが

できます。

3. Setや Dictionaryの類もまた、ストレージへのアクセスに補助的な配列を参
照しますが、これをハッシュテーブルとして用います。Bagは、内部の要素
そのものとその出現回数をキーと値のペアとして持つ補助的な Dictionary
を使います。

4. LinkedListは、標準的な片方向リンクの内部形式を取ります。

5. Intervalは、開始点、終了点、ステップサイズの三つの整数を格納します。

これらの基本的なクラスの他に、Array、Set、それから辞書の多くには、「Weak
」な変種があります。これらのコレクションは、要素を「弱く」保持します。

す な わ ち 要素のガーベージコレクションを禁止しないということです。Pharo
のバーチャルマシンはこれらのクラスを判別し、特別に扱います。

Smalltalkのコレクションについてより詳しく知りたい場合は、LaLonde
と Pughの素晴らしい本6 を参照してください。

9.4 主要なクラスの例

よく使う、あるいは重要なコレクションクラスについて、簡単なコード例で説

明しましょう。コレクションの主なプロトコルは以下の通りです: at: 、at:put:
—要素へのアクセス、add:、remove: —要素の追加または削除、size、isEmpty、
include: —コレクションについての情報の取得、do:、collect:、select: —コレク
ションについての繰り返し。それぞれのコレクションによって、これらのプロ

トコルを実装したりしなかったりしますし、実装しているときは、コレクショ

ンの意味論に沿うようにプロトコルを解釈します。それぞれのクラスをブラウ

ズしてみて、具体的でより上級のプロトコルを探してみるといいでしょう。

以下では、最もよく使う次のクラスを取り上げます: OrderedCollection、Set
、SortedCollection、Dictionary、Interval、Array。

よく使う生成プロトコル。 コレクションのインスタンスを作る方法はいくつ

かあります。new:メソッドと with:メソッドを使うのが一番汎用的なやり方で
しょう。new: anIntegerは、サイズが anIntegerですべての要素が nilであるコレ
クションを作ります。with: anObjectはコレクションを作り、それに anObjectを
追加します。コレクションごとにこの振る舞いの実装は異なるでしょう。

初期要素を持つコレクションを作るには、with:、with:with:などのメソッド
を使うことができます。with:with:...方式では、最大 6個の要素を持ったコレク
ションまで作ることができます。

Array with: 1 −→ #(1)
Array with: 1 with: 2 −→ #(1 2)
Array with: 1 with: 2 with: 3 −→ #(1 2 3)
Array with: 1 with: 2 with: 3 with: 4 −→ #(1 2 3 4)
Array with: 1 with: 2 with: 3 with: 4 with: 5 −→ #(1 2 3 4 5)
Array with: 1 with: 2 with: 3 with: 4 with: 5 with: 6 −→ #(1 2 3 4 5 6)

あるコレクションのすべての要素を別の種類のコレクションに加えるには、

addAll:が使えます:

(1 to: 5) asOrderedCollection addAll: ’678’; yourself −→ an OrderedCollection(1 2 3
4 5 $6 $7 $8)

addAll:は引数を返します。レシーバを返すわけではないのに注意してください!
withAll:や newFrom:で、いろいろなコレクションを作ることもできます。

6Wilf LaLonde and John Pugh, Inside Smalltalk: Volume 1. Prentice Hall, 1990, ISBN
0–13–468414–1.

Array withAll: #(7 3 1 3) −→ #(7 3 1 3)
OrderedCollection withAll: #(7 3 1 3) −→ an OrderedCollection(7 3 1 3)
SortedCollection withAll: #(7 3 1 3) −→ a SortedCollection(1 3 3 7)
Set withAll: #(7 3 1 3) −→ a Set(7 1 3)
Bag withAll: #(7 3 1 3) −→ a Bag(7 1 3 3)
Dictionary withAll: #(7 3 1 3) −→ a Dictionary(1->7 2->3 3->1 4->3 )

Array newFrom: #(7 3 1 3) −→ #(7 3 1 3)
OrderedCollection newFrom: #(7 3 1 3) −→ an OrderedCollection(7 3 1

3)
SortedCollection newFrom: #(7 3 1 3) −→ a SortedCollection(1 3 3 7)
Set newFrom: #(7 3 1 3) −→ a Set(7 1 3)
Bag newFrom: #(7 3 1 3) −→ a Bag(7 1 3 3)
Dictionary newFrom: {1 -> 7. 2 -> 3. 3 -> 1. 4 -> 3} −→ a Dictionary(1->7 2->3 3

->1 4->3 )

この二つのメソッドは同じではないことに注意してください。具体的には、

Dictionary class»withAll:では引数を値のコレクションと解釈しますが、Dictionary
class»newFrom:はアソシエーションのコレクションが与えられることを期待し
ています。

Array

Array (配列)は、サイズが固定のコレクションです。また配列の要素には、整数
の添字でアクセスできます。Cの流儀とは異なり、Smalltalkの配列の先頭要
素は位置 1です。0ではありません。配列の要素にアクセスする主なプロトコ
ルは、at:と at:put:です。at: anIntegerは、添字 anIntegerにある要素を返します。
at: anInteger put: anObjectは、anObjectを添字 anIntegerの位置に入れます。配列
はサイズ固定のコレクションなので、配列の末尾に要素を追加したり、末尾の

要素を削除したりはできません。以下のコードは、サイズ 5の配列を作り、最
初の三つの要素にそれぞれ値を入れ、先頭要素の値を返します。

anArray := Array new: 5.
anArray at: 1 put: 4.
anArray at: 2 put: 3/2.
anArray at: 3 put: ’ssss’.
anArray at: 1 −→ 4

Arrayクラスのインスタンスを作るにはいくつか方法があります。new:、with:、
それから #( )構文と { }構文を使うこともできます。

new: による生成。 new: anInteger は、サイズ anInteger の配列を生成します。
Array new: 5は、サイズ 5の配列を生成します。

with:による生成。 with:メソッドは、要素の値を引数として与えることがで
きます。次のコードでは、数値 4、分数 3/2、文字列 ’lulu’の三つの要素を持つ
配列を生成します。

Array with: 4 with: 3/2 with: ’lulu’ −→ {4. (3/2). ’lulu’}

#()を用いたリテラル生成。 #()は、静的な (または「リテラル」)要素を持つ
リテラル配列を生成します。静的というのは、要素の値が実行時ではなく式が

コンパイルされたときにわかっていなければならない、ということです。以下

のコードは、最初の要素が (リテラル)数値 1で次の要素が (リテラル)文字列
’here’であるサイズ 2の配列を生成します。

#(1 ’here’) size −→ 2

#(1+2)を評価した場合、数値 3だけを要素として持つ配列ではなく、#(1 #+ 2)
す な わ ち 数値 1、シンボル #+、数値 2の三つの要素を持つ配列になります。

#(1+2) −→ #(1 #+ 2)

これは、#()構文によって、配列に格納されている式がリテラルとしてコンパイ
ラに解釈されるからです。式はスキャンされて結果として得られた要素が新し

い配列に格納されます。リテラル配列は、数値、nil、true、false、シンボル、文
字列を格納できます。

{ }を用いた動的生成。 最後に {}構文を使って動的配列を作ることができま
す。{ a . b }は Array with: a with: bと等価です。つまるところ、{と }で囲まれた
式が実行されるということです。

{ 1 + 2 } −→ #(3)
{(1/2) asFloat} at: 1 −→ 0.5
{10 atRandom . 1/3} at: 2 −→ (1/3)

要素へのアクセス。 すべての配列は、at:および at:put:で要素にアクセスでき
ます。

anArray := #(1 2 3 4 5 6) copy.
anArray at: 3 −→ 3
anArray at: 3 put: 33.
anArray at: 3 −→ 33

リテラル配列を変更するコードには十分注意してください! リテラル配列につ
いて、コンパイラは領域を最初の一回だけ確保します。そのため配列をコピー

しない限り、上の例を二回目に評価したとき、「リテラル」配列の値は期待通

りにはならないでしょう。(コピーを取らなければ、二回目の実行時にはリテラ

ル配列 #(1 2 3 4 5 6)は実は #(1 2 33 4 5 6)になってしまっているのです!) 動的配
列にはこのような問題はありません7。

OrderedCollection

OrderedCollectionは、自動でサイズを拡張し、順序付きで要素を追加できるコレ
クションです。OrderedCollectionは、add:、addFirst:、addLast:、addAll:などの
多様なメソッドを提供します。

ordCol := OrderedCollection new.
ordCol add: ’Seaside’; add: ’SqueakSource’; addFirst: ’Monticello’.
ordCol −→ an OrderedCollection(’Monticello’ ’Seaside’ ’SqueakSource’)

要素を削除する。 メソッド remove: anObjectは、anObjectと一致する最初のオ
ブジェクトをコレクションから削除します。もしそのオブジェクトが存在しな

い場合はエラーを発生させます。

ordCol add: ’Monticello’.
ordCol remove: ’Monticello’.
ordCol −→ an OrderedCollection(’Seaside’ ’SqueakSource’ ’Monticello’)

remove:の変種に、remove:ifAbsent:があります。remove:ifAbsent:の第二引数に
は、削除しようとした要素がコレクションに存在しないときの動作をブロック

として渡すことができます。

res := ordCol remove: ’zork’ ifAbsent: [33].
res −→ 33

変換。 Array に対して (またはその他どんなコレクションに対しても)、
asOrderedCollectionメッセージを送信して、そのコレクションをOrderedCollection
に変換することができます:

#(1 2 3) asOrderedCollection −→ an OrderedCollection(1 2 3)
’hello’ asOrderedCollection −→ an OrderedCollection($h $e $l $l $o)

Interval

Interval クラスは数値の並びを表現します。例えば 1 から 100 までの数の並び
は、以下のように定義できます:

7訳注: 同一コンパイル単位内に字面がまったく同じリテラル配列が出現すると、それらは同一
のオブジェクトとして扱われます。

Interval from: 1 to: 100 −→ (1 to: 100)

この Intervalオブジェクトを printStringした結果を見ると、Numberクラスに to:
という便利メソッド (便利コンストラクタ)があって、これを使って Intervalオ
ブジェクトを生成することもできるとわかります。

(Interval from: 1 to: 100) = (1 to: 100) −→ true

Interval class»from:to:by:または Number»to:by:によって、以下のように二つの
値の間のステップを指定することができます。

(Interval from: 1 to: 100 by: 0.5) size −→ 199
(1 to: 100 by: 0.5) at: 198 −→ 99.5
(1/2 to: 54/7 by: 1/3) last −→ (15/2)

Dictionary

Dictionary (辞書)は、要素がキーによってアクセスされる重要なコレクションで
す。Dictionaryのメッセージの中でよく使うものとしては、at:、at:put:、at:ifAbsent:
、keys、valuesがあります。

colors := Dictionary new.
colors at: #yellow put: Color yellow.
colors at: #blue put: Color blue.
colors at: #red put: Color red.
colors at: #yellow −→ Color yellow
colors keys −→ a Set(#blue #yellow #red)
colors values −→ {Color blue. Color yellow. Color red}8

辞書は、キーを同値性によって比較します。 =で比較したときに真である
ならば、二つのキーは等しいとみなされます。よくあるけれど見つけにくいバ

グは、キーとして使うオブジェクトで =メソッドを再定義しているにもかかわ
らず、hashメソッドを再定義していないというものです。これら二つのメソッ
ドは、どちらも辞書の実装の中でオブジェクトの比較に用いられます。

コレクション階層はサブクラスの考え方に基づいており、サブタイプに基

づいているわけではありません。そのことは、Dictionaryクラスに明確に示され
ています。Dictionaryは Setのサブクラスですが、普通は Dictionaryを Setのよう
に使おうとは思わないでしょう。しかし実装を見ると、Dictionaryはアソシエー
ション (キーと値。->メッセージにより生成される)の集合であることが明ら
かです。そのため、Dictionaryはアソシエーションのコレクションから作ること
もできますし、辞書をアソシエーションの配列に変換することもできます。

8訳注: Pharo 1.3では rgb値が返ります。

colors := Dictionary newFrom: { #blue->Color blue . #red->Color red . #yellow->Color
yellow }.

colors removeKey: #blue.
colors associations −→ {#yellow->Color yellow. #red->Color red}9

IdentityDictionary。 辞書が二つのキーが同じであるかを判定するのに =
メッセージおよび hashメッセージを用いるのに対し、IdentityDictionaryは、キー
の同一性 (==メッセージ)を用います。す な わ ち 二つのキーが同じオブジェク
トであるときにだ け 、これらのキーは等しいとみなされます。

しばしば Symbol はキーとして用いられますが、そのような場合に
IdentityDictionary を使うのはごく自然なことです。なぜなら、Symbol はグロー
バルに一意であることが保証されているからです。一方、キーが Stringの場合
は、普通の Dictionaryクラスを使う方が賢明です。そうしないと問題が生じる
でしょう。

a := ’foobar’.
b := a copy.
trouble := IdentityDictionary new.
trouble at: a put: ’a’; at: b put: ’b’.
trouble at: a −→ ’a’
trouble at: b −→ ’b’
trouble at: ’foobar’ −→ ’a’

aと bはそれぞれ別のオブジェクトなので、異なるオブジェクトとして扱われ
ます。興味深いことですが、リテラル ’foobar’は一度しか確保されないので、a
とまったく同じオブジェクトになります10。このような微妙な振る舞いに依存
したコードを書きたいなんて思わないでしょう! 普通の Dictionaryは、’foobar’と
同値 (equal)なキーに対して同じ値を返します。

IdentityDictionary のキーとしては、グローバルに一意であるオブジェクト
(Symbol や SmallInteger など) だけを使ってください。String (やその他のオブ
ジェクト)は、普通の Dictionaryのキーとして使いましょう。

特筆すべきは、グローバル変数 Smalltalk11 が SystemDictionaryのインスタン
スであり、SystemDictionaryは IdentityDictionaryのサブクラスなので、Smalltalkの
すべてのキーは Symbol (実際は 8ビット文字しか格納できない ByteSymbol)だ
ということです。

Smalltalk keys collect: [ :each | each class ] −→ a Set(ByteSymbol)12

9訳注: Pharo 1.3では rgb値が返ります。
10訳注: 同一コンパイル単位内に限ります。
11訳注: Pharo 1.3では Smalltalk globals
12訳注: Pharo 1.3では長い Arrayが返ります。

keysまたは valuesメッセージを Dictionaryに送信すると結果は Setになります
が、このクラスについては次に説明します13。

Set

Setは、数学的な「集合」のように振る舞うコレクションです。す な わ ち 重複し
た要素を持たず、要素間には順序がありません。Setに要素を加えるには add:
メッセージを用いますが、at:によって要素にアクセスすることはできません。
Setに格納されるオブジェクトは、メソッド hashと =を実装していなければい
けません。

s := Set new.
s add: 4/2; add: 4; add:2.
s size −→ 2

Setを作るには、Set class»newFrom:または変換メッセージ Collection»asSet
を用いることができます。

(Set newFrom: #( 1 2 3 1 4 )) = #(1 2 3 4 3 2 1) asSet −→ true

asSetは、コレクションから重複を取り除くのに便利に使えます。

{ Color black. Color white. (Color red + Color blue + Color green) } asSet size −→ 2

Color red + Color blue + Color green = Color whiteであることに注意してください。

Bagは Setととても似ていますが、重複を許すという違いがあります。

{ Color black. Color white. (Color red + Color blue + Color green) } asBag size −→ 3

集合演算である union (和 集 合 )、intersection (積 集 合 )、要 素 を 含 む か の テ ス
ト は、Collectionメソッドの union:、intersection:、includes:として実装されてい
ます。レシーバが最初に Setに変換されるので、これらの演算はすべてのコレ
クションでちゃんと動くのです!
(1 to: 6) union: (4 to: 10) −→ a Set(1 2 3 4 5 6 7 8 9 10)
’hello’ intersection: ’there’ −→ ’he’
#Smalltalk includes: $k −→ true

後で説明するように、Setの各要素はイテレータでアクセスできます (9.5節
参照)。

SortedCollection

OrderedCollection とは対照的に SortedCollection は、ソート順を保ちつつ要素を
格納します。デフォルトでは、SortedCollectionはソート順を決めるのに <=メッ

13訳注: Pharo 1.3では Arrayが返ります。

セージを用いるので、抽象クラスMagnitudeのサブクラスのインスタンスをソー
トできます。Magnitudeは、比較プロトコル (<, =, >, >=, between:and:...) を定義
しています (第 8章参照)。

SortedCollectionを作るには、まず新しいインスタンスを生成し、要素を追加
していきます。

SortedCollection new add: 5; add: 2; add: 50; add: -10; yourself. −→ a
SortedCollection(-10 2 5 50)

しかし普通は、既にあるコレクションに変換メッセージ asSortedCollectionを送
信して SortedCollectionを作ります:

#(5 2 50 -10) asSortedCollection −→ a SortedCollection(-10 2 5 50)

この例は、次の FAQに対する答えとなります:

FAQ:コレクションをソートするにはどうすればいいでしょ
うか?
答え: asSortedCollectionメッセージを送信します。

’hello’ asSortedCollection −→ a SortedCollection($e $h $l $l $o)

ではソートした結果を Stringに戻すにはどうすればいいでしょう? 残念なこ
とに asStringが返す printString表現は、思った通りのものではありません。

’hello’ asSortedCollection asString −→ ’a SortedCollection($e $h $l $l $o)’

正解は、String class»newFrom:、String class»withAll:、Object»as:のいずれかを用い
ることです。

’hello’ asSortedCollection as: String −→ ’ehllo’
String newFrom: (’hello’ asSortedCollection) −→ ’ehllo’
String withAll: (’hello’ asSortedCollection) −→ ’ehllo’

すべての要素が互いに比較可能である限り、SortedCollection は異なった種
類の要素を持つことができます。例えば整数、浮動小数点数、分数と言った異

なった種類の数を混ぜることができます。

{ 5. 2/-3. 5.21 } asSortedCollection −→ a SortedCollection((-2/3) 5 5.21)

もしかしたら、<=メッセージを理解しないオブジェクトをソートしたかっ
たり、<=以外の基準でソートしたかったりするかもしれません。この場合は、
SortedCollectionの生成時にソートブロックと呼ばれる 2引数のブロックを与え
ることで、ソートが可能になります。例えば、Colorクラスは Magnitudeではあ
りませんし <=メソッドも実装していませんが、次のようにソートブロックを指
定することで、色を明度 (明るさの尺度)によってソートすることができます。

col := SortedCollection sortBlock: [:c1 :c2 | c1 luminance <= c2 luminance].
col addAll: { Color red. Color yellow. Color white. Color black }.
col −→ a SortedCollection(Color black Color red Color yellow Color white)

String

Smalltalkの String (文字列)は、Characterのコレクションとして表現されます。
このコレクションは順序付きで、添字付きで、変更可能で、同じ種類の要素す

なわち Character のインスタンスしか持つことができません。Array のように、
Stringは専用の文法を持ち、普通は Stringリテラルを一重引用符で囲むことによ
り直接生成しますが、通常のコレクション作成メソッドもちゃんと動きます。

’Hello’ −→ ’Hello’
String with: $A −→ ’A’
String with: $h with: $i with: $! −→ ’hi!’
String newFrom: #($h $e $l $l $o) −→ ’hello’

実際は、Stringは抽象クラスです。Stringをインスタンス化したときに実際
に得られるのは、8ビット文字の ByteStringか 32ビット文字のWideStringです。
話を簡単にするために、その差異は無視して単に Stringのインスタンスについ
て述べます。

Stringの二つのインスタンスは、カンマによって連結できます。

s := ’no’, ’ ’, ’worries’.
s −→ ’no worries’

文字列は変更可能なコレクションなので、at:put:メソッドを用いて変更する
こともできます。

s at: 4 put: $h; at: 5 put: $u.
s −→ ’no hurries’

カンマメソッドは Collection で定義されているので、どんな種類のコレク
ションでも使えることに注意してください!

(1 to: 3) , ’45’ −→ #(1 2 3 $4 $5)

以下に示すように、既にある文字列を replaceAll:with:または relpaceFrom:to:with:
を使って変更することもできます。後者の文字数と指定した間隔の長さが異な

るとエラーが発生するので、注意してください。

s replaceAll: $n with: $N.
s −→ ’No hurries’
s replaceFrom: 4 to: 5 with: ’wo’.
s −→ ’No worries’

上のメソッドと異なり、copyReplaceAll:は新たな文字列を生成します (興味
深いことに、引数は個別の文字というよりも部分文字列なので、サイズが一致

している必要はありません)。

s copyReplaceAll: ’rries’ with: ’mbats’ −→ ’No wombats’

実装をちょっと見てみれば、これらのメソッドは String専用ではなくどん
な SequenceableCollectionにも使えることがわかるので、例えば次のような例も
ちゃんと動きます。

(1 to: 6) copyReplaceAll: (3 to: 5) with: { ’three’. ’etc.’ } −→ #(1 2 ’three’ ’etc.’ 6)

文字列のマッチ。 match:メッセージを送信することによって、パターンと文
字列が一致するかどうかを調べることができます。パターンでは、*を任意の
長さの文字列へのマッチ、#を 1文字へのマッチの意味で使うことができます。
注意したいのは、match:はパターンに送信されるのであって、調べたい文字列
に送信されるのではないということです。

’Linux *’ match: ’Linux mag’ −→ true
’GNU/Linux #ag’ match: ’GNU/Linux tag’ −→ true

もう一つ便利なメソッドは findString:です。

’GNU/Linux mag’ findString: ’Linux’ −→ 5
’GNU/Linux mag’ findString: ’linux’ startingAt: 1 caseSensitive: false −→ 5

Perl互換のより高度のパターンマッチングの機能が必要なら、Regex パッ
ケージを使うこともできます。

文字列に対するテスト。 以下の例は、Stringだけでなく他の一般的なコレク
ションにも定義されているメッセージ: isEmpty、includes:、anySatisfy:を文字列
に使ったときの動作を示しています。

’Hello’ isEmpty −→ false
’Hello’ includes: $a −→ false
’JOE’ anySatisfy: [:c | c isLowercase] −→ false
’Joe’ anySatisfy: [:c | c isLowercase] −→ true

文字列のテンプレート化。 文字列のテンプレート化 に便利なメッセージが

三つあります: format:、expandMacros、expandMacrosWith:。

’{1} is {2}’ format: {’Pharo’ . ’cool’} −→ ’Pharo is cool’

expandMacros系のメッセージは、<n>によるキャリッジリターン展開、<t>
によるタブ展開、<1s>、<2s>、<3s>による引数展開 (<1p>、<2p>もほぼ同様で



202 コ レ ク シ ョ ン

すが、これらは文字列を一重引用符で囲みます)、<1?value1:value2>による条件
付き展開などをサポートしています。

’look-<t>-here’ expandMacros −→ ’look- -here’
’<1s> is <2s>’ expandMacrosWith: ’Pharo’ with: ’cool’ −→ ’Pharo is cool’
’<2s> is <1s>’ expandMacrosWith: ’Pharo’ with: ’cool’ −→ ’cool is Pharo’
’<1p> or <1s>’ expandMacrosWith: ’Pharo’ with: ’cool’ −→ ’’’Pharo’’ or Pharo’
’<1?Quentin:Thibaut> plays’ expandMacrosWith: true −→ ’Quentin plays’
’<1?Quentin:Thibaut> plays’ expandMacrosWith: false −→ ’Thibaut plays’

その他のユーティリティメソッド。 String クラスは他にもたくさんのユー
ティリティメソッドを提供しています。例えば、asLowercase 、asUppercase 、
capitalized。

’XYZ’ asLowercase −→ ’xyz’
’xyz’ asUppercase −→ ’XYZ’
’hilaire’ capitalized −→ ’Hilaire’
’1.54’ asNumber −→ 1.54
’this sentence is without a doubt far too long’ contractTo: 20 −→ ’this sent...too long’

printStringメッセージを送信してオブジェクトにその文字列表現を問い合わ
せたときと、asStringメッセージを送信してオブジェクトを文字列に変換したと
きの結果は、一般的に異なるということに注意してください。例えば次の通り。

#ASymbol printString −→ ’#ASymbol’
#ASymbol asString −→ ’ASymbol’

シンボルは文字列と似ていますが、グローバルに一意であることが保証さ

れる点で異なります。このため辞書の、とりわけ IdentifyDictionaryのインスタン
スのキーとして、シンボルは文字列より適しています。Stringと Symbolについ
ては、第 8章も参照してください。

9.5 コレクションイテレータ

Smalltalkでは、ループや条件分岐は、コレクションや整数やブロックのような
オブジェクトへの単なるメッセージ送信です (第 3章も参照のこと)。to:do:の
ような低水準のメッセージ—初期値から最終値まで変化する数値を引数とし
てブロックを評価する—に加え、Smalltalkのコレクション階層は様々な高水
準のイテレータを提供しています。このようなイテレータを使うことで、コー

ドがより堅牢でコンパクトになります。

繰り返し (do:)

do:メソッドは、基本的なコレクションイテレータです。do:は、引数 (1引数ブ
ロック)をレシーバの各要素に順に適用していきます。次の例では、レシーバ
に含まれるすべての文字列をトランスクリプトにプリントします。

#(’bob’ ’joe’ ’toto’) do: [:each | Transcript show: each; cr].

変種。 do:にはたくさんの変種が存在します。例えば do:without:、doWithIndex:
、 reverseDo: などです: 添字付きのコレクション (Array、OrderedCollection、
SortedCollection) 用に doWithIndex: メソッドがあって、これを使うとブロック
内から現在の添字にもアクセスできます。このメソッドは、Numberクラスで
定義されている to:do:に関係しています。

#(’bob’ ’joe’ ’toto’) doWithIndex: [:each :i | (each = ’joe’) ifTrue: [ ↑ i ] ] −→ 2

OrderedCollectionの reverseDo:は、要素を逆順に処理します。

以下のコードは、do:separatedBy:という興味深いメッセージについて示して
います。do:separatedBy:は、二つの要素の間においてだけ 2番目のブロックを
実行します。

res := ’’.
#(’bob’ ’joe’ ’toto’) do: [:e | res := res, e ] separatedBy: [res := res, ’.’].
res −→ ’bob.joe.toto’

ただしこのコードは、特に効率的だとは言えません。このように一時文字列を

作る代わりに、書き込みストリームを用いて結果をバッファした方がよいとい

うことに注意しましょう (第 10章参照):
String streamContents: [:stream | #(’bob’ ’joe’ ’toto’) asStringOn: stream delimiter: ’.’ ]

−→ ’bob.joe.toto’

Dictionary。 do:メッセージが辞書に送信されたときにチェックされる要素
は、「値」であって「アソシエーション」ではありません。keysDo:、valuesDo:、
associationsDo:を使う方がより適当です。それぞれキー、値、アソシエーショ
ンについて繰り返します14。

colors := Dictionary newFrom: { #yellow -> Color yellow. #blue -> Color blue. #red ->
Color red }.

colors keysDo: [:key | Transcript show: key; cr]. ”キ ー の 一 覧 を 表 示 す る ”
colors valuesDo: [:value | Transcript show: value;cr]. ”値 の 一 覧 を 表 示 す る ”
colors associationsDo: [:value | Transcript show: value;cr]. ”ア ソ シ エ ー シ ョ ン の 一 覧

を 表 示 す る ”

14訳注: do:と valuesDo:は、ほとんど同じ意味です。従って両者は同じ程度不便です。

結果を集める (collect:)

コレクションの要素に何かの処理を行い、その結果を新たなコレクションにし

たい場合、do:を用いるより collect:またはその他のイテレータメソッドを用い
た方がいいでしょう。これらのメソッドのほとんどは、Collectionやそのサブク
ラスの enumeratingプロトコルに分類されています。

例えば各要素を倍にしたコレクションを新たに作る場合を考えてみましょ

う。do:メソッドを使うと、次のように書かなければなりません:
double := OrderedCollection new.
#(1 2 3 4 5 6) do: [:e | double add: 2 * e].
double −→ an OrderedCollection(2 4 6 8 10 12)

collect:メソッドは、引数で渡されたブロックを各要素に対し実行し、その結果
を格納した新たなコレクションを返します。collect:を用いるとコードはとても
シンプルになります。

#(1 2 3 4 5 6) collect: [:e | 2 * e] −→ #(2 4 6 8 10 12)

do:に比べた collect:の利点は、次の例でよりいっそう劇的になります。整数
のコレクションを用意して、その絶対値からなるコレクションを生成します:
aCol := #( 2 -3 4 -35 4 -11).
result := aCol species new: aCol size.
1 to: aCol size do: [ :each | result at: each put: (aCol at: each) abs].
result −→ #(2 3 4 35 4 11)

上の例と、ずっとシンプルな以下の式を比べてみてください。

#( 2 -3 4 -35 4 -11) collect: [:each | each abs ] −→ #(2 3 4 35 4 11)

二つ目の解のさらに良い点は、Setや Bagでも同じように動くことです。

一般的に、コレクションの要素それぞれにメッセージを送信したいのでな

ければ、do:の使用は避けるべきです。

collect:メッセージを送信すると、レシーバと同じ種類のコレクションが返
ることに注意しましょう。以下のコードが失敗するのはこれが原因です (String
は整数を保持できません)。

’abc’ collect: [:ea | ea asciiValue ] ”エ ラ ー !”

こうする代わりに、最初に文字列を Arrayまたは OrderedCollectionに変換してお
かなければなりません:
’abc’ asArray collect: [:ea | ea asciiValue ] −→ #(97 98 99)

実は、collect:がレシーバと完全に同じクラスのコレクションを返すという
保証はなく、単に同じ 「 種 類 (species)」 を返すだけです。例えば、Intervalのイ

ンスタンスに collect:を送信すると、実際には Arrayのインスタンスが返ります。
Interval»speciesが Arrayを返すからです!

(1 to: 5) collect: [ :ea | ea * 2 ] −→ #(2 4 6 8 10)

要素の選別 (select:と reject:)

select:メソッドは、レシーバの要素の中で与えられた条件を満たすものを返し
ます:

(2 to: 20) select: [:each | each isPrime] −→ #(2 3 5 7 11 13 17 19)

reject:は、その反対です:

(2 to: 20) reject: [:each | each isPrime] −→ #(4 6 8 9 10 12 14 15 16 18 20)

detect: による要素の特定

detect:メソッドは、レシーバの要素の中で与えられたブロック引数を満たす最
初のものを返します。

’through’ detect: [:each | each isVowel] −→ $o

detect:ifNone:メソッドは detect:の変種です。detect:ifNone:の二つ目のブロッ
クは、最初のブロックを満たす要素が存在しないときに評価されます。

Smalltalk allClasses detect: [:each | ’*cobol*’ match: each asString] ifNone: [ nil ]
−→ nil

inject:into: を用いて結果を累積する

関数型言語はしばしば、foldや reduceと言った高階関数を用意しています。こ
れらはコレクションのすべての要素に二項演算子を繰り返し適用して、結果を

累積します。Pharoでこのようなことをするには、Collection»inject:into:が使え
ます。

第一引数は初期値です。第二引数は 2引数ブロックで、これまでの累積結
果と今回処理する要素を受け取ります。

inject:into:の簡単な例として、数のコレクションの総和を求めてみましょう。
ガウス少年のように、Pharoでは 1から 100までの自然数の和を以下のように
書けるのです:

(1 to: 100) inject: 0 into: [:sum :each | sum + each ] −→ 5050

別の例として、階乗を計算する 1引数ブロックは以下のように書けます。

factorial := [:n | (1 to: n) inject: 1 into: [:product :each | product * each ] ].
factorial value: 10 −→ 3628800

その他のメッセージ

count: count:メッセージは、与えられた条件を満たす要素の数を返します。条
件は、真偽値を返すブロックで表現します。

Smalltalk allClasses count: [:each | ’Collection*’ match: each asString ] −→ 315

includes: includes:メッセージは、引数がコレクションの中に含まれているか
どうかを調べます。

colors := {Color white . Color yellow. Color red . Color blue . Color orange}.
colors includes: Color blue. −→ true

anySatisfy: anySatisfy:メッセージは、引数で渡された条件を満足する要素がコ
レクション中に一つでもあれば trueを返します。

colors anySatisfy: [:c | c red > 0.5] −→ true

9.6 コレクションを使うときのヒント

add:を使うときのよくある間違い。 次のエラーは、Smalltalkで最も頻繁に
見られる間違いです。

collection := OrderedCollection new add: 1; add: 2.
collection −→ 2

変数 collectionは、新たに作られたコレクションではなく最後に追加された値を
保持しています。これは、add:メソッドが追加された要素を返し、レシーバを
返すわけではないことによるものです。

次のコードによって期待した結果を得ることができます。

collection := OrderedCollection new.
collection add: 1; add: 2.
collection −→ an OrderedCollection(1 2)

15訳注: Pharo 1.3では 5が返ります。

また、yourselfメッセージを用いることもできます。yourselfはカスケード
されたメッセージのレシーバを返します。

collection := OrderedCollection new add: 1; add: 2; yourself −→ an
OrderedCollection(1 2)

繰り返しを行っているコレクションから要素を削除する。 もう一つよくある

間違いは、繰り返しを行っているコレクションから要素を削除しようとするこ

とです。

range := (2 to: 20) asOrderedCollection.
range do: [:aNumber | aNumber isPrime ifFalse: [ range remove: aNumber ] ].
range −→ an OrderedCollection(2 3 5 7 9 11 13 15 17 19)

9と 15が除外されなければいけないので、この結果は明らかに間違っています!
解決策は、処理の前にコレクションをコピーしておくことです。

range := (2 to: 20) asOrderedCollection.
range copy do: [:aNumber | aNumber isPrime ifFalse: [ range remove: aNumber ] ].
range −→ an OrderedCollection(2 3 5 7 11 13 17 19)

=と hashは、両方再定義しなければならない。 発見が難しいのは、=を再定
義して hashを再定義していないというエラーです。Setに入れたはずの要素が
いつの間にかなくなっていたりとか、そういうおかしな振る舞いが症状として

現れます。ケント・ベックが提案した一つの解決策は、xor:を用いて hashを再
定義することです。例えば、著者もタイトルも同じ二つの本は「等しい」と考

えたくなるでしょうから、そのときは次のように、=だけでなく hashも再定義
します:

Method 9.1: = と hash を 再 定 義 す る

Book»= aBook
self class = aBook class ifFalse: [↑ false].
↑ title = aBook title and: [ authors = aBook authors]

Book»hash
↑ title hash xor: authors hash

また、変更可能なオブジェクト す な わ ち ハッシュ値が時間と共に変わって

いくようなオブジェクトを Setの要素、あるいは Dictionaryのキーとして用いる
と、別のたちの悪い問題が生じます。デバッグがお好きでなければ、そういう

ことは止めておきましょう!

9.7 まとめ

Smalltalkのコレクション階層は、異なった種類のコレクションを同じように
扱える共通の語彙を持っています。

• 主な違いは、SequenceableCollectionが要素を与えられた順序で保持するの
に対し、Dictionaryとそのサブクラスはキーと値のアソシエーションを保
持し、Setと Bagは順序を持たないことです。

• ほとんどのコレクションは、asArray、asOrderedCollectionなどのメッセー
ジを送信することで、他のコレクションに変換できます。

• コレクションをソートするには、asSortedCollection メッセージを送信し
ます。

• リテラルの Arrayは #( ... )という特別な文法で作ることができます。動的
な Arrayは { ... }という文法で作ることができます。

• Dictionaryはキーを同値性で比較します。この性質は、キーとして Stringの
インスタンスを用いるときに最も便利です。一方 IdentifyDictionaryはオブ
ジェクトの同一性でキーを判別します。この性質は、キーとして Symbol
を用いるときか、オブジェクトの参照を値にマップするときなどに適し

ています。

• Stringは、通常のコレクションメッセージを受け取ることができます。加
えて Stringは、簡単な形式のパターンマッチングもサポートしています。
より高度なパターンマッチングが必要な場合は、Regexパッケージを検
討してください。

• 繰り返し用の基本的なメッセージは、do:です。do:は命令調のコード、例
えばコレクションの各要素を変更するとか、各要素に同じメッセージを

送信するようなコードで有用です。

• Smalltalkでは、do:より collect:、select:、reject:、includes:、inject:into:その
他の高水準なメッセージを用いる方が普通です。これらによって、コレ

クションを統一的な方法で処理することができます。

• 繰り返しを行っているコレクションの要素を削除してはいけません。繰り
返し中にコレクションを変更したいなら、コピーに対して繰り返します。

• =の定義をオーバーライドしたなら、hashのオーバーライドもお忘れなく!



Chapter 10

ストリーム

ストリームは、要素の並びを繰り返し処理するために用いられます。ここで並

びとは、順序付きコレクション、ファイル、ネットワークストリームなどです。

ストリームは読み込み可能か、書き込み可能か、もしくはその両方です。読み込

みまたは書き込みは、常にストリーム内の現在の位置に対して相対的に行われ

ます。ストリームは簡単にコレクションに変換可能ですし、その逆も同様です。

10.1 二つの「要素の並び」

ストリームを理解するには、以下のような例えが有効です: ストリームは二つ
の「要素の並び」す な わ ち 、過去の要素の並びと未来の要素の並びとして表現

することができます。ストリームはこの二つの並びの間に位置します。このモ

デルを理解しておくことは大切です。Smalltalkのすべてのストリーム操作は
このモデルによるからです。このような理由で、ほとんどの Streamクラスは
PositionableStreamのサブクラスになっています。図 10.1は、5個の文字を含む
ストリームを表しています。このストリームは最初の位置にあり、す な わ ち 、

過去の要素は一つもありません。resetメッセージを使えば、いつでもこの位置
に戻ることができます。

Figure 10.1: 元の位置にあるストリーム。

概念的に、要素の読み込みは、未来の要素の並びから先頭を取り除き、過

去となる要素の並びの最後に付け加えることです。nextメッセージを使って一

つの要素を読むと、ストリームの状態は図 10.2のようになります。

Figure 10.2: 図 10.1のストリームに nextを送信した直後の状態: 文字 aは「過
去」に属するが b、c、d、eは「未来」に属する。

要素の書き込みは、未来の要素の並びの先頭を新しいものに置き換えて、置

き換えられた要素を過去に移動させることです。図 10.3は、nextPut: anElement
メッセージを使って図 10.2のストリームに xを書き込んだ直後の状態を示し
ています。

Figure 10.3: 図 10.2のストリームに xを書き込んだ直後の状態。

10.2 ストリーム対コレクション

コレクションのプロトコルは、要素の格納と除去と列挙をサポートしますが、

これらの操作を混ぜることはできません。例えば OrderedCollectionの要素を do:
メソッドによって処理する場合、do:ブロック内で要素の追加や削除はできま
せん。あるいはコレクションのプロトコルには、二つのコレクションについて

同時に繰り返し、1回ごとに片方を選んで前へ進めながらもう片方は止めてお
くようなものはありません。このような処理を行うには、走査用の添字か位置

の参照を、コレクション自身の外部で管理する必要があります: これはまさに
ReadStream、WriteStreamそして ReadWriteStreamの役割です。

これら三つのクラスは、コレクションをス ト リ ー ム 処 理 するために定義さ

れています。例えば、次のコードは Intervalオブジェクトに対するストリーム
を作り、ストリームから二つの要素を読み込みます。

r := ReadStream on: (1 to: 1000).
r next. −→ 1
r next. −→ 2
r atEnd. −→ false

WriteStreamを使えばコレクションにデータを書き込むことができます:

w := WriteStream on: (String new: 5).
w nextPut: $a.
w nextPut: $b.
w contents. −→ ’ab’

読み込みと書き込みの両方のプロトコルをサポートする ReadWriteStreamを
使うこともできます。

PharoのWriteStreamと ReadWriteStreamの主な問題は、配列と文字列しかサ
ポートしていないことです。現在開発中の Nileという新しいライブラリがこ
の状況を変えつつありますが、今のところそれ以外のコレクションをストリー

ム処理すると、エラーになります1:

w := WriteStream on: (OrderedCollection new: 20).
w nextPut: 12. −→ エラー

ストリームはコレクション専用ではありません。ファイルやソケットに対

しても使えます。次の例は test.txtというファイルを作り、二つの文字列を改行
で区切って書き込み、ファイルを閉じます。

StandardFileStream
fileNamed: ’test.txt’
do: [:str | str

nextPutAll: ’123’;
cr;
nextPutAll: ’abcd’].

次の節では、ストリームプロトコルについてさらに詳しく解説します。

10.3 コレクションに対するストリーム処理

ストリームはコレクションを扱うのにとても便利で、コレクションの要素の読

み込みと書き込みに使うことができます。コレクションに対するストリーム機

能について探っていきましょう。

コレクションの読み込み

この節ではコレクションの読み込みに使われる機能について見ていきます。コ

レクションの読み込みにストリームを使うことは、本質的にはコレクションへ

のポインタを得ることを意味します。ポインタは読み込みによって前へ進み、

また希望する場所へ自由に移動させることができます。ReadStream クラスを
使ってコレクションの要素を読み込みます。

1訳注: このコードはエラーにはなりません。

nextメソッドでコレクションから要素を 1個読み取り、next:メソッドで任
意個の要素を読み取ります。

stream := ReadStream on: #(1 (a b c) false).
stream next. −→ 1
stream next. −→ #(#a #b #c)
stream next. −→ false

stream := ReadStream on: ’abcdef’.
stream next: 0. −→ ’’
stream next: 1. −→ ’a’
stream next: 3. −→ ’bcd’
stream next: 2. −→ ’ef’

peekメッセージは、前へ進まずにストリームの次の要素を調べたいときに
使います。

stream := ReadStream on: ’-143’.
negative := (stream peek = $-). ”前 へ 進 ま ず に 最 初 の 要 素 を 調 べ る 。 ”
negative. −→ true
negative ifTrue: [stream next]. ”マ イ ナ ス 文 字 を 無 視 す る ”
number := stream upToEnd.
number. −→ ’143’

このコードは、ストリームに数の符合があるかどうかで変数 negativeに真偽値を
設定し、数の絶対値を numberに設定します。upToEndメソッドは、現在の位置
から最後までのストリームのすべてを返し、ストリームを最後の位置に設定し

ます。このコードは、peekFor:メソッドを使うことで単純化できます。peekFor:
は、ストリームの次の要素がパラメータと同じときは前へ進み、異なるときは

進みません。

stream := ’-143’ readStream.
(stream peekFor: $-) −→ true
stream upToEnd −→ ’143’

peekFor:はまた、パラメータと次の要素が等しかったかかどうかを示す真偽値
を返します。

上の例を見て、ストリームを作る新しい方法に気づいたかもしれません: 順
序付きコレクションに対して単に readStreamを送信すれば、そのコレクション
に対する読み込みストリームを作ることができます。

位置決め。 ストリームのポインタの位置を設定するメソッドがあります。も

し添字がわかっていれば、position:で直接その位置へ移動できます。positionで
現在の位置を求めることができます。ストリームは要素そのものを指している

わけではなく、要素と要素の間を指していることを覚えておいてください。ス

トリームの先頭に対応する添字は 0です。

図 10.4に描かれたストリームの状態を次のコードで得ることができます:

stream := ’abcde’ readStream.
stream position: 2.
stream peek −→ $c

Figure 10.4: 位置 2にあるストリーム

ストリームを先頭か末尾に移動させるには、それぞれ reset、setToEndを使
うことができます。skip:と skipTo:は、現在の位置を基準として前へ進む場合
に使います。skip:は引数に数を取ることができ、その数だけ要素をスキップし
ます。一方 skipTo:は、パラメータと同じ要素が見つかるまでストリームの要素
をすべてスキップします。このときストリームは、見つかった要素の後ろに位

置決めされることに注意してください。

stream := ’abcdef’ readStream.
stream next. −→ $a ”ス ト リ ー ム は 今 、 a の 真 後 ろ に 位 置 し て い る ”
stream skip: 3. ”ス ト リ ー ム は 今 、 d の 後 ろ ”
stream position. −→ 4
stream skip: -2. ”ス ト リ ー ム は b の 後 ろ ”
stream position. −→ 2
stream reset.
stream position. −→ 0
stream skipTo: $e. ”ス ト リ ー ム は 今 、 e の 真 後 ろ ”
stream next. −→ $f
stream contents. −→ ’abcdef’

ご覧のように文字 eはスキップされています。

contentsメソッドは、常にストリーム全体のコピーを返します。

テスト。 現在のストリームの状態をテストするためのメソッドがあります。

atEndは、これ以上読める要素がない場合のみ真を返します。isEmptyは、コレ
クションにまったく要素がない場合のみ真を返します。

atEndを使ったアルゴリズムの実装例を示します。これは二つのソート済み
のコレクションをパラメータとして取り、それらをマージして新たなソート済

みのコレクションにします:

stream1 := #(1 4 9 11 12 13) readStream.
stream2 := #(1 2 3 4 5 10 13 14 15) readStream.

”変 数 result に は ソ ー ト 済 み の コ レ ク シ ョ ン が 入 る 。 ”
result := OrderedCollection new.
[stream1 atEnd not & stream2 atEnd not]
whileTrue: [stream1 peek < stream2 peek
”両 方 の ス ト リ ー ム の 要 素 の う ち 小 さ い 方 を 取 り 除 き 、 result に 加 え る 。 ”
ifTrue: [result add: stream1 next]
ifFalse: [result add: stream2 next]].

”ど ち ら か の ス ト リ ー ム は 終 端 に 来 て い な い か も し れ な い 。 残 り す べ て を コ
ピ ー す る 。 ”

result
addAll: stream1 upToEnd;
addAll: stream2 upToEnd.

result. −→ an OrderedCollection(1 1 2 3 4 4 5 9 10 11 12 13 13 14 15)

コレクションへの書き込み

ReadStreamを使ってコレクションの要素を繰り返し読み取る方法は、既に見て
きました。次に、WriteStreamを使ってコレクションを作る方法を学びましょう。

WriteStreamは、コレクションの様々な場所にたくさんのデータを加えると
き便利です。次の例のように、静的な部分と動的な部分からなる文字列を生成

するときにしばしば用いられます:

stream := String new writeStream.
stream
nextPutAll: ’This Smalltalk image contains: ’;
print: Smalltalk allClasses size;
nextPutAll: ’ classes.’;
cr;
nextPutAll: ’This is really a lot.’.

stream contents. −→ ’This Smalltalk image contains: 2322 classes.
This is really a lot.’

このテクニック2は例えば、様々なクラスの printOn:メソッドの実装でも使
われています。

単に文字列を生成するためだけにストリームを用いる場合、より単純で効

率的な方法があります:

2訳注: 書き込みストリームにメッセージのカスケードを送信するテクニック。

string := String streamContents:
[:stream |
stream

print: #(1 2 3);
space;
nextPutAll: ’size’;
space;
nextPut: $=;
space;
print: 3. ].

string. −→ ’#(1 2 3) size = 3’

streamContents:メソッドは、コレクションとそのコレクションへの書き込み
ストリームを作ります。このメソッドは与えられたブロックを実行しますが、

このときこのストリームをパラメータとしてブロックに渡します。ブロックが

終了すると streamContents:は、コレクションの中身を返します。

このようなコンテキスト3では、以下の WriteStream のメソッドが特に便利
です:

nextPut: パラメータをストリームに追加する;

nextPutAll: パラメータとして渡されたコレクションの要素をそれぞれストリー
ムに追加する;

print: パラメータのテキスト表現をストリームに追加する。

特殊な文字をストリームに表示するのに便利な space 、tab 、cr (carriage
return)などのメソッドもあります。もう一つ便利なメソッドは、ensureASpace
です。これはストリームの最後の文字がスペースをあることを保証します;ス
トリームの最後の文字がスペースでない場合にはスペースを追加します。

文字 (列)の連結について。 WriteStreamに nextPut:、nextPutAll:を送信するこ
とは多くの場合、文字 (列)を連結する最良の方法になります。カンマ演算子 (,)
による結合は、これらに比べるとはるかに非効率的です:
[| temp |
temp := String new.
(1 to: 100000)
do: [:i | temp := temp, i asString, ’ ’]] timeToRun −→ 115176 ”(milliseconds)”

[| temp |
temp := WriteStream on: String new.
(1 to: 100000)
do: [:i | temp nextPutAll: i asString; space].
temp contents] timeToRun −→ 1262 ”(milliseconds)”

3訳注: 書き込みストリームにアクセス可能なコンテキスト。’This Smalltalk image contains: ...’の
コード例もその一つ。

ストリームを使った方がずっと効率的になり得るのは、カンマ演算子はレ

シーバと引数を連結した新たな文字列を作成するので、レシーバと引数の両

方をコピーしなければならないからです。もし同じレシーバに対してカンマ演

算子で連結を繰り返すと、連結するごとに文字列が長くなり、コピーしなけれ

ばならない文字数は指数関数的に増えてしまいます。またこの方法は、メモリ

上に回収しなければならないゴミをたくさん残します。カンマ演算子の代わり

にストリームを使うことは、最適化の技法としてよく知られています。実際、

streamContents:メソッド (215ページで述べました)が、ストリームを用いた文
字列の連結作業を助けてくれます:

String streamContents: [ :tempStream |
(1 to: 100000)

do: [:i | tempStream nextPutAll: i asString; space]]

読み込みと書き込みを同時に行う

ストリームを介してコレクションへの読み込みアクセス、書き込みアクセスを

同時に行うこともできます。ウェブブラウザの進む・戻るボタンの動作を管理

する History (履歴)クラスが作りたかったとしましょう。履歴は 10.5から 10.11
のような動作をします。

Figure 10.5: 新しい履歴は空。ウェブブラウザには何も表示されていない。

Figure 10.6: ユーザはページ 1を開く。

Figure 10.7: ユーザはページ 2へのリンクをクリックする。

Figure 10.8: ユーザはページ 3へのリンクをクリックする。

Figure 10.9: ユーザは戻るボタンをクリックする。今、再びページ 2 を見て
いる。

Figure 10.10: ユーザは再び戻るボタンをクリックする。ページ 1 が表示され
ている。

Figure 10.11: ユーザはページ 1からページ 4へのリンクをクリックする。ペー
ジ 2、ページ 3の履歴はなくなる。

この振る舞いは ReadWriteStreamを使って実装できます。

Object subclass: #History
instanceVariableNames: ’stream’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE-Streams’

History»initialize
super initialize.
stream := ReadWriteStream on: Array new.

ここでは特に難しいことはありません。ストリームを持つ新しいクラスを

定義します。このストリームは initializeメソッドで作られます。

次に、進むと戻るのメソッドが必要です:
History»goBackward
self canGoBackward ifFalse: [self error: ’Already on the first element’].
stream skip: -2.
↑ stream next.

History»goForward
self canGoForward ifFalse: [self error: ’Already on the last element’].
↑ stream next

ここまでのコードは実に単純明快です。次に、ユーザがリンクをクリック

したときに動作する goTo:メソッドに取りかかりましょう。例えば次のような
コードを思いつくかもしれません:
History»goTo: aPage
stream nextPut: aPage.

しかしこれは完全ではありません。ユーザがリンクをクリックした後は、そ

れ以上は進めるページがあってはいけないからです。す な わ ち 、進むボタンは

無効にならなければいけません。これを実現する最も簡単な方法は、すぐ後ろ

に nilを書き込んで履歴の最後を示すことです:
History»goTo: anObject
stream nextPut: anObject.
stream nextPut: nil.
stream back.

もはや実装しなければならないメソッドは、canGoBackwardと canGoForward
だけになりました。

ストリームは常に二つの要素の間に位置しています。戻るためには現在の

位置より前に二つのページがなければいけません: 一つは現在のページ、もう
一つは戻りたいページです。

History»canGoBackward
↑ stream position > 1

History»canGoForward
↑ stream atEnd not and: [stream peek notNil]

ストリームの中身を確認するメソッドを追加しましょう:
History»contents
↑ stream contents

これで履歴が予告通りに動きます:

History new
goTo: #page1;
goTo: #page2;
goTo: #page3;
goBackward;
goBackward;
goTo: #page4;
contents −→ #(#page1 #page4 nil nil)

10.4 ストリームをファイルアクセスに使う

コレクションをストリーム処理する方法については、既に見てきた通りです。

同じように、ハードディスクのファイルもストリーム処理することができま

す。いったん作成されると、ファイルに対するストリームはコレクションに対

するストリームとほとんど変わりません: 同じプロトコルでストリームの読み
込み、書き込み、位置決めができます。主な違いはストリームの作成方法にあ

ります。ファイルストリームの作成について、いくつかの異なる方法を見てい

きましょう。

ファイルストリームを作成する

ファイルストリームを作るには、以下のインスタンス作成メソッドのいずれか

を用いなければなりません。これらのメソッドは、FileStreamクラスが提供し
ます:

fileNamed: 与えられた名前のファイルを、読み込みと書き込みのために開きま
す。ファイルが既に存在していた場合、その内容は変更されたり置き換え

られたりしますが、ファイルを閉じるときに切り捨てられません。ファ

イル名にディレクトリ部が明示されていない場合、ファイルはデフォル

トディレクトリ内に作られます。

newFileNamed: 与えられた名前の新しいファイルを作り、そのファイルに書き
込むためのストリームを返します。ファイルが既に存在していた場合、

ユーザにどうするか尋ねます。

forceNewFileNamed: 与えられた名前の新しいファイルを作り、そのファイルに
書き込むためのストリームを返します。ファイルが既に存在していた場

合、黙ってそのファイルを削除した後に新しいファイルを作ります。

oldFileNamed: 与えられた名前の既存のファイル4を、読み込みと書き込みのた
めに開きます。ファイルが既に存在していた場合、その内容は変更され

4訳注: ファイルが存在しなかった場合、ユーザにどうするか尋ねます。

たり置き換えられたりしますが、ファイルを閉じるときに切り捨てられ

ません。ファイル名にディレクトリ部が明示されていない場合、ファイ

ルはデフォルトディレクトリ内に作られます。

readOnlyFileNamed: 与えられた名前の既存のファイルを、読み込みのために開
きます。

開いたファイルストリームは忘れずに閉じなければなりません。これには

closeメソッドを使います。

stream := FileStream forceNewFileNamed: ’test.txt’.
stream
nextPutAll: ’This text is written in a file named ’;
print: stream localName.

stream close.

stream := FileStream readOnlyFileNamed: ’test.txt’.
stream contents. −→ ’This text is written in a file named ’’test.txt’’’
stream close.

ファイルストリームに localNameを送信すると、パス名の最後の要素が返り
ます。fullNameを送信してフルパス名を得ることもできます。

すぐに、ファイルストリームを手作業で閉じるのが苦痛でエラーの元である

ことに気づくでしょう。forceNewFileNamed:do:は、この問題を解消するために
用意されています。このメソッドは、書き込みストリームをパラメータとする

ブロックを評価してすべての書き込みをすませた後、自動的にこのストリーム

を閉じます。同様に読み込みストリーム用に readOnlyFileNamed:do:があります。

FileStream
forceNewFileNamed: ’test.txt’
do: [:stream |
stream
nextPutAll: ’This text is written in a file named ’;
print: stream localName].

string := FileStream
readOnlyFileNamed: ’test.txt’
do: [:stream | stream contents].

string −→ ’This text is written in a file named ’’test.txt’’’

ファイルストリーム作成メソッドのうちブロックを引数に取るものは、ま

ずストリームを作成し、次にストリームを引数としてブロックを実行し、最後

にストリームを閉じます。これらのメソッドは、ブロックが返すものを返しま

す。つまりブロックの最後にある式の値を返します。前の例では、ブロックの

最後の式: stream contentsを評価してファイルの中身を得、その値を変数 string
に代入しています。

バイナリストリーム

デフォルトでファイルストリームは、テキストベースつまり文字の読み書き用

に作られます。ストリームでバイナリデータを扱う場合は、ストリームに binary
メッセージを送信しなければなりません。

ストリームがバイナリモードの場合、0から 255 (1バイト)の数値しか書き
込めません。nextPutAll:メソッドを使って一度に二つ以上の数値を書きたいと
きは、ByteArrayオブジェクトを引数として渡さなければなりません。

FileStream
forceNewFileNamed: ’test.bin’
do: [:stream |

stream
binary;
nextPutAll: #(145 250 139 98) asByteArray].

FileStream
readOnlyFileNamed: ’test.bin’
do: [:stream |

stream binary.
stream size. −→ 4
stream next. −→ 145
stream upToEnd. −→ #[250 139 98]

].

次の例では「test.pgm」という画像ファイルを作成します (PGM: portable
graymap)。作成したファイルは、お気に入りのドロープログラムで開くこと
ができます。

FileStream
forceNewFileNamed: ’test.pgm’
do: [:stream |
stream
nextPutAll: ’P5’; cr;
nextPutAll: ’4 4’; cr;
nextPutAll: ’255’; cr;
binary;
nextPutAll: #(255 0 255 0) asByteArray;
nextPutAll: #(0 255 0 255) asByteArray;
nextPutAll: #(255 0 255 0) asByteArray;
nextPutAll: #(0 255 0 255) asByteArray

]

これは図 10.12のような 4x4のチェッカーボード (市松模様)を作ります。

Figure 10.12: バイナリストリームで描くことができる 4x4のチェッカーボー
ド (市松模様)

10.5 まとめ

ストリームは、要素の並びをインクリメンタルに読み書きする場合に、コレク

ションよりも良い方法を提供します。ストリームとコレクションを相互に変換

する簡単な方法があります。

• ストリームは読み込み可能、書き込み可能、あるいは読み書きともに可
能です。

• コレクションをストリームに変換する場合、コレクション「に対する (on)」
ストリームを定義します (例 え ば ReadStream on: (1 to: 1000))。あるいはコ
レクションに readStreamなどのメッセージを送信します。

• ストリームをコレクションに変換する場合、contentsメッセージを送信し
ます。

• 大きなコレクションを連結する場合は、カンマ演算子を使うよりもスト
リームを使った方が効率的です。つまりストリームを作り、nextPutAll:で
コレクションをストリームに追加し、contentsを送信して結果を取り出し
ます。

• ファイルストリームはデフォルトで文字型です。これを明示的にバイナ
リにするには、binaryメッセージを送信します。



Chapter 11

Morphic

Morphicは、Pharoの GUIに付けられた名前です。Morphicは Smalltalkで
書かれており、様々な OSの間で完全な互換性があります。その結果、Pharo
は Unix, MacOS, Windowsのいずれにおいてもまったく同じ見た目となって
います。Morphicと他の多くのユーザ・インターフェース・ツールキットとの
違いは、インターフェースを「構成すること」と「実行すること」とのモードを

分けていないということです。画面上のすべてのグラフィック要素（graphical
elements）は、いつでもユーザが組み立てたり分解したりできます。1

11.1 Morphicの歴史

Morphicは、プログラミング言語 Self 向けとして、John Maloneyと Randy
Smithによって 1993年頃から開発が始められました。Maloneyは Squeakの
ために新しくMorphicを書き直しましたが、Self版の基本的な考え方は今でも
PharoのMorphicに受け継がれています。その基本的な考え方とは、直 接 性
と生 命 感 です。直接性とは、画面上の図形がマウスでクリックして直接調べた

り変更したりできるオブジェクトであるということです。生命感とは、ユーザ

インターフェースがいつでもユーザの操作に反応できることです。つまり、画

面上の情報は、それが表している世界が変わるたびに自動的に更新されます。

簡単な例を挙げれば、メニューから項目を取り外してボタンとして使うことが

できます。

World メ ニ ュ ー を 出 し ま す 。 そ の 上 で メ タ ク リ ッ ク を 一 回 す る と 、 Morphic
ハ ロ ー が 現 れ ま す 。

2
そ の 後 、 取 り 外 し た い メ ニ ュ ー 項 目 の 上 で さ ら に メ タ ク

1Hilaire Fernandesに謝意を表します。彼のフランス語で書かれたオリジナル版をこの章のベー
スとすることを許諾してくれました。

2プリファレンス・ブラウザで、 halosEnabledを設定することを忘れないでください。または、
ワークスペース上で Preferences enable: #halosEnabledを実行します。

リ ッ ク を す る と 、 メ ニ ュ ー 項 目 の 上 に ハ ロ ー が 現 れ ま す 。 図 11.1 で 示 す よ う
に 、 黒 い ハ ン ド ル を 掴 ん で そ の 項 目 を 画 面 の 適 当 な 場 所 に ド ラ ッ グ し ま す 。

Figure 11.1: Workspaceメニューのモーフを取り外して独立したボタンにして
いるところ。

Pharoの起動後、画面上にあるすべてのオブジェクトは、モ ー フ と呼ばれま
す。つまり、これらのオブジェクトはMorphクラスのインスタンスです。Morph
クラスは多くのメソッドを持つ大きなクラスであるため、少ないコードで面白

い動作をするサブクラスを作れます。あらゆるオブジェクトを表すモーフを作

れますが、どのくらい良い表現が得られるかはオブジェクトによります。

文 字 列 オ ブ ジ ェ ク ト を 表 す モ ー フ を 作 る に は 、 ワ ー ク ス ペ ー ス で 以 下 の

コ ー ド を 実 行 し ま す 。

’Morph’ asMorph openInWorld

これは、’Morph’という文字列を表すモーフを作り、「ワールド」上で開きま
す。（つまり、表示します）「ワールド」とは、Pharoの画面のことです。これに
よりメタクリックで操作できるグラフィック要素—モーフ—が得られます。
もちろん、今見ているものより面白い表現を持つモーフを定義することも可

能です。class Objectクラスにある asMorphメソッドのデフォルトの実装では、
単に StringMorph を作るだけです。例えば、Color tan asMorphは、単に Color
tan printStringの実行結果をラベルに持つ StringMorphを返します。それでは、
これを変更して色の付いた長方形を代わりに返すようにしましょう。

ブ ラ ウ ザ を 開 き Colorク ラ ス に 以 下 の メ ソ ッ ド を 追 加 し ま す 。

Method 11.1: Colorの イ ン ス タ ン ス を 表 す モ ー フ を 返 す
Color»asMorph
↑ Morph new color: self

ワークスペースで Color orange asMorph openInWorldを実行すれば、文字列のモー
フの代わりにオレンジ色の長方形が現れます。

11.2 モーフを操作する

モーフはオブジェクトです。つまり、Smalltalkの他のオブジェクトと同様に操
作できます。メッセージを送ることで属性を変えたり、新しいMorphのサブク
ラスを作ったりできます。どんなモーフも（たとえ画面に表示されていなくと

も）位置と大きさを持っています。便宜上、すべてのモーフは画面上の四角形

の領域を占めていると考えます。不規則な形状の場合であっても、その位置と

大きさは、そのモーフを囲む最小の四角形の「箱」となります。この箱を、モー

フのバウンディングボックス、あるいは単に「bounds」と言います。position
メソッドは、モーフの左上の座標（またはモーフのバウンディングボックスの

左上の座標）を表す Pointオブジェクトを返します。この座標系の原点は画面の
左上の端にあります。y座標は画面の下 方 向 に 増えていき、x座標は右方向に
増えていきます。extentメソッドも Pointオブジェクトを返しますが、位置では
なくモーフの幅と高さからなる座標値を返します。

ワ ー ク ス ペ ー ス に 以 下 の コ ー ド を 入 力 し do it し ま す 。

joe := Morph new color: Color blue.
joe openInWorld.
bill := Morph new color: Color red .
bill openInWorld.

さらに、joe positionと入力してprint itします。joeを動かすために、joe position: (
joe position + (10@3))を繰り返して実行します。

大きさについても同様のことが可能です。joe extentは joeの大きさを返し
ます。joeを大きくするには、joe extent: (joe extent * 1.1)を実行します。モーフの
色を変えるには、好きな Colorオブジェクトを引数にして color:メッセージを送
ります。例えば、joe color: Color orangeのようにします。透明度を設定するには、
joe color: (Color orange alpha: 0.5)を試してください。

bill に joe の 後 を 追 い か け さ せ る に は 、 以 下 の コ ー ド を 繰 り 返 し て 実 行 し
ま す 。

bill position: (joe position + (100@0))

マウスで joeを移動した後にこのコードを実行すれば、billは joeの 100ピク
セル右側に移動するでしょう。

11.3 モーフの組み込み

新しいグラフィカルな表現を作る一つの方法は、モーフの内側に別のモーフを

置くことです。これをモーフの組 み 込 み と言います。モーフはどんな深さにも

組み込めます。モーフを別のモーフに組み込むには、addMorph:というメッセー
ジを入れ物となるモーフに送ります。

あ る モ ー フ を 別 の モ ー フ に 加 え ま し ょ う 。

star := StarMorph new color: Color yellow.
joe addMorph: star.
star position: joe position.

最後の行では、星を joeと同じ座標に置いています。組み込まれるモーフの座
標は、入れ物となるモーフを基準にするのではなく画面を基準とすることに気

をつけてください。モーフの位置を設定するための多くのメソッドがあります。

Morphクラスの geometry をブラウズすれば自分自身で確認できます。例えば、
joeの中心に星を置くには、star center: joe centerを実行します。

Figure 11.2: 青色の半透明なモーフである joeが星を含んでいる。

マウスで星を掴もうとすると実際には joeを掴み、二つのモーフが一緒に
動きます。星は joeの内側に埋 め 込 ま れ て います。joeの内側にもっと多くの
モーフを埋め込むことができます。プログラムによる埋め込みに加え、直接的

な操作によっても埋め込むことができます。

11.4 自由にモーフを作り、描く

モーフの組み込みによりたくさんの便利で面白いグラフィカルな表現を作れま

すが、いずれ今までとまったく違う何かを作りたいと思うでしょう。そのため

には、Morphのサブクラスを作り drawOn:メソッドをオーバーライドすれば、見
た目を変更できます。

Morphicのフレームワークは、モーフの再描画が必要となった際に、モーフ
に対して drawOn:メッセージを送ります。drawOn:メッセージの引数は、Canvas
クラス（またはそのサブクラス）のインスタンスで、期待される振る舞いはモー

フが自分自身をキャンバスの bounds内に描くことです。この知識を使って十
字モーフを作ってみましょう。

ブ ラ ウ ザ を 開 き 、 Morphを 継 承 し た CrossMorph ク ラ ス を 新 し く 定 義 し ま す 。

Class 11.2: CrossMorphを 定 義 す る
Morph subclass: #CrossMorph
instanceVariableNames: ’’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE-Morphic’

以下のように drawOn:メソッドを定義できます。

Method 11.3: CrossMorphを 描 く
drawOn: aCanvas
| crossHeight crossWidth horizontalBar verticalBar |
crossHeight := self height / 3.0 .
crossWidth := self width / 3.0 .
horizontalBar := self bounds insetBy: 0 @ crossHeight.
verticalBar := self bounds insetBy: crossWidth @ 0.
aCanvas fillRectangle: horizontalBar color: self color.
aCanvas fillRectangle: verticalBar color: self color

Figure 11.3: Haloを伴ったCrossMorph。自由に大きさを変更できる。

モーフに boundsメッセージを送ると、Rectangleクラスのインスタンスであ
るバウンディングボックスが返されます。Rectangleクラスのインスタンスは数
多くのメッセージを理解し、四角形に関連する他の四角形を作れます。この例

では、引数に点を指定した insetBy:メッセージを使って高さを低くした四角形を
作り、さらに幅を縮めた四角形を作っています。

新 し い モ ー フ を テ ス ト す る た め に 、 CrossMorph new openInWorld を 実 行 し
ま す 。

実行すると図 11.3のようになります。しかし、マウスに反応する場所（つ
まりモーフを掴むのにクリックするところ）がバウンディングボックス全体に

なっていることに気づくと思います。これを解決しましょう。

Morphicフレームワークは、モーフがカーソルの下にあるかどうか知る必
要があるとき、マウスポインタの下にバウンディングボックスがあるすべての

モーフに対して、containsPoint:メッセージを送ります。つまり、マウスに反応す
る場所をモーフの十字部分に制限するには、containsPoint:メソッドをオーバー
ライドします。

CrossMorphク ラ ス に 以 下 の メ ソ ッ ド を 定 義 し ま す 。

Method 11.4: CrossMorphの マ ウ ス に 反 応 す る 場 所 を 設 定 す る 。
containsPoint: aPoint
| crossHeight crossWidth horizontalBar verticalBar |
crossHeight := self height / 3.0.
crossWidth := self width / 3.0.
horizontalBar := self bounds insetBy: 0 @ crossHeight.
verticalBar := self bounds insetBy: crossWidth @ 0.
↑ (horizontalBar containsPoint: aPoint)
or: [verticalBar containsPoint: aPoint]

このメソッドは drawOn:と同じロジックを使っています。つまり、
containsPoint:が trueと答える領域は、drawOnによって色が付けられた領域と一致
すると確実に言えます。Rectangleクラスの containsPoint:を活用すれば、この面
倒な仕事をやりとげることができます。

メソッド Method 11.3 と 11.4のコードには二つの問題点があります。明
らかな点は、コードが重複していることです。これはとても重大な過ちです。

というのも、horizonatalBarや verticalBarの計算方法を変える必要があるときに、
二箇所にあるコードの一方を修正するのを簡単に忘れてしまうからです。解決

方法は、これらの計算を新たな二つのメソッドに抽出することです。どちらも

privateプロトコルに格納します。

Method 11.5: horizontalBar.
horizontalBar
| crossHeight |
crossHeight := self height / 3.0.
↑ self bounds insetBy: 0 @ crossHeight

Method 11.6: verticalBar.
verticalBar
| crossWidth |
crossWidth := self width / 3.0.
↑ self bounds insetBy: crossWidth @ 0

これらのメソッドを使って drawOn:と containsPoint:の両方を書き直します。

Figure 11.4: 十字の中央が２度描かれ
る。

Figure 11.5: 塗られていない部分のあ
る十字モーフ。

Method 11.7: リ フ ァ ク タ リ ン グ 後 の CrossMorph»drawOn:.
drawOn: aCanvas
aCanvas fillRectangle: self horizontalBar color: self color.
aCanvas fillRectangle: self verticalBar color: self color

Method 11.8: リ フ ァ ク タ リ ン グ 後 の CrossMorph»containsPoint:.
containsPoint: aPoint
↑ (self horizontalBar containsPoint: aPoint)
or: [self verticalBar containsPoint: aPoint]

プライベートメソッドに意味のある名前を付けたおかげで、このコードは

より簡単に理解できるようになりました。実際、すぐに２番目の問題が明らか

になります。それは、十字の中央の部分が horizontalBarと verticalBarの両者
の下になっており、２度描かれてしまうという問題です。不透明な色を塗る場

合には問題になりませんが、図 11.4に示すような半透明な色で描くとすぐにこ
のバグが露見します。

ワ ー ク ス ペ ー ス で 、 以 下 の コ ー ド を １ 行 ず つ 実 行 し ま す 。

m := CrossMorph new bounds: (0@0 corner: 300@300).
m openInWorld.
m color: (Color blue alpha: 0.3).

この解決法は、verticalBarを三つの部分に分け、上側と下側の部分だけ塗りつ
ぶすということです。Rectangleクラスでこの面倒な仕事をしてくれるメソッド
がまた見つかります。r1 areasOutside: r2は、r2の外側に出ている r1を構成する四
角形の配列を返します。以下が改良されたコードになります。

Method 11.9: drawOn:メ ソ ッ ド の 改 良 版 。 十 字 の 中 心 を 一 度 だ け 塗 り つ ぶ す 。
drawOn: aCanvas
| topAndBottom |
aCanvas fillRectangle: self horizontalBar color: self color.

topAndBottom := self verticalBar areasOutside: self horizontalBar.
topAndBottom do: [ :each | aCanvas fillRectangle: each color: self color]

このコードは正しく動作するように見えますが、何回か十字の大きさを変

更しようとすると、図 11.5のように、ある大きさのときに１ピクセル分の幅
の線が十字の下側と残りの部分を分けてしまうことに気づきます。これは丸

め誤差のせいです。塗りつぶそうとする四角形の大きさが整数でないときに、

fillRectangle: color:が不適切に数値を丸めてしまい、１ピクセル分だけ塗りつぶ
されなくなるからです。このバグは、四角形の大きさを計算する際に明示的に

四捨五入することで解決できます。

Method 11.10: 明 示 的 に 四 捨 五 入 し た CrossMorph»horizontalBar

horizontalBar
| crossHeight |
crossHeight := (self height / 3.0) rounded.
↑ self bounds insetBy: 0 @ crossHeight

Method 11.11: 明 示 的 に 四 捨 五 入 し た CrossMorph»verticalBar

verticalBar
| crossWidth |
crossWidth := (self width / 3.0) rounded.
↑ self bounds insetBy: crossWidth @ 0

11.5 対話とアニメーション

モーフを使って、いきいきとしたユーザインターフェースを実現するには、マ

ウスやキーボードでモーフと対話できなければなりません。さらに、モーフは

自分の見た目や位置を変えることで（つまり自分自身のアニメーションによっ

て）、ユーザからの入力に反応する必要があります。

マウスイベント

マウスボタンが押されたとき、Morphicはマウスポインタの下にある各モーフ
に対して handlesMouseDown:メッセージを送ります。もし、モーフが trueを返す
と、Morphicは直ちにそのモーフに対してmouseDown:メッセージを送ります。
ユーザがマウスボタンを離した際にも mouseUp:メッセージを送ります。すべ
てのモーフが falseを返すと、Morphicはドラッグ＆ドロップの操作を開始しま
す。後述するように、mouseDown:と mouseUp:メッセージは、MouseEventオブ
ジェクトの引数を伴って送られます。この引数は、マウスの動作の詳細を表し

ています。

CrossMorphクラスを拡張して、マウスイベントを扱うようにしましょう。ま
ず、すべての十字モーフが handlesMouseDown:メッセージに確実に trueを返す
ことから始めます。

CrossMorphに 以 下 の メ ソ ッ ド を 追 加 し ま す 。

Method 11.12: CrossMorphが マ ウ ス ク リ ッ ク に 反 応 す る よ う に 宣 言 す る 。
CrossMorph»handlesMouseDown: anEvent
↑true

十字モーフのクリックで赤色に変わり、アクションクリックで黄色に変わ

るようにしましょう。これはMethod 11.13によって実現できます。

Method 11.13: マ ウ ス ク リ ッ ク に 反 応 し て モ ー フ の 色 を 変 え る よ う に す る 。
CrossMorph»mouseDown: anEvent
anEvent redButtonPressed ”click”
ifTrue: [self color: Color red].

anEvent yellowButtonPressed ”action-click”
ifTrue: [self color: Color yellow].

self changed

このメソッドでは、モーフの色を変えた後で self changedを送っていること
に注意してください。これによって、Morphicは直ちに drawOn:を送るように
なります。なお、いったんモーフがマウスイベントを処理すると、もはやモー

フをマウスで掴んで動かすことができなくなることに気をつけてください。そ

の代わりにHaloを使う必要があります。モーフ上のメタクリックでHaloを
出し、モーフの上部にある茶色の か、赤の のいずれかでモーフを掴みます。

mouseDown:の引数の anEventは、MouseEventクラスのインスタンスです。こ
れは MorphicEventクラスのサブクラスです。MouseEventでは、redButtonPressed
と yellowButtonPressedメソッドを定義しています。マウスイベントを扱うメソッ
ドにどんなものがあるか調べるには、このクラスをブラウズします。

キーボードイベント

キーボードイベントを得るには、次の３ステップを必要とします。

1. 「キーボードフォーカス」を特定のモーフに与えます。例えば、マウス
がモーフの上にあるときに、フォーカスを与えることができます。

2. handleKeystroke:メソッドを使って、モーフがキーボードイベントを処理
するようにします。このメッセージは、ユーザがキーを押した際に、キー

ボードフォーカスを持っているモーフへ送られます。

3. マウスがモーフの上にないときに、キーボードフォーカスを解放します。

キー入力に反応するように CrossMorphクラスを拡張します。まず、マウスが
モーフ上にあることを通知するように変更します。これは、handlesMouseOver:
メッセージに trueを返すことで実現できます。

モ ー フ が マ ウ ス ポ イ ン タ の 下 に あ る と き に 、 CrossMorphが 対 応 で き る よ う
宣 言 し ま す 。

Method 11.14: 「 マ ウ ス オ ー バ ー 」 イ ベ ン ト を 扱 え る よ う に す る 。
CrossMorph»handlesMouseOver: anEvent
↑true

マウスの位置の取り扱いは handlesMouseDown:の場合と同様です。マウスポイ
ンタがモーフに入る、または離れる際に、mouseEnter:と mouseLeave:メッセー
ジがモーフに送られます。

CrossMorphク ラ ス が キ ー ボ ー ド フ ォ ー カ ス の 獲 得 と 解 放 を 行 え る よ う に 、
二 つ の メ ソ ッ ド を 定 義 し ま す 。 三 つ 目 の メ ソ ッ ド で 実 際 に キ ー 入 力 を 処 理 し

ま す 。

Method 11.15: マ ウ ス が モ ー フ に 入 っ た と き に キ ー ボ ー ド フ ォ ー カ ス を 獲 得
す る 。

CrossMorph»mouseEnter: anEvent
anEvent hand newKeyboardFocus: self

Method 11.16: マ ウ ス ポ イ ン タ が モ ー フ か ら 離 れ た と き に 、 キ ー ボ ー ド フ ォ ー
カ ス を 解 放 す る 。

CrossMorph»mouseLeave: anEvent
anEvent hand newKeyboardFocus: nil

Method 11.17: キ ー ボ ー ド イ ベ ン ト を 受 け 取 っ て 処 理 す る 。
CrossMorph»handleKeystroke: anEvent
| keyValue |
keyValue := anEvent keyValue.
keyValue = 30 ”up arrow”
ifTrue: [self position: self position - (0 @ 1)].

keyValue = 31 ”down arrow”
ifTrue: [self position: self position + (0 @ 1)].

keyValue = 29 ”right arrow”
ifTrue: [self position: self position + (1 @ 0)].

keyValue = 28 ”left arrow”
ifTrue: [self position: self position - (1 @ 0)]

矢印キーを使ってモーフを動かすことができるようなメソッドを書きまし

た。マウスがモーフの上にないときには、handleKeystroke: メッセージが送ら

れないため、モーフはキーボードに反応しなくなることに注意してください。

キーの値を知りたいときは、トランスクリプトウィンドウを出し、Transcript
show: anEvent keyValueをMethod 11.17に加えてください。handleKeystroke:の引
数の anEventは、KeyboardEventのインスタンスで、MorphicEventクラスのサブク
ラスです。このクラスをブラウズすれば、キーボードイベントについてさらに

学習できます。

Morphicアニメーション

Morphicは主に二つのメソッドからなる単純なアニメーション機能を提供しま
す。stepが一定の間隔でモーフに送られ、stepTimeは stepの間隔をミリ秒単位
で指定します。3加えて、startSteppingは stepを開始させ、stopSteppingはそれ
を停止させます。stepが始まっているかどうかを知るには、isSteppingを使い
ます。

以 下 の よ う な メ ソ ッ ド を 定 義 し て 、 CrossMorphを 点 滅 さ せ ま す 。

Method 11.18: ア ニ メ ー シ ョ ン の 間 隔 を 設 定 す る 。
CrossMorph»stepTime
↑ 100

Method 11.19: ア ニ メ ー シ ョ ン の step を 作 る 。
CrossMorph»step
(self color diff: Color black) < 0.1
ifTrue: [self color: Color red]
ifFalse: [self color: self color darker]

アニメーションを開始させるには、CrossMorph上でインスペクタを開いて（Halo
のデバッグハンドル を使います）、下部の小さなワークスペースペインに

self startSteppingと入力し、do itを実行します。代わりに handleKeystroke:メソッ
ドを変更して、+キーと −キーで stepを開始、停止させることもできます。

以 下 の コ ー ド を Method 11.17 に 加 え ま す 。

keyValue = $+ asciiValue
ifTrue: [self startStepping].

keyValue = $- asciiValue
ifTrue: [self stopStepping].

3実際には、stepTimeは step間の最 小 時間を指定します。仮に stepTimeを 1 にしても、Pharoの
負荷が高くなりすぎて、それほど頻繁に step をモーフに送れないので注意してください。

11.6 対話機能

ユーザに入力を促すために、UIManagerクラスはすぐに使えるダイアログボッ
クスを多く提供しています。例えば、request:initialAnswer:メソッドは、ユーザ
によって入力された文字列を返します。（図 11.6）
UIManager default request: ’What’’s your name?’ initialAnswer: ’no name’

Figure 11.6: 入力ダイアログ

Figure 11.7: ポップアップメニュー

ポップアップメニューを出すには、様々な chooseFrom:メソッドを使います。
（図 11.7）
UIManager default
chooseFrom: #(’circle’ ’oval’ ’square’ ’rectangle’ ’triangle’)
lines: #(2 4) message: ’Choose a shape’

UIManager ク ラ ス を ブ ラ ウ ズ し て 、 提 供 さ れ て い る 対 話 メ ソ ッ ド を 試 し て
み ま し ょ う 。

11.7 ドラッグ＆ドロップ

Morphicはドラッグ＆ドロップもサポートします。受け入れ側のモーフとド
ロップされるモーフの二つのモーフからなる簡単な例について考えてみましょ

う。受け入れ側のモーフは、ドロップされるモーフが条件を満たしたときにだ

けドロップを受け入れます。この例ではモーフが青色であることを条件としま

す。受け入れが拒否された場合、ドロップされるモーフが何をするか決めます。

ま ず 受 け 入 れ 側 の モ ー フ を 定 義 し ま し ょ う 。

Class 11.20: 他 の モ ー フ を 受 け 入 れ る モ ー フ を 定 義 す る 。
Morph subclass: #ReceiverMorph
instanceVariableNames: ’’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE-Morphic’

い つ も の や り 方 で 初 期 化 メ ソ ッ ド を 定 義 し ま す 。

Method 11.21: ReceiverMorphを 初 期 化 す る 。
ReceiverMorph»initialize
super initialize.
color := Color red.
bounds := 0 @ 0 extent: 200 @ 200

受け入れ側のモーフがドロップを受け入れるか否かは、どのように判断さ

れるのでしょうか？ 一般的に、モーフの両者がドロップに同意する必要があ

ります。受け入れ側は wantsDroppedMorph:event: メッセージに反応することで
決定します。最初の引数はドロップされるモーフ、２番目の引数はマウスイベ

ントです。例えば、受け入れ側はドロップ時に何らかの修飾キーが押されたか

どうかを調べられます。ドロップされるモーフも、wantsToBeDroppedInto:メッ
セージが送られたときに、ドロップされる先のモーフが気に入るか調べるチャ

ンスがあります。このメソッドのデフォルトの実装（Morphクラスで定義され
ている）では、trueを返します。

Method 11.22: 色 を 基 準 に し て ド ロ ッ プ を 受 け 入 れ る 。
ReceiverMorph»wantsDroppedMorph: aMorph event: anEvent
↑ aMorph color = Color blue

受け入れ側のモーフがドロップを望まない場合には、ドロップされるモーフ

に何が起こるのでしょうか?デフォルトの動作は、何もしないことです。つまり、
ドロップされずに受け入れ側のモーフの上に置かれるだけです。より直感的な

動作としては、ドロップされるモーフが元の位置に戻ることでしょう。これは、

ドロップされるモーフを拒否する際に、受け入れ側のモーフが repelsMorph:event:
メッセージに対して trueを返すことで実現できます。

Method 11.23: ド ロ ッ プ 拒 否 時 の モ ー フ の 振 る 舞 い を 変 え る 。
ReceiverMorph»repelsMorph: aMorph event: ev
↑ (self wantsDroppedMorph: aMorph event: ev) not

以上が、受け入れ側のモーフで必要なことすべてです。

ワ ー ク ス ペ ー ス で ReceiverMorph と EllipseMorph の イ ン ス タ ン ス を 作 り ま す 。

ReceiverMorph new openInWorld.
EllipseMorph new openInWorld.

黄色い EllipseMorphを受け入れ側のモーフにドラッグ＆ドロップしましょう。
モーフはドロップを拒否され、元の位置に戻るはずです。

こ の 動 作 を 変 え る に は 、 イ ン ス ペ ク タ を 使 っ て 楕 円 の モ ー フ の 色 を Color
blueに 変 え ま す 。 青 い モ ー フ は ReceiverMorphに 受 け 入 れ ら れ る は ず で す 。

では、DroppedMorphと名付けた Morphクラスの特別なサブクラスを作って、
もう少し経験を積みましょう。

Class 11.24: ReceiverMorphに ド ラ ッ グ ＆ ド ロ ッ プ 可 能 な モ ー フ を 定 義 す る 。
Morph subclass: #DroppedMorph
instanceVariableNames: ’’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE-Morphic’

Method 11.25: DroppedMorphを 初 期 化 す る 。
DroppedMorph»initialize
super initialize.
color := Color blue.
self position: 250@100

ドロップされるモーフが受け入れ側に拒否されたときにすべきことを設定

できます。ここでは、モーフをマウスポインタに付いたままにします。

Method 11.26: ド ロ ッ プ を 拒 否 さ れ た モ ー フ の 反 応 を 定 義 す る 。
DroppedMorph»rejectDropMorphEvent: anEvent
| h |
h := anEvent hand.
WorldState
addDeferredUIMessage: [h grabMorph: self].

anEvent wasHandled: true

イベントに対して handメッセージを送ると、HandMorphのインスタンスで
あるハ ン ド が返されます。このオブジェクトはマウスポインタを表していて、

マウスポインタに関係するものは何でも持っています。ここでは、ドロップを

拒否されたモーフである selfをハンドが掴むようにWorldに指示しています。

DroppedMorphの 二 つ の イ ン ス タ ン ス を 作 成 し 、 受 け 入 れ 側 の モ ー フ に ド
ラ ッ グ ＆ ド ロ ッ プ し ま す 。

ReceiverMorph new openInWorld.
(DroppedMorph new color: Color blue) openInWorld.
(DroppedMorph new color: Color green) openInWorld.

緑色のモーフはドロップを拒否されるため、マウスポインタに付いたままにな

ります。

11.8 例題

サイコロを転がすモーフを作りましょう。4モーフをクリックするとすべての
面が高速に表示され、再度クリックするとアニメーションが止まります。

Figure 11.8: Morphicで作られたサイコロ

外 枠 を 描 く た め 、 Morphク ラ ス の 代 わ り に BorderedMorph の サ ブ ク ラ ス と
し て サ イ コ ロ を 定 義 し ま す 。

Class 11.27: サ イ コ ロ モ ー フ を 定 義 す る 。
BorderedMorph subclass: #DieMorph
instanceVariableNames: ’faces dieValue isStopped’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE-Morphic’

インスタンス変数の facesは、サイコロの面の個数を記録します。ここで実
装するサイコロモーフは、９までの面を許すことにします！ dieValueは現在表
示されている面の値を表し、isStoppedはサイコロのアニメーションが停止して
いるときに trueとなります。サイコロのインスタンスを作るために、DieMorph
のク ラ ス 側に faces: nメソッドを定義し、n個の面を持つ新しいサイコロを生成
するようにします。

Method 11.28: 好 き な 個 数 の 面 を 持 つ 新 し い サ イ コ ロ を 生 成 す る 。
DieMorph class»faces: aNumber
↑ self new faces: aNumber

initializeメソッドは今までと同じやり方でインスタンス側に定義します。な
お、newは新しく生成されたインスタンスに initializeメッセージを送ることを覚
えておいてください。

4

Method 11.29: DieMorphの イ ン ス タ ン ス を 初 期 化 す る 。
DieMorph»initialize
super initialize.
self extent: 50 @ 50.
self useGradientFill; borderWidth: 2; useRoundedCorners.
self setBorderStyle: #complexRaised.
self fillStyle direction: self extent.
self color: Color green.
dieValue := 1.
faces := 6.
isStopped := false

サイコロの見た目をよくするために、BorderedMorphクラスのメソッドをい
くつか使います。浮き出る効果を得るために枠線を太くしたり、角を丸めたり、

表示面にグラデーションをかける等をします。インスタンス側の faces:メソッ
ドでは、正しい引数が与えられているかをチェックするように定義します。

Method 11.30: サ イ コ ロ の 面 の 個 数 を 設 定 す る 。
DieMorph»faces: aNumber

”Set the number of faces”
(aNumber isInteger

and: [aNumber > 0]
and: [aNumber <= 9])

ifTrue: [faces := aNumber]

サイコロが生成されたときにメッセージが送られる順序を知るのは良いこ

とです。例えば、DieMorph faces: 9を評価したと仮定します。

1. クラスメソッドの DieMorph class»faces:では、DieMorph classクラスに new
メッセージを送ります。

2. （Behaviorクラスから継承された DieMorph classクラスの）newメソッドで、
新しいインスタンスが生成され、そのインスタンスに initializeメッセージ
が送られます。

3. DieMorphの initializeメソッドが、facesインスタンス変数に初期値 6を設定
します。

4. DieMorph class»newメソッドの結果できたインスタンスが、クラスメソッ
ドの DieMorph class»faces:に返り、その新しいインスタンスに faces: 9メッ
セージが送られます。

5. インスタンスメソッドの DieMorph»faces:が実行され、facesインスタンス
変数を 9に設定します。

drawOn:を定義する前に、表示面に描く目の位置を求めるメソッドをいくつ
か定義します。

Methods 11.31: サ イ コ ロ の 目 の 位 置 を 求 め る 九 つ の メ ソ ッ ド を 定 義 す る 。
DieMorph»face1
↑{0.5@0.5}

DieMorph»face2
↑{0.25@0.25 . 0.75@0.75}

DieMorph»face3
↑{0.25@0.25 . 0.75@0.75 . 0.5@0.5}

DieMorph»face4
↑{0.25@0.25 . 0.75@0.25 . 0.75@0.75 . 0.25@0.75}

DieMorph»face5
↑{0.25@0.25 . 0.75@0.25 . 0.75@0.75 . 0.25@0.75 . 0.5@0.5}

DieMorph»face6
↑{0.25@0.25 . 0.75@0.25 . 0.75@0.75 . 0.25@0.75 . 0.25@0.5 . 0.75@0.5}

DieMorph»face7
↑{0.25@0.25 . 0.75@0.25 . 0.75@0.75 . 0.25@0.75 . 0.25@0.5 . 0.75@0.5 . 0.5@0

.5}
DieMorph »face8
↑{0.25@0.25 . 0.75@0.25 . 0.75@0.75 . 0.25@0.75 . 0.25@0.5 . 0.75@0.5 . 0.5@0.5

. 0.5@0.25}
DieMorph »face9
↑{0.25@0.25 . 0.75@0.25 . 0.75@0.75 . 0.25@0.75 . 0.25@0.5 . 0.75@0.5 . 0.5@0.5

. 0.5@0.25 . 0.5@0.75}

上記のメソッドは、サイコロの各面について目の座標をコレクションの形

で定義しています。それぞれの座標は 1× 1の範囲に収めているので、拡大し
て実際の目の場所を求めます。

drawOn:メソッドは、サイコロの背景を描くための super送信と、目を描くと
いう二つの仕事を行います。

Method 11.32: サ イ コ ロ モ ー フ を 描 く 。
DieMorph»drawOn: aCanvas
super drawOn: aCanvas.
(self perform: (’face’ , dieValue asString) asSymbol)
do: [:aPoint | self drawDotOn: aCanvas at: aPoint]

このメソッドの後半は、Smalltalkのリフレクション機能を使っています。
サイコロの面に対応する faceXメソッドから得た座標のコレクションに対して、
順に drawDotOn:at:メッセージを送ることでサイコロの目を描いています。適切
な faceXメソッドを呼び出すために、perform:メソッドを使います。このメソッ
ドが、(’face’, dieValue asString) asSymbolにより作られた文字列をメッセージとし
て送ります。このような perform:の使い方は至るところで見られます。

Method 11.33: サ イ コ ロ の 目 を 描 く 。
DieMorph»drawDotOn: aCanvas at: aPoint
aCanvas
fillOval: (Rectangle

center: self position + (self extent * aPoint)
extent: self extent / 6)

color: Color black

座標が [0:1]の範囲で正規化されているため、サイコロの大きさに合わせる
ために、self extent * aPointで拡大します。

ワ ー ク ス ペ ー ス で サ イ コ ロ の イ ン ス タ ン ス を 作 る こ と が で き ま す 。

(DieMorph faces: 6) openInWorld.

表示面を変えるために、myDie dieValue: 4のように使えるアクセサを作ります。

Method 11.34: サ イ コ ロ の 表 示 面 を 設 定 す る 。
DieMorph»dieValue: aNumber
(aNumber isInteger

and: [aNumber > 0]
and: [aNumber <= faces])

ifTrue:
[dieValue := aNumber.
self changed]

サイコロの面を高速に表示するためにアニメーション機能を使います。

Methods 11.35: サ イ コ ロ の ア ニ メ ー シ ョ ン を 作 る 。
DieMorph»stepTime
↑ 100

DieMorph»step
isStopped ifFalse: [self dieValue: (1 to: faces) atRandom]

サイコロが回ってますね！

サイコロの回転をクリックで始めたり止めたりするには、前述したマウスイ

ベントのやり方を使います。まず、マウスイベントを受け入れるようにします。

Methods 11.36: ア ニ メ ー シ ョ ン の 開 始 と 停 止 の た め に マ ウ ス ク リ ッ ク を 処 理
す る 。

DieMorph»handlesMouseDown: anEvent
↑ true

DieMorph»mouseDown: anEvent
anEvent redButtonPressed
ifTrue: [isStopped := isStopped not]

以上により、クリックでサイコロを振ったり止めたりできます。

11.9 キャンバスに関する追加情報

drawOn:メソッドは、Canvasクラスのインスタンスを引数として一つだけ取り
ます。このキャンバスとは、モーフ自身を描く領域のことです。キャンバスの

描画メソッドを使うことで、モーフに自由な外観を与えられます。Canvasクラ
スの階層構造をブラウズすれば、いくつもの変形を見ることができます。通常

用いられる Canvasクラスの変形は FormCanvasクラスなので、Canvasクラスと
FormCanvasクラスの中から主要な描画メソッドを見つけられます。これらのメ
ソッドには、点や線、多角形や四角形、楕円やテキスト、画像などの描画や、

回転や拡大・縮小などの機能があります。

透明なモーフやその他の描画メソッド、アンチエイリアスなどを実現する

ために、他の種類のキャンバスも利用できます。これらの特徴を使うために

は、AlphaBlendingCanvasクラスや BalloonCanvasクラスが必要となるでしょう。
しかし、drawOn:が引数として FormCanvasのインスタンスを受け取るとしたら、
drawOn:の中でどうやって他のキャンバスを得ればよいのでしょうか？ 幸いに
も、あるキャンバスを別の種類に変換することが可能です。

DieMorphの ア ル フ ァ チ ャ ン ネ ル 透 過 率 を 0.5 に す る キ ャ ン バ ス を 使 う よ う
に 、 drawOn:を 再 定 義 し ま す 。

Method 11.37: 透 明 な サ イ コ ロ を 描 く 。

DieMorph»drawOn: aCanvas
| theCanvas |
theCanvas := aCanvas asAlphaBlendingCanvas: 0.5.
super drawOn: theCanvas.
(self perform: (’face’ , dieValue asString) asSymbol)
do: [:aPoint | self drawDotOn: theCanvas at: aPoint]

以上で終わりです。

Figure 11.9: アルファチャンネル透過率を設定したサイコロ。

11.10 まとめ

Morphicは、グラフィカルなフレームワークで、動的に GUI要素を組み合わ
せることができます。

• asMorph openInWorldメッセージをオブジェクトに送ると、モーフに変換し
て画面に表示できます。

• モーフをメタクリックすると、ハンドルが表示されモーフを操ることが
できるようになります。（ハンドルの機能は個々のバルーンヘルプの説明

を見ましょう)

• ドラッグ＆ドロップによるか addMorph:メッセージを送ることで、モーフ
を別のモーフに組み込めます。

• 既にあるクラスのサブクラスを作り、initializeや drawOn:といった主要なメ
ソッドを再定義できます。

• handlesMouseDown:メソッドや handlesMouseOver:メソッドなどを再定義す
れば、マウスやキーボードに反応するようにモーフを制御できます。

• stepメソッド (何をするか)と stepTimeメソッド (ミリ秒単位の step間隔)
を定義すれば、モーフのアニメーションを作れます。

• PopUpMenuや FillInTheBlankといった多くの定義済みのモーフを使って、
ユーザと対話できます。



Chapter 12

Seaside by Example

Seasideは SmalltalkでWebアプリケーションを作成するためのフレームワー
クです。2002年に Avi Bryantによって開発が始まりました。一度マスターす
れば、デスクトップアプリケーションと同じようにWebアプリケーションを
容易に作ることができるようになります。

Seasideの有名な事例としては SqueakSource1 と Dabble DB2があります。
徹底したオブジェクト指向が Seasideの特徴です。XHTML生成のためのテン
プレート、Webページ間の複雑な遷移の定義、状態を埋め込んだ URLといっ
たものは Seasideでは不要です。オブジェクトにメッセージを送るだけなので
す。素敵です。

12.1 なぜSeaside?

最近のWebアプリケーションは、ユーザとのやりとりをデスクトップアプリ
ケーションと同じように行おうとします。アプリケーションはユーザに問いか

け、ユーザはフォームを記入したり、ボタンを押したりしてこれに応えます。

ところがWebの仕組みは実際には逆なのです。Webブラウザがサーバーにリ
クエストをすると、新しいページがサーバーから返ってきます。そのためWeb
アプリケーション開発 フレームワークは多くの問題に直面することになりま

す。とりわけ問題なのは、この逆になっている制御フローの取り扱いです。そ

こで、多くのWebアプリケーションはブラウザの「戻る」ボタンを押すこと
を禁じています。押されてしまうとセッションの状態を保つのが困難になるか

らです。Webページ間の複雑な制御の流れを表現するのは、やっかいなもので
す。さらに制御が分岐したりすると、表現するのが極めて難しくなるか、場合

によっては不可能になってしまいます。

1http://SqueakSource.com
2http://DabbleDB.com

http://SqueakSource.com
http://DabbleDB.com

SeasideではWebアプリケーション開発を容易にするため、コンポーネント
ベースのフレームワークを採用しています。まず、制御の流れはメッセージ送

信によって自然な形で表すことができます。Seasideは、Webアプリケーショ
ンの実行状態を、表示したページと共に覚えています。そのため「戻る」ボタ

ンを押しても問題が生じません。

また、状態は Seasideが自動的に管理します。バックトラック機能が有効な
場合、「戻る」ボタンを押すと変更した状態が復元します。トランザクション機

能を使えば、戻るボタンを押したときに変更した状態が復元されないようにす

ることもできます。URLに状態を埋め込んだりする必要もありません。URL
生成も Seasideが行ってくれます。

Webページはコンポーネント群によって構成されます。それぞれが独立し
た制御フローを持つことができます。XHTMLのテンプレートはなく、 簡単
な Smalltalkのメッセージ送信によって正しい XHMLがプログラム的に生成
されます。Seasideは CSSCSSをサポートしています。そのためページ内容と
レイアウトは明確に分離されます。

最後に、Seasideは、Webブラウザ上での便利な開発ツールを提供してい
ます。これによりインタラクティブな形で簡単にWebアプリケーションを開
発、デバッグできます。サーバーを動作させながら、リコンパイルしてアプリ

ケーションを拡張することだってできるのです。

12.2 はじめに

SeasideのWebサイト3から “Seaside One-Click Experience”をダウンロード
すると、すぐに Seasideを始めることができます。ファイルにはWindows, Mac
OSX, Linux用の Seaside 2.8の環境一式が同梱されています。4Seasideのサイ
トには他にもドキュメントやチュートリアルなど、たくさんのリンクがありま

す。Seasideは長年にわたって進化を続けているものなので、すべてが最新の
Seasideについてのものであるとは限らないので注意してください。

SeasideはWebサーバーを含んでいます。8080ポートで動かすにはWAKom
startOn: 8080を実行します。サーバーを終了するときは、WAKom stopです。デ
フォルトの管理者ユーザ名は adminであり、パスワードは seasideになってい
ます。WADispatcherEditor initializeと実行すると、新たなユーザとパスワードを
聞いてくるようになります。

Seaside を 開 始 し て Web ブ ラ ウ ザ で http:// localhost:8080/seaside/ を 開 き ま
し ょ う 。

図 12.1のようなページが表示されます。
examples. counter ペ ー ジ を ク リ ッ ク し て く だ さ い 。 (図 12.2)

3http://seaside.st
4訳注：Seaside は頻繁にアップデートが行われており、翻訳時点では既に 3.0 が公開されてい

ます

http://localhost:8080/seaside/
http://seaside.st

Figure 12.1: Seasideの初期画面

Figure 12.2: カウンターのアプリケーション

これは小さな Seasideアプリケーションのサンプルです。++や ––のリンクを
クリックするとカウンターの値を変えることができます。

リ ン ク を ク リ ッ ク し て 値 を 変 え て み ま し ょ う 。 Web ブ ラ ウ ザ の 「 戻 る 」 ボ
タ ン で 前 の 状 態 に 戻 っ て か ら ++ の リ ン ク を 再 び ク リ ッ ク し て み て く だ さ い 。
カ ウ ン タ ー は 、 表 示 さ れ て い た 値 か ら 増 え て い ま す 。 決 し て 「 戻 る 」 ボ タ ン を

押 し 始 め た と き の 値 か ら で は あ り ま せ ん 。

図 12.1のwebページの下の方にあるツールバーに注目してみてください。
Seasideは「セッション」の概念をサポートしており、ユーザごとにアプリケー
ションの状態を管理します。New Session を押すと、カウンターアプリケーショ

ンの新たなセッションが始まります。Configure ではアプリケーションの各種
設定をWebブラウザ経由で行うことができます。

(Configureの画面を閉じるには、右上の x リンクをクリックします)。
Toggle Halos を押すと、Seasideで動作しているアプリケーションの状態が調
べられます。 Profiler と Memory ボタンはアプリケーション実行時のパフォー
マンスに関する詳細な情報を見るためのものです。

XHTML は生成されたWebページの XHTMLをバリデートするときに使い
ます。W3Cのバリデータを使っているため、対象とするページが外部からア
クセス可能になっていないとこの機能は動作しません。

Seasideのアプリケーションは付け外しの可能な「コンポーネント」からで
きています。コンポーネントはありふれた Smalltalkオブジェクトであり、他
と異なる点と言えばフレームワークが提供するWAComponentを継承したクラス
のインスタンスであるということだけです。Pharoのイメージからコンポーネ
ントやクラスを探索できますし、Webブラウザからハローを通じて直接たどっ
ていくこともできます。

Figure 12.3: ハロー

Toggle Halos を 押 し て み ま し ょ う 。 図 12.3 の よ う な 画 面 に な る は ず で す 。
左 上 に WACounterと 表 示 さ れ て い ま す が 、 こ れ は 現 在 の ペ ー ジ の 構 成 要 素 と な っ
て い る コ ン ポ ー ネ ン ト の ク ラ ス 名 を 表 し て い ま す 。 そ の 隣 に は 三 つ の ボ タ ン ア

イ コ ン が あ り ま す 。 最 初 の 鉛 筆 の ア イ コ ン は 、 こ の コ ン ポ ー ネ ン ト の ク ラ ス ブ

ラ ウ ザ を 開 く た め の も の で す 。 2 番 目 の 虫 眼 鏡 ア イ コ ン で は 、 現 在 ア ク テ ィ ブ
に な っ て い る WACounterの イ ン ス タ ン ス の イ ン ス ペ ク タ を 開 く こ と が で き ま す 。
3 番 目 の 色 の 付 い た 丸 の ボ タ ン を 押 す と 、 こ の コ ン ポ ー ネ ン ト に 対 す る CSS ス
タ イ ル シ ー ト エ デ ィ タ が 表 示 さ れ ま す 。 右 上 に は R と S の リ ン ク が あ り ま す 。
そ れ ぞ れ ク リ ッ ク す る と コ ン ポ ー ネ ン ト が 通 常 表 示 と ソ ー ス 表 示 と に 切 り 替 わ

り ま す 。 こ う し た リ ン ク を す べ て 試 し て み ま し ょ う 。 ソ ー ス ビ ュ ー の 表 示 の と

き に も 、 ++ や – は 有 効 に な っ て い ま す 。 フ ォ ー マ ッ ト も さ れ て い る た め 、 Web
ブ ラ ウ ザ の ソ ー ス 表 示 よ り も 見 や す い か も し れ ま せ ん 。

Seasideのクラスブラウザやオブジェクト・インスペクタは、サーバーが別
のマシンで動作していて、画面がなかったり、リモートの環境だったりすると

きには非常に便利です。とはいえ、最初に Seasideのアプリケーションを開発
するときには、ローカルでサーバーを動かすと思います。その場合は Pharoの
イメージが提供する普段の開発ツールの方が使いやすいでしょう。

Figure 12.4: カウンターアプリケーションを haltさせる

Web ブ ラ ウ ザ の オ ブ ジ ェ ク ト ・ イ ン ス ペ ク タ の リ ン ク を 使 っ て カ ウ ン タ ー
の イ ン ス ペ ク タ を 開 き 、 self haltを 実 行 し て み ま し ょ う 。 Web ペ ー ジ は ロ ー ド 中
の 状 態 に な り ま す 。 Seaside の イ メ ー ジ の 方 に 移 っ て み て く だ さ い 。 ノ ー テ ィ
フ ァ イ ア が 上 が っ て い て (図 12.4)、 WACounter の オ ブ ジ ェ ク ト が halt の 実 行 で
止 ま っ て い る こ と が 確 認 で き ま す 。 さ ら に デ バ ッ ガ を 開 い て Proceed を 押 し て
み ま し ょ う 。 Web ブ ラ ウ ザ に 戻 る と カ ウ ン タ ー は 再 び 動 作 し て い ま す 。

Seasideのコンポーネントは、異なるコンテキストで何度でも初期化でき
ます。

Web ブ ラ ウ ザ で http:// localhost:8080/seaside/examples/multicounter を 開 い て
み て く だ さ い 。 い く つ も の 独 立 し た カ ウ ン タ ー か ら な る ア プ リ ケ ー シ ョ ン が 立

ち 上 が り ま す 。 カ ウ ン タ ー を い く つ か 選 ん で 値 を 変 え て み ま し ょ う 。 「 戻 る 」 ボ

タ ン を 押 し て も 問 題 な く 動 作 し ま す 。 ハ ロ ー を 表 示 し て み る と 、 入 れ 子 に な っ

た コ ン ポ ー ネ ン ト 群 か ら ア プ リ ケ ー シ ョ ン が で き て い る こ と が わ か り ま す 。 ク

ラ ス ブ ラ ウ ザ を 使 っ て 、 WAMultiCounterの 実 装 が ど う な っ て い る か 見 て み ま し ょ
う 。 ク ラ ス メ ソ ッ ド が 三 つ (canBeRoot、 description、 initialize)、 イ ン ス タ ン ス メ
ソ ッ ド も 三 つ (children、 initialize、 renderContentOn:) 定 義 さ れ て い る こ と が わ か
り ま す 。 ア プ リ ケ ー シ ョ ン は 、 コ ン ポ ー ネ ン ト 階 層 の ル ー ト に な れ る と い う だ

け で 、 単 な る コ ン ポ ー ネ ン ト で す 。 ク ラ ス メ ソ ッ ド の canBeRootで trueを 返 す
こ と で 、 ル ー ト に な る た め の 宣 言 を 行 い ま す 。

SeasideのWebインターフェースを使うと、アプリケーション (す な わ ち
ルートになるコンポーネント)を設定したり、コピーしたり、登録解除したり
できます。設定を少しいじってみましょう。

http://localhost:8080/seaside/examples/multicounter

Figure 12.5: 独立したサブコンポーネント

Web ブ ラ ウ ザ で http:// localhost:8080/seaside/config を 開 き ま す 。 ユ ー ザ ID と
パ ス ワ ー ド を 入 れ て 、 ロ グ イ ン し て く だ さ い 。 (デ フ ォ ル ト で は admin / seaside と
な っ て い ま す )。 “examples” の 隣 に あ る Configure を ク リ ッ ク し ま す 。 “Add entry
point” の と こ ろ で 、 “counter2” と 名 前 を 入 れ 、 タ イ プ は Application の ま ま
Add ボ タ ン を ク リ ッ ク し ま す (図 12.6 を 参 照 )。 次 の 画 面 で 、 Root Component
を WACounter に し て 、 Save を 押 し 、 Close で 閉 じ ま す 。 こ れ で カ ウ ン タ ー の
新 し い ア プ リ ケ ー シ ョ ン が http:// localhost:8080/seaside/examples/counter2 で 起 動
す る よ う に な り ま し た 。 同 様 に 、 こ の 設 定 ツ ー ル 画 面 を 通 じ て 、 登 録 し た ア プ

リ ケ ー シ ョ ン の 解 除 を 行 う こ と も で き ま す 。

Seasideは二つのモードで動作します。一つは今まで見てきた 開 発 モードで
す。 デ プ ロ イ モードにすると、画面下のツールバーが非表示となります。デプロ

イモードにするには、全体設定を行うツールのページからアプリケーションを選

び、Configureをクリックするか、ツールバーに表示されているConfigureをクリッ
クします。どちらの場合も、デプロイモードのプルダウンボックスが表示される

ので、trueに設定します。この設定は新たなセッションを開始したタイミングで
有効になります。Seaside全体でモードを切り替えるにはWAGlobalConfiguration
setDeploymentModeや、WAGlobalConfiguration setDevelopmentModeを実行します。

設定ツール自体も Seasideのアプリケーションです。そのため同じように
Webブラウザから各種の設定ができます。もしも誤って “config”のアプリケー
ションを削除してしまった場合は、WADispatcherEditor initializeとすると元に戻
ります。

http://localhost:8080/seaside/config
http://localhost:8080/seaside/examples/counter2

Figure 12.6: 新たなアプリケーションの設定

12.3 Seasideのコンポーネント

前の節で示したように、Seasideのアプリケーションはコ ン ポ ー ネ ン ト から成
り立っています。Hello Worldのコンポーネントを作成し、Seasideの仕組みを
さらに詳しく見てみましょう。

コンポーネントを定義するにはWAComponentを直接あるいは間接的に継承
するようにします (図 12.8)。

WAComponentを 継 承 し た WAHelloWorldク ラ ス を 定 義 し て み ま し ょ う 。

コンポーネントでは自身の表示の仕方を定める必要があります。このため

には renderContentOn:をオーバーライドします。このメソッドには、引数として
XHTMLの生成方法を知っているWAHtmlCanvasのインスタンスが渡されます。

rendering プ ロ ト コ ル に 以 下 の よ う に メ ソ ッ ド を 定 義 し て み ま し ょ う 。

WAHelloWorld»renderContentOn: html
html text: ’hello world’

次に Seasideに対して、コンポーネントがスタンドアローンのアプリケーショ
ンであるという宣言を行います。

WAHelloWorld の ク ラ ス 側 に 以 下 の メ ソ ッ ド を 定 義 し ま し ょ う 。

WAHelloWorld class»canBeRoot
↑ true

これでほとんどできあがりです。

Web ブ ラ ウ ザ で http:// localhost:8080/seaside/config を 開 き 、 “hello” の エ ン
ト リ ー ポ イ ン ト を 追 加 し て 、 ル ー ト の コ ン ポ ー ネ ン ト を WAHelloWorldに 設 定 し
ま す 。 そ し て http:// localhost:8080/seaside/hello を 開 く と 、 図 12.7 の よ う に 表 示
さ れ ま す !

Figure 12.7: Seasideの “Hello World”

状態のバックトラックと “Counter”アプリケーション

“hello world”に比べると「カウンター」のアプリケーションは、ほんの少し
だけ難しいものになっています。

WACounterはスタンドアローンのアプリケーションなので WACounter class
では canBeRootを実装して、trueを返す必要があります。さらに、図 12.8にあ
るように、アプリケーションとしての登録をクラスメソッド initializeの中で行っ
ています。

http://localhost:8080/seaside/config
http://localhost:8080/seaside/hello

Figure 12.8: counterアプリケーションはWACounterクラスにより実装されてい
ます。アンダーラインの引いてあるメソッドはクラスメソッドで、普通のテキ

ストで書かれたメソッドはインスタンスメソッドです。

WACounterには increaseと decreaseメソッドがあります。それぞれWebペー
ジで ++、––のリンクをクリックしたときに呼び出されます。カウンターの値
を保持するため、インスタンス変数として countが定義してあります。「戻る」
ボタンを押してもカウンターの値がブラウザの値とずれないようにするには、

WACounterの状態も元に戻せるようにしなければなりません。Seafsideは状態
を元に戻すためのメカニズムを備えていますが、アプリケーションのどの状態

に対して行うかは開発者が指定する必要があります。

インスタンスメソッドの statesを実装すると、どの状態を元に戻す対象にす
るかを定めることができます。statesでは対象とすべきオブジェクト群を配列
の形で返すようにします。カウンターアプリケーションの場合、Array with: self
としています。こうするとWACounterのオブジェクト自身が元に戻す対象とし
て Seasideに登録されることになります。

注意 状態を戻すオブジェクトを宣言する場合、ちょっとした注意点がありま

す。Seasideは states配列で宣言されたオブジェクトの copyを取ることで、過
去の状態を保持します。この実際の処理は WASnapshot によって行われます。
WASnapshotは IdentityDictionary のサブクラスで、キーとして元に戻す対象とな
るオブジェクト群、値として各オブジェクトのシャローコピーを持つ辞書です。

特定のスナップショットにアプリケーションの状態を戻す必要があるときには、

辞書に格納された値を用いてオブジェクトの変数の値を上書きします。

ここで注意すべき点があります。先ほどのWACounterの場合、バックトラッ
クの対象は、インスタンス変数 countの値だけでよいのではと思うかもしれませ
ん。しかし、statesメソッドで Array with: countのように書いてもうまく動作し
ません。これは countの中身が整数で、整数は不変のオブジェクトだからです。
increaseや decreaseメソッドでは、整数オブジェクトの状態を 0から 1に変え
たり、 3から 2に変えたりしているわけではありません。その代わりに、 count
変数の中身が代入されて別の整数オブジェクトに置き換えられているのです。

整数が増えたり減ったりするたびに、 countが指すオブジェクトは別のものに
成 り 代 わ っ て います。このため、 WACounter»statesでは Array with: selfと書く必
要があるのです。WACounterが以前の状態に戻るときには、各インスタンス変
数の値がコピーされたWACounterから復元されます。これで countの値は正しく
戻ることになります。

12.4 XHTMLのレンダリング

Webアプリケーションでは、Webページを動的に生成して表示 (レンダリング)
を行います。 12.3節の節で述べたように、Seasideのコンポーネントは、それ
ぞれが自身をレンダリングするようになっています。仕組みを知るために、カ

ウンターのコンポーネントがどのようにレンダリングを行っているか、見てい

きましょう。

カウンターのレンダリング

カウンターのレンダリングは比較的単純です。図 12.8のコードを見てくださ
い。カウンターの現在の値が XHTMLの見出しとして表示されます。値の増
減の操作はアンカー (リンク)として実装され、コールバックがブロックの形で
仕掛けられています。アンカーをクリックすると increaseや decreaseがカウン
ターオブジェクトに対して送られるようになっています。

レンダリングで送るメッセージ群についてはすぐ後で詳しく解説しますが、

その前にマルチカウンターの方もどうなっているか見てみましょう。

CounterからMultiCounterへ

まず図 12.9で示すように、WAMultiCounterもスタンドアローンのアプリケーショ
ンなので、 canBeRootを実装して trueを返すようにしています。

マルチカウンターのコンポーネントは、サブコンポーネントを含んだ コ ン

ポ ジ ッ ト となっています。そのため childrenメソッドを実装して、中に含まれ
るコンポーネント群を配列の形で返しています。

レンダリングでは、サブコンポーネント群を、水平線で区切りを入れて表

示するように書いています。そのほか初期化を行うためのインスタンスメソッ

ド、クラスメソッドがありますが、基本的にはこれだけでよいのです!

Figure 12.9: WAMultiCounter

レンダリングの詳細

今までの例で見てきたように、SeasideではWebページの生成にテンプレート
を使いません。その代わり、Smalltalkプログラムによって XHTMLを生成し
ます。Seasideのコンポーネントは renderContentOn: をオーバーライドする決
まりになっています。このメソッドはコンポーネントのレンダリングのときに

フレームワークの側から呼び出されます。renderContentOn:は引数を一つ取り、
html キ ャ ン バ ス オブジェクトが渡されます。これを使ってコンポーネントは
レンダリングを行います。慣習によりキャンバスの引数は html という名前に
なっています。htmlキャンバスはMorphicや別の GUIフレームワークでの
キャンバスに似たものです。デバイス固有となる描画処理の詳細を、隠す役割

を持っています。

基本的なレンダリングメッセージをいくつか挙げてみます。

html text: ’hello world’. ”テ キ ス ト の 表 示 ”
html html: ’&ndash;’. ”XHTMLの 直 接 指 定 ”
html render: 1. ”オ ブ ジ ェ ク ト の レ ン ダ リ ン グ ”

任意のオブジェクトを表示するには render: anObjectを使うことができます。
通常はサブコンポーネントの表示に使われます。引数となった anObject には
renderContentOn:のメッセージが内部的に送られます。これはマルチカウンター
の例で使いました (図 12.9を参照)。

ブラシの利用

キャンバス上で描画を行うために、たくさんの ブ ラ シ を使うことができます。

XHTMLの各要素 (段落、テーブル、リストなど)に対応したブラシがありま
す。ブラシやそれに関連する便利メソッドを知るには、WACanvasとそのサブ

クラスをブラウズするとよいでしょう。renderContentOn:の引数は、WACanvas
を継承した WARenderCanvasのインスタンスになっています。

既にカウンターとマルチカウンターの例で、以下のブラシを使いました。

html horizontalRule.

Figure 12.10: レンダリングのデモ

図 12.10は、Seasideが提供する基本的なブラシをいろいろと使ってみた例
です。 5SeasideDemoのルートコンポーネントは単にサブコンポーネントの表示

5Method 12.1のソースコードは http://www.squeaksource.com/PharoByExampleプロジェクトの PBE

http://www.squeaksource.com/PharoByExample

を行うだけです。サブコンポーネントはMethod 12.1を読むとわかりますが、
SeasideHtmlDemo、SeasideFormDemo、SeasideEditCallDemo、SeasideDialogDemo
となっています。

Method 12.1: SeasideDemo»renderContentOn:
SeasideDemo»renderContentOn: html
html heading: ’Rendering Demo’.
html heading
level: 2;
with: ’Rendering basic HTML: ’.

html div
class: ’subcomponent’;
with: htmlDemo.

”残 り を レ ン ダ リ ン グ ...”

コンポーネントはサブコンポーネントを返すというのを覚えているでしょうか。

Seasideではレンダリングのたびにサブコンポーネントを再利用します。

SeasideDemo»children
↑ { htmlDemo . formDemo . editDemo . dialogDemo }

上記のコードでは heading ブラシの使い方が少し違っていることに注意し
てください。heading:は、ブラシの生成と同時にテキストをメッセージの引数
でセットする書き方になっています。次に出てくるのは、headingメッセージを
送ってブラシを生成した後で、様々なプロパティをカスケードでセットしてい

くというものです。ブラシはたいていこの 2種類の書き方ができるようになっ
ています。

もしも カスケード を使って、ブラシに with: を送るなら、
with:は必ず最 後 に 書く必要があります。with:によってブラ
シのレンダリングが始まるからです。

Method 12.1のメソッドでは、最初の見出しはデフォルトでレベル 1にな
ります。次の見出しはレベル 2を明示的に指定しています。サブコンポーネン
トは XHTMLで divの中に囲まれる形になります。divには “subcomponent”
という CSS のクラスが付与されます。(CSSの使い方については 12.5 節で解
説します) with:メッセージの引数は、文字列とは限りません。この例ではコン
ポーネントを渡しています。次の例でも出てきますが、レンダリングの操作を

記述したブロックでもよいのです。

SeasideHtmlDemoコンポーネントは基本的なブラシをいろいろと使ったデモ
になっています。コードを見れば大体の利用の仕方がわかるでしょう。

SeasideHtmlDemo»renderContentOn: html

-SeasideDemoパッケージ内にあります。

self renderParagraphsOn: html.
self renderListsAndTablesOn: html.
self renderDivsAndSpansOn: html.
self renderLinkWithCallbackOn: html

上のように、レンダリングのメソッドが長くなった場合には細かいメソッ

ド群に分割することがよく行われます。

一つのメソッド内に長々とレンダリング処理を書くやり方は

お勧めできません。render*On:という形でヘルパーメソッド
に分割しましょう。そうしたメソッド群は renderingメソッ
ドカテゴリに置いておくようにします。また、コンポーネ

ントに renderContentOn:を直接送ってはいけません。代わり
に render:を送るようにしてください。

さらにコードを見ていきます。最初のヘルパーメソッド SeasideHtmlDemo»
renderParagraphsOn:は、XHTMLの段落、通常のテキスト、強調表示されたテ
キスト、画像の表示の例になっています。Seasideでは単純な要素なら、ただ
文字列を指定するだけでレンダリングできますし、複雑なものであればブロッ

クを使います。レンダリングのコードを書くときの指針として覚えておくとよ

いでしょう。

SeasideHtmlDemo»renderParagraphsOn: html
html paragraph: ’A plain text paragraph.’.
html paragraph: [
html
text: ’A paragraph with plain text followed by a line break. ’;
break;
emphasis: ’Emphasized text ’;
text: ’followed by a horizontal rule.’;
horizontalRule;
text: ’An image URI: ’.

html image
url: self squeakImageUrl;
width: ’50’]

次のヘルパーメソッド SeasideHtmlDemo»renderListsAndTablesOn:は、リスト
とテーブルの表示例になっています。テーブルではネストしたブロックを使い、

列と列内のセルを表示します。

SeasideHtmlDemo»renderListsAndTablesOn: html
html orderedList: [
html listItem: ’An ordered list item’].

html unorderedList: [
html listItem: ’An unordered list item’].

html table: [

html tableRow: [
html tableData: ’A table with one data cell.’]]

次の例は、divや spanの要素に、CSSの classや id属性を指定するやり方を
示しています。もちろん class:や id:といったメッセージは、div span以外のブラ
シにも送ることができます。SeasideDemoWidget»styleのメソッドで、XHTML
の要素が実際にどのようなスタイルで表示されるかを定義しています (12.5節
を参照)。

SeasideHtmlDemo»renderDivsAndSpansOn: html
html div
id: ’author’;
with: [
html text: ’Raw text within a div with id ’’author’’. ’.
html span
class: ’highlight’;
with: ’A span with class ’’highlight’’.’]

最後に、リンクの簡単な例を見てみましょう。アンカー (リンク)にコール
バックを設定しています。リンクをクリックするたびに、文字列が “true”と
“false”に切り替わりますが、これはインスタンス変数の toggleValueの値がコー
ルバックの実行により変化するからです。

SeasideHtmlDemo»renderLinkWithCallbackOn: html
html paragraph: [
html text: ’An anchor with a local action: ’.
html span with: [
html anchor
callback: [toggleValue := toggleValue not];
with: ’toggle boolean:’].

html space.
html span
class: ’boolean’;
with: toggleValue ]

アプリケーションの状態を変えるような処理は、コールバッ

ク内に書くようにします。レンダリング中に、直接そのよ

うなコードを書くべきではありません。

フォーム

フォームも今までの例と同様のやり方で表示できます。図 12.10 の
SeasideFormDemoコンポーネントのコードを示します。

SeasideFormDemo»renderContentOn: html

| radioGroup |
html heading: heading.
html form: [
html span: ’Heading: ’.
html textInput on: #heading of: self.
html select
list: self colors;
on: #color of: self.

radioGroup := html radioGroup.
html text: ’Radio on:’.
radioGroup radioButton
selected: radioOn;
callback: [radioOn := true].

html text: ’off:’.
radioGroup radioButton
selected: radioOn not;
callback: [radioOn := false].

html checkbox on: #checked of: self.
html submitButton
text: ’done’ ]

フォーム要素は複雑なので、ブロックを使ってレンダリングします。状態

の変更に関わるコードはすべてコールバック内で書かれており、レンダリング

部分には書かれないことに注意してください。

on:of:メッセージは Seasideの特徴の一つとも言えるもので、特に説明が必
要でしょう。上の例ではテキスト入力フィールドに、自身のインスタンス変

数 headingを対応させるために使われています。アンカーやボタンもこのメッ
セージに答えられます。最初の引数はインスタンス変数の名前です。2番目の
引数はそのインスタンス変数を保持するオブジェクトを指定します。通常の名

前付けルールに従って、インスタンス変数のアクセサ (heading)とミューテイ
タ (heading:)がクラスに定義されていることが前提となっています。上記のテ
キスト入力フィールドの例では、デフォルト値の設定や、値を更新するための

コールバックを書く手間を省くのに役立っています。on: #heading of: selfを使う
ことで heading変数の値はユーザの入力に応じて自動的に更新されます。

すぐ後にコンボボックスで color変数を指定した別の例があります。その次
のチェックボックスの例では、 checked変数の値が更新されるようになってい
ます。Seasideの機能テストのコードを見ると、さらに様々なフォームの例が見
つかります。Seaside-Tests-Functionalのクラスカテゴリを見てみましょう。また
は、Webブラウザで http://localhost:8080/seaside/tests/alltestsを開き、WAInputTest
を選んで Restart ボタンを押すと、フォームのほとんどの機能を知ることがで
きます。6

Webブラウザの Toggle Halosボタンを押せば、対応するソースコードにすぐ
にアクセスできるという Seasideの機能を、ここで活用してください。

6訳注：Seaside 3.0では、URLが http://localhost:8080/tests/functional/WAInputPostFunctionalTestに、
メニューがWAInputGetFunctionalTest(WAInputPostFunctionalTest)に変わっています。

http://localhost:8080/seaside/tests/alltests
http://localhost:8080/tests/functional/WAInputPostFunctionalTest

12.5 CSS:カスケーディング・スタイル・シート

CSS (カスケーディング・スタイル・シート) 7は、Webアプリケーションで、内
容と見栄えを分離する標準的な方法になっています。Seasideでは、見栄えを
意識してレンダリングのコードが煩雑になるのを、CSSの技術を使って防いで
います。

styleというメソッドをコンポーネントに定義することで、使用するスタイル
シートを決めることができます。メソッドの中身は、コンポーネントが使うス

タイルについて定めた単なる CSSの文字列です。Webページに配置されるコ
ンポーネントのスタイルシートは結合されて一つになりますが、コンポーネン

トは自身のスタイルを自ら定義できます。抽象クラスを定義して、アプリケー

ションで使う共通のスタイルをメソッドとして定義しておくのもよいでしょう。

とはいえデプロイ時には、外部ファイルとしてスタイルシートを用意する

ことも広く行われます。こうするとコンポーネントの機能的な部分と、見栄え

の分離がさらに進み、完全に分離されることになります。( WAFileLibraryクラ
スを使うと、別途サーバーを用意せずに、外部のファイルを利用できます)。

CSSをよく知っているのでしたら、これで説明は終わりです。念のため、以
後は簡単な CSSの紹介を書いておきます。

CSSを使うと、段落やテキストの表示属性を直接Webページに埋め込む代
わりに、要素の分類や表示にまつわるあれこれを独立したスタイルシートに定

義できます。段落に該当する要素が divで、インラインのテキスト要素は span
と呼ばれます。例えば自分で “highlight”といった形で、要素に名前を付け、ハ
イライトされるテキストを定義することができます (例を参照のこと)。スタイ
ルシートの中では、ハイライトがどのように表示されるべきかを記述します。

基本的に CSSは、XHTMLの要素をどのようなフォーマットで表示するか
についてのルールを記述したものです。ルールは二つの部分から成り立ってい

ます。セ レ ク タ と呼ばれる部分では、どの XHTML要素にルールを適用するか
を指定し、宣 言 の部分では、その要素に適用される属性の値を定めます。

図 12.11は、図 12.10で使われているスタイルシートのメソッドを示したも
のです。最初のルールは body要素で使われるフォントを指定しています。さら
に、レベル 2の見出し (h2)、テーブル (table)、セル (td)についての定義が続き
ます。

残りのルールは、特定の “class”や “id”を持つ XHTML要素についてです。
クラス属性の CSSのセレクタは “.”で始まり、id属性のセレクタは “#”で始ま
ります。classと idの違いは、ページ内で id属性の要素は一つしか存在できな
いのに対して (ID の た め )、クラス属性の要素は複数あってもよいということで
す。highlightのようなクラス属性は、ページ内に何度出てきてもかまいません。
id属性はページ内で 唯 一 の要素、例えばメニュー、更新日、作者の名前など
で使われます。また、XHTMLの各要素には、複数のクラス属性が指定できま
す。その場合は指定された順番に沿って、表示の属性が適用されていきます。

7http://www.w3.org/Style/CSS/

http://www.w3.org/Style/CSS/

SeasideDemoWidget»style
↑ ’

body {
font: 10pt Arial, Helvetica, sans-serif, Times New Roman;

}
h2 {
font-size: 12pt;
font-weight: normal;
font-style: italic;

}
table { border-collapse: collapse; }
td {
border: 2px solid #CCCCCC;
padding: 4px;

}
#author {
border: 1px solid black;
padding: 2px;
margin: 2px;

}
.subcomponent {
border: 2px solid lightblue;
padding: 2px;
margin: 2px;

}
.highlight { background-color: yellow; }
.boolean { background-color: lightgrey; }
.field { background-color: lightgrey; }
’

Figure 12.11: SeasideDemoWidget共通スタイルシート

セレクタは組み合わせることができます。 div.subcomponentというセレクタ
は、divで、クラス属性が “subcomponent”の要素のみを選択することになり
ます。

あまり必要になることはありませんが、ネストした要素を指定することも可

能です。例えば “p span”は、段落内 (p)の spanを選択します。div内の span
にマッチすることはありません。

CSSについては多くの本やWebページがあります。CSSでどれだけのこ
とができるかを知るには、CSS Zen Garden8のサイトを見てみるとよいでしょ
う。CSSのスタイルシートを変えるだけで、同じ内容の文書をまったく異なる
見栄えにできるということがわかります。

8http://www.csszengarden.com/

http://www.csszengarden.com/

12.6 制御フローの管理

Seasideでは、複雑なフローを持つWebアプリケーションを、非常に簡単に開
発できるようになっており、二つの基本的な仕組みが提供されています。

1. コンポーネントは、caller call: calleeのように書くことで、他のコンポーネ
ントを呼 び 出 す ことができます。呼び出し側は一時的に表示されなくな

り、呼ばれた側に制御が移ります。answer:によって、制御を元に返すこ
とができます。呼び出し側は通常 selfですが、現在表示されているコン
ポーネントであればどれでもかまいません。

2. 全体のワークフローは タ ス ク の仕組みで定義できます。タスクは、
(WAComponent でなく)WATask を継承する特別なコンポーネントです。
タスクは内部にコンポーネントを持たず、renderContentOn:を実装する代
わりに、goメソッドを定義し、コンポーネントを順番に call:する形で全
体の流れを書いていきます。

call:と answer:

call:と anuswer:は単純なダイアログ形式の UIを実現するために使われます。
call:と answer:を使った簡単な例は、図 12.10のレンダリングのデモに含ま

れています。SeasideEditCallDemoのコンポーネントは、テキストフィールドと
editリンクを表示しています。editのリンクにはコールバックが仕掛けられて
いて、テキストフィールドの値で初期化した上で SeasideEditAnswerDemoのイン
スタンスを呼び出すようになっています。コールバックではさらに、呼び出し

たコンポーネントから受け取った値を、テキストフィールドに反映させるよう

になっています。

(以下のコード例では目立たせるために call:と answer:のメッセージにアン
ダーラインを引いています。)
SeasideEditCallDemo»renderContentOn: html
html span
class: ’field’;
with: self text.

html space.
html anchor
callback: [self text: (self call: (SeasideEditAnswerDemo new text: self text))];
with: ’edit’

次に表示すべきWebページ全体を意識する部分がまったくない、というと
ころがエレガントです。実行すると SeasideEditCallDemoコンポーネントによっ
て表示されている部分が SeasideEditAnswerDemoコンポーネントに置き換わり、
ページ全体の表示が行われます。外側のコンポーネントや隣接するコンポーネ

ントはそのままです。

call:や answer:はレンダリング中に直接記述してはいけませ
ん。コールバックや、タスクの goメソッドの中で書くよ
うにします。

SeasideEditAnswerDemo コンポーネントも非常にシンプルです。テキスト
フィールドの初期値を表示し、送信ボタンのコールバックで、更新されたテキ

ストフィールドの値を返すように書いてあるだけです。

SeasideEditAnswerDemo»renderContentOn: html
html form: [
html textInput
on: #text of: self.

html submitButton
callback: [ self answer: self text ];
text: ’ok’.

]

わずかこれだけでページの遷移が実現できます。

Seaside は適切なコンポーネントに制御を移し、レンダリングを自動的に
行ってくれます。Webブラウザの「戻る」ボタンも問題なく動作します。 (た
だし副作用をロールバックできるようにするにはもう少しコードが必要になり

ます)。

便利なメソッド群

call/answerを使ったやりとりには非常によく出てくるパターンがあります。
そのため Seasideは、SeasideEditAnswerDemoのような些細なコンポーネントを
わざわざ書かずにすむように、便利なメソッドを提供しています。便利メソッ

ドによって生成されるダイアログを図 12.12 に示しました。これらの便利メ
ソッドの動作は SeasideDialogDemo»renderContentOn:で確認できます。

request:メッセージは、編集可能なテキストフィールド付きのコンポーネン
トを開きます。ユーザが入力したテキストが呼び出し側に返るようになってい

ます。ラベルやデフォルト値も指定できます。

SeasideDialogDemo»renderContentOn: html
html anchor
callback: [ self request: ’edit this’ label: ’done’ default: ’some text’ ];
with: ’self request:’.

...

inform:メッセージは、指定した文字列を表示するコンポーネントを開きま
す。“ok”ボタンを押すとコンポーネントが閉じ、制御が呼び出し側に戻りま
す。このコンポーネントは selfを返すだけです。

...
html space.
html anchor
callback: [ self inform: ’yes!’ ];
with: ’self inform:’.

...

confirm: メッセージを使うと “Yes”/“No” の質問ダイアログとなり、ユー
ザはどちらかを選ぶことになります。真偽値を呼び出し側に返すので、それに

よって次の動作を決めることができます。

...
html space.
html anchor
callback: [
(self confirm: ’Are you happy?’)
ifTrue: [ self inform: ’:-)’ ]
ifFalse: [ self inform: ’:-(’ ]

];
with: ’self confirm:’.

WAComponentの convenienceプロトコルには、 chooseFrom:caption:のような
役に立つメソッドがさらに定義してあります。

タスク

WATaskを継承すると、タスクを定義できます。タスク自身はレンダリングを
行わず、単に制御フローに沿って他のコンポーネントを call:していくだけです。

aBoolean
selfaString

Figure 12.12: 標準ダイアログ

goに、制御の流れを書いていきます。

Seaside-Tests-Functionalクラスカテゴリに定義されているWAConvenienceTest
は、タスクの簡単な使用例になっています。動作を確認するには、Webブラウ
ザで http://localhost:8080/seaside/tests/alltestsのURLを開き、WAConvenienceTest
を選択して Restartボタンを押します。9

WAConvenienceTest»go
[ self chooseCheese.
self confirmCheese ] whileFalse.
self informCheese

このタスクでは三つのコンポーネントの呼び出しを順次行っています。ま

ず chooseFrom: caption:の便利メソッドで、WAChoiceDialogのダイアログが開き
ます。ここでユーザは好きなチーズを選択します。

WAConvenienceTest»chooseCheese
cheese := self
chooseFrom: #(’Greyerzer’ ’Tilsiter’ ’Sbrinz’)
caption: ’What’’s your favorite Cheese?’.

cheese isNil ifTrue: [ self chooseCheese ]

次に WAYesOrNoDialogでユーザの選択を確認します (ダイアログは confirm:
で生成されたものです)。

WAConvenienceTest»confirmCheese
↑self confirm: ’Is ’, cheese, ’ your favorite cheese?’

最後に便利メソッド inform:によって、WAFormDialogが呼び出されます。

WAConvenienceTest»informCheese
self inform: ’Your favorite cheese is ’, cheese, ’.’

生成されるダイアログを図 12.13に示します。

トランザクション

10

12.3節の節で、 Seasideではバックトラック対象となるオブジェクトを登
録することで、コンポーネントの状態をWebページごとに記録できるという
話をしました。コンポーネント側で必要なのは、statesメソッドを定義し、バッ
クトラック対象のオブジェクトを配列の形で返すことだけでした。

9訳注：Seaside 3.0では、WAConvenienceTestではなく、WAFlowConvenienceFunctionalTestに変わっ
ています。動作確認では、http://localhost:8080/tests/functionalを開き、WAFlowConvenienceFunctionalTest
を選びます。

10訳注：トランザクションは Seaside 3.0 で未サポートとなったため、この節の内容は意味を持
たなくなりました。

http://localhost:8080/seaside/tests/alltests
http://localhost:8080/tests/functional

no

yes

Figure 12.13: 単純なタスクの例

しかし時にはバックトラックをしたくないときもあります。ユーザが行っ

た操作を、意図せず取り消してしまうことを防 ぎ た い ときがあるのです。これ

は「ショッピングカート問題」などと呼ばれます。ユーザがショッピングカー

トの精算をすませた後は、Webブラウザのボタンで戻って、カートに商品を追
加できてしまってはまずいのです。

Seasideでは、一連の操作をひとまとめにしてタスク内でト ラ ン ザ ク シ ョ ン
として定義することにより、この問題を解決します。トランザクション内では

バックトラックができますが、トランザクションが終了すると戻ることはでき

ません。トランザクションを越えて戻ろうとしても、前のページは 無 効 になり

ます。無効なページの表示時に Seasideが警告を出し、直近の有効なページへ
と自動的にリダイレクトします。

Figure 12.14: 寿司屋デモ

寿 司 屋 デ モ は、Seasideの多くの特徴を示すサンプルアプリケーションで

す。もちろんトランザクションも使われています。このサンプルは Seasideに
含まれており、Webブラウザで http://localhost:8080/seaside/examples/storeのURL
を指定すると起動します。11

寿司屋デモのワークフローは以下のようになっています。

1. サイトを訪れる
2. 寿司をブラウズ、または検索する
3. ショッピングカートに寿司を追加する
4. 注文を開始する
5. カートの内容を確認する
6. 届け先の住所を入力する
7. 住所を確認する
8. 支払い情報を入力する
9. 寿司を待つ

ハローを表示してみると、トップレベルのコンポーネントはWAStoreとなっ
ていることがわかります。このコンポーネントはタイトルバーの表示を行って

いるだけで、WAStoreTaskのインスタンスである taskにレンダリングを委譲し
ています。

WAStore»renderContentOn: html
”... タ イ ト ル 部 分 の レ ン ダ リ ン グ ...”
html div id: ’body’; with: task

ワークフローの順番を定義しているのが WAStoreTask です。一度サブミッ
トした情報を、ユーザが戻って変更できないようにしていることが重要です。

い く つ か 寿 司 を 購 入 し た 後 で 、 カ ー ト の 中 に さ ら に 寿 司 を 追 加 す る た め 、

Web ブ ラ ウ ザ の 「 戻 る 」 ボ タ ン を 押 し て み ま し ょ う 。 “That page has expired.”
と 出 て 追 加 で き ま せ ん 。

Seasideでは、プログラマがワークフローの特定部分をトランザクションと
して定義できるようになっています。トランザクションが終了すると、ユーザ

はそこに戻って操作をやり直すことはできません。トランザクションを定義す

るには、isolate:メッセージをタスクに送ります。引数でトランザクション部分
ブロックで指定します。寿司屋デモでは以下のようにしています。

WAStoreTask»go
| shipping billing creditCard |
cart := WAStoreCart new.
self isolate:

11もしも同梱されていない場合は SqueakSource のページ (http://www.squeaksource.com/
SeasideExamples/) からロードできます。

http://localhost:8080/seaside/examples/store
http://www.squeaksource.com/SeasideExamples/
http://www.squeaksource.com/SeasideExamples/

[[self fillCart.
self confirmContentsOfCart]
whileFalse].

self isolate:
[shipping := self getShippingAddress.
billing := (self useAsBillingAddress: shipping)

ifFalse: [self getBillingAddress]
ifTrue: [shipping].

creditCard := self getPaymentInfo.
self shipTo: shipping billTo: billing payWith: creditCard].

self displayConfirmation.

ここでは二つのトランザクションが定義されていることがわかります。最

初のトランザクションは、カートに寿司を入れて内容を確認する部分です。(
fillCart等のヘルパーメソッドは、内部でサブコンポーネントを生成したり呼び
出したりしています。)一度カートの内容を確認してしまうと戻ることはでき
ません。2番目のトランザクションは、住所と支払い方法を入れる部分です。
支払い方法を確定するまでは、このトランザクション内で戻ることもできます。

しかし、トランザクションを終えてしまうと、戻ろうとしても常に警告が出て

失敗します。

トランザクションをネストすることもできます。WANestedTransactionで簡
単なデモを見ることができます。最初の isolate:のブロック内に、さらに別の
isolate:ブロックが含まれています。

WANestedTransaction»go
self inform: ’Before parent txn’.
self isolate:

[self inform: ’Inside parent txn’.
self isolate: [self inform: ’Inside child txn’].
self inform: ’Outside child txn’].

self inform: ’Outside parent txn’

http:// localhost:8080/seaside/ tests/alltests を 開 い て WATransactionTest の メ ニ
ュ ー を 選 び 、 Restart ボ タ ン を 押 し て み ま し ょ う 。 親 側 と 子 側 の ト ラ ン ザ ク シ ョ
ン 内 で OK ボ タ ン や ブ ラ ウ ザ の 戻る ボ タ ン を 使 い 、 行 っ た り 来 た り し て み ま
し ょ う 。 ト ラ ン ザ ク シ ョ ン 終 了 後 は 、 戻 っ て OK ボ タ ン を 押 し て も 、 エ ラ ー が
出 て 終 了 後 の ペ ー ジ に 必 ず リ ダ イ レ ク ト さ れ ま す 。

http://localhost:8080/seaside/tests/alltests

12.7 サンプルアプリ作成によるチュートリアル

ここでは Seaside のアプリケーションを一から作成してみることにします。
12RPN (逆ポーランド記法) 計算機を Seasideのアプリケーションとして作る
ことにしましょう。単純なスタックマシンを実装して、モデル部分で使うこと

にします。この計算機アプリでは二種類の表示方法を提供します。一つ目はス

タックの先頭の値のみを表示するタイプで、もう一つはスタックの内容をすべ

て表示するタイプです。図 12.15は、二つの表示方法を切り替える様子を表し
ています。

Figure 12.15: スタックマシンを使った RPN計算機

スタックマシンの実装を、テストを書きながら始めていきましょう。

contentsと い う イ ン ス タ ン ス 変 数 を 持 つ MyStackMachineと い う ク ラ ス を 定
義 し て く だ さ い 。 contentsは OrderedCollectionで 初 期 化 し ま す 。

MyStackMachine»initialize
super initialize.
contents := OrderedCollection new.

スタックマシンは値のプッシュやポップをサポートしなければなりません。

またスタックの先頭を見る機能や、積んだ値に対して加減乗除を計算する機能

も必要です。

ス タ ッ ク 操 作 の テ ス ト を 書 い て か ら 実 装 し て み ま し ょ う 。 以 下 は テ ス ト の

サ ン プ ル で す 。

12このチュートリアルには二、三時間かかります。すべてのソースコードを見たい場合は
SqueakSource のプロジェクト http://www.squeaksource.com/PharoByExample から入手できます。
PBE-SeasideRPN のパッケージを選んでロードします。クラス名を少し変えてあるので、すべて
読み込んでも、本書に従ってアプリケーションを作っていくことができます。

http://www.squeaksource.com/PharoByExample

MyStackMachineTest»testDiv
stack
push: 3;
push: 4;
div.

self assert: stack size = 1.
self assert: stack top = (4/3).

加減乗除では、事前にスタックに値が二つ積まれているかをチェックする

ヘルパーメソッドが必要でしょう。条件が整っていない場合はエラーとなるよ

うにします。13これができてしまえば、各メソッドの実装はほぼ一行か二行で
すみます。

デバッグしやすくするため、MyStackMachine»printOn:も実装しておくとよい
でしょう。インスペクタを使って、スタックマシンの状態を観察できるように

なります。(ヒント: contentsに表示を委譲してしまうのが楽です)。

dup (先 頭 の 値 を コ ピ ー し て プ ッ シ ュ す る )、 exch (ス タ ッ ク の 上 位 二 つ の
値 を 入 れ 替 え る )、 rotUp (ス タ ッ ク の 内 容 を 上 下 逆 に す る — 先 頭 の 要 素 が 最 後
に な る )、 を MyStackMachine に 実 装 し 、 ス タ ッ ク マ シ ン を 完 成 さ せ ま し ょ う 。

スタックマシンの実装がすんだら、次はいよいよ Seaside上に RPN計算機
を作っていきます。

クラスを五つ定義することにします:

• MyRPNWidget— RPN計算機のコンポーネントで共通に使う、振る舞いや
CSSを定義するための抽象クラスです。WAComponentのサブクラスで、以
下の四つのクラスのスーパークラスになります。

• MyCalculator—ルートコンポーネントです。クラスメソッドでアプリケー
ションの登録を行います。サブコンポーネントを生成し、レンダリング

を行わせます。またバックトラック用に状態の登録も行います。

• MyKeypad—計算機のキーパッドを表示します。

• MyDisplay—スタックの先頭の値と、詳細なビューに切り替えるためのボ
タンを表示します。

• MyDisplayStack—スタックの詳細なビューと、元のビューに戻るためのボ
タンを表示します。MyDisplayのサブクラスです。

MyCalculatorク ラ ス カ テ ゴ リ に MyRPNWidgetを 定 義 し て み ま し ょ う 。 ア プ
リ ケ ー シ ョ ン で 共 通 に 使 う style メ ソ ッ ド を 書 く よ う に し ま す 。

アプリケーション用に CSS を定義します。もっと凝ってみてもかまいま
せん。

13事前条件を指定するには Object»assert:を使うとよいでしょう。不正な状態でスタックマシン
を使おうとしているときには AssertionFailureが起こるというわけです。

MyRPNWidget»style
↑ ’table.keypad { float: left; }

td.key {
border: 1px solid grey;
background: lightgrey;
padding: 4px;
text-align: center;

}
table.stack { float: left; }
td.stackcell {
border: 2px solid white;
border-left-color: grey;
border-right-color: grey;
border-bottom-color: grey;
padding: 4px;
text-align: right;

}
td.small { font-size: 8pt; }’

ル ー ト コ ン ポ ー ネ ン ト MyCalculatorを 定 義 し て 、 ア プ リ ケ ー シ ョ ン と し て 登
録 し ま し ょ う 。 (ク ラ ス メ ソ ッ ド canBeRoot と initialize を 実 装 し ま す )MyCalculator
»renderContentOn: で は 、 ク ラ ス 名 の 表 示 な ど の 簡 単 な レ ン ダ リ ン グ を 書 い て 、
Web ブ ラ ウ ザ で 表 示 さ れ る か を 確 認 し て み ま し ょ う 。

MyCalculatorでは、MyStackMachine、MyKeypad、MyDisplayのインスタンスを
保持することにします。

MyRPNWidget の サ ブ ク ラ ス と し て MyKeypadと MyDisplayを 定 義 し ま し ょ
う 。 コ ン ポ ー ネ ン ト は ど れ も ス タ ッ ク マ シ ン の イ ン ス タ ン ス を 共 有 す る の で 、

イ ン ス タ ン ス 変 数 stackMachine と 初 期 化 用 の メ ソ ッ ド setMyStackMachine: を
MyRPNWidgetに 用 意 し ま す 。 MyCalculatorに は keypad と display の イ ン ス タ ン ス
変 数 を 定 義 し 、 MyCalculator»initializeで 初 期 化 す る こ と に し ま す 。 (super initialize
を 送 る の を 忘 れ な い よ う に !)

initialize メ ソ ッ ド で は 、 共 有 す る stackMachine を keypad と display に 渡 す
よ う に し ま す 。 レ ン ダ リ ン グ メ ソ ッ ド MyCalculator»renderContentOn: の 実 装 は 、
単 に keypad と display に 委 譲 す る だ け で す 。 サ ブ コ ン ポ ー ネ ン ト を 正 し く 表 示
す る に は 、 MyCalculator»children を 実 装 し て 、 keypad と display か ら な る 配 列 を
返 す よ う に し な け れ ば な り ま せ ん 。 keypad と display に も レ ン ダ リ ン グ メ ソ ッ
ド を 作 り ま す が 、 中 身 は 空 に し て お き ま す 。 ま ず は 計 算 機 が ち ゃ ん と サ ブ コ ン

ポ ー ネ ン ト を 表 示 し て い る か を 確 認 し て み ま し ょ う 。

次は displayのレンダリングメソッドの中身を書き、スタックの先頭を表
示できるようにします。

レ ン ダ リ ン グ で は 、 “keypad” と い う CSS ク ラ ス 名 の テ ー ブ ル を 書 き 、 行
に は “stackcell” と い う ク ラ ス 名 の セ ル を 一 つ だ け 入 れ る こ と に し ま す 。 ま た 、

ス タ ッ ク が 空 の 場 合 、 先 頭 に 0 を push す る よ う に し て お き ま す 。 (MyKeypad
»ensureMyStackMachineNotEmptyと い う メ ソ ッ ド を 定 義 し て 使 う こ と に し ま す )。
さ ら に “keypad” の ク ラ ス 名 で 空 の テ ー ブ ル を 表 示 さ せ ま す 。 こ れ で 、 計 算 機
は 0 を 含 ん だ 一 つ の セ ル を 表 示 す る よ う に な り ま す 。 ハ ロ を 表 示 さ せ て み る と
図 12.16 の よ う に な る は ず で す 。

Figure 12.16: スタックの先頭を表示

今度はスタックとやりとりするための UIを書いていくことにしましょう。

ま ず 、 UI を 簡 単 に 書 け る よ う に す る た め に 、 以 下 の ヘ ル パ ー メ ソ ッ ド を
定 義 す る こ と に し ま す 。

MyKeypad»renderStackButton: text callback: aBlock colSpan: anInteger on: html
html tableData
class: ’key’;
colSpan: anInteger;
with:

[html anchor
callback: aBlock;
with: [html html: text]]

MyKeypad»renderStackButton: text callback: aBlock on: html
self
renderStackButton: text
callback: aBlock
colSpan: 1
on: html

これらのメソッドは、キーパッドのボタンを表示しコールバックを定義す

るために使います。ボタンの中には複数のカラムを占めるものもありますが、

基本はカラムを一つ使うことにします。

以 下 の よ う に 二 つ の ヘ ル パ ー メ ソ ッ ド を 使 っ て キ ー パ ッ ド の レ ン ダ リ ン

グ を 書 い て み ま し ょ う 。 (ヒ ン ト : 数 字 キ ー と “Enter” キ ー か ら 初 め て 、 計 算 用
の キ ー は 後 か ら 付 け る と 楽 で す )。

MyKeypad»renderContentOn: html
self ensureStackMachineNotEmpty.
html table
class: ’keypad’;
with: [
html tableRow: [
self renderStackButton: ’+’ callback: [self stackOp: #add] on: html.
self renderStackButton: ’&ndash;’ callback: [self stackOp: #min] on: html.
self renderStackButton: ’&times;’ callback: [self stackOp: #mul] on: html.
self renderStackButton: ’&divide;’ callback: [self stackOp: #div] on: html.
self renderStackButton: ’&plusmn;’ callback: [self stackOp: #neg] on: html ].
html tableRow: [
self renderStackButton: ’1’ callback: [self type: ’1’] on: html.
self renderStackButton: ’2’ callback: [self type: ’2’] on: html.
self renderStackButton: ’3’ callback: [self type: ’3’] on: html.
self renderStackButton: ’Drop’ callback: [self stackOp: #pop]
colSpan: 2 on: html ].

”な ど な ど ”
html tableRow: [
self renderStackButton: ’0’ callback: [self type: ’0’] colSpan: 2 on: html.
self renderStackButton: ’C’ callback: [self stackClearTop] on: html.
self renderStackButton: ’Enter’
callback: [self stackOp: #dup. self setClearMode]
colSpan: 2 on: html ]]

キーパッドがブラウザできちんと表示されるか確認しましょう。当然です

が、クリックしてみてもまだ計算機としては動作しません。

MyKeypad»type: を 実 装 し て 、 タ イ プ し た 数 字 が ス タ ッ ク の 先 頭 に 反 映 さ れ
る よ う に し ま し ょ う 。 入 力 さ れ た 値 は 文 字 列 で 得 ら れ る の で 、 ス タ ッ ク の 値 を

文 字 列 に し て 連 結 し 、 整 数 に し て か ら ス タ ッ ク に 戻 す 必 要 が あ り ま す 。 具 体 的

に は 以 下 の よ う な コ ー ド に な り ま す 。

MyKeypad»type: aString
stackMachine push: (stackMachine pop asString, aString) asNumber.

数字キーを打つと、表示が更新されるようになりました。(MyStackMachine»pop
でポップした値を返すようにきちんと実装していないと、動作しません!)

次 に MyKeypad»stackOp:を 書 く こ と に し ま し ょ う 。 以 下 は ち ょ っ と し た コ
ツ を 使 っ て い ま す 。

MyKeypad»stackOp: op
[ stackMachine perform: op ] on: AssertionFailure do: [ ].

ポイントはすべての操作が成功するわけではないと言うことです。例えば

二つの数値がスタックに積まれていないと、加算できません。当面は単にエ

ラーを無視するように書いておきます。もっと機能を充実させたい場合は、エ

ラーハンドリングのブロックにユーザに注意を促すように書いていくことがで

きます。

逆 ポ ー ラ ン ド 記 法 の 計 算 機 は こ れ で 一 応 動 く も の に な り ま し た 。 数 字 キ ー

を 押 し た 後 に Enter キ ー を 押 し て 、 い く つ か の 数 値 を プ ッ シ ュ し て み ま し ょ う 。
+ を 押 し て 二 つ の 数 値 が 足 さ れ る こ と を 確 認 し て く だ さ い 。

数値の入力が予想した動きと若干違うのではないでしょうか。新 た な 数値

を入力しているのか、既存の数値を更新しているのかをモードによって区別で

きないとまずいようです。

MyKeypad»type:を 修 正 し 、 入 力 モ ー ド に 応 じ て 動 作 を 変 え る よ う に し て み
ま し ょ う 。 modeと い う イ ン ス タ ン ス 変 数 を 定 義 し て 、 三 つ の 値 の い ず れ か を 取
る よ う に し ま す 。 数 値 を 入 力 し て い る と き は #typing で す 。 計 算 ボ タ ン を 押 し た
後 は 、 #push モ ー ド と な り 、 入 力 し て い た 値 を 強 制 的 に プ ッ シ ュ す る よ う に し
ま す 。 Enter を 押 し た 後 は 、 #clear と な り 、 次 の 入 力 の 前 に ス タ ッ ク の 先 頭 を ク
リ ア し ま す 。 type:メ ソ ッ ド を 書 き 直 し た の が 以 下 の コ ー ド で す 。

MyKeypad»type: aString
self inPushMode ifTrue: [
stackMachine push: stackMachine top.
self stackClearTop ].

self inClearMode ifTrue: [ self stackClearTop ].
stackMachine push: (stackMachine pop asString, aString) asNumber.

”
前よりは入力が便利になりましたが、スタックの先頭の値を見ることがで

きない点がまだ不満です。

MyDisplayの サ ブ ク ラ ス と し て MyDisplayStack を 定 義 し ま し ょ う 。 MyDisplay
の レ ン ダ リ ン グ メ ソ ッ ド に ボ タ ン の 描 画 を 追 加 し て 、 コ ー ル バ ッ ク で

MyDisplayStackの イ ン ス タ ン ス を call:す る よ う に し ま す 。 HTML の ア ン カ ー を
使 っ て 、 以 下 の よ う に 書 け る で し ょ う 。

html anchor
callback: [ self call: (MyDisplayStack new setMyStackMachine: stackMachine)];
with: ’open’

コールバックが呼ばれると、MyDisplayコンポーネントの表示は、一時的に
MyDisplayStackコンポーネントへと切り替わり、スタックのすべての情報を表示
するようになります。self answerによって、このコンポーネントが終了して閉
じると、制御は元の MyDisplayへと戻ります。

MyDisplayStackの レ ン ダ リ ン グ メ ソ ッ ド を 定 義 し て 、 ス タ ッ ク の す べ て の
内 容 を 表 示 す る よ う に し ま し ょ う 。 (ス タ ッ ク マ シ ン の contentsの 値 を 返 す ア ク

セ サ を 用 意 す る か 、 MyStackMachine»do:を 定 義 し て 、 ス タ ッ ク の 値 を イ テ レ ー
ト で き る よ う に す る と よ い で し ょ う )。 “close” ボ タ ン も 付 け る こ と に し ま す 。
コ ー ル バ ッ ク で は 単 に self answerを 実 行 す る だ け で す 。

html anchor
callback: [ self answer];
with: ’close’

これで openと closeで表示を切り替えられるようになりました。
とはいえ、まだ考慮していなかったこともあります。いくつか操作をして

みた後で、Webブラウザの「戻る」ボタンで戻って、操作をしてみるとどうな
るでしょうか。(例えばopen でスタックの全体を表示させ、1、 Enterと 2回入
力し、+を押します。この時点でスタックの表示は “2”と “1”となっているは
ずです。ここで「戻る」ボタンを押してみましょう。スタックは “1”を三つ表
示した状態になり、+を押すと、“3”が表示されます。つまりバックトラック
が動作していないのです。

MyCalculator»statesメ ソ ッ ド を 実 装 し 、 ス タ ッ ク マ シ ン の contentsを 返 す よ
う に し ま し ょ う 。 今 度 は ち ゃ ん と バ ッ ク ト ラ ッ ク が 効 い て い ま す 。

さて、これで動くアプリケーションができあがりました。乾杯!

12.8 AJAXの利用

AJAX (Asynchronous JavaScript and XML)はクライアント側で JavaScriptを
活用することで、、Webアプリケーションをさらにインタラクティブにする技
術です。

よく知られた JavaScript のライブラリとしては Prototype (http://www.
prototypejs.org) と script.aculo.us (http://script.aculo.us) があります。Prototype
は、JavaScriptを簡潔に書けるようにするためのフレームワークです。script.ac-
ulo.usは、Prototype上に、アニメーションや、ドラッグ&ドロップなどをサ
ポートする機能を追加したものです。Seasideでは “Scriptaculous”というパッ
ケージを通じて、これらのフレームワークを利用することができます。

Seaside のインストールイメージには、Scriptaculous パッケージが最初
から入っています。最新のバージョンは http://www.squeaksource.com/Seasideか
ら入手可能です。http://scriptaculous.seasidehosting.st でデモを見ることができ
ます。イメージに Scriptaculousが入っていれば、http://localhost:8080/javascript/
scriptaculousでも、デモを試せます。

Scriptaculousパッケージでは Seasideでおなじみのアプローチが取られて
おり、Smalltalkのオブジェクトを使ってアプリケーションをモデリングすれ
ば、JavaScriptのコードが自動生成されます。
簡単な例として、クライアント側で JavaScriptを使うことで、RPN計算機

http://www.prototypejs.org
http://www.prototypejs.org
http://script.aculo.us
http://www.squeaksource.com/Seaside
http://scriptaculous.seasidehosting.st
http://localhost:8080/javascript/scriptaculous
http://localhost:8080/javascript/scriptaculous

をより自然な UIに変えていきましょう。今までは、キーを打つたびにページ
全体がリフレッシュされていました。ページはそのままで、クライアントが操

作した部分のみが更新されるようにしたいと思います。

JavaScript か ら 表 示 部 分 を 特 定 す る に は 、 id を 振 る 必 要 が あ り ま す 。 計 算
機 の レ ン ダ リ ン グ メ ソ ッ ド を 以 下 の よ う に 変 え ま し ょ う 。

14

MyCalculator»renderContentOn: html
html div id: ’keypad’; with: keypad.
html div id: ’display’; with: display.

キ ー パ ッ ド を 押 し た と き に 再 描 画 さ れ る よ う に す る に は 、 キ ー パ ッ

ド 側 で 表 示 コ ン ポ ー ネ ン ト を 知 る 必 要 が あ り ま す 。 イ ン ス タ ン ス 変 数

displayを MyKeypadに 追 加 し ま す 。 初 期 化 用 の メ ソ ッ ド と し て MyKeypad»
setDisplay:を 用 意 し 、 MyCalculator>>initializeで 初 期 化 し ま し ょ う 。 こ れ で ボ タ ン
に JavaScript の コ ー ド を 割 り 当 て る こ と が で き る よ う に な り ま す 。 MyKeypad»
renderStackButton:callback:colSpan:on: を 以 下 の よ う に 変 え ま し ょ う 。

MyKeypad»renderStackButton: text callback: aBlock colSpan: anInteger on: html
html tableData
class: ’key’;
colSpan: anInteger;
with: [
html anchor
callback: aBlock;
onClick: ”JavaScriptの イ ベ ン ト 処 理 ”
(html updater
id: ’display’;
callback: [ :r |
aBlock value.
r render: display ];

return: false);
with: [ html html: text ] ]

onClick: で JavaScript のイベントハンドラを指定しています。html updater
によって SUUpdaterのインスタンスが返ってきます。これは JavaScript の

Ajax.Updater オブジェクト (http://www.prototypejs.org/api/ajax/updater) に該当す
る Smalltalkのオブジェクトです。このオブジェクトはAJAXリクエストの送
信を行い、返ってきたテキストを使い特定部分の更新を行います。id:によって、
updaterに XHTMLのどの DOM要素を更新すべきかを指定しています。こ
の例の場合は’display’の idが振られた div要素になります。callback:で、ユー
ザがボタンを押したときのコールバックをブロックの形で指定します。ブロッ

クの引数 rは、新たに生成されるレンダラで、これを使って displayのコンポー
14チュートリアルの例を自分で実装していない場合は、http://www.squeaksource.com/PharoByExample

から、PBE-SeasideRPN パッケージをロードして、コードを修正していけばよいでしょう。クラ
スの接頭辞は本書と違い、My*ではなく RPN*になっています。

http://www.prototypejs.org/api/ajax/updater
http://www.squeaksource.com/PharoByExample

ネントの表示を行います。(コード上はメソッドの引数となっている htmlレン
ダラにもアクセスできますが、コールバックのブロックが評価されるときには

無効になっているので注意してください。)コールバック内では displayのコ
ンポーネントをレンダリングする前に、aBlockを評価しています。適切な処理
を外から渡して行わせるためです。

return: falseでは、リンクに設定されていた元々のコールバックを呼ばない
ように JavaScript側に指定しています。元のコールバックが呼ばれると、ペー
ジ全体がリフレッシュされてしまうからです。リンクの callback:部分を削除し
て、コールバックを設定しないようにもできますが、JavaScriptが無効になっ
ていても動作するように、あえてこのようにしています。

計 算 機 の 数 字 キ ー を 押 し て み て 、 ま だ ペ ー ジ 全 体 の リ フ レ ッ シ ュ が 行 わ れ

て し ま う こ と を 確 認 し て み て く だ さ い 。 (Web ペ ー ジ の URL が ボ タ ン を 押 す た
び に 変 わ り ま す )。

クライアント側の JavaScriptの振る舞いを実装しましたが、まだ無効なま
まです。では、JavaScriptのイベント処理を有効にしましょう。

計 算 機 の 下 の ツ ー ル バ ー に 表 示 さ れ て い る Configure リ ン ク を ク リ ッ ク
し て く だ さ い 。 “Add Library:” の 部 分 で SULibraryを 選 び 、 Add ボ タ ン を 押 し て
Close で 閉 じ れ ば 完 了 で す 。

ライブラリを手で追加する代わりに、プログラム的に行うこともできます。

Seasideにアプリケーションを登録する際に以下のようにして指定します。

MyCalculator class»initialize
(self registerAsApplication: ’rpn’)
addLibrary: SULibrary

も う 一 度 RPN 計 算 機 を 試 し て み ま し ょ う 。 前 よ り も 自 然 な 形 で 操 作 で き
ま す 。 キ ー ボ タ ン を 押 し て も URL が 変 わ ら な く な り ま し た 。

動作は確認できましたが、仕 組 み が ど う な っ て い る の か 気になるかもしれ

ません。図 12.17で、AJAXなしの場合と、有効にした場合とで、RPN計算機
のメッセージの流れを示しました。AJAXの場合は、基本的にレンダリング処
理を飛ばして、displayコンポーネントの み の更新を行います。JavaScriptは
リクエストの送信と、対応する DOM要素の更新の両方を受け持っています。
生成される JavaScriptのソースを見てみましょう。

new Ajax.Updater(
’display’,
’http://localhost/seaside/RPN+Calculator’,
{’evalScripts’: true,
’parameters’: [’_s=zcdqfonqwbeYzkza’, ’_k=jMORHtqr’,’9’].join(’&’)});

return false

Figure 12.17: Seasideでの AJAXの処理 (簡略化したもの)

より高度なサンプルについては http://localhost:8080/javascript/scriptaculous を
見るとよいでしょう。

Tips サーバー側のデバッグは通常の Smalltalkのデバッガを使うことができ
ます。クライアント側のデバッグは FireFox (http://www.mozilla.com)を入れて、
FireBug (http://www.getfirebug.com/)プラグインを有効にしておくと便利です。

12.9 この章のまとめ

• Seasideを始めるには “Seaside One-Click Experience”イメージを http:
//seaside.stからダウンロードするのが最も簡単です。

• WAKom startOn: 8080でサーバーが開始し、 WAKom stopで終了します。

http://localhost:8080/javascript/scriptaculous
http://www.mozilla.com
http://www.getfirebug.com/
http://seaside.st
http://seaside.st

• 管理用の IDとパスワードはWADispatcherEditor initializeで初期化できます。

• Toggle Halos をクリックすると、コンポーネントのソースコードやオブ
ジェクトの状態、CSSや XHTMLを見ることができます。

• WAGlobalConfiguration setDeploymentMode で開発用のツールバーを隠せま
す。

• Seasideのアプリケーションはコンポーネントからできています。コン
ポーネントとは WAComponentを継承したクラスのインスタンスのこと
です。

• Seaside の Web 管理ツールからアプリケーションとして登録できるコ
ンポーネントを、ルートコンポーネントと呼びます。クラスメソッドの

canBeRoot を実装して true を返すようにします。initialize のクラスメ
ソッドで、 self registerAsApplication: を自身に送ることで、登録すること
もできます。引数は ア プ リ ケ ー シ ョ ン の パ ス になります。description を
オーバーライドするとWeb上のコンフィギュレーションエディタにアプ
リケーションの説明を表示させることも可能です。

• 状態をバックトラックするには、コンポーネントで statesメソッドを定
義して、復元したいオブジェクトの配列を返すように実装する必要があ

ります。ユーザが「戻る」ボタンを押しても状態が正常に保たれます。

• コンポーネントは renderContentOn:を実装することで、自身をWebペー
ジ内に表示させることができます。引数として XHTMLレンダリング用
のキ ャ ン バ ス がわたってきます。(通常は htmlという名前になっています)

• コンポーネントは、自身が含むサブコンポーネントを self render: subcom-
ponentで表示させることができます。

• XHTMLは ブ ラ シ にメッセージを送ることで生成されます。キャンバス
に対して、 paragraphや divなどのメッセージを送ると、XHTML要素に
対応したブラシが返ってきます。

• ブラシに with:を含んだメッセージを連続して送るときは、必ず with:が最
後になるようにします。with:によって XHTML生成が終わり、その結果
がレンダリングされるからです。

• 状態を操作するアクションはコールバックの中で書くようにします。レ
ンダリングメソッド内でコンポーネントの状態を変えてはいけません。

• ブラシに on: イ ン ス タ ン ス 変 数 of: オ ブ ジ ェ ク ト を送ることで、フォーム
の要素やリンクに対し、インスタンス変数の値を簡単に結び付けること

ができます。

• スタイルシートの文字列を返す styleメソッドを実装すると、コンポーネ
ントが使う CSSをクラス階層中に定義できます。(通常デプロイ時には、
外部の CSSを固定の URLで参照させるようにします。)

• 制御フローは、x call: yのように書くことで表せます。この場合、xのコン
ポーネントは yのコンポーネントへと置き換えられます。 yは answer:で
制御を xに戻すことができます。call:のレシーバはたいていは selfですが、
それ以外のコンポーネントであってもかまいません。

• 制御フローは タ ス ク として表すこともできます。タスクはWATaskのサブ
クラスとして定義します。 goメソッドの中に、ワークフローを連続した
call:のメッセージ送信として書いていきます。

• ユーザと簡単なやりとりを行うため、WAComponentsの便利メソッドとし
て、request:、 inform:、 confirm:、chooseFrom:caption:が用意されています。

• ユーザが「戻る」ボタンを押して以前の状態にアプリケーションを戻し
てしまうのを防ぐには、isolate:を使います。ブロックでワークフローを分
割することで、ブロックを越えて過去の状態に戻ることができなくなり

ます。





Part III

Advanced Pharo





Chapter 13

クラスとメタクラス

第 5章で見てきたように、Smalltalkではすべてがオブジェクトであり、いずれ
も何らかのクラスのインスタンスです。クラスも例外ではありません。つまり、

クラスはオブジェクトであり、クラスオブジェクトは他のクラスのインスタン

スとなっています。このオブジェクトモデルは、簡潔で、シンプルかつエレガ

ントに統一されており、オブジェクト指向プログラミングの本質を捉えていま

す。しかし、不慣れなうちはこの一貫性のために混乱してしまうかもしれませ

ん。この章では、それが複雑ではなく、“魔法”でも特別なものでもない、ただ
シンプルな規則が一貫して適用されているということを示します。これらの規

則を追えば、システムがなぜこのように成り立っているのかがわかるでしょう。

13.1 クラスとメタクラスのルール

Smalltalkのオブジェクトモデルは、一貫した数少ないコンセプトの上に成り
立っています。Smalltalkの設計者は、オッカムの剃刀と呼ばれる規則を使っ
て、モデルが必要以上に複雑になる要因を削ぎ落としました。

第 5章で見た、オブジェクトモデルのルールを思い出しましょう。

Rule 1. すべてのものはオブジェクトである。

Rule 2. すべてのオブジェクトは、あるクラスのインスタンスである。

Rule 3. すべてのクラスは、スーパークラスを持つ。

Rule 4. すべての出来事は、メッセージを送ることで生じる。

Rule 5. メソッド探索は継承チェーンに沿って行われる。

この章の冒頭で述べたように、Rule 1の規則よりク ラ ス も オ ブ ジ ェ ク ト で あ
る が成り立ちます。そして、Rule 2より、クラスも何らかのクラスのインスタ

ンスであることがわかります。このクラスのクラスをメ タ ク ラ ス といいます。

あるクラスのメタクラスは、そのクラスを定義した際に自動的に作られます。

普段はメタクラスのことを気にする必要はないでしょう。しかし、ブラウザで

“クラス側”を選んだときには、実は異なるクラスを見ているということを思い
返してください。クラスとそのメタクラスは、前者が後者のインスタンスであ

るということを除けば、それぞれ別々のクラスであるということです。

クラスとメタクラスを正確に説明するには、第 5章のルールに付け足す必
要があります。

Rule 6. すべてのクラスはメタクラスのインスタンスである。

Rule 7. メタクラスの階層はクラスの階層と並列に存在する。

Rule 8. すべてのメタクラスは Classと Behaviorを継承している。

Rule 9. すべてのメタクラスは Metaclassのインスタンスである。

Rule 10. Metaclassのメタクラスは、Metaclassのインスタンスである。

これら 10個のルールが、Smalltalkのオブジェクトモデルのすべてです。
以下では、まず第 5章で紹介した五つのルールを小さな例を使いながら説

明します。そして、同じ例を用いて、新しいルールを深く見ていきます。

13.2 Smalltalkオブジェクトモデルの復習

すべてがオブジェクトなので、「青色オブジェクト」も Smalltalkではオブジェ
クトです。

Color blue −→ Color blue

すべてのオブジェクトはあるクラスのインスタンスです。青色オブジェクトの

クラスは Colorです。

Color blue class −→ Color

興味深いことに、alpha（透明度）を指定すると、異なるクラス（TranslucentColor
）のインスタンスが得られます。

(Color blue alpha: 0.4) class −→ TranslucentColor

Morphを作り、その色としてこの半透明オブジェクトを指定することができ
ます。

EllipseMorph new color: (Color blue alpha: 0.4); openInWorld

これを実行すると、図 13.1のようになります。

Figure 13.1: 半透明の楕円モーフ。

Rule 3より、どのクラスもスーパークラスを持ちます。TranslucentColorの
スーパークラスは Colorです。そして、Colorのスーパークラスは Objectです。

TranslucentColor superclass −→ Color
Color superclass −→ Object

Rule 4「すべての出来事は、メッセージを送ることで生じる」より、以下の
ことが導きだせます。blueは Colorへのメッセージ、classと alpha:は青色オブ
ジェクトへのメッセージ、openInWorldは楕円Morphへのメッセージ、そして、
superclassは TranslucentColorと Colorへのメッセージです。すべてがオブジェク
トなので、どのケースでもメッセージのレシーバはオブジェクトです。ただし、

いくつかのケースではレシーバがクラスとなっています。

Rule 5「メソッド探索は継承チェーンに沿って行われる」より、Color blue
alpha: 0.4で得られたオブジェクトに classメッセージを送ると、図 13.2で示す
ように、メッセージは対応するメソッドが見つかる Objectで取り扱われます。

この図は is-aの継承関係を示したものです。図のように (Color blue alpha: 0.4
で得られた)translucentBlueオブジェクトは、TranslucentColorのインスタンス
で あ る わけですが、同時に Colorオブジェクトで あ る とも、Objectオブジェクト
で あ る とも言えるのは、これらのクラスで定義されたすべてのメッセージに応

答することができるからです。さらに補足すると、isKindOf:メッセージをオブ
ジェクトに送ることで、レシーバが引数で与えられたクラスとで あ る (is a)の
関係にあるかどうかを調べることができます。

translucentBlue := Color blue alpha: 0.4.
translucentBlue isKindOf: TranslucentColor −→ true
translucentBlue isKindOf: Color −→ true
translucentBlue isKindOf: Object −→ true

Figure 13.2: 半透明色オブジェクトへのメッセージ送信。

13.3 すべてのクラスはメタクラスのインスタンスで

ある

13.1節で述べたように、そのインスタンスがクラスであるようなクラスはメタ
クラスと呼ばれます。

メタクラスは暗黙的な存在である メタクラスはクラスを定義すると自動的に

作成されます。これはプログラマが気にする必要はないので、暗 黙 的 であると

言えます。作成されるクラスごとに、そのクラスを唯一のインスタンスとして

持つ暗黙的なメタクラスが作成されます。

通常のクラスはグローバル変数によって参照できますが、メタクラスは匿

名の存在です。しかし、インスタンスであるクラスを通して、いつでもメタク

ラスを参照できます。例えば、Colorのクラスは Color classのインスタンスであ
り、Objectのクラスは、Object classのインスタンスです。

Color class −→ Color class
Object class −→ Object class

図 13.3は、クラスがその (匿名の)メタクラスのインスタンスであることを示
しています。

クラスもまたオブジェクトであることから、メッセージを送って情報を得

ることも簡単にできます。以下を見てみましょう。

Color subclasses −→ {TranslucentColor}
TranslucentColor subclasses −→ #()
TranslucentColor allSuperclasses −→ an OrderedCollection(Color Object

ProtoObject)

Figure 13.3: TranslucentColorのメタクラスと、そのスーパークラス。

TranslucentColor instVarNames −→ #(’alpha’)
TranslucentColor allInstVarNames −→ #(’rgb’ ’cachedDepth’ ’cachedBitPattern’ ’

alpha’)
TranslucentColor selectors −→ an IdentitySet(#pixelWord32

#asNontranslucentColor #privateAlpha #pixelValueForDepth: #isOpaque
#isTranslucentColor #storeOn: #pixelWordForDepth: #scaledPixelValue32 #alpha
#bitPatternForDepth: #hash #isTransparent #isTranslucent
#balancedPatternForDepth: #setRgb:alpha: #alpha: #storeArrayValuesOn:)

13.4 メタクラスの階層はクラスの階層と並列に存在

する

Rule 7は、メタクラスのスーパークラスに任意のクラスがなれるわけではない
ことを示しています。メタクラスのスーパークラスとして許されるのは、メタ

クラスの唯一のインスタンスであるクラスのスーパークラスのメタクラスです。

TranslucentColor class superclass −→ Color class
TranslucentColor superclass class −→ Color class

これが、「メタクラス階層はクラス階層と並列に存在する」ことの意味です。

図 13.4は、TranslucentColorの階層において、どのようになっているかを示して
います。

TranslucentColor class −→ TranslucentColor class
TranslucentColor class superclass −→ Color class
TranslucentColor class superclass superclass −→ Object class

Figure 13.4: メタクラス階層と並列に存在するクラス階層。

クラスとオブジェクトの一貫性 ここで一息入れて考え直してみると、オブ

ジェクトであろうとクラスであろうとメッセージを送るのに違いがないことは

興味深いことです。どちらのケースも、メソッド探索はメッセージのレシーバ

クラスから始まり、継承チェーンをたどっていきます。

従って、クラスに送ったメッセージの探索は、メタクラスの階層チェーンをた

どらなければなりません。例えば、blueメソッドは Colorのクラス側に定義され
ています。TranslucentColorに blueメッセージを送信した場合は、他のメッセージ
とまったく同様にメソッドを探索します。すなわち、探索はTranslucentColor class
でまず行われ、対応するメソッドが見つかるまでメタクラス階層を Color class
までたどっていきます。(図 13.5を参照)

TranslucentColor blue −→ Color blue

TranslucentColorのインスタンスではなく、普通の Color blueオブジェクトが得ら
れることに注意してください—不思議なところはどこにもありません！

Figure 13.5: クラスに対するメッセージ探索は、通常のオブジェクトの場合と
同じ。

すなわち、Smalltalkでは統一された方法でメソッド探索が行われている
と言えます。クラスもまた単なるオブジェクトであり、他のオブジェクトと同

じように振る舞います。クラスは、新しいインスタンスを作る能力を持ちます

が、これは、クラスがたまたま newというメッセージに反応することができて、
newに対するメソッドが新しいインスタンスの作り方を知っているからにすぎ
ません。通常はクラス以外のオブジェクトはこのメッセージに応答できません

が、そうする理由があるなら、newメソッドをメタクラス以外に追加すること
を止めるものは何もありません。

クラスもオブジェクトなので、インスペクトすることもできます。

Color blueと Colorを イ ン ス ペ ク ト し て み ま し ょ う 。

前者は Colorのインスタンスを、後者は Color自身をインスペクトしていることに
注意してください。どちらの場合もインスペクタのタイトルバーにはオブジェ

クトのク ラ ス が表示されるので、少し混乱するかもしれません。

図 13.6 にあるように、Colorのインスペクタによって、スーパーク
ラス (superclass)、インスタンス変数 (instanceVariables)、メソッド辞書
(methodDict)などを見ることできます。

Figure 13.6: クラスもオブジェクトである。

Figure 13.7: メタクラスは、Classと Behaviorを継承している。

13.5 すべてのメタクラスは Classと Behaviorを継承
している

どのメタクラスもクラスである (is-a)ため、Classを継承しています。Classの
スーパークラスはClassDescriptionであり、さらにそのスーパークラスは Behavior
です。Smalltalkではすべてがオブジェクトである (is-a)ため、結局これらのク
ラスもObjectを継承していることになります。図 13.7に、完全な図を示します。

newはどこで定義されているのか？ newがどこで定義されており、どのよう
に探索されるのかについて考えてみることが、メタクラスが Classと Behaviorを
継承しているということの重要性を理解する手助けになります。newメッセー
ジがクラスに送られるとき、図 13.8のように、メタクラスの継承チェーンをた
どって、スーパークラスである Class、ClassDescription、そして Behaviorの順に
メソッド探索が行われます。

「 ど こ に newが 定 義 さ れ て い る の か 」 という問いは決定的に重要です。new
は、まず Behaviorで定義され、必要に応じてそのサブクラスで再定義されます。
これには、プログラマが定義したクラスのメタクラスも含まれます。newメッ
セージをクラスに送った場合、まずは通常通りそのクラスのメタクラスから探

索が開始され、そこで見つからなければ、newを再定義しているクラスまたは
最終的には Behaviorまで、順にスーパークラスをたどって探索されます。

TranslucentColor newを評価した結果は TranslucentColor のインスタンスであ
り、たとえメソッドが Behaviorで見つかったとしても Behaviorのインスタンス

Figure 13.8: newは普通のメッセージで、メタクラスチェーンをたどって探索
される。

になるわけではな い ことに注意してください。newは常にそのメッセージを受
け取ったクラスである selfのインスタンスを返します。これは newが他のクラ
スで実装されている場合でも同じです。

TranslucentColor new class −→ TranslucentColor ”Behaviorの イ ン ス タ ン ス で
は な い ！ ”

よくある間違いは、レシーバとなっているクラスのスーパークラスで newを
探すことです。同じことは、指定された大きさのオブジェクトを生成するため

の標準的なメッセージである、new:についても言えます。例えば、Array new: 4
は、四つの要素を持つ配列を生成します。Arrayやそのスーパークラスで、この
メソッドの定義を見つけることはできないでしょう。代わりに、実際のメソッ

ド探索が行われる Array classとそのスーパークラスの階層に目を向ける必要が
あります。

Behavior、ClassDescriptionおよび Classの責務について Behaviorは、インス
タンスを持つオブジェクトのための必要最小限の状態を保持しています。すな

わち、スーパークラスへのリンク、メソッド辞書、およびクラスのフォーマッ

トです。クラスのフォーマットは、ポインタか否かの区別、コンパクトクラス

か否かの区別、およびインスタンスの基本的な大きさをコード化したものです。

Behaviorは Objectを継承しているので、それ自身とすべてのサブクラスは、普通
のオブジェクトとして振る舞うことができます。

Behaviorは、コンパイラへの基本的なインターフェースも提供します。メ
ソッド辞書の生成、メソッドのコンパイル、インスタンスの生成（す な わ ち new,

basicNew, new:, basicNew:）、クラス階層の操作（す な わ ち superclass:, addSubclass:
）、メソッドへのアクセス（す な わ ち selectors, allSelectors, compiledMethodAt:）、
インスタンス、およびインスタンス変数へのアクセス（す な わ ち allInstances
, instVarNames …）、クラス階層へのアクセス（す な わ ち superclass, subclasses
）、および問い合わせ（す な わ ち hasMethods, includesSelector, canUnderstand:,
inheritsFrom:, isVariable）のメソッドを提供します。

ClassDescription は抽象クラスであり、Class と Metaclass という直接のサブ
クラスが必要とする機能を提供しています。ClassDescriptionは Behaviorが提供
する基礎に、以下のような多くの機能を加えます。それらは、インスタンス変

数の名前の管理、プロトコルによるメソッドの分類、チェンジセットの管理と

変更の記録、そして変更のファイルアウトに必要なほとんどの機能などです。

Classは、すべてのクラスに共通する振る舞いを提供し、クラス名、メソッ
ドのコンパイル、メソッドの保存、インスタンス変数を提供します。また、ク

ラス変数名、共有プール変数（addClassVarName:, addSharedPool:, initialize）の実
際のインターフェースも提供します。メタクラスは、唯一のインスタンス (す
な わ ち ,メタクラスでないクラス)に対するクラスであり、クラスとしてインス
タンスにサービスを提供するため、すべてのメタクラスは最終的に Classを継
承しています。

13.6 すべてのメタクラスはMetaclassのインスタンス
である

メタクラスもまたオブジェクトです。図 13.9 が示すように、メタクラスは
Metaclass のインスタンスです。Metaclassのインスタンスは匿名のメタクラス
で、それぞれ唯一のインスタンスを持ちます。つまり、それがクラスというわ

けです。

Metaclassは、メタクラスに共通する振る舞いを提供し、メタクラスの唯一
のインスタンスに対して初期化されたインスタンスを生成するインスタンス生

成（subclassOf:）、クラス変数の初期化、メタクラスのインスタンス、メソッド
のコンパイル、およびクラスの情報（継承関係、インスタンス変数、など）の

メソッドを提供します。

13.7 MetaclassのメタクラスはMetaclassのインスタ
ンスである

残る質問は一つだけです。Metaclass classのクラスは何でしょうか？

Figure 13.9: すべてのメタクラスは Metaclassのインスタンスである。

Figure 13.10: すべてのメタクラスはMetaclassのインスタンスである。Metaclass
のメタクラスも同様。

答えは簡単でメタクラスです。もちろんシステム内の他のすべてのメタク

ラスと同様に、Metaclassのインスタンスであるべきです。(図 13.10を参照)

この図は、すべてのメタクラスがMetaclassのインスタンスであり、Metaclass
のメタクラスもまた同様であるということを示しています。13.9と 13.10を比
べると、Object classに至るまでメタクラスの階層がクラス階層の完全な鏡像に
なっていることがわかるでしょう。

以下の例は、クラス階層の問い合わせによって、図 13.10が正確であるこ
とを示しています。（現実にはちょっとしたウソが混じっており、Object class
superclassの結果は ProtoObject class であって Classではありません。Pharo で
は、Classに至るため、もう一つ上のスーパークラスに行く必要があります。）

Example 13.1: ク ラ ス 階 層
TranslucentColor superclass −→ Color
Color superclass −→ Object

Example 13.2: ク ラ ス 階 層 に 並 列 す る メ タ ク ラ ス 階 層
TranslucentColor class superclass −→ Color class
Color class superclass −→ Object class
Object class superclass superclass −→ Class ”NB: skip ProtoObject class”
Class superclass −→ ClassDescription
ClassDescription superclass −→ Behavior
Behavior superclass −→ Object

Example 13.3: Metaclass の イ ン ス タ ン ス
TranslucentColor class class −→ Metaclass
Color class class −→ Metaclass
Object class class −→ Metaclass
Behavior class class −→ Metaclass

Example 13.4: Metaclass class は Metaclass の イ ン ス タ ン ス で あ る
Metaclass class class −→ Metaclass
Metaclass superclass −→ ClassDescription

13.8 この章のまとめ

クラスがどのように組織されており、一貫性のあるオブジェクトモデルがどの

ような効果を及ぼしているのかについて、よく理解できたのではないでしょう

か。忘れたり混乱したりしたなら、いつでもメッセージ・パッシングが鍵であ

ることを思い出してください。メッセージのレシーバのクラスからメソッドを

探せばよいのです。このことはど ん な レシーバについても同様です。レシーバ

のクラスでメソッドが見つからなければ、そのスーパークラスを探せばよいの

です。

• すべてのクラスはメタクラスのインスタンスである。メタクラスは暗黙
的である。メタクラスはクラスを定義するとき、そのクラスを唯一のイ

ンスタンスとするように自動的に作成されます。

• メタクラスの階層はクラスの階層と並列に存在する。クラスのメソッド
探索は、普通のオブジェクトのメソッド探索と並行しており、メタクラ

スの継承チェーンに沿って行われます。

• すべてのメタクラスは Classと Behaviorを継承している。すべてのクラ
スは Classのインスタンスです (is-a)。メタクラスはクラスでもあるため
Classを継承しています。Behaviorはインスタンスを持つすべての存在に共
通した振る舞いを提供します。

• すべてのメタクラスは Metaclassのインスタンスである。ClassDescription
は Classおよび Metaclassに共通することがらすべてを提供します。

• Metaclassのメタクラスは Metaclassのインスタンスである。クラス-イン
スタンスの関係（instance-of）は閉じたループを形成します。すなわち、
Metaclass class class −→ Metaclassということです。





Chapter 14

リフレクション

Smalltalkはリフレクティブなプログラミング言語です。つまり、プログラム
は自分自身の実行や構造を「（ランタイムシステムに）反映する (リフレクト
する)」ことができます。ランタイムシステムのメ タ オ ブ ジ ェ ク ト を通常のオ
ブジェクトとして具 現 化 (reify))することができるので、メタオブジェクトに
問い合わせを行ったり、インスペクトして内容を見たりすることができます。

Smalltalkのメタオブジェクトには、クラス、メタクラス、メソッド辞書、コ
ンパイル済みメソッド、実行時スタックなどがあります。このようなリフレク

ションはイ ン ト ロ ス ペ ク シ ョ ン とも呼ばれ、多くのモダンなプログラミング言

語でサポートされています。

Figure 14.1: 具現化とリフレクション。

Smalltalkでは逆の操作も可能で、具現化されたメタオブジェクトを変更し
て、ランタイムシステムに反 映 させることもできます (図 14.1参照)。このよう
なリフレクションはイ ン タ ー セ ッ シ ョ ン とも呼ばれ、主に動的なプログラミン

グ言語でサポートされていますが、静的な言語では極めて限られています。

他のプログラム（または自分自身）を操作するようなプログラムをメ タ プ ロ

グ ラ ム と呼びます。リフレクティブなプログラミング言語では、イントロスペ

クションとインターセッションがどちらもサポートされているべきです。イ

ントロスペクションとは、言語自身を定義するデータ構造（オブジェクト、クラ

ス、メソッド、実行時スタックなど）を検 査 する機能です。インターセッショ

ンとは、そのようなデータ構造を変 更 できること、言い換えれば、言語の意味

論やプログラムの振る舞いを、プログラム自身から変更できる機能です。ラン

タイムシステムの構造を検査・変更することを構 造 的 リ フ レ ク シ ョ ン と呼び、

それらの構造の解釈を変更することを振 る 舞 い 的 リ フ レ ク シ ョ ン と呼びます。

この章では、主に ind構造的リフレクションを扱います。実用的な例を多
数挙げながら、 Smalltalkがどのようにイントロスペクションとメタプログラ
ミングをサポートしているのかを探っていきます。

14.1 イントロスペクション

インスペクタを使うと、オブジェクトの内容を見たり、インスタンス変数の値

を変更したり、オブジェクトにメッセージを送ったりすることができます。

次 の コ ー ド を ワ ー ク ス ペ ー ス で 評 価 し て み ま し ょ う :

w := Workspace new.
w openLabel: ’My Workspace’.
w inspect

このコードを評価すると、インスペクタとワークスペースがもう一つ開き

ます。このインスペクタは新しく開いたワークスペースの内部状態を示してお

り、左ペインにはインスタンス変数（dependents, contents, bindingsなど）、右ペ
インには選択したインスタンス変数の値が表示されます。contentsインスタン
ス変数は、ワークスペースのテキストエリアに表示されている内容を表します

ので、この時点で contentsを選択すると、インスペクタの右ペインには空文字
列が表示されます。

こ の 空 文 字 列 の と こ ろ に ’hello’と 入 力 し て 、 accept し て み ま し ょ う 。

これで contents変数の値が変わりますが、まだワークスペースウィンドウは
変化に気づかないので再描画されません。ウィンドウを更新させるには、イン

スペクタの下ペインで self contentsChangedを評価します。

インスタンス変数へのアクセス

インスペクタはどのように動いているのでしょうか？ Smalltalkでは、インス
タンス変数はすべて保護されています。理論上は、アクセサメソッドがそのク

Figure 14.2: !ワークスペース!をインスペクトしている様子。

ラスに定義されていない限り、他のオブジェクトからインスタンス変数にアク

セスすることはできません。しかし、実際にはインスペクタはアクセサがなく

てもインスタンス変数にアクセスできます。これは Smalltalkがリフレクショ
ンの機能を持っているからです。Smalltalkでは、クラスはインスタンス変数
を名前または数値の添字で定義します。インスペクタは、Objectクラスで定義
されているメソッド instVarAt: indexまたは instVarNamed: aStringを使ってインス
タンス変数にアクセスします。数値の添字 index で位置を指定するか、インス
タンス変数名 aString を指定します。インスタンス変数に新しい値を代入する
には、instVarAt:put:か instVarNamed:put:を使います。

例えば次のコードを評価すると、最初のワークスペースで変数 wに束縛さ
れていたオブジェクトの内部状態を変更できます。

w instVarNamed: ’contents’ put: ’howdy’; contentsChanged

注 意 : これらのメソッドは開発ツールを作る分には便利で
すが、通常のアプリケーション記述には使わない方がよい

でしょう。リフレクティブなメソッドはオブジェクトのカ

プセル化境界をまたいでしまうので、理解も保守も難しい

コードの原因となってしまいます。

instVarAt:と instVarAt:put:はどちらもプリミティブメソッドです。プリミティ
ブメソッドは、バーチャルマシンの原始的操作として実装されています。これ

らのメソッドのコードを調べれば、特殊なプラグマ構文 <primitive: N>（Nは整
数）が使われていることがわかるでしょう。

Object»instVarAt: index
”プ リ ミ テ ィ ブ 。 オ ブ ジ ェ ク ト 内 の 固 定 長 配 列 で 管 理 さ れ て い る イ ン ス タ ン
ス 変 数 を 返 す 。 ”

<primitive: 73>
”固 定 長 配 列 の 範 囲 を 超 え て ア ク セ ス す る 。 ”
↑self basicAt: index - self class instSize

通常はプリミティブ呼び出しの行以降のコードは実行されません。その部

分はプリミティブが失敗した場合にのみ実行されます。この例の場合では、存

在しないインスタンス変数にアクセスしようとすると、プリミティブに続く

コードがバックアップとして試みられます。この仕組みにより、プリミティブ

メソッドが失敗したときにデバッガを起動することもできます。プリミティブ

メソッドのコードは変更できますが、Pharoシステムの安定性を脅かしかねな
いため注意してください。

Figure 14.3: ある Workspaceオブジェクトのすべてのインスタンス変数を表示
している様子。

図 14.3は、Workspaceクラスのあるインスタンス（w）のすべてのインスタ
ンス変数を表示する方法を示しています。allInstVarNamesメソッドは、与えら
れたクラスにあるすべてのインスタンス変数の名前を返します。

同様の方法で、同じ特徴を持つインスタンスを集めることもできます。例

えば SketchMorphクラスのインスタンスのうち、ownerインスタンス変数の内容
がワールドモーフであるもの（す な わ ち 現在表示されているすべてのイメージ）

を集めてみましょう。次の式を試してみてください。

SketchMorph allInstances select: [:c | (c instVarNamed: ’owner’) isWorldMorph]

インスタンス変数に対する繰り返し処理

instanceVariableValuesメッセージについて調べてみましょう。このメソッドは、
継承したものを除くすべてのインスタンス変数の値をコレクションとして返し

ます。次に例を示します:
(1@2) instanceVariableValues −→ an OrderedCollection(1 2)

このメソッドは Objectクラスで次のように実装されています:

Object»instanceVariableValues
”レ シ ー バ の ク ラ ス で 追 加 さ れ た イ ン ス タ ン ス 変 数 の 値 を 集 め た コ レ ク シ ョ
ン を 返 す 。 ”

| c |
c := OrderedCollection new.
self class superclass instSize + 1
to: self class instSize
do: [ :i | c add: (self instVarAt: i)].

↑ c

このメソッドは、スーパークラスで使われている最後のインスタンス変数

の添字の直後から開始して、レシーバのクラスが定義しているインスタンス変

数の添字に対して繰り返し処理をしています。（instSizeメソッドは、クラスが
定義している名前付きインスタンス変数の数を返します）。

クラスやインターフェースへの問い合わせ

Pharoの開発ツール（コードブラウザ、デバッガ,インスペクタなど）はすべ
て、これまでに見て来たリフレクティブな機能を使っています。

もういくつか、開発ツールを作る際に役立つメッセージを挙げましょう:
isKindOf: aClassは、レシーバが aClassそのいずれかのスーパークラスのイ

ンスタンスであれば真を返します。例えば:

1.5 class −→ Float
1.5 isKindOf: Number −→ true
1.5 isKindOf: Integer −→ false

respondsTo: aSymbol は、セレクタが aSymbol であるメソッドをレシーバが
持っていれば真を返します。例えば:

1.5 respondsTo: #floor −→ true ”Numberク ラ ス は floorを 実 装 し て い ま す ”
1.5 floor −→ 1
Exception respondsTo: #, −→ true ”例 外 ク ラ ス は グ ル ー プ 化 で き ま す ”

警 告 :ここで紹介した機能は、開発ツールの記述には大変便
利ですが、通常のアプリケーション作成には適していませ

ん。オブジェクトが属するクラスや理解できるメッセージ

を調べようとするのは、カプセル化の原則に反しており、設

計に問題がある兆候です。しかしながら、開発ツールは普

通のアプリケーションではありません。開発ツールの対象

はソフトウェアそのものなので、コードの内部を詳細に調

べる権利があると言えます。

コード・メトリクス

Smalltalkのイントロスペクションを使って、いろいろなコード・メトリクス
が簡単に調べられるという例を見てみましょう。コード・メトリクスとは、継

承の階層の深さ、直接的・間接的なサブクラスの数、各クラスのメソッドやイ

ンスタンス変数の数、クラスで新しく定義されたメソッドやインスタンス変数

の数などを計測することです。以下に Morphクラスのメトリクスをいくらか示
します。Morphクラスは Pharoのすべてのグラフィカル・オブジェクトのスー
パークラスなので、それ自身が巨大で、かつ巨大な継承階層の根となっている

ことがわかります。もしかするとリファクタリングが必要かも！

Morph allSuperclasses size. −→ 2 ”継 承 階 層 の 深 さ ”
Morph allSelectors size. −→ 1378 ”メ ソ ッ ド の 数 ”
Morph allInstVarNames size. −→ 6 ”イ ン ス タ ン ス 変 数 の 数 ”
Morph selectors size. −→ 998 ”Morphで 新 し く 定 義 さ れ た メ ソ ッ ド の

数 ”
Morph instVarNames size. −→ 6 ”Morphで 新 し く 定 義 さ れ た イ ン ス タ ン

ス 変 数 の 数 ”
Morph subclasses size. −→ 45 ”直 接 の サ ブ ク ラ ス の 数 ”
Morph allSubclasses size. −→ 326 ”サ ブ ク ラ ス の 総 数 ”
Morph linesOfCode. −→ 5968 ”総 コ ー ド 行 数 ”

オブジェクト指向言語において最も興味深いメトリクスの一つは、スーパー

クラスから継承したメソッドに拡張を加えているメソッドの数です。このメト

リクスから、クラスとそのスーパークラスの関係に関する情報が得られます。

次の節以降では、このような疑問に対する答えを見つけるために、ランタイム

の情報に関する知識をどのように使えばよいのかを見ていきます。

14.2 コードをブラウズする

Smalltalkでは、すべてがオブジェクトです。特にクラスもまたオブジェクト
であり、そのインスタンスを探索するための便利な機能を提供しています。こ

れから挙げるメッセージはほとんど Behaviorで実装されているので、どのクラ
スも理解できます。

前に挙げたように、クラスに#someInstanceメッセージを送ると、そのクラ
スのあるインスタンスが得られます。

Point someInstance −→ 0@0

すべてのインスタンスを集めるには#allInstancesを、メモリ中にあるインス
タンスの数を得るには#instanceCountを使うことができます。

ByteString allInstances −→ #(’collection’ ’position’ ...)
ByteString instanceCount −→ 104565
String allSubInstances size −→ 101675

以下の機能は、クラスから特定の条件を満たすメソッドを抽出できるので、

アプリケーションをデバッグするときに役立ちます。

• whichSelectorsAccess:は、指定したインスタンス変数を読み書きするすべ
てのメソッドのセレクタを返します。

• whichSelectorsStoreInto:は、指定したインスタンス変数の値を変更するす
べてのメソッドのセレクタを返します。

• whichSelectorsReferTo:は、指定したメッセージを送るすべてのメソッドの
セレクタを返します。

• crossReference各メッセージを、そのメッセージを送るメソッドの集合に
関連づける

Point whichSelectorsAccess: ’x’ −→ #(#’\\’ #= #scaleBy: ...)
Point whichSelectorsStoreInto: ’x’ −→ #(#setX:setY: ...)
Point whichSelectorsReferTo: #+ −→ an IdentitySet(#rotateBy:about: ...)
Point crossReference −→ an Array(

an Array(’*’ an IdentitySet(#rotateBy:about: ...))
an Array(’+’ an IdentitySet(#rotateBy:about: ...))
...)

次のメッセージは継承階層も考慮した結果を返します:

• whichClassIncludesSelector:与えられたメッセージを実装しているクラスを
返します。

• unreferencedInstanceVariablesレシーバのクラスでもサブクラスでも使われ
ていないインスタンス変数をリストにして返します。

Rectangle whichClassIncludesSelector: #inspect −→ Object
Rectangle unreferencedInstanceVariables −→ #()

SystemNavigationは、システム内のソースコードのブラウズや問い合わせを
行うのに便利なメソッドを持つファサードです。SystemNavigation defaultを評価
して得られるインスタンスを使ってシステム情報を調べられます。例えば:

SystemNavigation default allClassesImplementing: #yourself −→ {Object}

次のメッセージの説明は不要でしょう:

SystemNavigation default allSentMessages size −→ 24930
SystemNavigation default allUnsentMessages size −→ 6431
SystemNavigation default allUnimplementedCalls size −→ 270

中には実装されているのに使われていないメッセージもありますが、perform:
を使うなどして、ソースコードに現れない形で送られることもあるので、必ずし

も不要とは限りません。ただし、送られているのに実装されていないメッセー

ジは、送信したときに失敗することになるので、問題がある可能性が高いと言

えます。このようなメッセージは、不完全な実装、廃止されたAPI、ライブラ
リ不足を示している可能性があります。

SystemNavigation default allCallsOn: #Pointは、ソースコード中で Pointを明示的
にレシーバとしているすべてのメッセージを返します。

以上の機能はすべて、Pharoのプログラミング環境（特にコードブラウザ）
に統合されています。既にご存知だとは思いますが、メッセージをブラウズする

際に使える便利なキーボードショートカットがあります。それらは、メッセー

ジの実装を取得するための cmd–m 、メッセージの送信者を取得するための
cmd–nなどです。一方で、このような問い合わせ機能が多数 SystemNavigation
クラスの browsing プロトコルに属するメソッドとして実装されていることは、
それほど知られていないかもしれません。例えば、次の式を評価すると ifTrue:
メッセージのすべての実装をプログラム的にブラウズできます:

SystemNavigation default browseAllImplementorsOf: #ifTrue:

Figure 14.4: すべての#ifTrue:の実装をブラウズしている様子。

特に便利なメソッドは browseAllSelect:と browseMethodsWithSourceString:の二
つです。以下にシステム中のすべてのメソッドから super送信を行うものをブ
ラウズする方法を示します。最初の方法は総当たり的ですが、二つ目の方法で

はより正確で、検索結果から不要なものを多少減らすことができます。

SystemNavigation default browseMethodsWithSourceString: ’super’.
SystemNavigation default browseAllSelect: [:method | method sendsToSuper ].

14.3 クラス、メソッド辞書、メソッド

クラスもオブジェクトなので、他のオブジェクトと同様にインスペクト・エク

スプロアできます。

Point exploreを 評 価 し て み ま し ょ う 。

図 14.5では、エクスプローラが Pointクラスの構造を表示しています。こ
のクラスがメソッドを辞書として保持しており、この辞書のキーがメソッドの

セレクタとなっていることがわかります。セレクタ #*をキーとしている要素
として、Point»*が逆コンパイルされたバイトコードとして表示されています。

Figure 14.5: Pointクラスと#*メソッドのバイトコードをエクスプロアする。

クラスとメソッドの関係について考えてみましょう。図 14.6を見ると、ク
ラスとメタクラスは共通するスーパークラスとして Behaviorを持っていること
がわかります。Behaviorが、newなどのようにクラスに関するメソッドとして重
要なものが定義されている場所となっています。どのクラスもメソッドセレク

タとコンパイル済みメソッド を関連づけるメソッド辞書を持っています。コ

ンパイル済みメソッドは自分が所属するクラスを知っています。図 14.5から、
literal5literal5にキーをシンボル Point、値を Pointクラスとするアソシエーション
として所属クラスの情報が含まれていることがわかります。

システムに関する情報を問い合わせる際にクラスとメソッドの関係を利用

することができます。例えば、あるクラスのメソッドのうち新しく定義された

もの、つまりスーパークラスのメソッドをオーバーライドしていないものを見

つけるには、次のようにクラスからメソッド辞書をたどります:
[:aClass| aClass methodDict keys select: [:aMethod |
(aClass superclass canUnderstand: aMethod) not ]] value: SmallInteger

Figure 14.6: クラス、メソッド辞書、コンパイル済みメソッド

−→ an IdentitySet(#threeDigitName #printStringBase:nDigits: ...)

コンパイル済みメソッドは、メソッドのバイトコードを保持しているだけ

ではなく、システムに問い合わせを行う便利なメソッドも数多く提供していま

す。isAbstractメソッド（subclassResponsibilityを送るかどうか）はそのようなメ
ソッドの一つです。isAbstractメソッドを使って、抽象クラスのすべての抽象メ
ソッドを同定することができます。

[:aClass| aClass methodDict keys select: [:aMethod |
(aClass>>aMethod) isAbstract ]] value: Number
−→ #(#storeOn:base: #printOn:base: #+ #- #* #/ ...)

このコードでは、セレクタに対応するコンパイル済みメソッドを取得するため

の>>メッセージが使われています。

ある継承階層内で使われている super送信、例えば Collectionクラスのサ
ブクラスで super送信を行っているメソッドの一覧をブラウズするには、以下
のようにいくらか高度な問い合わせを行います。

class := Collection.
SystemNavigation default
browseMessageList: (class withAllSubclasses gather: [:each |
each methodDict associations
select: [:assoc | assoc value sendsToSuper]
thenCollect: [:assoc | MethodReference class: each selector: assoc key]])

name: ’Supersends of ’ , class name , ’ and its subclasses’

目的のメソッドを同定するため、クラスからメソッド辞書を経由してコンパイ

ル済みメソッドへと探索している過程に注目してください。MethodReferenceは
コンパイル済みメソッドのための軽量プロキシで、多くのツールで使われてい

ます。CompiledMethod»methodReference はコンパイル済みメソッドへのメソッ
ド参照（MethodReferenceのインスタンス）を返す便利なメソッドです。

(Object>>#=) methodReference methodSymbol −→ #=

14.4 ブラウザ環境

SystemNavigation はシステムコードに対してプログラム的に問い合わせを行っ
たりブラウズしたりする便利な機能を提供しますが、もっと良い方法もありま

す。Pharoに統合されているリファクタリングブラウザを使うと、複雑な問い
合わせを対話的にもプログラム的にも行うことができます。

例えば、Collectionクラス階層のメソッドで、メソッド自身と異なるメッセー
ジを superに送っているメソッドに興味があるとしましょう。このようなコー
ドは、普通であれば super送信を self送信に置き換えるべきなので、悪いコー
ドの臭いがすると考えられます。（考えてみてください。 superを使う必要が
あるのは、メソッドをオーバーライドして拡張するときだけです。継承された

メソッドにアクセスするのであれば、 selfにメッセージを送ればすみます。）

リファクタリングブラウザは、問い合わせを関心のあるクラスやメソッド

に限定するエレガントな方法を提供しています。

ブ ラ ウ ザ を 開 い て Collectionク ラ ス を 見 て み ま し ょ う 。 ク ラ ス 名 を ア ク シ ョ
ン ク リ ッ ク し て refactoring scope>subclasses with を 選 択 す る と 、 Collectionク ラ ス
階 層 に 限 定 さ れ た ブ ラ ウ ザ 環 境 が 開 き ま す 。 こ の 限 定 さ れ た ス コ ー プ の 中 で

refactoring scope>super-sends を 選 択 す る と 、 Collectionク ラ ス 階 層 に あ る super
送 信 を 行 う す べ て の メ ソ ッ ド を 列 挙 し た 新 し い ウ ィ ン ド ウ が 開 き ま す 。 こ の

中 か ら あ る メ ソ ッ ド を ク リ ッ ク し て 、 refactor>code critics を 選 ん で く だ さ い 。
Lint checks>Possible bugs>Sends different super message と 選 ん で ゆ き 、 browse を
ア ク シ ョ ン ク リ ッ ク し て み ま し ょ う 。

図 14.7にあるように、Collectionクラス階層にはそのようなメソッドが 19個
あって、その中には super printOn:を送っている Collection»printNameOn:も含まれ
ていることがわかります。

このブラウザ環境はプログラムからでも用意できます。例えば、次のコー

ドでは Collectionとそのサブクラスを扱う BrowserEnvironmentを生成し、そこか
ら super送信を行うメソッドのみを抽出して、その結果を表示するウィンドウ
を開きます。

Figure 14.7: 自身のセレクタとは異なる super送信を行うメソッドを探す様子。

((BrowserEnvironment new forClasses: (Collection withAllSubclasses))
selectMethods: [:method | method sendsToSuper])
label: ’Collection methods sending super’;
open.

このコードは先に挙げた SystemNavigationを使った例と同等ですが、よりコ
ンパクトになっていることに注目してください。

最終的には次のようにして、自身のメソッドセレクタと異なる super送信を
行うメソッドをプログラム的に探せます:

((BrowserEnvironment new forClasses: (Collection withAllSubclasses))
selectMethods: [:method |
method sendsToSuper
and: [(method parseTree superMessages includes: method selector) not]])

label: ’Collection methods sending different super’;
open

ここではメソッドセレクタと異なる superメッセージを探すために、コンパイ
ル済みメソッドのそれぞれから、（リファクタリングブラウザが使用する）構文

木を取得しています。RBProgramNodeクラスの queryingプロトコルを見て、構
文木に何を尋ねることができるのかを確認してください。

14.5 実行時コンテキストにアクセスする

これまで Smalltalk のリフレクティブな機能を使ったオブジェクト、クラス、
メソッドへの問い合わせの仕方やエクスプロアの方法を見てきました。では、

実行時環境についてはどうでしょう?

メソッドコンテキスト

実際のところ、実行中のメソッドに対応する実行時コンテキストはバーチャル

マシンが管理しています—仮想イメージの中にはないのです！しかし、デバッ
ガは明らかに実行時コンテキストにアクセスして動作しています。さらには、

実行時コンテキストを探ることも、他のオブジェクトと同様に何の問題もあり

ません。どうなっているいるのでしょうか？

実は、デバッガには何の不思議なこともありません。この秘密は、これま

で軽く触れただけだった thisContextという擬似変数にあります。メソッド中で
thisContextが参照されたときにはいつでも、そのメソッドのコンテキストが具現
化され、リンクした一連の MethodContextオブジェクトとしてイメージから操
作できるようになります。

この仕組みは簡単に実験できます。

Integer»factorialの 実 装 に 下 線 部 の 式 を 挿 入 し て し て く だ さ い :

Integer»factorial
”レ シ ー バ の 階 乗 を 返 す 。 ”
self = 0 ifTrue: [thisContext explore. self halt. ↑ 1].
self > 0 ifTrue: [↑ self * (self - 1) factorial].
self error: ’Not valid for negative integers’

3 factorialを ワ ー ク ス ペ ー ス で 評 価 し ま す 。 図 14.8 の よ う に デ バ ッ ガ ウ ィ
ン ド ウ と エ ク ス プ ロ ー ラ の 両 方 が 表 示 さ れ る は ず で す 。

貧相なデバッガへようこそ！ ここでエクスプロアされたオブジェクトのク

ラスをブラウズすると（す な わ ち エクスプローラの下のペインで self browseを
評価すると）、thisContextが、MethodContextのインスタンスであることがわかる
でしょう。senderの連鎖からたどれるすべてのオブジェクトも同様です。

thisContextは日常的なプログラミングで使うものではありませんが、デバッ
ガのようなツールを実装したり、コールスタックの情報にアクセスしたりする

際には必要不可欠です。次の式を評価すると、thisContextを使っているメソッ
ドを見つけられます。

SystemNavigation default browseMethodsWithSourceString: ’thisContext’

実は、メッセージのセンダを取得することが thisContextの最も一般的な用途
となっています。次に示すコードは典型的な例です:

Figure 14.8: thisContextをエクスプロアしている様子。

Object»subclassResponsibility
”こ の メ ッ セ ー ジ に よ り 、 サ ブ ク ラ ス の 挙 動 に 関 す る を フ レ ー ム ワ ー ク を 設
定 し ま す 。

サ ブ ク ラ ス が こ の メ ッ セ ー ジ を 実 装 す べ き で あ る と 宣 言 し ま す 。 ”

self error: ’My subclass should have overridden ’, thisContext sender selector
printString

Smalltalkの慣習では、self subclassResponsibilityを送るメソッドは抽象メソッ
ドとみなされます。しかし、Object»subclassResponsibilityはどの抽象メソッドが
呼ばれたのかという情報を含むエラーメッセージをどうやって提供するので

しょうか？ 答えは簡単。thisContextにセンダを尋ねるだけです。

賢いブレークポイント

Smalltalkでブレークポイントを設定するには、メソッド内の関心のある箇所
で self haltを評価するコードを書きます。この式が実行されると thisContextが具
現化され、そのブレークポイントにおけるデバッガのウィンドウが開きます。

しかしながら、システムで集中的に使われるメソッドにブレークポイントを設

定すると問題が起きてしまいます。

例えば、OrderedCollection»add:の実行を解析したいとしましょう。このメソッ
ド内にブレークポイントを設定すると困ったことになります。

新しい仮 想 イ メ ー ジ を 用 意 し て 、 次 の ブ レ ー ク ポ イ ン ト を 設 定 し ま す :

OrderedCollection»add: newObject
self halt.
↑self addLast: newObject

仮想イメージがデバッガウィンドウすら開かずに直ちにフリーズしてしまい

ます！次の点を理解すれば、問題がはっきりとわかります。 (i) OrderedCollection
»add:はシステムのあちこちで使われているので、変更をアクセプトしてごく間
もない時点でブレークポイントが起動します。 (ii)しかし、起動したデ バ ッ ガ
自 身 が add:メッセージを OrderedCollectionのインスタンスに送るので、そのデ
バッガが邪魔されてしまいます！ このような問題を避けるには、デバッグし

たい状況でのみ条 件 的 に 停 止 する方法が必要です。その用途にぴったりなのが

Object»haltIf:です。

では add:を、例えば OrderedCollectionTest»testAddのコンテキストから呼ばれ
たときにだけ停止させたいとしましょう。

ま た 新 し い 仮 想 イ メ ー ジ を 用 意 し て 、 次 の ブ レ ー ク ポ イ ン ト を 設 定 し て み

ま し ょ う :

OrderedCollection»add: newObject
self haltIf: #testAdd.
↑self addLast: newObject

OrderedCollectionTestを実行してみましょう。今度はフリーズしません。
（OrderedCollectionTestは CollectionsTests-Sequenceableカテゴリにあります）

どう動いているのでしょうか？ Object»haltIf:を見てみましょう:

Object»haltIf: condition
| cntxt |
condition isSymbol ifTrue: [

”指 定 さ れ た セ レ ク タ ・ シ ン ボ ル の メ ソ ッ ド が セ ン ダ の 呼 び 出 し チ ェ ー ン
に 含 ま れ て い る 場 合 の み 停 止 す る ”
cntxt := thisContext.
[cntxt sender isNil] whileFalse: [
cntxt := cntxt sender.
(cntxt selector = condition) ifTrue: [Halt signal]. ].

↑self.
].
...

haltIf:は実行時スタックを thisContextから遡りながら、呼び出し側のメソッド
の名前が引数と一致するか確認します。一致したら例外を発生させます。この

例外はデフォルトの挙動として、デバッガを起動します。

haltIf:の引数に真偽値や真偽値を返すブロックを渡してもよいですが、それ
らの場合は単純な処理になるので thisContextは使わずにすみます。

14.6 理解されないメッセージをインターセプトする

ここまで、Smalltalkのリフレクティブな機能をオブジェクト、クラス、メソッ
ド、実行時スタックに問い合わせを行ったりエクスプロアしたりするのに主に

使ってきました。ここからは Smalltalkシステムの構造に関する知識を、メッ
セージをインターセプトしたり振る舞いを実行時に変えたりするためにどう

使っていけるのかを見ていきます。

オブジェクトはメッセージを受け取ると、まず自身のクラスのメソッド辞書

からメッセージに対応するメソッドを探します。対応するメソッドが見つから

なければ、Objectにたどり着くまでクラス階層を遡りつつ探します。それでも見
つからなければ、オブジェクトは doesNotUnderstand:メッセージを、元々のメッ
セージセレクタとメッセージへの引数を引数として自 分 自 身 に 送 り ま す 。メソッ

ド探索の処理が doesNotUnderstand:の送信に対しても Object»doesNotUnderstand:
が見つかるまで繰り返され、その結果としてデバッガが起動します。

しかし、もし doesNotUnderstand:がメソッド検索パスにある Objectのサブク
ラスでオーバーライドされていたらどうなるでしょう？ これはある種の非常

に動的な振る舞いを実現するのに便利な方法でです。doesNotUnderstand:をオー
バーライドすることで、オブジェクトは理解できないメッセージに応答するた

めの代替的な戦略を持つことができます。

この手法の最も一般的な応用例は、(1)軽量プロキシの実装と (2)足りない
コードの動的なコンパイルまたはロード、です。

軽量プロキシ

最初の例として、既存のオブジェクトのプロキシとして振る舞う「最小限のオ

ブジェクト 」を紹介します。このプロキシは固有のメソッドを実質的に一つ

も持たないので、どんなメッセージも doesNotUnderstand:で捕捉されることにな
ります。このメッセージを実装すれば、プロキシは実体のオブジェクトにメッ

セージを移譲する前に特別な動作を行うことができます。

これがどのように実装できるのかを見てみましょう。1。

まず、LoggingProxyを次のように定義します:

ProtoObject subclass: #LoggingProxy
instanceVariableNames: ’subject invocationCount’
classVariableNames: ’’
poolDictionaries: ’’
category: ’PBE-Reflection’

Objectではなく ProtoObjectのサブクラスとして定義していることに注意してく
ださい。Objectが持つ 400個以上のメソッド (!)を継承したくないからです。

1PBE-Reflectionを http://www.squeaksource.com/PharoByExample/ からロードできます

http://www.squeaksource.com/PharoByExample/

Object methodDict size −→ 408

このプロキシには二つのインスタンス変数があります。 subjectはプロキシ
の実体であり、invocationCountはプロキシがインターセプトしたメッセージの数
です。二つのインスタンス変数を初期化し、インターセプトしたメッセージの

数を返すアクセサを提供します。初期化時点では、subject変数はプロキシオブ
ジェクト自身を指しています。

LoggingProxy»initialize
invocationCount := 0.
subject := self.

LoggingProxy»invocationCount
↑ invocationCount

この例では、理解できないすべてのメッセージをインターセプトし、トラ

ンスクリプトに表示し、メッセージ受信回数を更新し、メッセージを実体へ転

送します。

LoggingProxy»doesNotUnderstand: aMessage
Transcript show: ’performing ’, aMessage printString; cr.
invocationCount := invocationCount + 1.
↑ aMessage sendTo: subject

ここでちょっとした魔法の登場です。Pointオブジェクトと LoggingProxyオブ
ジェクトを生成し、プロキシに「その Pointオブジェクトになれ（become:）」と
伝えます。

point := 1@2.
LoggingProxy new become: point.

これにより、イメージ内にある、この Pointオブジェクトへのすべての参照
をプロキシへの参照へと入れ替えます（逆も同様です）。重要なのは、これに

よってプロキシの subjectインスタンス変数もこの Pointオブジェクトを参照す
るようになることです！

point invocationCount −→ 0
point + (3@4) −→ 4@6
point invocationCount −→ 1

これはたいていの場合うまく動作するのですが、足りない点もあります:

point class −→ LoggingProxy

興味深いことに、classメソッドは ProtoObjectで実装されておらず（Objectで実
装されています）、LoggingProxyは Objectを継承していません！ classはメッセー

ジとして送られるのではなく、仮想マシンによって直接処理されます。2

このような特別なメッセージ送信を無視するとしても、この手法では克服

不可能な、self送信をインターセプトできないという根本的な問題があります。

point := 1@2.
LoggingProxy new become: point.
point invocationCount −→ 0
point rect: (3@4) −→ 1@2 corner: 3@4
point invocationCount −→ 1

このプロキシは rect:メソッド中の二度の self送信を数え損ねています:

Point»rect: aPoint
↑ Rectangle origin: (self min: aPoint) corner: (self max: aPoint)

この方法のプロキシを使えばメッセージをインターセプトできますが、プ

ロキシに起因する制限に気をつけてください。14.7節では、メッセージをイン
ターセプトするためのより一般的な方法を見ていきます。

欠けているメソッドを生成する

理解されないメッセージをインターセプトする手法の一般的な応用例として、

未実装のメソッドの動的なロードまたは生成が挙げられます。多数のメソッド

を持つ巨大なクラスライブラリを考えてみましょう。ライブラリ全体をロード

する代わりに各クラスのスタブをロードしてもよいでしょう。このスタブは、

そのクラスで定義されたすべてのメソッドのソースコードがどこにあるかを

知っています。スタブは理解できないメッセージをすべて捕捉し、欠けている

メソッドを必要に応じて動的にロードします。この処理をある時点で無効化す

れば、それまでにロードされたコードが特定のクライアントアプリケーション

に最低限必要なものとして保存することができます。

この方法のシンプルな応用例を見てみましょう。次に示すクラスは必要に

応じてインスタンス変数へのアクセサを自動的に追加します:

DynamicAcccessors»doesNotUnderstand: aMessage
| messageName |
messageName := aMessage selector asString.
(self class instVarNames includes: messageName)
ifTrue: [

2yourselfも同様に、決して送られません。レシーバによって仮想マシンが直接解釈する可能
性のあるメッセージを次に示します: +- < > <= >= =∼= * / == @ bitShift: // bitAnd: bitOr: at: at:put: size
next nextPut: atEnd blockCopy: value value: do: new new: x y 次に示すセレクタは決して送られません。
コンパイラが比較とジャンプのバイトコードにインライン展開します: ifTrue: ifFalse: ifTrue:ifFalse:
ifFalse:ifTrue: and: or: whileFalse: whileTrue: whileFalse whileTrue to:do: to:by:do: caseOf: caseOf:otherwise:
ifNil: ifNotNil: ifNil:ifNotNil: ifNotNil:ifNil: これらのメッセージを論理値ではないオブジェクトに送ろう
とした場合も、レシーバが mustBeBooleanをオーバーライドするか、NonBooleanReceiver例外を捕捉
して適切な論理値を返せば実行を継続できます。

self class compile: messageName, String cr, ’ ↑ ’, messageName.
↑ aMessage sendTo: self ].

↑ super doesNotUnderstand: aMessage

理解できないメッセージはすべてここで捕捉されます。送られたメッセージと

同名のインスタンス変数があれば、自身のクラスにアクセサをコンパイルする

ように依頼します。その後、メッセージをもう一度送信します。

DynamicAccessorsクラスに（初期化されていない）インスタンス変数 xがあ
り、xのアクセサが定義されていないとしましょう。次の式を評価すると、アク
セサが動的に生成されて xの値を取得できます:

myDA := DynamicAccessors new.
myDA x −→ nil

DynamicAccessorsオブジェクトに最初に xメッセージが送られたときに何が
起きるか、順に見ていきましょう（図 14.9を参照）。

Figure 14.9: アクセサを動的に生成する。

(1) xを myDAに送ると (2) メソッドがクラスのメソッド辞書から探索
されますが、(3) クラス階層をたどっても見つかりません。(4) 従って self
doesNotUnderstand: #xがオブジェクトに送り返され、(5)改めてメソッド探索が
行われます。DynamicAccessorsではすぐに doesNotUnderstand:が見つかり、(6)文
字列’x ↑ x’をコンパイルするよう依頼します。 (7) compileメソッドが探索され、
(8) 最終的に Behaviorクラスで見つかります。(9-10) コンパイル済みメソッド
が DynamicAccessorsのメソッド辞書に追加されます。(11-13)最後に再び xメッ
セージが再送されますが、今度は対応するメソッドが見つかります。

この方法はインスタンス変数のセッターや、その他のお約束コード、例え

ば Visitorパターンで使うメソッドの生成などに使えます。

ステップ (13)で使われている Object»perform:に注目してください。このメ
ソッドを使うと、実行時に動的にメッセージを生成して送れます。

5 perform: #factorial −→ 120
6 perform: (’fac’, ’torial’) asSymbol −→ 720
4 perform: #max: withArguments: (Array with: 6) −→ 6

14.7 メソッドラッパーとしてのオブジェクト

これまで見てきたように、Smalltalkではコンパイル済みメソッドも通常のオブ
ジェクトであり、プログラマが使うことのできる多くの問い合わせ用のメソッ

ドを用意しています。ただし、ど ん な オブジェクトでもコンパイル済みメソッ

ドの役を演じることができるということはちょっと意外かもしれません。その

オブジェクトは、run:with:in:およびその他いくつかの重要なメッセージに応答し
さえすればいいのです。

空 の ク ラ ス Demoを 定 義 し ま す 。 Demo new answer42を 評 価 し て 、 い つ も の
“Message Not Understood” エ ラ ー の 発 生 を 確 認 し て く だ さ い 。

ここで、普通のオブジェクトを Demoクラスのメソッド辞書にインストール
します。

Demo methodDict at: #answer42 put: ObjectsAsMethodsExample new. を 評 価 し 、
再 び Demo new answer42の 結 果 を 表 示 し て く だ さ い 。 今 度 は 42と い う 回 答 を 得
ら れ ま す 。

ObjectsAsMethodsExampleクラス定義を調べると次の二つのメソッドが見つ
かります:

answer42
↑42

run: oldSelector with: arguments in: aReceiver
↑self perform: oldSelector withArguments: arguments

Demoのインスタンスが answer42メッセージを受け取るといつも通りにメ
ソッド探索が行われますが、仮想マシンは普通のオブジェクトがコンパイル済

みメソッド役を演じようとしていることを検出します。そこで、仮想マシンは

ObjectsAsMethodsExampleオブジェクトに run:with:in:メッセージを送ります。こ
のメッセージの引数は最初に送られたメソッドセレクタ、引数、レシーバです。

ObjectsAsMethodsExampleは run:with:in:メソッドを実装しているので、メッセージ
をインターセプトして自身に移譲します。

次のようにすれば、この偽のメソッドを削除できます:

Demo methodDict removeKey: #answer42 ifAbsent: []



メ ソ ッ ド ラ ッ パ ー と し て の オ ブ ジ ェ ク ト 317

ObjectsAsMethodsExampleをさらに詳しく調べると、スーパークラスが
flushCache、methodClass:、selector:のメソッドを定義していることがわかりま
すが、定義だけで何も実装されていません。これらのメッセージはコンパイル

済みメソッドに送られる可能性があるので、コンパイル済みメソッドのふりをす

るために実装しておく必要があります（flushCacheはこの中で最も重要なメソッ
ドです。他のメソッドは Behavior»addSelector:withMethod:またはMethodDictionary
»at:put:でメソッドをインストールする場合に必要になります）。

メソッドラッパーをテストカバレッジの評価に使う

メソッドラッパーはメッセージをインターセプトする有名な方法です3。オリ
ジナルの実装4では、メソッドラッパーは CompiledMethodのサブクラスのインス
タンスです。インストールされると、メソッドラッパーは元のメソッドの起動

前後に特別な動作を実行できます。アンインストールすると、元のメソッドが

メソッド辞書の然るべき位置に戻されます。

Pharoではメソッドラッパーを簡単に実装できます。CompiledMethodのサ
ブクラスを作る代わりに run:with:in:を実装します。実際、メソッドラッパーと
してのオブジェクトの、軽量の実装5が存在します。しかし、これを書いている
時点では、この実装は標準の Pharoには組み込まれていません。
いずれにせよ、Pharoのテストランナーはまさにこの手法を使ってテスト

カバレッジを評価しています。テストランナーがどのように動作するのか見て

みましょう。

テストカバレッジの開始点は TestRunner»runCoverageです:

TestRunner»runCoverage
| packages methods |
... ”カ バ レ ッ ジ を 測 定 す る メ ソ ッ ド を 決 定 す る ”
self collectCoverageFor: methods

TestRunner»collectCoverageFor:メソッドに、カバレッジを調べるアルゴリズ
ムがわかりやすく書かれています:

TestRunner»collectCoverageFor: methods
| wrappers suite |
wrappers := methods collect: [ :each | TestCoverage on: each ].
suite := self
reset;
suiteAll.

[ wrappers do: [ :each | each install ].
[ self runSuite: suite ] ensure: [ wrappers do: [ :each | each uninstall ] ] ]
valueUnpreemptively.

3John Brant  et al., Wrappers to the Rescue. In Proceedings European Conference on Object
Oriented Programming (ECOOP’98). Volume 1445, Springer-Verlag 1998.

4http://www.squeaksource.com/MethodWrappers.html
5http://www.squeaksource.com/ObjectsAsMethodsWrap.html

wrappers := wrappers reject: [ :each | each hasRun ].
wrappers isEmpty
ifTrue:
[ UIManager default inform: ’Congratulations. Your tests cover all code under

analysis.’ ]
ifFalse: ...

カバレッジを調べたいメソッドごとにラッパーが生成され、それらがインス

トールされます。ここでテストが実行され、テスト実行後にラッパーがアンイ

ンストールされます。最終的にカバーされていないメソッドに関するフィード

バックを得られます。

ラッパー自身はどのように動くのでしょうか？ TestCoverageラッパーには
3個のインスタンス変数（hasRun、reference、method）があり、次のように初期
化されます。

TestCoverage class»on: aMethodReference
↑ self new initializeOn: aMethodReference

TestCoverage»initializeOn: aMethodReference
hasRun := false.
reference := aMethodReference.
method := reference compiledMethod

installメソッドと uninstallメソッドは単純な方法でメソッド辞書を更新し
ます:

TestCoverage»install
reference actualClass methodDictionary
at: reference methodSymbol
put: self

TestCoverage»uninstall
reference actualClass methodDictionary
at: reference methodSymbol
put: method

run:with:in:メソッドは hasRunインスタンス変数を更新し、ラッパーをアンイン
ストールし（カバレッジが検証されたので不要になります）、元のメソッドに

メッセージを再送します。

run: aSelector with: anArray in: aReceiver
self mark; uninstall.
↑ aReceiver withArgs: anArray executeMethod: method

mark
hasRun := true

（ProtoObject»withArgs:executeMethod:を調べて、メソッド辞書から取り除かれた
メソッドがどう起動されるのか確認してください。）

たったこれだけです！

メソッドラッパーを使えば、あらゆる種類の動作をメソッドの通常処理の

前後で実行することができます。典型的な応用例は計測（メソッドの呼び出し

パターンに関する統計情報の収集）、選択的な事前・事後条件の確認、メモ化

（メソッドの処理結果を選択的にキャッシュすること）です。

14.8 プラグマ

プ ラ グ マ とはプログラムに付与する注釈の一種で、プログラムに関連するデー

タを追加することができます。プラグマはメソッド実行に対して直接的な影響

は及ぼしません。プラグマには以下のようなものを含むいくつもの使い道があ

ります:

• コンパイラへの情報: プラグマはコンパイラにメソッド呼び出しをプリ
ミティブ関数として扱うよう指示します。プリミティブ関数は仮想マシ

ンか外部プラグインで定義する必要があります。

• 実行時処理: プラグマによって付加された情報を実行時に検証することが
できます。

プラグマはメソッド宣言にのみ適用できます。メソッドには複数のプラグ

マを宣言でき、Smalltalk文より前に宣言する必要があります。プラグマはリ
テラルを引数に取る静的なメッセージ送信と言えます。

この章の最初でプリミティブを紹介したときにプラグマが出てきました。

プリミティブはプラグマ宣言以上のものではありません。instVarAt:メソッド内
の<primitive: 73>を見てください。このプラグマのセレクタは primitive:で、引数
はリテラル 73です。

おそらくコンパイラはプラグマの大口ユーザです。SUnitも注釈を活用す
るツールです。SUnitはテスト単体からアプリケーションのカバレッジを推定
することができます。しかし、カバレッジから一部のメソッドを除外したくな

ることもあるでしょう。SplitJointTest classの documentationメソッドはそのよう
な例となっています:

SplitJointTest class»documentation
<ignoreForCoverage>
”self showDocumentation”

↑ ’This package provides function.... ”

メソッドに<ignoreForCoverage>プラグマを加えるだけで、カバレッジの適用
範囲を制御できます。

プラグマは Pragmaのインスタンスであり、1級のオブジェクトです。コンパ
イル済みメソッドに pragmasメッセージを送ると、プラグマの配列を返します。

(SplitJoinTest class >> #showDocumentation) pragmas.
−→ an Array(<ignoreForCoverage>)

(Float>>#+) pragmas −→ an Array(<primitive: 41>)

(以下の allNamed:in:メソッドにより)特定のプラグマを定義しているメソッ
ドをクラスから取得することもできます。SplitJoinTestには<ignoreForCoverage>
と注釈されたクラスメソッドがあります:

Pragma allNamed: #ignoreForCoverage in: SplitJoinTest class −→ an Array(<
ignoreForCoverage> <ignoreForCoverage> <ignoreForCoverage>)

allNamed:in:の変種が、Pragmaのクラスメソッドとして定義されています。

プラグマは自身が定義されたメソッド（methodを使います）、そのメソッド
名（selector）、そのメソッドを持つクラス（methodClass）、引数の数（numArgs）、
リテラルを引数として持つかどうか（hasLiteral:と hasLiteralSuchThat:）を知って
います。

14.9 章のまとめ

リフレクションとは、ランタイムシステムのメタオブジェクトに対して普通の

オブジェクトと同様に問い合わせたり、検証したり、変更すらできる機能です。

• インスペクタは instVarAt:や関連するメソッドを使って、オブジェクトの
「プライベートな」インスタンス変数に問い合わせたり、変数の内容を変

更したりします。

• クラスの全インスタンスを取得したいときは、Behavior»allInstancesを送り
ます。

• class、isKindOf:、respondsTo:などのメソッドはメトリクスの収集や開発ツー
ルの実装には便利ですが、通常のアプリケーションでは避けるべきです:
これらのメソッドはオブジェクトのカプセル化の原則に反しており、コー

ドの理解や保守の邪魔になります。

• SystemNavigationはクラス階層の探索や閲覧に便利な多くの問い合わせ
方法を持つユーティリティクラスです。例えば、ある文字列をソース

コードに含むメソッドをすべて見つけ出すには SystemNavigation default
browseMethodsWithSourceString: ’pharo’.を使います（遅いですが、完璧で
す！）。

• どのクラスも MethodDictionaryのインスタンスを参照しています。メソッ
ド辞書はセレクタとコンパイル済みメソッドのインスタンスを結び付け

ています。コンパイル済みメソッドは自身が所属するクラスを知ってお

り、お互いの参照が循環しています。

• MethodReferenceはコンパイル済みメソッドのための軽量プロキシです。
便利なメソッドを提供しており、多くの Smalltalkのツールに使われて
います。

• リファクタリングブラウザの基盤の一部である BrowserEnvironmentを使え
ば、SystemNavigationよりも洗練されたインターフェースでシステムに問
い合わせられます。問い合わせた結果を、次の新しい問い合わせの対象

範囲として使えます。GUIにとプログラム的なインターフェースのどち
らも利用できます。

• thisContextは仮想マシンの実行時スタックを具現化した擬似変数です。主
にデバッガで使われ、スタックの対話的な操作を動的に実現しています。

メッセージのセンダを動的に調べる場合にも便利です。

• 賢いブレークポイントを設定するには haltIf:を使います。このメソッドは
引数にメソッドセレクタを取り、実行時スタックのセンダが同名のメソッ

ドである場合のみ停止します。

• 対象へのメッセージをインターセプトする一般的な方法は「最小限のオブ
ジェクト」をプロキシとして使うことです。プロキシが実装するメソッド

をできるだけ減らし、doesNotunderstand:を実装してすべてのメッセージ
送信を捕捉します。これで追加のアクションを実行してから対象にメッ

セージを転送できます。

• プロキシと主体のような二つのオブジェクトへの参照を交換するには
become:を使います。

• classや yourselfのように、実際には送られず仮想マシンによって解釈され
るメッセージに気をつけてください。他にも+-、ifTrue:などのメッセージ
は、レシーバによっては仮想マシンに直接解釈されたり、インライン展

開されたりする可能性があります。

• 欠けているメソッドの遅延ロードまたは遅延コンパイルのために
doesNotUnderstand:をオーバーライドすることもよくあります。

• doesNotUnderstand:は self送信を捕捉できません。

• より正確にメッセージをインターセプトするには、オブジェクトをメソッ
ドラッパーとして使います。メソッドラッパーはコンパイル済みメソッ

ドの代わりにメソッド辞書にインストールされます。メソッドラッパー

は run:with:in:を実装すべきです。このメッセージは仮想マシンがメソッド
辞書からコンパイル済みメソッドの代わりに普通のオブジェクトを検出

したときに送られます。この方法は SUnitテストランナーでカバレッジ
データを集めるのに使われています。

Appendix A

よくある質問

A.1 はじめに

FAQ 1 最 新 の Pharo は ど こ で 入 手 で き ま す か ?

Answer http://www.pharo-project.org/

FAQ 2 こ の 本 で は ど の Pharo の イ メ ー ジ を 使 う べ き で す か ?

Answer どの Pharoイメージを使っても大丈夫ですが、Pharo by Example
のwebサイト (http://PharoByExample.org)に置いてあるイメージを使うことをお
勧めします。ほとんどの場合、他のイメージも使用できますが、演習によって

は驚くほど動作が異なることがあります。

A.2 コレクション

FAQ 3 OrderedCollection を 並 べ 替 え る に は ど う す れ ば い い で す か ?

Answer インスタンスに asSortedCollectionメッセージを送ります。

#(7 2 6 1) asSortedCollection −→ a SortedCollection(1 2 6 7)

FAQ 4 文 字 の 配 列 を 文 字 列 (String) に 変 換 す る に は ど う す れ ば い い で す か ?

http://www.pharo-project.org/
http://PharoByExample.org

Answer

String streamContents: [:str | str nextPutAll: ’hello’ asSet] −→ ’hleo’

A.3 システムブラウザ

FAQ 5 ブ ラ ウ ザ が こ の 本 で 使 用 さ れ て い る も の と は 違 い ま す 。 ど う す れ ば い
い で す か ?

Answer おそらくデフォルトのブラウザとして、異なるバージョンの Om-
niBrowser がインストールされたイメージを使用しているからです。この本
では、デフォルトとして OmniBrowser の (Package Browser) がインストール
されていることを前提としています。デフォルトブラウザはブラウザのメニ

ューバーをクリックすると変更できます。ウィンドウの右上隅にある灰色の角

丸長方形のボタンをクリックし、“Choose new default Browser”を選択し、
O2PackageBrowserAdaptorを選んでください。次回以降開かれるブラウザは
パッケージブラウザになります。

(a) Choose new default browserを選択 (b) O2PackageBrowserAdaptorを選択

Figure A.1: デフォルトブラウザの変更

FAQ 6 ク ラ ス を 検 索 す る に は ど う す れ ば い い で す か ?

Answer クラス名の文字列を選択して cmd–b (browse)か、ブラウザのカテ
ゴリペイン上にマウスカーソルがある状態で cmd–fです。

FAQ 7 super に メ ッ セ ー ジ を 送 っ て い る す べ て の メ ソ ッ ド を 検 索 し た り ブ ラ ウ
ズ す る に は ど う す れ ば い い で す か ?

Answer 二つ目の手法の方が高速です:
SystemNavigation default browseMethodsWithSourceString: ’super’.
SystemNavigation default browseAllSelect: [:method | method sendsToSuper ].

FAQ 8 あ る ク ラ ス 階 層 内 で 、 super に メ ッ セ ー ジ を 送 っ て い る メ ソ ッ ド を す べ
て ブ ラ ウ ズ す る に は ど う す れ ば い い で す か ?

Answer
browseSuperSends := [:aClass | SystemNavigation default
browseMessageList: (aClass withAllSubclasses gather: [ :each |
(each methodDict associations
select: [ :assoc | assoc value sendsToSuper ])
collect: [ :assoc | MethodReference class: each selector: assoc key ] ])

name: ’Supersends of ’ , aClass name , ’ and its subclasses’].
browseSuperSends value: OrderedCollection.

FAQ 9 サ ブ ク ラ ス で 新 た に 追 加 さ れ た メ ソ ッ ド (親 ク ラ ス の メ ソ ッ ド を オ ー
バ ー ラ イ ド し て い な い メ ソ ッ ド ) を 見 つ け る に は 、 ど う す れ ば い い で す か ?

Answer Trueクラスで追加されたメソッドを調べる例です:
newMethods := [:aClass| aClass methodDict keys select:
[:aMethod | (aClass superclass canUnderstand: aMethod) not ]].

newMethods value: True −→ an IdentitySet(#asBit #xor:)

FAQ 10 ク ラ ス の 抽 象 メ ソ ッ ド は ど う す れ ば わ か り ま す か ?

Answer
abstractMethods :=
[:aClass | aClass methodDict keys select:
[:aMethod | (aClass>>aMethod) isAbstract ]].

abstractMethods value: Collection −→ an IdentitySet(#remove:ifAbsent: #add: #do:)

FAQ 11 AST (抽 象 構 文 木 ) の ビ ュ ー を 生 成 す る に は ど う す れ ば い い で す か ?

Answer squeaksource.comから ASTパッケージをロードし、
(RBParser parseExpression: ’3+4’) explore

を実行します。(または explore itします)

FAQ 12 シ ス テ ム に あ る す べ て の ト レ イ ト を 見 つ け る に は ど う す れ ば い い で
す か ?

Answer

Smalltalk allTraits

FAQ 13 ト レ イ ト を 使 用 し て い る ク ラ ス を 探 す に は ど う す れ ば い い で す か ?

Answer

Smalltalk allClasses select: [:each | each hasTraitComposition and: [each
traitComposition notEmpty ]]

A.4 Monticelloと SqueakSourceを使う

FAQ 14 SqueakSource か ら プ ロ ジ ェ ク ト を ロ ー ド す る に は ど う す れ ば い い で
す か ?

Answer

1. ロードしたいプロジェクトを http://squeaksource.comから探します

2. プロジェクトのリポジトリ情報をクリップボードにコピーします

3. open .Monticello browser を選択してMonticelloブラウザを開きます

4. +Repository .HTTP を選択します

5. リポジトリ情報をペーストし、必要ならパスワードを入力してOKします

6. できたリポジトリを選択して Open を選択します

7. 最新のバージョンのファイル名を選択してロードします

FAQ 15 SqueakSource に プ ロ ジ ェ ク ト を 新 規 作 成 す る に は ど う す れ ば い い で
す か ?

Answer

1. ウェブブラウザで http://squeaksource.comを開きます

2. メンバ登録を行います

3. プロジェクトを登録します (名前＝カテゴリとなります)

4. リポジトリ情報をクリップボードにコピーします

http://squeaksource.com
http://squeaksource.com

5. open .Monticello browser を選択してMonticelloブラウザを開きます

6. +Package を選択してカテゴリを登録します

7. パッケージを選択します

8. +Repository .HTTP を選択します

9. リポジトリ情報をペーストし、必要ならパスワードを入力してOKします

10. Saveボタンで最初のバージョンを保存します

FAQ 16 既 存 の Number ク ラ ス に 追 加 し た Number»chf メ ソ ッ ド を 、 自 分 の Money
プ ロ ジ ェ ク ト の パ ッ ケ ー ジ に 入 れ る に は ど う す れ ば い い で す か ?

Answer そのメソッドのメソッドカテゴリを *Moneyにします。Monticello
は‘*package’のような名前で始まるカテゴリのメソッドをまとめて、そのパッ
ケージに入れます。

A.5 ツール

FAQ 17 プ ロ グ ラ ム ソ ー ス 中 か ら SUnit TestRunner を 開 く に は ど う す れ ば い い
で す か ?

Answer TestRunner openを実行します

FAQ 18 Refactoring Browser は ど こ に あ り ま す か ?

Answer squeaksource.com から AST パッケージをロードし、その後リフ
ァクタリングエンジンをロードしてください: http://www.squeaksource.com/AST
http://www.squeaksource.com/RefactoringEngine

FAQ 19 デ フ ォ ル ト の ブ ラ ウ ザ を 設 定 す る に は ど う す れ ば い い で す か ?

Answer ブラウザウィンドウの右上隅にある灰色のボタンをクリックします。

A.6 正規表現と構文解析

FAQ 20 RegEx(正 規 表 現 ) パ ッ ケ ー ジ に 関 す る ド キ ュ メ ン ト は ど こ に あ り ま
す か ?

http://www.squeaksource.com/AST
http://www.squeaksource.com/RefactoringEngine

Answer VB-Regex カテゴリにある RxParser classクラスの DOCUMENTATION
プロトコルを参照してください。

FAQ 21 パ ー サ を 書 く た め の 何 か い い ツ ー ル は あ り ま す か ?

Answer SmaCC (the Smalltalk Compiler Compiler)を使ってください。そ
の場合少なくとも SmaCC-lr.13以降をインストールする必要があります。http:
//www.squeaksource.com/SmaccDevelopment.htmlからロードできます。ここにはと
てもよくできたオンラインチュートリアルかあります: http://www.refactory.com/
Software/SmaCC/Tutorial.html

FAQ 22 パ ー サ を 書 く た め に は SqueakSource の SmaccDevelopment に あ る ど の
パ ッ ケ ー ジ を ロ ー ド す れ ば い い で す か ?

Answer SmaCCDevの最新バージョンをロードしてください。ランタイム
も含まれています (SmaCC-Developmentは Squeak 3.8用です)。

http://www.squeaksource.com/SmaccDevelopment.html
http://www.squeaksource.com/SmaccDevelopment.html
http://www.refactory.com/Software/SmaCC/Tutorial.html
http://www.refactory.com/Software/SmaCC/Tutorial.html

